

Functional Data Structures

Tobias Nipkow

December 12, 2021

Abstract

A collection of verified functional data structures. The emphasis is on conciseness of algorithms and succinctness of proofs, more in the style of a textbook than a library of efficient algorithms.

For more details see [13].

Contents

1	Sorting	4
2	Creating Almost Complete Trees	13
3	Three-Way Comparison	19
4	Lists Sorted wrt $<$	20
5	List Insertion and Deletion	21
6	Specifications of Set ADT	24
7	Unbalanced Tree Implementation of Set	26
8	Association List Update and Deletion	29
9	Specifications of Map ADT	33
10	Unbalanced Tree Implementation of Map	34
11	Augmented Tree (Tree2)	36
12	Function <i>isin</i> for Tree2	37
13	Interval Trees	37
14	AVL Tree Implementation of Sets	44

15 Function <i>lookup</i> for Tree2	54
16 AVL Tree Implementation of Maps	55
17 AVL Tree with Balance Factors (1)	59
18 AVL Tree with Balance Factors (2)	64
19 Height-Balanced Trees	69
20 Red-Black Trees	77
21 Red-Black Tree Implementation of Sets	78
22 Alternative Deletion in Red-Black Trees	86
23 Red-Black Tree Implementation of Maps	89
24 2-3 Trees	92
25 2-3 Tree Implementation of Sets	93
26 2-3 Tree Implementation of Maps	102
27 2-3 Tree from List	105
28 2-3-4 Trees	109
29 2-3-4 Tree Implementation of Sets	110
30 2-3-4 Tree Implementation of Maps	123
31 1-2 Brother Tree Implementation of Sets	127
32 1-2 Brother Tree Implementation of Maps	140
33 AA Tree Implementation of Sets	145
34 AA Tree Implementation of Maps	156
35 Join-Based Implementation of Sets	162
36 Join-Based Implementation of Sets via RBTs	171
37 Braun Trees	177
38 Arrays via Braun Trees	182

39 Tries via Functions	197
40 Tries via Search Trees	198
41 Binary Tries and Patricia Tries	201
42 Queue Specification	207
43 Queue Implementation via 2 Lists	209
44 Priority Queue Specifications	211
45 Heaps	211
46 Leftist Heap	214
47 Binomial Heap	219
48 Time functions for various standard library operations	233
49 The Median-of-Medians Selection Algorithm	235
50 Bibliographic Notes	260

1 Sorting

theory *Sorting*

imports

Complex_Main

HOL-Library.Multiset

begin

hide_const *List.insort*

declare *Let_def* [*simp*]

1.1 Insertion Sort

fun *insort* :: '*a*::*linorder* \Rightarrow '*a* list \Rightarrow '*a* list **where**

insort *x* [] = [*x*] |

insort *x* (*y*#*ys*) =

(if $x \leq y$ then *x*#*y*#*ys* else *y*#(*insort* *x* *ys*))

fun *isort* :: '*a*::*linorder* list \Rightarrow '*a* list **where**

isort [] = [] |

isort (*x*#*xs*) = *insort* *x* (*isort* *xs*)

1.1.1 Functional Correctness

lemma *mset_insort*: $mset (insort\ x\ xs) = \{\#x\} + mset\ xs$

apply(*induction* *xs*)

apply *auto*

done

lemma *mset_isort*: $mset (isort\ xs) = mset\ xs$

apply(*induction* *xs*)

apply *simp*

apply (*simp* *add*: *mset_insort*)

done

lemma *set_insort*: $set (insort\ x\ xs) = \{x\} \cup set\ xs$

by(*simp* *add*: *mset_insort* *flip*: *set_mset_mset*)

lemma *sorted_insort*: $sorted (insort\ a\ xs) = sorted\ xs$

apply(*induction* *xs*)

apply(*auto* *simp* *add*: *set_insort*)

done

lemma *sorted_isort*: $sorted (isort\ xs)$

```

apply(induction xs)
apply(auto simp: sorted_insort)
done

```

1.1.2 Time Complexity

We count the number of function calls.

$$\text{insort } x \ [] = [x] \text{ insort } x (y\#ys) = (\text{if } x \leq y \text{ then } x\#y\#ys \text{ else } y\#(\text{insort } x \ ys))$$

```

fun T_insort :: 'a::linorder  $\Rightarrow$  'a list  $\Rightarrow$  nat where
  T_insort x [] = 1 |
  T_insort x (y#ys) =
    (if  $x \leq y$  then 0 else T_insort x ys) + 1
  isort [] = [] isort (x#xs) = insort x (isort xs)

```

```

fun T_isort :: 'a::linorder list  $\Rightarrow$  nat where
  T_isort [] = 1 |
  T_isort (x#xs) = T_isort xs + T_insort x (isort xs) + 1

```

```

lemma T_insort_length: T_insort x xs  $\leq$  length xs + 1
apply(induction xs)
apply auto
done

```

```

lemma length_insort: length (insort x xs) = length xs + 1
apply(induction xs)
apply auto
done

```

```

lemma length_isort: length (isort xs) = length xs
apply(induction xs)
apply (auto simp: length_insort)
done

```

```

lemma T_isort_length: T_isort xs  $\leq$  (length xs + 1) ^ 2
proof(induction xs)
  case Nil show ?case by simp
next
  case (Cons x xs)
  have T_isort (x#xs) = T_isort xs + T_insort x (isort xs) + 1 by simp
  also have ...  $\leq$  (length xs + 1) ^ 2 + T_insort x (isort xs) + 1
  using Cons.IH by simp

```

also have $\dots \leq (\text{length } xs + 1) ^ 2 + \text{length } xs + 1 + 1$
using T_insert_length [of x $isort$ xs] **by** ($simp$ add: $length_isort$)
also have $\dots \leq (\text{length}(x\#xs) + 1) ^ 2$
by ($simp$ add: $power2_eq_square$)
finally show $?case$.
qed

1.2 Merge Sort

fun $merge :: 'a::linorder\ list \Rightarrow 'a\ list \Rightarrow 'a\ list$ **where**
 $merge []\ ys = ys$ |
 $merge\ xs\ [] = xs$ |
 $merge\ (x\#xs)\ (y\#ys) = (\text{if } x \leq y \text{ then } x \# merge\ xs\ (y\#ys) \text{ else } y \# merge\ (x\#xs)\ ys)$

fun $msort :: 'a::linorder\ list \Rightarrow 'a\ list$ **where**
 $msort\ xs = (\text{let } n = \text{length } xs \text{ in}$
 $\text{if } n \leq 1 \text{ then } xs$
 $\text{else } merge\ (msort\ (\text{take } (n\ \text{div } 2)\ xs))\ (msort\ (\text{drop } (n\ \text{div } 2)\ xs)))$

declare $msort.simps$ [$simp\ del$]

1.2.1 Functional Correctness

lemma $mset_merge: mset(merge\ xs\ ys) = mset\ xs + mset\ ys$
by($induction\ xs\ ys$ rule: $merge.induct$) $auto$

lemma $mset_msort: mset\ (msort\ xs) = mset\ xs$

proof($induction\ xs$ rule: $msort.induct$)

case ($1\ xs$)

let $?n = \text{length } xs$

let $?ys = \text{take } (?n\ \text{div } 2)\ xs$

let $?zs = \text{drop } (?n\ \text{div } 2)\ xs$

show $?case$

proof $cases$

assume $?n \leq 1$

thus $?thesis$ **by**($simp$ add: $msort.simps$ [of xs])

next

assume $\neg ?n \leq 1$

hence $mset\ (msort\ xs) = mset\ (msort\ ?ys) + mset\ (msort\ ?zs)$

by($simp$ add: $msort.simps$ [of xs] $mset_merge$)

also have $\dots = mset\ ?ys + mset\ ?zs$

using $\langle \neg ?n \leq 1 \rangle$ **by**($simp$ add: $1.IH$)

also have $\dots = mset\ (?ys\ @\ ?zs)$ **by** ($simp\ del: append_take_drop_id$)

```

    also have ... = mset xs by simp
    finally show ?thesis .
qed
qed

```

Via the previous lemma or directly:

```

lemma set_merge: set(merge xs ys) = set xs ∪ set ys
by (metis mset_merge set_mset_mset set_mset_union)

```

```

lemma set(merge xs ys) = set xs ∪ set ys
by(induction xs ys rule: merge.induct) (auto)

```

```

lemma sorted_merge: sorted (merge xs ys)  $\longleftrightarrow$  (sorted xs ∧ sorted ys)
by(induction xs ys rule: merge.induct) (auto simp: set_merge)

```

```

lemma sorted_msort: sorted (msort xs)

```

```

proof(induction xs rule: msort.induct)
  case (1 xs)
  let ?n = length xs
  show ?case
  proof cases
    assume ?n ≤ 1
    thus ?thesis by(simp add: msort.simps[of xs] sorted01)
  next
    assume ¬ ?n ≤ 1
    thus ?thesis using 1.IH
    by(simp add: sorted_merge msort.simps[of xs])
  qed
qed

```

1.2.2 Time Complexity

We only count the number of comparisons between list elements.

```

fun C_merge :: 'a::linorder list ⇒ 'a list ⇒ nat where
  C_merge [] ys = 0 |
  C_merge xs [] = 0 |
  C_merge (x#xs) (y#ys) = 1 + (if x ≤ y then C_merge xs (y#ys) else
  C_merge (x#xs) ys)

```

```

lemma C_merge_ub: C_merge xs ys ≤ length xs + length ys
by (induction xs ys rule: C_merge.induct) auto

```

```

fun C_msort :: 'a::linorder list ⇒ nat where
  C_msort xs =

```

```

(let n = length xs;
  ys = take (n div 2) xs;
  zs = drop (n div 2) xs
in if n ≤ 1 then 0
  else C_msort ys + C_msort zs + C_merge (msort ys) (msort zs))

```

```

declare C_msort.simps [simp del]

```

```

lemma length_merge: length(merge xs ys) = length xs + length ys
apply (induction xs ys rule: merge.induct)
apply auto
done

```

```

lemma length_msort: length(msort xs) = length xs
proof (induction xs rule: msort.induct)
  case (1 xs)
  show ?case
  by (auto simp: msort.simps [of xs] 1 length_merge)
qed

```

Why structured proof? To have the name "xs" to specialize `msort.simps` with `xs` to ensure that `msort.simps` cannot be used recursively. Also works without this precaution, but that is just luck.

```

lemma C_msort_le: length xs = 2^k ⇒ C_msort xs ≤ k * 2^k
proof(induction k arbitrary: xs)
  case 0 thus ?case by (simp add: C_msort.simps)
next
  case (Suc k)
  let ?n = length xs
  let ?ys = take (?n div 2) xs
  let ?zs = drop (?n div 2) xs
  show ?case
  proof (cases ?n ≤ 1)
    case True
    thus ?thesis by(simp add: C_msort.simps)
  next
  case False
  have C_msort(xs) =
    C_msort ?ys + C_msort ?zs + C_merge (msort ?ys) (msort ?zs)
  by (simp add: C_msort.simps msort.simps)
  also have ... ≤ C_msort ?ys + C_msort ?zs + length ?ys + length
  ?zs
  using C_merge_ub[of msort ?ys msort ?zs] length_msort[of ?ys]
  length_msort[of ?zs]

```

```

    by arith
  also have ... ≤ k * 2^k + C_msort ?zs + length ?ys + length ?zs
    using Suc.IH[of ?ys] Suc.prem by simp
  also have ... ≤ k * 2^k + k * 2^k + length ?ys + length ?zs
    using Suc.IH[of ?zs] Suc.prem by simp
  also have ... = 2 * k * 2^k + 2 * 2^k
    using Suc.prem by simp
  finally show ?thesis by simp
qed
qed

```

lemma *C_msort_log*: $\text{length } xs = 2^k \implies C_msort\ xs \leq \text{length } xs * \log 2 (\text{length } xs)$
using *C_msort_le*[of *xs k*] **apply** (*simp add: log_nat_power algebra_simps*)
by (*metis (mono_tags) numeral_power_eq_of_nat_cancel_iff of_nat_le_iff of_nat_mult*)

1.3 Bottom-Up Merge Sort

```

fun merge_adj :: ('a::linorder) list list ⇒ 'a list where
merge_adj [] = [] |
merge_adj [x] = [x] |
merge_adj (xs # ys # zss) = merge xs ys # merge_adj zss

```

For the termination proof of *merge_all* below.

lemma *length_merge_adjacent*[*simp*]: $\text{length } (\text{merge_adj } xs) = (\text{length } xs + 1) \text{ div } 2$
by (*induction xs rule: merge_adj.induct*) *auto*

```

fun merge_all :: ('a::linorder) list list ⇒ 'a list where
merge_all [] = [] |
merge_all [x] = x |
merge_all xss = merge_all (merge_adj xss)

```

definition *msort_bu* :: ('a::linorder) list ⇒ 'a list **where**
msort_bu xs = merge_all (map (λx. [x]) xs)

1.3.1 Functional Correctness

abbreviation *mset_mset* :: 'a list list ⇒ 'a multiset **where**
mset_mset xss ≡ ∑ # (image_mset mset (mset xss))

lemma *mset_merge_adj*:
mset_mset (merge_adj xss) = mset_mset xss

by(*induction xss rule: merge_adj.induct*) (*auto simp: mset_merge*)

lemma *mset_merge_all*:

mset (merge_all xss) = mset_mset xss

by(*induction xss rule: merge_all.induct*) (*auto simp: mset_merge mset_merge_adj*)

lemma *mset_msort_bu*: *mset (msort_bu xs) = mset xs*

by(*simp add: msort_bu_def mset_merge_all multiset.map_comp comp_def*)

lemma *sorted_merge_adj*:

$\forall xs \in \text{set } xss. \text{sorted } xs \implies \forall xs \in \text{set } (\text{merge_adj } xss). \text{sorted } xs$

by(*induction xss rule: merge_adj.induct*) (*auto simp: sorted_merge*)

lemma *sorted_merge_all*:

$\forall xs \in \text{set } xss. \text{sorted } xs \implies \text{sorted } (\text{merge_all } xss)$

apply(*induction xss rule: merge_all.induct*)

using $[[\text{simp_depth_limit}=3]]$ **by** (*auto simp add: sorted_merge_adj*)

lemma *sorted_msort_bu*: *sorted (msort_bu xs)*

by(*simp add: msort_bu_def sorted_merge_all*)

1.3.2 Time Complexity

fun *C_merge_adj* :: (*'a::linorder*) *list list* \Rightarrow *nat* **where**

C_merge_adj [] = 0 |

C_merge_adj [xs] = 0 |

C_merge_adj (xs # ys # zss) = *C_merge* xs ys + *C_merge_adj* zss

fun *C_merge_all* :: (*'a::linorder*) *list list* \Rightarrow *nat* **where**

C_merge_all [] = 0 |

C_merge_all [xs] = 0 |

C_merge_all xss = *C_merge_adj* xss + *C_merge_all* (merge_adj xss)

definition *C_msort_bu* :: (*'a::linorder*) *list* \Rightarrow *nat* **where**

C_msort_bu xs = *C_merge_all* (map ($\lambda x. [x]$) xs)

lemma *length_merge_adj*:

$[[\text{even}(\text{length } xss); \forall xs \in \text{set } xss. \text{length } xs = m]]$

$\implies \forall xs \in \text{set } (\text{merge_adj } xss). \text{length } xs = 2*m$

by(*induction xss rule: merge_adj.induct*) (*auto simp: length_merge*)

lemma *C_merge_adj*: $\forall xs \in \text{set } xss. \text{length } xs = m \implies \text{C_merge_adj } xss < m * \text{length } xss$

proof(*induction xss rule: C_merge_adj.induct*)

```

  case 1 thus ?case by simp
next
  case 2 thus ?case by simp
next
  case (3 x y) thus ?case using C_merge_ub[of x y] by (simp add: algebra_simps)
qed

```

lemma C_merge_all : $[\forall xs \in set\ xss. length\ xs = m; length\ xss = 2^k]$
 $\implies C_merge_all\ xss \leq m * k * 2^k$

proof (induction xss arbitrary: k m rule: C_merge_all.induct)

```

  case 1 thus ?case by simp
next
  case 2 thus ?case by simp
next
  case (3 xs ys xss)
  let ?xss = xs # ys # xss
  let ?xss2 = merge_adj ?xss
  obtain k' where k': k = Suc k' using 3.prem2
  by (metis length_Cons nat.inject nat_power_eq_Suc_0_iff nat.exhaust)
  have even (length ?xss) using 3.prem2 k' by auto
  from length_merge_adj[OF this 3.prem1]
  have *:  $\forall x \in set(merge\_adj\ ?xss). length\ x = 2*m$  .
  have **:  $length\ ?xss2 = 2^{k'}$  using 3.prem2 k' by auto
  have C_merge_all ?xss = C_merge_adj ?xss + C_merge_all ?xss2 by
  simp
  also have ...  $\leq m * 2^k + C\_merge\_all\ ?xss2$ 
  using 3.prem2 C_merge_adj[OF 3.prem1] by (auto simp: algebra_simps)
  also have ...  $\leq m * 2^k + (2*m) * k' * 2^{k'}$ 
  using 3.IH[OF * **] by simp
  also have ...  $= m * k * 2^k$ 
  using k' by (simp add: algebra_simps)
  finally show ?case .
qed

```

corollary C_msort_bu : $length\ xs = 2^k \implies C_msort_bu\ xs \leq k * 2^k$

using C_merge_all [of map ($\lambda x. [x]$) xs 1] **by** (simp add: C_msort_bu_def)

1.4 Quicksort

```

fun quicksort :: ('a::linorder) list  $\Rightarrow$  'a list where
quicksort [] = [] |

```

$quicksort\ (x\#\!xs) = quicksort\ (filter\ (\lambda y. y < x)\ xs)\ @\ [x]\ @\ quicksort\ (filter\ (\lambda y. x \leq y)\ xs)$

lemma *mset_quicksort*: $mset\ (quicksort\ xs) = mset\ xs$
apply (*induction xs rule: quicksort.induct*)
apply (*auto simp: not_le*)
done

lemma *set_quicksort*: $set\ (quicksort\ xs) = set\ xs$
by(*rule mset_eq_setD[OF mset_quicksort]*)

lemma *sorted_quicksort*: $sorted\ (quicksort\ xs)$
apply (*induction xs rule: quicksort.induct*)
apply (*auto simp add: sorted_append set_quicksort*)
done

1.5 Insertion Sort w.r.t. Keys and Stability

Note that *insort_key* is already defined in theory *HOL.List*. Thus some of the lemmas are already present as well.

fun *insort_key* :: (*'a* \Rightarrow *'k::linorder*) \Rightarrow *'a list* \Rightarrow *'a list* **where**
insort_key *f* [] = [] |
insort_key *f* (*x* # *xs*) = *insort_key* *f* *x* (*insort_key* *f* *xs*)

1.5.1 Standard functional correctness

lemma *mset_insort_key*: $mset\ (insort_key\ f\ x\ xs) = \{\#x\#\} + mset\ xs$
by(*induction xs simp_all*)

lemma *mset_isort_key*: $mset\ (isort_key\ f\ xs) = mset\ xs$
by(*induction xs (simp_all add: mset_insort_key)*)

lemma *set_isort_key*: $set\ (isort_key\ f\ xs) = set\ xs$
by (*rule mset_eq_setD[OF mset_isort_key]*)

lemma *sorted_insort_key*: $sorted\ (map\ f\ (insort_key\ f\ a\ xs)) = sorted\ (map\ f\ xs)$
by(*induction xs (auto simp: set_insort_key)*)

lemma *sorted_isort_key*: $sorted\ (map\ f\ (isort_key\ f\ xs))$
by(*induction xs (simp_all add: sorted_insort_key)*)

1.5.2 Stability

lemma *insort_is_Cons*: $\forall x \in \text{set } xs. f a \leq f x \implies \text{insort_key } f a xs = a \# xs$
by (*cases xs*) *auto*

lemma *filter_insort_key_neg*:
 $\neg P x \implies \text{filter } P (\text{insort_key } f x xs) = \text{filter } P xs$
by (*induction xs*) *simp_all*

lemma *filter_insort_key_pos*:
 $\text{sorted } (\text{map } f xs) \implies P x \implies \text{filter } P (\text{insort_key } f x xs) = \text{insort_key } f x (\text{filter } P xs)$
by (*induction xs*) (*auto, subst insort_is_Cons, auto*)

lemma *sort_key_stable*: $\text{filter } (\lambda y. f y = k) (\text{isort_key } f xs) = \text{filter } (\lambda y. f y = k) xs$

proof (*induction xs*)

case Nil **thus** *?case* **by** *simp*

next

case (*Cons a xs*)

thus *?case*

proof (*cases f a = k*)

case False **thus** *?thesis* **by** (*simp add: Cons.IH filter_insort_key_neg*)

next

case True

have $\text{filter } (\lambda y. f y = k) (\text{isort_key } f (a \# xs))$

$= \text{filter } (\lambda y. f y = k) (\text{insort_key } f a (\text{isort_key } f xs))$ **by** *simp*

also have $\dots = \text{insort_key } f a (\text{filter } (\lambda y. f y = k) (\text{isort_key } f xs))$

by (*simp add: True filter_insort_key_pos sorted_isort_key*)

also have $\dots = \text{insort_key } f a (\text{filter } (\lambda y. f y = k) xs)$ **by** (*simp add: Cons.IH*)

also have $\dots = a \# (\text{filter } (\lambda y. f y = k) xs)$ **by** (*simp add: True insort_is_Cons*)

also have $\dots = \text{filter } (\lambda y. f y = k) (a \# xs)$ **by** (*simp add: True*)

finally show *?thesis* .

qed

qed

end

2 Creating Almost Complete Trees

theory *Balance*

```

imports
  HOL-Library.Tree_Real
begin

fun bal :: nat ⇒ 'a list ⇒ 'a tree * 'a list where
  bal n xs = (if n=0 then (Leaf,xs) else
    (let m = n div 2;
      (l, ys) = bal m xs;
      (r, zs) = bal (n-1-m) (tl ys)
      in (Node l (hd ys) r, zs)))

declare bal.simps[simp del]
declare Let_def[simp]

definition bal_list :: nat ⇒ 'a list ⇒ 'a tree where
  bal_list n xs = fst (bal n xs)

definition balance_list :: 'a list ⇒ 'a tree where
  balance_list xs = bal_list (length xs) xs

definition bal_tree :: nat ⇒ 'a tree ⇒ 'a tree where
  bal_tree n t = bal_list n (inorder t)

definition balance_tree :: 'a tree ⇒ 'a tree where
  balance_tree t = bal_tree (size t) t

lemma bal_simps:
  bal 0 xs = (Leaf, xs)
  n > 0 ⇒
    bal n xs =
      (let m = n div 2;
        (l, ys) = bal m xs;
        (r, zs) = bal (n-1-m) (tl ys)
        in (Node l (hd ys) r, zs))
by(simp_all add: bal.simps)

lemma bal_inorder:
  [ n ≤ length xs; bal n xs = (t,zs) ]
  ⇒ xs = inorder t @ zs ∧ size t = n
proof(induction n arbitrary: xs t zs rule: less_induct)
case (less n) show ?case
proof cases
  assume n = 0 thus ?thesis using less.premis by (simp add: bal_simps)
next

```

```

assume [arith]:  $n \neq 0$ 
let  $?m = n \text{ div } 2$  let  $?m' = n - 1 - ?m$ 
from less.prems(2) obtain  $l\ r\ ys$  where
   $b1: \text{bal } ?m\ xs = (l, ys)$  and
   $b2: \text{bal } ?m'\ (tl\ ys) = (r, zs)$  and
   $t: t = \langle l, \text{hd } ys, r \rangle$ 
by(auto simp: bal_simps split: prod.splits)
have IH1:  $xs = \text{inorder } l @ ys \wedge \text{size } l = ?m$ 
using b1 less.prems(1) by(intro less.IH) auto
have IH2:  $tl\ ys = \text{inorder } r @ zs \wedge \text{size } r = ?m'$ 
using b2 IH1 less.prems(1) by(intro less.IH) auto
show ?thesis using t IH1 IH2 less.prems(1) hd_Cons_tl[of ys] by
fastforce
qed
qed

```

```

corollary inorder_bal_list[simp]:
   $n \leq \text{length } xs \implies \text{inorder}(\text{bal\_list } n\ xs) = \text{take } n\ xs$ 
unfolding bal_list_def
by (metis (mono_tags) prod.collapse[of bal n xs] append_eq_conv_conj
bal_inorder length_inorder)

```

```

corollary inorder_balance_list[simp]:  $\text{inorder}(\text{balance\_list } xs) = xs$ 
by(simp add: balance_list_def)

```

```

corollary inorder_bal_tree:
   $n \leq \text{size } t \implies \text{inorder}(\text{bal\_tree } n\ t) = \text{take } n\ (\text{inorder } t)$ 
by(simp add: bal_tree_def)

```

```

corollary inorder_balance_tree[simp]:  $\text{inorder}(\text{balance\_tree } t) = \text{inorder } t$ 
by(simp add: balance_tree_def inorder_bal_tree)

```

The length/size lemmas below do not require the precondition $n \leq \text{length } xs$ (or $n \leq \text{size } t$) that they come with. They could take advantage of the fact that $\text{bal } xs\ n$ yields a result even if $\text{length } xs < n$. In that case the result will contain one or more occurrences of $\text{hd } []$. However, this is counter-intuitive and does not reflect the execution in an eager functional language.

```

lemma bal_length:  $\llbracket n \leq \text{length } xs; \text{bal } n\ xs = (t, zs) \rrbracket \implies \text{length } zs = \text{length } xs - n$ 
using bal_inorder by fastforce

```

```

corollary size_bal_list[simp]:  $n \leq \text{length } xs \implies \text{size}(\text{bal\_list } n\ xs) = n$ 
unfolding bal_list_def using bal_inorder prod.exhaust_sel by blast

```

corollary *size_balance_list*[simp]: $\text{size}(\text{balance_list } xs) = \text{length } xs$
by (*simp add: balance_list_def*)

corollary *size_bal_tree*[simp]: $n \leq \text{size } t \implies \text{size}(\text{bal_tree } n \ t) = n$
by(*simp add: bal_tree_def*)

corollary *size_balance_tree*[simp]: $\text{size}(\text{balance_tree } t) = \text{size } t$
by(*simp add: balance_tree_def*)

lemma *min_height_bal*:

$\llbracket n \leq \text{length } xs; \text{bal } n \ xs = (t, zs) \rrbracket \implies \text{min_height } t = \text{nat}(\lfloor \log 2 (n + 1) \rfloor)$

proof(*induction n arbitrary: xs t zs rule: less_induct*)

case (*less n*)

show *?case*

proof *cases*

assume $n = 0$ **thus** *?thesis* **using** *less.prem1* **by** (*simp add: bal_simps*)

next

assume [*arith*]: $n \neq 0$

let $?m = n \text{ div } 2$ **let** $?m' = n - 1 - ?m$

from *less.prem1* **obtain** $l \ r \ ys$ **where**

$b1: \text{bal } ?m \ xs = (l, ys)$ **and**

$b2: \text{bal } ?m' \ (tl \ ys) = (r, zs)$ **and**

$t: t = \langle l, \text{hd } ys, r \rangle$

by(*auto simp: bal_simps split: prod.splits*)

let $?hl = \text{nat}(\text{floor}(\log 2 (?m + 1)))$

let $?hr = \text{nat}(\text{floor}(\log 2 (?m' + 1)))$

have *IH1*: $\text{min_height } l = ?hl$ **using** *less.IH[OF _ _ b1]* *less.prem1*

by *simp*

have *IH2*: $\text{min_height } r = ?hr$

using *less.prem1* *bal_length[OF _ b1]* *b2* **by**(*intro less.IH*) *auto*

have $(n+1) \text{ div } 2 \geq 1$ **by** *arith*

hence $0: \log 2 ((n+1) \text{ div } 2) \geq 0$ **by** *simp*

have $?m' \leq ?m$ **by** *arith*

hence $le: ?hr \leq ?hl$ **by**(*simp add: nat_mono floor_mono*)

have $\text{min_height } t = \min ?hl ?hr + 1$ **by** (*simp add: t IH1 IH2*)

also have $\dots = ?hr + 1$ **using** le **by** (*simp add: min_absorb2*)

also have $?m' + 1 = (n+1) \text{ div } 2$ **by** *linarith*

also have $\text{nat}(\text{floor}(\log 2 ((n+1) \text{ div } 2))) + 1$
 $= \text{nat}(\text{floor}(\log 2 ((n+1) \text{ div } 2) + 1))$

using 0 **by** *linarith*

also have $\dots = \text{nat}(\text{floor}(\log 2 (n + 1)))$

using *floor_log2_div2[of n+1]* **by** (*simp add: log_mult*)

finally show *?thesis* .

qed
qed

lemma *height_bal*:

$\llbracket n \leq \text{length } xs; \text{bal } n \text{ } xs = (t, zs) \rrbracket \implies \text{height } t = \text{nat } \lceil \log 2 (n + 1) \rceil$

proof (*induction n arbitrary: xs t zs rule: less_induct*)

case (*less n*) **show** *?case*

proof *cases*

assume $n = 0$ **thus** *?thesis*

using *less.prem*s **by** (*simp add: bal_simps*)

next

assume [*arith*]: $n \neq 0$

let $?m = n \text{ div } 2$ **let** $?m' = n - 1 - ?m$

from *less.prem*s **obtain** $l \ r \ ys$ **where**

$b1: \text{bal } ?m \ xs = (l, ys)$ **and**

$b2: \text{bal } ?m' \ (tl \ ys) = (r, zs)$ **and**

$t: t = \langle l, hd \ ys, r \rangle$

by (*auto simp: bal_simps split: prod.splits*)

let $?hl = \text{nat } \lceil \log 2 (?m + 1) \rceil$

let $?hr = \text{nat } \lceil \log 2 (?m' + 1) \rceil$

have *IH1*: $\text{height } l = ?hl$ **using** *less.IH*[*OF* $_ _ b1$] *less.prem*s(1) **by**

simp

have *IH2*: $\text{height } r = ?hr$

using $b2 \text{ bal_length}$ [*OF* $_ b1$] *less.prem*s(1) **by** (*intro less.IH*) *auto*

have $0: \log 2 (?m + 1) \geq 0$ **by** *simp*

have $?m' \leq ?m$ **by** *arith*

hence $le: ?hr \leq ?hl$

by (*simp add: nat_mono ceiling_mono del: nat_ceiling_le_eq*)

have $\text{height } t = \max ?hl ?hr + 1$ **by** (*simp add: t IH1 IH2*)

also have $\dots = ?hl + 1$ **using** le **by** (*simp add: max_absorb1*)

also have $\dots = \text{nat } \lceil \log 2 (?m + 1) + 1 \rceil$ **using** 0 **by** *linarith*

also have $\dots = \text{nat } \lceil \log 2 (n + 1) \rceil$

using ceiling_log2_div2 [*of* $n+1$] **by** (*simp*)

finally show *?thesis* .

qed

qed

lemma *acomplete_bal*:

assumes $n \leq \text{length } xs \text{ bal } n \text{ } xs = (t, ys)$ **shows** *acomplete t*

unfolding *acomplete_def*

using height_bal [*OF* *assms*] min_height_bal [*OF* *assms*]

by *linarith*

lemma *height_bal_list*:

$n \leq \text{length } xs \implies \text{height } (\text{bal_list } n \text{ } xs) = \text{nat } \lceil \log 2 (n + 1) \rceil$
unfolding *bal_list_def* **by** (*metis height_bal prod.collapse*)

lemma *height_balance_list*:
 $\text{height } (\text{balance_list } xs) = \text{nat } \lceil \log 2 (\text{length } xs + 1) \rceil$
by (*simp add: balance_list_def height_bal_list*)

corollary *height_bal_tree*:
 $n \leq \text{size } t \implies \text{height } (\text{bal_tree } n \text{ } t) = \text{nat } \lceil \log 2 (n + 1) \rceil$
unfolding *bal_list_def bal_tree_def*
by (*metis bal_list_def height_bal_list length_inorder*)

corollary *height_balance_tree*:
 $\text{height } (\text{balance_tree } t) = \text{nat } \lceil \log 2 (\text{size } t + 1) \rceil$
by (*simp add: bal_tree_def balance_tree_def height_bal_list*)

corollary *acomplete_bal_list[simp]*: $n \leq \text{length } xs \implies \text{acomplete } (\text{bal_list } n \text{ } xs)$
unfolding *bal_list_def* **by** (*metis acomplete_bal prod.collapse*)

corollary *acomplete_balance_list[simp]*: $\text{acomplete } (\text{balance_list } xs)$
by (*simp add: balance_list_def*)

corollary *acomplete_bal_tree[simp]*: $n \leq \text{size } t \implies \text{acomplete } (\text{bal_tree } n \text{ } t)$
by (*simp add: bal_tree_def*)

corollary *acomplete_balance_tree[simp]*: $\text{acomplete } (\text{balance_tree } t)$
by (*simp add: balance_tree_def*)

lemma *wbalanced_bal*: $\llbracket n \leq \text{length } xs; \text{bal } n \text{ } xs = (t, ys) \rrbracket \implies \text{wbanced } t$
proof(*induction n arbitrary: xs t ys rule: less_induct*)

case (*less n*)

show *?case*

proof *cases*

assume $n = 0$

thus *?thesis using less.prem(2)* **by**(*simp add: bal_simps*)

next

assume [*arith*]: $n \neq 0$

with *less.prem* **obtain** $l \text{ } ys \text{ } r \text{ } zs$ **where**

rec1: $\text{bal } (n \text{ div } 2) \text{ } xs = (l, ys)$ **and**

rec2: $\text{bal } (n - 1 - n \text{ div } 2) \text{ } (tl \text{ } ys) = (r, zs)$ **and**

t: $t = \langle l, \text{hd } ys, r \rangle$

by(*auto simp add: bal_simps split: prod.splits*)

```

have l: wbalanced l using less.IH[OF ___ rec1] less.prem1 by linarith
have wbalanced r
using rec1 rec2 bal_length[OF _ rec1] less.prem1 by (intro less.IH)
auto
with l t bal_length[OF _ rec1] less.prem1 bal_inorder[OF _ rec1]
bal_inorder[OF _ rec2]
show ?thesis by auto
qed
qed

```

An alternative proof via *wbalanced ?t \implies acomplete ?t*:

```

lemma [ $n \leq \text{length } xs$ ;  $\text{bal } n \text{ } xs = (t, ys)$ ]  $\implies$  acomplete t
by (rule acomplete_if_wbalanced[OF wbalanced_bal])

```

```

lemma wbalanced_bal_list[simp]:  $n \leq \text{length } xs \implies \text{wbalanced } (\text{bal\_list } n \text{ } xs)$ 
by (simp add: bal_list_def) (metis prod.collapse wbalanced_bal)

```

```

lemma wbalanced_balance_list[simp]:  $\text{wbalanced } (\text{balance\_list } xs)$ 
by (simp add: balance_list_def)

```

```

lemma wbalanced_bal_tree[simp]:  $n \leq \text{size } t \implies \text{wbalanced } (\text{bal\_tree } n \text{ } t)$ 
by (simp add: bal_tree_def)

```

```

lemma wbalanced_balance_tree:  $\text{wbalanced } (\text{balance\_tree } t)$ 
by (simp add: balance_tree_def)

```

```

hide_const (open) bal

```

```

end

```

3 Three-Way Comparison

```

theory Cmp
imports Main
begin

```

```

datatype cmp_val = LT | EQ | GT

```

```

definition cmp :: 'a:: linorder  $\Rightarrow$  'a  $\Rightarrow$  cmp_val where
cmp x y = (if x < y then LT else if x=y then EQ else GT)

```

```

lemma
LT[simp]:  $\text{cmp } x \text{ } y = \text{LT} \iff x < y$ 

```

and $EQ[simp]: cmp\ x\ y = EQ \longleftrightarrow x = y$
and $GT[simp]: cmp\ x\ y = GT \longleftrightarrow x > y$
by (*auto simp: cmp_def*)

lemma $case_cmp_if[simp]: (case\ c\ of\ EQ \Rightarrow e \mid LT \Rightarrow l \mid GT \Rightarrow g) =$
(if $c = LT$ *then* l *else if* $c = GT$ *then* g *else* e *)*
by(*simp split: cmp_val.split*)

end

4 Lists Sorted wrt $<$

theory *Sorted_Less*
imports *Less_False*
begin

hide_const *sorted*

Is a list sorted without duplicates, i.e., wrt $<?$.

abbreviation $sorted :: 'a::linorder\ list \Rightarrow bool$ **where**
 $sorted \equiv sorted_wrt (<)$

lemmas $sorted_wrt_Cons = sorted_wrt.simps(2)$

The definition of *sorted_wrt* relates each element to all the elements after it. This causes a blowup of the formulas. Thus we simplify matters by only comparing adjacent elements.

declare

$sorted_wrt.simps(2)[simp\ del]$
 $sorted_wrt1[simp]\ sorted_wrt2[OF\ transp_less,\ simp]$

lemma $sorted_cons: sorted\ (x\#\ xs) \Longrightarrow sorted\ xs$
by(*simp add: sorted_wrt_Cons*)

lemma $sorted_cons': ASSUMPTION\ (sorted\ (x\#\ xs)) \Longrightarrow sorted\ xs$
by(*rule ASSUMPTION_D [THEN sorted_cons]*)

lemma $sorted_snoc: sorted\ (xs\ @\ [y]) \Longrightarrow sorted\ xs$
by(*simp add: sorted_wrt_append*)

lemma $sorted_snoc': ASSUMPTION\ (sorted\ (xs\ @\ [y])) \Longrightarrow sorted\ xs$
by(*rule ASSUMPTION_D [THEN sorted_snoc]*)

lemma $sorted_mid_iff:$

$sorted(xs @ y \# ys) = (sorted(xs @ [y]) \wedge sorted(y \# ys))$
by(*fastforce simp add: sorted_wrt_Cons sorted_wrt_append*)

lemma *sorted_mid_iff2*:
 $sorted(x \# xs @ y \# ys) =$
 $(sorted(x \# xs) \wedge x < y \wedge sorted(xs @ [y]) \wedge sorted(y \# ys))$
by(*fastforce simp add: sorted_wrt_Cons sorted_wrt_append*)

lemma *sorted_mid_iff'*: *NO_MATCH [] ys* \implies
 $sorted(xs @ y \# ys) = (sorted(xs @ [y]) \wedge sorted(y \# ys))$
by(*rule sorted_mid_iff*)

lemmas *sorted_lems = sorted_mid_iff' sorted_mid_iff2 sorted_cons' sorted_snoc'*

Splay trees need two additional *sorted* lemmas:

lemma *sorted_snoc_le*:
 $ASSUMPTION(sorted(xs @ [x])) \implies x \leq y \implies sorted(xs @ [y])$
by (*auto simp add: sorted_wrt_append ASSUMPTION_def*)

lemma *sorted_Cons_le*:
 $ASSUMPTION(sorted(x \# xs)) \implies y \leq x \implies sorted(y \# xs)$
by (*auto simp add: sorted_wrt_Cons ASSUMPTION_def*)

end

5 List Insertion and Deletion

theory *List_Ins_Del*
imports *Sorted_Less*
begin

5.1 Elements in a list

lemma *sorted_Cons_iff*:
 $sorted(x \# xs) = ((\forall y \in set xs. x < y) \wedge sorted xs)$
by(*simp add: sorted_wrt_Cons*)

lemma *sorted_snoc_iff*:
 $sorted(xs @ [x]) = (sorted xs \wedge (\forall y \in set xs. y < x))$
by(*simp add: sorted_wrt_append*)

lemmas *isin_simps = sorted_mid_iff' sorted_Cons_iff sorted_snoc_iff*

5.2 Inserting into an ordered list without duplicates:

fun *ins_list* :: 'a::linorder \Rightarrow 'a list \Rightarrow 'a list **where**
ins_list x [] = [x] |
ins_list x (a#xs) =
 (if x < a then x#a#xs else if x=a then a#xs else a # *ins_list* x xs)

lemma *set_ins_list*: set (*ins_list* x xs) = set xs \cup {x}
by(*induction xs*) *auto*

lemma *sorted_ins_list*: sorted xs \Longrightarrow sorted(*ins_list* x xs)
by(*induction xs rule: induct_list012*) *auto*

lemma *ins_list_sorted*: sorted (xs @ [a]) \Longrightarrow
ins_list x (xs @ a # ys) =
 (if x < a then *ins_list* x xs @ (a#ys) else xs @ *ins_list* x (a#ys))
by(*induction xs*) (*auto simp: sorted_lems*)

In principle, sorted (?xs @ [?a]) \Longrightarrow *ins_list* ?x (?xs @ ?a # ?ys) = (if ?x < ?a then *ins_list* ?x ?xs @ ?a # ?ys else ?xs @ *ins_list* ?x (?a # ?ys)) suffices, but the following two corollaries speed up proofs.

corollary *ins_list_sorted1*: sorted (xs @ [a]) \Longrightarrow a \leq x \Longrightarrow
ins_list x (xs @ a # ys) = xs @ *ins_list* x (a#ys)
by(*auto simp add: ins_list_sorted*)

corollary *ins_list_sorted2*: sorted (xs @ [a]) \Longrightarrow x < a \Longrightarrow
ins_list x (xs @ a # ys) = *ins_list* x xs @ (a#ys)
by(*auto simp: ins_list_sorted*)

lemmas *ins_list_simps* = sorted_lems *ins_list_sorted1 ins_list_sorted2*

Splay trees need two additional *ins_list* lemmas:

lemma *ins_list_Cons*: sorted (x # xs) \Longrightarrow *ins_list* x xs = x # xs
by (*induction xs*) *auto*

lemma *ins_list_snoc*: sorted (xs @ [x]) \Longrightarrow *ins_list* x xs = xs @ [x]
by(*induction xs*) (*auto simp add: sorted_mid_iff2*)

5.3 Delete one occurrence of an element from a list:

fun *del_list* :: 'a \Rightarrow 'a list \Rightarrow 'a list **where**
del_list x [] = [] |
del_list x (a#xs) = (if x=a then xs else a # *del_list* x xs)

lemma *del_list_idem*: x \notin set xs \Longrightarrow *del_list* x xs = xs

by (*induct xs simp_all*)

lemma *set_del_list*:

$sorted\ xs \implies set\ (del_list\ x\ xs) = set\ xs - \{x\}$

by(*induct xs (auto simp: sorted_Cons_iff)*)

lemma *sorted_del_list*: $sorted\ xs \implies sorted\ (del_list\ x\ xs)$

apply(*induction xs rule: induct_list012*)

apply *auto*

by (*meson order.strict_trans sorted_Cons_iff*)

lemma *del_list_sorted*: $sorted\ (xs\ @\ a\ \#\ ys) \implies$

$del_list\ x\ (xs\ @\ a\ \#\ ys) = (if\ x < a\ then\ del_list\ x\ xs\ @\ a\ \#\ ys\ else\ xs$
 $@\ del_list\ x\ (a\ \#\ ys))$

by(*induction xs*)

(*fastforce simp: sorted_lems sorted_Cons_iff intro!: del_list_idem*)+

In principle, $sorted\ (?xs\ @\ ?a\ \#\ ?ys) \implies del_list\ ?x\ (?xs\ @\ ?a\ \#\ ?ys)$
 $= (if\ ?x < ?a\ then\ del_list\ ?x\ ?xs\ @\ ?a\ \#\ ?ys\ else\ ?xs\ @\ del_list\ ?x\ (?a$
 $\#\ ?ys))$ suffices, but the following corollaries speed up proofs.

corollary *del_list_sorted1*: $sorted\ (xs\ @\ a\ \#\ ys) \implies a \leq x \implies$

$del_list\ x\ (xs\ @\ a\ \#\ ys) = xs\ @\ del_list\ x\ (a\ \#\ ys)$

by (*auto simp: del_list_sorted*)

corollary *del_list_sorted2*: $sorted\ (xs\ @\ a\ \#\ ys) \implies x < a \implies$

$del_list\ x\ (xs\ @\ a\ \#\ ys) = del_list\ x\ xs\ @\ a\ \#\ ys$

by (*auto simp: del_list_sorted*)

corollary *del_list_sorted3*:

$sorted\ (xs\ @\ a\ \#\ ys\ @\ b\ \#\ zs) \implies x < b \implies$

$del_list\ x\ (xs\ @\ a\ \#\ ys\ @\ b\ \#\ zs) = del_list\ x\ (xs\ @\ a\ \#\ ys)\ @\ b\ \#\ zs$

by (*auto simp: del_list_sorted sorted_lems*)

corollary *del_list_sorted4*:

$sorted\ (xs\ @\ a\ \#\ ys\ @\ b\ \#\ zs\ @\ c\ \#\ us) \implies x < c \implies$

$del_list\ x\ (xs\ @\ a\ \#\ ys\ @\ b\ \#\ zs\ @\ c\ \#\ us) = del_list\ x\ (xs\ @\ a\ \#\ ys\ @$
 $b\ \#\ zs)\ @\ c\ \#\ us$

by (*auto simp: del_list_sorted sorted_lems*)

corollary *del_list_sorted5*:

$sorted\ (xs\ @\ a\ \#\ ys\ @\ b\ \#\ zs\ @\ c\ \#\ us\ @\ d\ \#\ vs) \implies x < d \implies$

$del_list\ x\ (xs\ @\ a\ \#\ ys\ @\ b\ \#\ zs\ @\ c\ \#\ us\ @\ d\ \#\ vs) =$

$del_list\ x\ (xs\ @\ a\ \#\ ys\ @\ b\ \#\ zs\ @\ c\ \#\ us)\ @\ d\ \#\ vs$

by (*auto simp: del_list_sorted sorted_lems*)

lemmas *del_list_simps = sorted_lems*

del_list_sorted1

del_list_sorted2

del_list_sorted3

del_list_sorted4

del_list_sorted5

Splay trees need two additional *del_list* lemmas:

lemma *del_list_notin_Cons: sorted (x # xs) \implies del_list x xs = xs*

by(*induction xs*)(*fastforce simp: sorted_Cons_iff*)+

lemma *del_list_sorted_app:*

sorted(xs @ [x]) \implies del_list x (xs @ ys) = xs @ del_list x ys

by (*induction xs*) (*auto simp: sorted_mid_iff2*)

end

6 Specifications of Set ADT

theory *Set_Specs*

imports *List_Ins_Del*

begin

The basic set interface with traditional *set*-based specification:

locale *Set =*

fixes *empty :: 's*

fixes *insert :: 'a \Rightarrow 's \Rightarrow 's*

fixes *delete :: 'a \Rightarrow 's \Rightarrow 's*

fixes *isin :: 's \Rightarrow 'a \Rightarrow bool*

fixes *set :: 's \Rightarrow 'a set*

fixes *invar :: 's \Rightarrow bool*

assumes *set_empty: set empty = {}*

assumes *set_isin: invar s \implies isin s x = (x \in set s)*

assumes *set_insert: invar s \implies set(insert x s) = set s \cup {x}*

assumes *set_delete: invar s \implies set(delete x s) = set s - {x}*

assumes *invar_empty: invar empty*

assumes *invar_insert: invar s \implies invar(insert x s)*

assumes *invar_delete: invar s \implies invar(delete x s)*

lemmas (**in** *Set*) *set_specs =*

set_empty set_isin set_insert set_delete invar_empty invar_insert invar_delete

The basic set interface with *inorder*-based specification:

```

locale Set_by_Ordered =
fixes empty :: 't
fixes insert :: 'a::linorder ⇒ 't ⇒ 't
fixes delete :: 'a ⇒ 't ⇒ 't
fixes isin :: 't ⇒ 'a ⇒ bool
fixes inorder :: 't ⇒ 'a list
fixes inv :: 't ⇒ bool
assumes inorder_empty: inorder empty = []
assumes isin: inv t ∧ sorted(inorder t) ⇒
  isin t x = (x ∈ set (inorder t))
assumes inorder_insert: inv t ∧ sorted(inorder t) ⇒
  inorder(insert x t) = ins_list x (inorder t)
assumes inorder_delete: inv t ∧ sorted(inorder t) ⇒
  inorder(delete x t) = del_list x (inorder t)
assumes inorder_inv_empty: inv empty
assumes inorder_inv_insert: inv t ∧ sorted(inorder t) ⇒ inv(insert x t)
assumes inorder_inv_delete: inv t ∧ sorted(inorder t) ⇒ inv(delete x t)

```

begin

It implements the traditional specification:

```

definition set :: 't ⇒ 'a set where
set = List.set o inorder

```

```

definition invar :: 't ⇒ bool where
invar t = (inv t ∧ sorted (inorder t))

```

sublocale Set

```

  empty insert delete isin set invar
proof(standard, goal_cases)
  case 1 show ?case by (auto simp: inorder_empty set_def)
next
  case 2 thus ?case by(simp add: isin invar_def set_def)
next
  case 3 thus ?case by(simp add: inorder_insert set_ins_list set_def in-
var_def)
next
  case (4 s x) thus ?case
  by (auto simp: inorder_delete set_del_list invar_def set_def)
next
  case 5 thus ?case by(simp add: inorder_empty inorder_inv_empty in-
var_def)
next
  case 6 thus ?case by(simp add: inorder_insert inorder_inv_insert sorted_ins_list

```

```

invar_def)
next
  case 7 thus ?case by (auto simp: inorder_delete inorder_inv_delete
sorted_del_list invar_def)
qed

end

Set2 = Set with binary operations:

locale Set2 = Set
  where insert = insert for insert :: 'a ⇒ 's ⇒ 's +
  fixes union :: 's ⇒ 's ⇒ 's
  fixes inter :: 's ⇒ 's ⇒ 's
  fixes diff :: 's ⇒ 's ⇒ 's
  assumes set_union:  [ invar s1; invar s2 ] ⇒ set(union s1 s2) = set s1
  ∪ set s2
  assumes set_inter:  [ invar s1; invar s2 ] ⇒ set(inter s1 s2) = set s1
  ∩ set s2
  assumes set_diff:   [ invar s1; invar s2 ] ⇒ set(diff s1 s2) = set s1 -
  set s2
  assumes invar_union: [ invar s1; invar s2 ] ⇒ invar(union s1 s2)
  assumes invar_inter: [ invar s1; invar s2 ] ⇒ invar(inter s1 s2)
  assumes invar_diff:  [ invar s1; invar s2 ] ⇒ invar(diff s1 s2)

end

```

7 Unbalanced Tree Implementation of Set

```

theory Tree_Set
imports
  HOL-Library.Tree
  Cmp
  Set_Specs
begin

definition empty :: 'a tree where
  empty = Leaf

fun isin :: 'a::linorder tree ⇒ 'a ⇒ bool where
  isin Leaf x = False |
  isin (Node l a r) x =
    (case cmp x a of
      LT ⇒ isin l x |
      EQ ⇒ True |

```

$GT \Rightarrow \text{isin } r \ x$)

hide_const (open) insert

fun insert :: 'a::linorder \Rightarrow 'a tree \Rightarrow 'a tree **where**

insert x Leaf = Node Leaf x Leaf |

insert x (Node l a r) =

(case cmp x a of

LT \Rightarrow Node (insert x l) a r |

EQ \Rightarrow Node l a r |

GT \Rightarrow Node l a (insert x r))

Deletion by replacing:

fun split_min :: 'a tree \Rightarrow 'a * 'a tree **where**

split_min (Node l a r) =

(if l = Leaf then (a,r) else let (x,l') = split_min l in (x, Node l' a r))

fun delete :: 'a::linorder \Rightarrow 'a tree \Rightarrow 'a tree **where**

delete x Leaf = Leaf |

delete x (Node l a r) =

(case cmp x a of

LT \Rightarrow Node (delete x l) a r |

GT \Rightarrow Node l a (delete x r) |

EQ \Rightarrow if r = Leaf then l else let (a',r') = split_min r in Node l a' r')

Deletion by joining:

fun join :: ('a::linorder)tree \Rightarrow 'a tree \Rightarrow 'a tree **where**

join t Leaf = t |

join Leaf t = t |

join (Node t1 a t2) (Node t3 b t4) =

(case join t2 t3 of

Leaf \Rightarrow Node t1 a (Node Leaf b t4) |

Node u2 x u3 \Rightarrow Node (Node t1 a u2) x (Node u3 b t4))

fun delete2 :: 'a::linorder \Rightarrow 'a tree \Rightarrow 'a tree **where**

delete2 x Leaf = Leaf |

delete2 x (Node l a r) =

(case cmp x a of

LT \Rightarrow Node (delete2 x l) a r |

GT \Rightarrow Node l a (delete2 x r) |

EQ \Rightarrow join l r)

7.1 Functional Correctness Proofs

lemma isin_set: sorted(inorder t) \Longrightarrow isin t x = (x \in set (inorder t))

by (*induction t*) (*auto simp: isin_simps*)

lemma *inorder_insert*:

$sorted(inorder\ t) \implies inorder(insert\ x\ t) = ins_list\ x\ (inorder\ t)$

by(*induction t*) (*auto simp: ins_list_simps*)

lemma *split_minD*:

$split_min\ t = (x, t') \implies t \neq Leaf \implies x \# inorder\ t' = inorder\ t$

by(*induction t arbitrary: t' rule: split_min.induct*)

(*auto simp: sorted_lems split: prod.splits if_splits*)

lemma *inorder_delete*:

$sorted(inorder\ t) \implies inorder(delete\ x\ t) = del_list\ x\ (inorder\ t)$

by(*induction t*) (*auto simp: del_list_simps split_minD split: prod.splits*)

interpretation *S*: *Set_by_Ordered*

where *empty* = *empty* **and** *isin* = *isin* **and** *insert* = *insert* **and** *delete* = *delete*

and *inorder* = *inorder* **and** *inv* = $\lambda_.$ *True*

proof (*standard, goal_cases*)

case 1 show ?case by (*simp add: empty_def*)

next

case 2 thus ?case by(*simp add: isin_set*)

next

case 3 thus ?case by(*simp add: inorder_insert*)

next

case 4 thus ?case by(*simp add: inorder_delete*)

qed (*rule TrueI*)+

lemma *inorder_join*:

$inorder(join\ l\ r) = inorder\ l\ @\ inorder\ r$

by(*induction l r rule: join.induct*) (*auto split: tree.split*)

lemma *inorder_delete2*:

$sorted(inorder\ t) \implies inorder(delete2\ x\ t) = del_list\ x\ (inorder\ t)$

by(*induction t*) (*auto simp: inorder_join del_list_simps*)

interpretation *S2*: *Set_by_Ordered*

where *empty* = *empty* **and** *isin* = *isin* **and** *insert* = *insert* **and** *delete* = *delete2*

and *inorder* = *inorder* **and** *inv* = $\lambda_.$ *True*

proof (*standard, goal_cases*)

case 1 show ?case by (*simp add: empty_def*)

```

next
  case 2 thus ?case by(simp add: isin_set)
next
  case 3 thus ?case by(simp add: inorder_insert)
next
  case 4 thus ?case by(simp add: inorder_delete2)
qed (rule TrueI)+

end

```

8 Association List Update and Deletion

```

theory AList_Upd_Del
imports Sorted_Less
begin

```

abbreviation $sorted1\ ps \equiv sorted(map\ fst\ ps)$

Define own map_of function to avoid pulling in an unknown amount of lemmas implicitly (via the simpset).

hide_const (open) map_of

```

fun map_of :: ('a*'b)list  $\Rightarrow$  'a  $\Rightarrow$  'b option where
map_of [] = ( $\lambda x.$  None) |
map_of ((a,b)#ps) = ( $\lambda x.$  if x=a then Some b else map_of ps x)

```

Updating an association list:

```

fun upd_list :: 'a::linorder  $\Rightarrow$  'b  $\Rightarrow$  ('a*'b) list  $\Rightarrow$  ('a*'b) list where
upd_list x y [] = [(x,y)] |
upd_list x y ((a,b)#ps) =
  (if x < a then (x,y)#(a,b)#ps else
   if x = a then (x,y)#ps else (a,b) # upd_list x y ps)

```

```

fun del_list :: 'a::linorder  $\Rightarrow$  ('a*'b)list  $\Rightarrow$  ('a*'b)list where
del_list x [] = [] |
del_list x ((a,b)#ps) = (if x = a then ps else (a,b) # del_list x ps)

```

8.1 Lemmas for map_of

```

lemma map_of_ins_list: map_of (upd_list x y ps) = (map_of ps)(x := Some y)
by(induction ps) auto

```

```

lemma map_of_append: map_of (ps @ qs) x =

```

(case map_of ps x of None \Rightarrow map_of qs x | Some y \Rightarrow Some y)
by(induction ps)(auto)

lemma map_of_None: sorted (x # map fst ps) \Longrightarrow map_of ps x = None
by (induction ps) (fastforce simp: sorted_lems sorted_wrt_Cons)+

lemma map_of_None2: sorted (map fst ps @ [x]) \Longrightarrow map_of ps x = None
by (induction ps) (auto simp: sorted_lems)

lemma map_of_del_list: sorted1 ps \Longrightarrow
 map_of (del_list x ps) = (map_of ps)(x := None)
by(induction ps) (auto simp: map_of_None sorted_lems fun_eq_iff)

lemma map_of_sorted_Cons: sorted (a # map fst ps) \Longrightarrow x < a \Longrightarrow
 map_of ps x = None
by (simp add: map_of_None sorted_Cons_le)

lemma map_of_sorted_snoc: sorted (map fst ps @ [a]) \Longrightarrow a \leq x \Longrightarrow
 map_of ps x = None
by (simp add: map_of_None2 sorted_snoc_le)

lemmas map_of_sorteds = map_of_sorted_Cons map_of_sorted_snoc
lemmas map_of_simps = sorted_lems map_of_append map_of_sorteds

8.2 Lemmas for upd_list

lemma sorted_upd_list: sorted1 ps \Longrightarrow sorted1 (upd_list x y ps)
apply(induction ps)
apply simp
apply(case_tac ps)
apply auto
done

lemma upd_list_sorted: sorted1 (ps @ [(a,b)]) \Longrightarrow
 upd_list x y (ps @ (a,b) # qs) =
 (if x < a then upd_list x y ps @ (a,b) # qs
 else ps @ upd_list x y ((a,b) # qs))
by(induction ps) (auto simp: sorted_lems)

In principle, sorted1 (?ps @ [(?a, ?b)]) \Longrightarrow upd_list ?x ?y (?ps @ (?a, ?b) # ?qs) = (if ?x < ?a then upd_list ?x ?y ?ps @ (?a, ?b) # ?qs else ?ps @ upd_list ?x ?y ((?a, ?b) # ?qs)) suffices, but the following two corollaries speed up proofs.

corollary *upd_list_sorted1*: $\llbracket \text{sorted } (\text{map } \text{fst } ps \text{ @ } [a]); x < a \rrbracket \implies$
 $\text{upd_list } x \ y \ (ps \text{ @ } (a,b) \# qs) = \text{upd_list } x \ y \ ps \text{ @ } (a,b) \# qs$
by (*auto simp: upd_list_sorted*)

corollary *upd_list_sorted2*: $\llbracket \text{sorted } (\text{map } \text{fst } ps \text{ @ } [a]); a \leq x \rrbracket \implies$
 $\text{upd_list } x \ y \ (ps \text{ @ } (a,b) \# qs) = ps \text{ @ } \text{upd_list } x \ y \ ((a,b) \# qs)$
by (*auto simp: upd_list_sorted*)

lemmas *upd_list_simps* = *sorted_lems upd_list_sorted1 upd_list_sorted2*

Splay trees need two additional *upd_list* lemmas:

lemma *upd_list_Cons*:
 $\text{sorted1 } ((x,y) \# xs) \implies \text{upd_list } x \ y \ xs = (x,y) \# xs$
by (*induction xs*) *auto*

lemma *upd_list_snoc*:
 $\text{sorted1 } (xs \text{ @ } [(x,y)]) \implies \text{upd_list } x \ y \ xs = xs \text{ @ } [(x,y)]$
by(*induction xs*) (*auto simp add: sorted_mid_iff2*)

8.3 Lemmas for *del_list*

lemma *sorted_del_list*: $\text{sorted1 } ps \implies \text{sorted1 } (\text{del_list } x \ ps)$
apply(*induction ps*)
apply *simp*
apply(*case_tac ps*)
apply (*auto simp: sorted_Cons_le*)
done

lemma *del_list_idem*: $x \notin \text{set}(\text{map } \text{fst } xs) \implies \text{del_list } x \ xs = xs$
by (*induct xs*) *auto*

lemma *del_list_sorted*: $\text{sorted1 } (ps \text{ @ } (a,b) \# qs) \implies$
 $\text{del_list } x \ (ps \text{ @ } (a,b) \# qs) =$
 $(\text{if } x < a \text{ then } \text{del_list } x \ ps \text{ @ } (a,b) \# qs$
 $\text{else } ps \text{ @ } \text{del_list } x \ ((a,b) \# qs))$
by(*induction ps*)
(fastforce simp: sorted_lems sorted_wrt_Cons intro!: del_list_idem)+

In principle, $\text{sorted1 } (?ps \text{ @ } (?a, ?b) \# ?qs) \implies \text{del_list } ?x \ (?ps \text{ @ } (?a, ?b) \# ?qs) = (\text{if } ?x < ?a \text{ then } \text{del_list } ?x \ ?ps \text{ @ } (?a, ?b) \# ?qs \text{ else } ?ps \text{ @ } \text{del_list } ?x \ ((?a, ?b) \# ?qs))$ suffices, but the following corollaries speed up proofs.

corollary *del_list_sorted1*: $\text{sorted1 } (xs \text{ @ } (a,b) \# ys) \implies a \leq x \implies$
 $\text{del_list } x \ (xs \text{ @ } (a,b) \# ys) = xs \text{ @ } \text{del_list } x \ ((a,b) \# ys)$

by (*auto simp: del_list_sorted*)

lemma *del_list_sorted2*: $\text{sorted1 } (xs @ (a,b) \# ys) \implies x < a \implies$
 $\text{del_list } x (xs @ (a,b) \# ys) = \text{del_list } x xs @ (a,b) \# ys$

by (*auto simp: del_list_sorted*)

lemma *del_list_sorted3*:

$\text{sorted1 } (xs @ (a,a') \# ys @ (b,b') \# zs) \implies x < b \implies$
 $\text{del_list } x (xs @ (a,a') \# ys @ (b,b') \# zs) = \text{del_list } x (xs @ (a,a') \#$
 $ys) @ (b,b') \# zs$

by (*auto simp: del_list_sorted sorted_lems*)

lemma *del_list_sorted4*:

$\text{sorted1 } (xs @ (a,a') \# ys @ (b,b') \# zs @ (c,c') \# us) \implies x < c \implies$
 $\text{del_list } x (xs @ (a,a') \# ys @ (b,b') \# zs @ (c,c') \# us) = \text{del_list } x (xs$
 $@ (a,a') \# ys @ (b,b') \# zs) @ (c,c') \# us$

by (*auto simp: del_list_sorted sorted_lems*)

lemma *del_list_sorted5*:

$\text{sorted1 } (xs @ (a,a') \# ys @ (b,b') \# zs @ (c,c') \# us @ (d,d') \# vs) \implies$
 $x < d \implies$
 $\text{del_list } x (xs @ (a,a') \# ys @ (b,b') \# zs @ (c,c') \# us @ (d,d') \# vs)$
 $=$

$\text{del_list } x (xs @ (a,a') \# ys @ (b,b') \# zs @ (c,c') \# us) @ (d,d') \# vs$

by (*auto simp: del_list_sorted sorted_lems*)

lemmas *del_list_simps = sorted_lems*

del_list_sorted1

del_list_sorted2

del_list_sorted3

del_list_sorted4

del_list_sorted5

Splay trees need two additional *del_list* lemmas:

lemma *del_list_notin_Cons*: $\text{sorted } (x \# \text{map } \text{fst } xs) \implies \text{del_list } x xs =$
 xs

by(*induction xs*)(*fastforce simp: sorted_wrt_Cons*)+

lemma *del_list_sorted_app*:

$\text{sorted}(\text{map } \text{fst } xs @ [x]) \implies \text{del_list } x (xs @ ys) = xs @ \text{del_list } x ys$
by (*induction xs*) (*auto simp: sorted_mid_iff2*)

end

9 Specifications of Map ADT

```
theory Map_Specs
imports AList_Upd_Del
begin
```

The basic map interface with traditional *set*-based specification:

```
locale Map =
fixes empty :: 'm
fixes update :: 'a ⇒ 'b ⇒ 'm ⇒ 'm
fixes delete :: 'a ⇒ 'm ⇒ 'm
fixes lookup :: 'm ⇒ 'a ⇒ 'b option
fixes invar :: 'm ⇒ bool
assumes map_empty: lookup empty = (λ_. None)
and map_update: invar m ⇒ lookup(update a b m) = (lookup m)(a :=
Some b)
and map_delete: invar m ⇒ lookup(delete a m) = (lookup m)(a := None)
and invar_empty: invar empty
and invar_update: invar m ⇒ invar(update a b m)
and invar_delete: invar m ⇒ invar(delete a m)
```

```
lemmas (in Map) map_specs =
  map_empty map_update map_delete invar_empty invar_update invar_delete
```

The basic map interface with *inorder*-based specification:

```
locale Map_by_Ordered =
fixes empty :: 't
fixes update :: 'a::linorder ⇒ 'b ⇒ 't ⇒ 't
fixes delete :: 'a ⇒ 't ⇒ 't
fixes lookup :: 't ⇒ 'a ⇒ 'b option
fixes inorder :: 't ⇒ ('a * 'b) list
fixes inv :: 't ⇒ bool
assumes inorder_empty: inorder empty = []
and inorder_lookup: inv t ∧ sorted1 (inorder t) ⇒
  lookup t a = map_of (inorder t) a
and inorder_update: inv t ∧ sorted1 (inorder t) ⇒
  inorder(update a b t) = upd_list a b (inorder t)
and inorder_delete: inv t ∧ sorted1 (inorder t) ⇒
  inorder(delete a t) = del_list a (inorder t)
and inorder_inv_empty: inv empty
and inorder_inv_update: inv t ∧ sorted1 (inorder t) ⇒ inv(update a b t)
and inorder_inv_delete: inv t ∧ sorted1 (inorder t) ⇒ inv(delete a t)

begin
```

It implements the traditional specification:

```

definition invar :: 't ⇒ bool where
invar t == inv t ∧ sorted1 (inorder t)

sublocale Map
  empty update delete lookup invar
proof(standard, goal_cases)
  case 1 show ?case by (auto simp: inorder_lookup inorder_empty in-
order_inv_empty)
next
  case 2 thus ?case
  by(simp add: fun_eq_iff inorder_update inorder_inv_update map_of_ins_list
inorder_lookup
  sorted_upd_list invar_def)
next
  case 3 thus ?case
  by(simp add: fun_eq_iff inorder_delete inorder_inv_delete map_of_del_list
inorder_lookup
  sorted_del_list invar_def)
next
  case 4 thus ?case by(simp add: inorder_empty inorder_inv_empty in-
var_def)
next
  case 5 thus ?case by(simp add: inorder_update inorder_inv_update
sorted_upd_list invar_def)
next
  case 6 thus ?case by (auto simp: inorder_delete inorder_inv_delete
sorted_del_list invar_def)
qed

end

end

```

10 Unbalanced Tree Implementation of Map

```

theory Tree_Map
imports
  Tree_Set
  Map_Specs
begin

fun lookup :: ('a::linorder*'b) tree ⇒ 'a ⇒ 'b option where

```

```

lookup Leaf x = None |
lookup (Node l (a,b) r) x =
  (case cmp x a of LT => lookup l x | GT => lookup r x | EQ => Some b)

```

```

fun update :: 'a::linorder => 'b => ('a*'b) tree => ('a*'b) tree where
update x y Leaf = Node Leaf (x,y) Leaf |
update x y (Node l (a,b) r) = (case cmp x a of
  LT => Node (update x y l) (a,b) r |
  EQ => Node l (x,y) r |
  GT => Node l (a,b) (update x y r))

```

```

fun delete :: 'a::linorder => ('a*'b) tree => ('a*'b) tree where
delete x Leaf = Leaf |
delete x (Node l (a,b) r) = (case cmp x a of
  LT => Node (delete x l) (a,b) r |
  GT => Node l (a,b) (delete x r) |
  EQ => if r = Leaf then l else let (ab',r') = split_min r in Node l ab' r')

```

10.1 Functional Correctness Proofs

```

lemma lookup_map_of:
  sorted1(inorder t) ==> lookup t x = map_of (inorder t) x
by (induction t) (auto simp: map_of_simps split: option.split)

```

```

lemma inorder_update:
  sorted1(inorder t) ==> inorder(update a b t) = upd_list a b (inorder t)
by(induction t) (auto simp: upd_list_simps)

```

```

lemma inorder_delete:
  sorted1(inorder t) ==> inorder(delete x t) = del_list x (inorder t)
by(induction t) (auto simp: del_list_simps split_minD split: prod.splits)

```

```

interpretation M: Map_by_Ordered
where empty = empty and lookup = lookup and update = update and
delete = delete
and inorder = inorder and inv =  $\lambda\_.$  True
proof (standard, goal_cases)
  case 1 show ?case by (simp add: empty_def)
next
  case 2 thus ?case by(simp add: lookup_map_of)
next
  case 3 thus ?case by(simp add: inorder_update)
next
  case 4 thus ?case by(simp add: inorder_delete)

```

qed *auto*

end

11 Augmented Tree (Tree2)

theory *Tree2*

imports *HOL-Library.Tree*

begin

This theory provides the basic infrastructure for the type $('a \times 'b)$ *tree* of augmented trees where $'a$ is the key and $'b$ some additional information.

IMPORTANT: Inductions and cases analyses on augmented trees need to use the following two rules explicitly. They generate nodes of the form $\langle l, (a, b), r \rangle$ rather than $\langle l, a, r \rangle$ for trees of type $'a$ *tree*.

lemmas *tree2_induct* = *tree.induct*[**where** $'a = 'a * 'b$, *split_format(complete)*]

lemmas *tree2_cases* = *tree.exhaust*[**where** $'a = 'a * 'b$, *split_format(complete)*]

fun *inorder* :: $('a * 'b)$ *tree* \Rightarrow $'a$ *list* **where**

inorder *Leaf* = [] |

inorder (*Node* l ($a, _$) r) = *inorder* l @ a # *inorder* r

fun *set_tree* :: $('a * 'b)$ *tree* \Rightarrow $'a$ *set* **where**

set_tree *Leaf* = {} |

set_tree (*Node* l ($a, _$) r) = { a } \cup *set_tree* l \cup *set_tree* r

fun *bst* :: $('a::linorder * 'b)$ *tree* \Rightarrow *bool* **where**

bst *Leaf* = *True* |

bst (*Node* l ($a, _$) r) = $((\forall x \in \text{set_tree } l. x < a) \wedge (\forall x \in \text{set_tree } r. a < x)) \wedge \text{bst } l \wedge \text{bst } r$

lemma *finite_set_tree*[*simp*]: *finite*(*set_tree* t)

by (*induction* t) *auto*

lemma *eq_set_tree_empty*[*simp*]: *set_tree* $t = \{\}$ \longleftrightarrow $t = \text{Leaf}$

by (*cases* t) *auto*

lemma *set_inorder*[*simp*]: *set* (*inorder* t) = *set_tree* t

by (*induction* t) *auto*

lemma *length_inorder*[*simp*]: *length* (*inorder* t) = *size* t

by (*induction* t) *auto*

end

12 Function *isin* for Tree2

theory *Isin2*

imports

Tree2

Cmp

Set_Specs

begin

fun *isin* :: ('a::linorder*'b) tree \Rightarrow 'a \Rightarrow bool **where**

isin Leaf $x = \text{False}$ |

isin (Node l (a,_) r) $x =$

(case *cmp* x a of

LT \Rightarrow *isin* l x |

EQ \Rightarrow True |

GT \Rightarrow *isin* r x)

lemma *isin_set_inorder*: sorted(*inorder* t) \Longrightarrow *isin* t $x = (x \in \text{set}(\text{inorder}$
 $t))$

by (*induction* t rule: *tree2_induct*) (*auto simp: isin_simps*)

lemma *isin_set_tree*: *bst* $t \Longrightarrow$ *isin* t $x \longleftrightarrow x \in \text{set_tree } t$

by(*induction* t rule: *tree2_induct*) *auto*

end

13 Interval Trees

theory *Interval_Tree*

imports

HOL-Data_Structures.Cmp

HOL-Data_Structures.List_Ins_Del

HOL-Data_Structures.Isin2

HOL-Data_Structures.Set_Specs

begin

13.1 Intervals

The following definition of intervals uses the **typedef** command to define the type of non-empty intervals as a subset of the type of pairs p where *fst*

$p \leq \text{snd } p$:

typedef (overloaded) $'a::\text{linorder } \text{ivl} =$
 $\{p :: 'a \times 'a. \text{fst } p \leq \text{snd } p\}$ **by** *auto*

More precisely, $'a \text{ ivl}$ is isomorphic with that subset via the function Rep_ivl . Hence the basic interval properties are not immediate but need simple proofs:

definition $\text{low} :: 'a::\text{linorder } \text{ivl} \Rightarrow 'a$ **where**
 $\text{low } p = \text{fst } (\text{Rep_ivl } p)$

definition $\text{high} :: 'a::\text{linorder } \text{ivl} \Rightarrow 'a$ **where**
 $\text{high } p = \text{snd } (\text{Rep_ivl } p)$

lemma ivl_is_interval : $\text{low } p \leq \text{high } p$
by (*metis Rep_ivl high_def low_def mem_Collect_eq*)

lemma ivl_inj : $\text{low } p = \text{low } q \implies \text{high } p = \text{high } q \implies p = q$
by (*metis Rep_ivl_inverse high_def low_def prod_eqI*)

Now we can forget how exactly intervals were defined.

instantiation $\text{ivl} :: (\text{linorder}) \text{ linorder}$ **begin**

definition ivl_less : $(x < y) = (\text{low } x < \text{low } y \mid (\text{low } x = \text{low } y \wedge \text{high } x < \text{high } y))$

definition ivl_less_eq : $(x \leq y) = (\text{low } x < \text{low } y \mid (\text{low } x = \text{low } y \wedge \text{high } x \leq \text{high } y))$

instance proof

fix $x y z :: 'a \text{ ivl}$

show a : $(x < y) = (x \leq y \wedge \neg y \leq x)$

using $\text{ivl_less } \text{ivl_less_eq}$ **by** *force*

show b : $x \leq x$

by (*simp add: ivl_less_eq*)

show c : $x \leq y \implies y \leq z \implies x \leq z$

using ivl_less_eq **by** *fastforce*

show d : $x \leq y \implies y \leq x \implies x = y$

using $\text{ivl_less_eq } a \text{ ivl_inj } \text{ivl_less}$ **by** *fastforce*

show e : $x \leq y \vee y \leq x$

by (*meson ivl_less_eq leI not_less_iff_gr_or_eq*)

qed end

definition $\text{overlap} :: ('a::\text{linorder}) \text{ ivl} \Rightarrow 'a \text{ ivl} \Rightarrow \text{bool}$ **where**
 $\text{overlap } x y \longleftrightarrow (\text{high } x \geq \text{low } y \wedge \text{high } y \geq \text{low } x)$

definition *has_overlap* :: ('a::linorder) ivl set \Rightarrow 'a ivl \Rightarrow bool **where**
has_overlap S y \longleftrightarrow ($\exists x \in S. \text{overlap } x \ y$)

13.2 Interval Trees

type_synonym 'a ivl_tree = ('a ivl * 'a) tree

fun *max_hi* :: ('a::order_bot) ivl_tree \Rightarrow 'a **where**
max_hi Leaf = bot |
max_hi (Node _ (_,m) _) = m

definition *max3* :: ('a::linorder) ivl \Rightarrow 'a \Rightarrow 'a \Rightarrow 'a **where**
max3 a m n = max (high a) (max m n)

fun *inv_max_hi* :: ('a::{linorder,order_bot}) ivl_tree \Rightarrow bool **where**
inv_max_hi Leaf \longleftrightarrow True |
inv_max_hi (Node l (a, m) r) \longleftrightarrow (m = max3 a (max_hi l) (max_hi r))
 \wedge *inv_max_hi* l \wedge *inv_max_hi* r

lemma *max_hi_is_max*:
inv_max_hi t \Longrightarrow a \in set_tree t \Longrightarrow high a \leq max_hi t
by (induct t, auto simp add: max3_def max_def)

lemma *max_hi_exists*:
inv_max_hi t \Longrightarrow t \neq Leaf \Longrightarrow $\exists a \in \text{set_tree } t. \text{high } a = \text{max_hi } t$
proof (induction t rule: tree2_induct)
 case Leaf
 then show ?case **by** auto
next
 case N: (Node l v m r)
 then show ?case
 proof (cases l rule: tree2_cases)
 case Leaf
 then show ?thesis
 using N.prem1(1) N.IH(2) **by** (cases r, auto simp add: max3_def
max_def le_bot)
 next
 case Nl: Node
 then show ?thesis
 proof (cases r rule: tree2_cases)
 case Leaf
 then show ?thesis
 using N.prem1(1) N.IH(1) Nl **by** (auto simp add: max3_def max_def)

```

le_bot)
  next
  case Nr: Node
  obtain p1 where p1: p1 ∈ set_tree l high p1 = max_hi l
  using N.IH(1) N.prem(1) Nl by auto
  obtain p2 where p2: p2 ∈ set_tree r high p2 = max_hi r
  using N.IH(2) N.prem(1) Nr by auto
  then show ?thesis
  using p1 p2 N.prem(1) by (auto simp add: max3_def max_def)
qed
qed
qed

```

13.3 Insertion and Deletion

definition *node* where

[simp]: $\text{node } l \ a \ r = \text{Node } l \ (a, \max3 \ a \ (\max_hi \ l) \ (\max_hi \ r)) \ r$

fun *insert* :: '*a*::{linorder,order_bot} ivl ⇒ '*a* ivl_tree ⇒ '*a* ivl_tree **where**
insert *x* *Leaf* = *Node* *Leaf* (*x*, *high* *x*) *Leaf* |
insert *x* (*Node* *l* (*a*, *m*) *r*) =
 (case *cmp* *x* *a* of
 EQ ⇒ *Node* *l* (*a*, *m*) *r* |
 LT ⇒ *node* (*insert* *x* *l*) *a* *r* |
 GT ⇒ *node* *l* *a* (*insert* *x* *r*))

lemma *inorder_insert*:

$\text{sorted } (\text{inorder } t) \implies \text{inorder } (\text{insert } x \ t) = \text{ins_list } x \ (\text{inorder } t)$
by (*induct* *t* *rule*: *tree2_induct*) (*auto simp*: *ins_list_simps*)

lemma *inv_max_hi_insert*:

$\text{inv_max_hi } t \implies \text{inv_max_hi } (\text{insert } x \ t)$
by (*induct* *t* *rule*: *tree2_induct*) (*auto simp add*: *max3_def*)

fun *split_min* :: '*a*::{linorder,order_bot} ivl_tree ⇒ '*a* ivl × '*a* ivl_tree
where

split_min (*Node* *l* (*a*, *m*) *r*) =
 (if *l* = *Leaf* then (*a*, *r*)
 else let (*x*, *l'*) = *split_min* *l* in (*x*, *node* *l'* *a* *r*))

fun *delete* :: '*a*::{linorder,order_bot} ivl ⇒ '*a* ivl_tree ⇒ '*a* ivl_tree **where**
delete *x* *Leaf* = *Leaf* |

delete *x* (*Node* *l* (*a*, *m*) *r*) =
 (case *cmp* *x* *a* of

```

LT ⇒ node (delete x l) a r |
GT ⇒ node l a (delete x r) |
EQ ⇒ if r = Leaf then l else
      let (a', r') = split_min r in node l a' r')

```

lemma *split_minD*:

```

split_min t = (x,t') ⇒ t ≠ Leaf ⇒ x # inorder t' = inorder t
by (induct t arbitrary: t' rule: split_min.induct)
      (auto simp: sorted_lems split: prod.splits if_splits)

```

lemma *inorder_delete*:

```

sorted (inorder t) ⇒ inorder (delete x t) = del_list x (inorder t)
by (induct t)
      (auto simp: del_list_simps split_minD Let_def split: prod.splits)

```

lemma *inv_max_hi_split_min*:

```

[[ t ≠ Leaf; inv_max_hi t ]] ⇒ inv_max_hi (snd (split_min t))
by (induct t) (auto split: prod.splits)

```

lemma *inv_max_hi_delete*:

```

inv_max_hi t ⇒ inv_max_hi (delete x t)
apply (induct t)
apply simp
using inv_max_hi_split_min by (fastforce simp add: Let_def split: prod.splits)

```

13.4 Search

Does interval x overlap with any interval in the tree?

```

fun search :: 'a::{linorder,order_bot} avl_tree ⇒ 'a avl ⇒ bool where
search Leaf x = False |
search (Node l (a, m) r) x =
  (if overlap x a then True
   else if l ≠ Leaf ∧ max_hi l ≥ low x then search l x
   else search r x)

```

lemma *search_correct*:

```

inv_max_hi t ⇒ sorted (inorder t) ⇒ search t x = has_overlap (set_tree t) x
proof (induction t rule: tree2_induct)
  case Leaf
  then show ?case by (auto simp add: has_overlap_def)
next
  case (Node l a m r)
  have search_l: search l x = has_overlap (set_tree l) x

```

```

    using Node.IH(1) Node.premis by (auto simp: sorted_wrt_append)
  have search_r: search r x = has_overlap (set_tree r) x
    using Node.IH(2) Node.premis by (auto simp: sorted_wrt_append)
  show ?case
  proof (cases overlap a x)
    case True
    thus ?thesis by (auto simp: overlap_def has_overlap_def)
  next
    case a_disjoint: False
    then show ?thesis
    proof cases
      assume [simp]: l = Leaf
      have search_eval: search (Node l (a, m) r) x = search r x
        using a_disjoint overlap_def by auto
      show ?thesis
        unfolding search_eval search_r
        by (auto simp add: has_overlap_def a_disjoint)
    next
      assume l ≠ Leaf
      then show ?thesis
      proof (cases max_hi l ≥ low x)
        case max_hi_l_ge: True
        have inv_max_hi l
          using Node.premis(1) by auto
        then obtain p where p: p ∈ set_tree l high p = max_hi l
          using ⟨l ≠ Leaf⟩ max_hi_exists by auto
        have search_eval: search (Node l (a, m) r) x = search l x
          using a_disjoint ⟨l ≠ Leaf⟩ max_hi_l_ge by (auto simp: over-
lap_def)
        show ?thesis
        proof (cases low p ≤ high x)
          case True
          have overlap p x
            unfolding overlap_def using True p(2) max_hi_l_ge by auto
          then show ?thesis
            unfolding search_eval search_l
            using p(1) by (auto simp: has_overlap_def overlap_def)
        next
          case False
          have ¬overlap x rp if asm: rp ∈ set_tree r for rp
          proof -
            have low p ≤ low rp
              using asm p(1) Node(4) by (fastforce simp: sorted_wrt_append
ivl_less)

```

```

    then show ?thesis
      using False by (auto simp: overlap_def)
    qed
  then show ?thesis
    unfolding search_eval search_l
    using a_disjoint by (auto simp: has_overlap_def overlap_def)
  qed
next
case False
have search_eval: search (Node l (a, m) r) x = search r x
  using a_disjoint False by (auto simp: overlap_def)
have ¬overlap x lp if asm: lp ∈ set_tree l for lp
  using asm False Node.prem1 max_hi_is_max
  by (fastforce simp: overlap_def)
then show ?thesis
  unfolding search_eval search_r
  using a_disjoint by (auto simp: has_overlap_def overlap_def)
qed
qed
qed
qed

```

definition *empty* :: 'a ivl_tree where
empty = Leaf

13.5 Specification

locale *Interval_Set* = Set +
 fixes *has_overlap* :: 't ⇒ 'a::linorder ivl ⇒ bool
 assumes *set_overlap*: *invar* s ⇒ *has_overlap* s x = Interval_Tree.*has_overlap* (set s) x

fun *invar* :: ('a::{linorder, order_bot}) ivl_tree ⇒ bool where
invar t = (*inv_max_hi* t ∧ sorted(*inorder* t))

interpretation *S*: *Interval_Set*
 where *empty* = Leaf and *insert* = insert and *delete* = delete
 and *has_overlap* = search and *isin* = isin and *set* = set_tree
 and *invar* = invar

proof (*standard*, *goal_cases*)
 case 1
 then show ?case by auto
next
 case 2

```

    then show ?case by (simp add: isin_set_inorder)
next
  case 3
  then show ?case by (simp add: inorder_insert set_ins_list_flip: set_inorder)
next
  case 4
  then show ?case by (simp add: inorder_delete set_del_list_flip: set_inorder)
next
  case 5
  then show ?case by auto
next
  case 6
  then show ?case by (simp add: inorder_insert inv_max_hi_insert sorted_ins_list)
next
  case 7
  then show ?case by (simp add: inorder_delete inv_max_hi_delete sorted_del_list)
next
  case 8
  then show ?case by (simp add: search_correct)
qed

end

```

14 AVL Tree Implementation of Sets

```

theory AVL_Set_Code
imports
  Cmp
  Isin2
begin

```

14.1 Code

```

type_synonym 'a tree_ht = ('a*nat) tree

```

```

definition empty :: 'a tree_ht where
empty = Leaf

```

```

fun ht :: 'a tree_ht  $\Rightarrow$  nat where
ht Leaf = 0 |
ht (Node l (a,n) r) = n

```

```

definition node :: 'a tree_ht  $\Rightarrow$  'a  $\Rightarrow$  'a tree_ht  $\Rightarrow$  'a tree_ht where
node l a r = Node l (a, max (ht l) (ht r) + 1) r

```

definition *balL* :: 'a tree_ht ⇒ 'a ⇒ 'a tree_ht ⇒ 'a tree_ht **where**

balL *AB c C* =

(if *ht AB* = *ht C* + 2 then

case *AB* of

Node *A* (*a*, _) *B* ⇒

if *ht A* ≥ *ht B* then node *A a* (node *B c C*)

else

case *B* of

Node *B*₁ (*b*, _) *B*₂ ⇒ node (node *A a B*₁) *b* (node *B*₂ *c C*)

else node *AB c C*)

definition *balR* :: 'a tree_ht ⇒ 'a ⇒ 'a tree_ht ⇒ 'a tree_ht **where**

balR *A a BC* =

(if *ht BC* = *ht A* + 2 then

case *BC* of

Node *B* (*c*, _) *C* ⇒

if *ht B* ≤ *ht C* then node (node *A a B*) *c C*

else

case *B* of

Node *B*₁ (*b*, _) *B*₂ ⇒ node (node *A a B*₁) *b* (node *B*₂ *c C*)

else node *A a BC*)

fun *insert* :: 'a::linorder ⇒ 'a tree_ht ⇒ 'a tree_ht **where**

insert *x Leaf* = Node *Leaf* (*x*, 1) *Leaf* |

insert *x* (Node *l* (*a*, *n*) *r*) = (case *cmp* *x a* of

EQ ⇒ Node *l* (*a*, *n*) *r* |

LT ⇒ *balL* (*insert* *x l*) *a r* |

GT ⇒ *balR* *l a* (*insert* *x r*))

fun *split_max* :: 'a tree_ht ⇒ 'a tree_ht * 'a **where**

split_max (Node *l* (*a*, _) *r*) =

(if *r* = *Leaf* then (*l*, *a*) else let (*r'*, *a'*) = *split_max* *r* in (*balL* *l a r'*, *a'*))

lemmas *split_max_induct* = *split_max.induct*[*case_names* Node *Leaf*]

fun *delete* :: 'a::linorder ⇒ 'a tree_ht ⇒ 'a tree_ht **where**

delete _ *Leaf* = *Leaf* |

delete *x* (Node *l* (*a*, *n*) *r*) =

(case *cmp* *x a* of

EQ ⇒ if *l* = *Leaf* then *r*

else let (*l'*, *a'*) = *split_max* *l* in *balR* *l' a' r* |

LT ⇒ *balR* (*delete* *x l*) *a r* |

GT ⇒ *balL* *l a* (*delete* *x r*))

14.2 Functional Correctness Proofs

Very different from the AFP/AVL proofs

14.2.1 Proofs for insert

lemma *inorder_balL*:

$inorder (balL\ l\ a\ r) = inorder\ l\ @\ a\ \# \in inorder\ r$

by (*auto simp: node_def balL_def split:tree.splits*)

lemma *inorder_balR*:

$inorder (balR\ l\ a\ r) = inorder\ l\ @\ a\ \# \in inorder\ r$

by (*auto simp: node_def balR_def split:tree.splits*)

theorem *inorder_insert*:

$sorted(inorder\ t) \implies inorder(insert\ x\ t) = ins_list\ x\ (inorder\ t)$

by (*induct t*)

(*auto simp: ins_list_simps inorder_balL inorder_balR*)

14.2.2 Proofs for delete

lemma *inorder_split_maxD*:

$\llbracket split_max\ t = (t',a); t \neq Leaf \rrbracket \implies$

$inorder\ t' @ [a] = inorder\ t$

by(*induction t arbitrary: t' rule: split_max.induct*)

(*auto simp: inorder_balL split: if_splits prod.splits tree.split*)

theorem *inorder_delete*:

$sorted(inorder\ t) \implies inorder\ (delete\ x\ t) = del_list\ x\ (inorder\ t)$

by(*induction t*)

(*auto simp: del_list_simps inorder_balL inorder_balR inorder_split_maxD split: prod.splits*)

end

14.3 Invariant

theory *AVL_Set*

imports

AVL_Set_Code

HOL-Number_Theory.Fib

begin

fun *avl* :: 'a *tree_ht* \Rightarrow *bool* **where**

avl Leaf = True |

$avl (Node\ l\ (a,n)\ r) =$
 $(abs(int(height\ l) - int(height\ r)) \leq 1 \wedge$
 $n = max\ (height\ l)\ (height\ r) + 1 \wedge avl\ l \wedge avl\ r)$

14.3.1 Insertion maintains AVL balance

declare *Let_def* [*simp*]

lemma *ht_height*[*simp*]: $avl\ t \implies ht\ t = height\ t$
by (*cases\ t\ rule: tree2_cases*) *simp_all*

First, a fast but relatively manual proof with many lemmas:

lemma *height_balL*:
 $\llbracket avl\ l; avl\ r; height\ l = height\ r + 2 \rrbracket \implies$
 $height\ (balL\ l\ a\ r) \in \{height\ r + 2, height\ r + 3\}$
by (*auto\ simp:node_def\ balL_def\ split:tree.split*)

lemma *height_balR*:
 $\llbracket avl\ l; avl\ r; height\ r = height\ l + 2 \rrbracket \implies$
 $height\ (balR\ l\ a\ r) : \{height\ l + 2, height\ l + 3\}$
by(*auto\ simp\ add:node_def\ balR_def\ split:tree.split*)

lemma *height_node*[*simp*]: $height(node\ l\ a\ r) = max\ (height\ l)\ (height\ r) + 1$
by (*simp\ add: node_def*)

lemma *height_balL2*:
 $\llbracket avl\ l; avl\ r; height\ l \neq height\ r + 2 \rrbracket \implies$
 $height\ (balL\ l\ a\ r) = 1 + max\ (height\ l)\ (height\ r)$
by (*simp_all\ add: balL_def*)

lemma *height_balR2*:
 $\llbracket avl\ l; avl\ r; height\ r \neq height\ l + 2 \rrbracket \implies$
 $height\ (balR\ l\ a\ r) = 1 + max\ (height\ l)\ (height\ r)$
by (*simp_all\ add: balR_def*)

lemma *avl_balL*:
 $\llbracket avl\ l; avl\ r; height\ r - 1 \leq height\ l \wedge height\ l \leq height\ r + 2 \rrbracket \implies$
 $avl(balL\ l\ a\ r)$
by(*auto\ simp: balL_def\ node_def\ split!: if_split\ tree.split*)

lemma *avl_balR*:
 $\llbracket avl\ l; avl\ r; height\ l - 1 \leq height\ r \wedge height\ r \leq height\ l + 2 \rrbracket \implies$
 $avl(balR\ l\ a\ r)$

by(*auto simp: balR_def node_def split!: if_split tree.split*)

Insertion maintains the AVL property. Requires simultaneous proof.

theorem *avl_insert:*

avl t \implies avl(insert x t)

avl t \implies height (insert x t) \in {height t, height t + 1}

proof (*induction t rule: tree2_induct*)

case (*Node l a _ r*)

case 1

show *?case*

proof(*cases x = a*)

case *True with 1 show ?thesis by simp*

next

case *False*

show *?thesis*

proof(*cases x < a*)

case *True with 1 Node(1,2) show ?thesis by (auto intro!:avl_balL)*

next

case *False with 1 Node(3,4) $\langle x \neq a \rangle$ show ?thesis by (auto intro!:avl_balR)*

qed

qed

case 2

show *?case*

proof(*cases x = a*)

case *True with 2 show ?thesis by simp*

next

case *False*

show *?thesis*

proof(*cases x < a*)

case *True*

show *?thesis*

proof(*cases height (insert x l) = height r + 2*)

case *False with 2 Node(1,2) $\langle x < a \rangle$ show ?thesis by (auto simp: height_balL2)*

next

case *True*

hence (*height (balL (insert x l) a r) = height r + 2*) \vee

(*height (balL (insert x l) a r) = height r + 3*) (**is** *?A \vee ?B*)

using 2 *Node(1,2) height_balL[OF _ _ True]* **by** *simp*

thus *?thesis*

proof

assume *?A with 2 $\langle x < a \rangle$ show ?thesis by (auto)*

next

```

      assume ?B with 2 Node(2) True ⟨x < a⟩ show ?thesis by (simp)
arith
  qed
  qed
next
  case False
  show ?thesis
  proof (cases height (insert x r) = height l + 2)
    case False with 2 Node(3,4) ⟨¬x < a⟩ show ?thesis by (auto simp:
height_balR2)
  next
    case True
    hence (height (balR l a (insert x r)) = height l + 2) ∨
      (height (balR l a (insert x r)) = height l + 3) (is ?A ∨ ?B)
      using 2 Node(3) height_balR[OF _ _ True] by simp
    thus ?thesis
    proof
      assume ?A with 2 ⟨¬x < a⟩ show ?thesis by (auto)
    next
      assume ?B with 2 Node(4) True ⟨¬x < a⟩ show ?thesis by (simp)
arith
  qed
  qed
  qed
  qed
qed simp_all

```

Now an automatic proof without lemmas:

```

theorem avl_insert_auto: avl t ⇒
  avl(insert x t) ∧ height (insert x t) ∈ {height t, height t + 1}
apply (induction t rule: tree2_induct)

  apply (auto simp: balL_def balR_def node_def max_absorb2 split!: if_split
tree.split)
done

```

14.3.2 Deletion maintains AVL balance

```

lemma avl_split_max:
  [[ avl t; t ≠ Leaf ]] ⇒
  avl (fst (split_max t)) ∧
  height t ∈ {height(fst (split_max t)), height(fst (split_max t)) + 1}
by(induct t rule: split_max_induct)
  (auto simp: balL_def node_def max_absorb2 split!: prod.split if_split

```

tree.split)

Deletion maintains the AVL property:

theorem *avl_delete*:

avl t \implies *avl*(*delete x t*)

avl t \implies *height t* \in {*height* (*delete x t*), *height* (*delete x t*) + 1}

proof (*induct t rule: tree2_induct*)

case (*Node l a n r*)

case 1

show ?*case*

proof(*cases x = a*)

case *True* **thus** ?*thesis*

using 1 *avl_split_max*[*of l*] **by** (*auto intro!*: *avl_balR split: prod.split*)

next

case *False* **thus** ?*thesis*

using *Node 1* **by** (*auto intro!*: *avl_balL avl_balR*)

qed

case 2

show ?*case*

proof(*cases x = a*)

case *True* **thus** ?*thesis* **using** 2 *avl_split_max*[*of l*]

by(*auto simp: balR_def max_absorb2 split!: if_splits prod.split tree.split*)

next

case *False*

show ?*thesis*

proof(*cases x < a*)

case *True*

show ?*thesis*

proof(*cases height r = height* (*delete x l*) + 2)

case *False*

thus ?*thesis* **using** 2 *Node(1,2)* $\langle x < a \rangle$ **by**(*auto simp: balR_def*)

next

case *True*

thus ?*thesis* **using** *height_balR*[*OF* __ *True*, *of a*] 2 *Node(1,2)* $\langle x$

$< a \rangle$ **by** *simp linarith*

qed

next

case *False*

show ?*thesis*

proof(*cases height l = height* (*delete x r*) + 2)

case *False*

thus ?*thesis* **using** 2 *Node(3,4)* $\langle \neg x < a \rangle$ $\langle x \neq a \rangle$ **by**(*auto simp:*

balL_def)

next

```

      case True
      thus ?thesis
      using height_balL[OF _ _ True, of a] 2 Node(3,4) <¬x < a> <x ≠
a> by simp linarith
      qed
      qed
      qed
    qed simp_all

```

A more automatic proof. Complete automation as for insertion seems hard due to resource requirements.

```

theorem avl_delete_auto:
  avl t ⇒ avl(delete x t)
  avl t ⇒ height t ∈ {height (delete x t), height (delete x t) + 1}
proof (induct t rule: tree2_induct)
  case (Node l a n r)
  case 1
  thus ?case
  using Node avl_split_max[of l] by (auto intro!: avl_balL avl_balR split:
prod.split)
  case 2
  show ?case
  using 2 Node avl_split_max[of l]
  by auto
  (auto simp: balL_def balR_def max_absorb1 max_absorb2 split!:
tree.splits prod.splits if_splits)
  qed simp_all

```

14.4 Overall correctness

```

interpretation S: Set_by_Ordered
where empty = empty and isin = isin and insert = insert and delete =
delete
and inorder = inorder and inv = avl
proof (standard, goal_cases)
  case 1 show ?case by (simp add: empty_def)
next
  case 2 thus ?case by(simp add: isin_set_inorder)
next
  case 3 thus ?case by(simp add: inorder_insert)
next
  case 4 thus ?case by(simp add: inorder_delete)
next
  case 5 thus ?case by (simp add: empty_def)

```

```

next
  case 6 thus ?case by (simp add: avl_insert(1))
next
  case 7 thus ?case by (simp add: avl_delete(1))
qed

```

14.5 Height-Size Relation

Any AVL tree of height n has at least $\text{fib}(n+2)$ leaves:

theorem *avl_fib_bound*:

$\text{avl } t \implies \text{fib}(\text{height } t + 2) \leq \text{size1 } t$

proof (*induction rule: tree2_induct*)

case (*Node l a h r*)

have 1: $\text{height } l + 1 \leq \text{height } r + 2$ **and** 2: $\text{height } r + 1 \leq \text{height } l + 2$

using *Node.prem*s **by** *auto*

have $\text{fib}(\max(\text{height } l)(\text{height } r) + 3) \leq \text{size1 } l + \text{size1 } r$

proof *cases*

assume $\text{height } l \geq \text{height } r$

hence $\text{fib}(\max(\text{height } l)(\text{height } r) + 3) = \text{fib}(\text{height } l + 3)$

by(*simp add: max_absorb1*)

also have $\dots = \text{fib}(\text{height } l + 2) + \text{fib}(\text{height } l + 1)$

by (*simp add: numeral_eq_Suc*)

also have $\dots \leq \text{size1 } l + \text{fib}(\text{height } l + 1)$

using *Node* **by** (*simp*)

also have $\dots \leq \text{size1 } r + \text{size1 } l$

using *Node fib_mono[OF 1]* **by** *auto*

also have $\dots = \text{size1}(\text{Node } l(a,h) r)$

by *simp*

finally show *?thesis*

by (*simp*)

next

assume $\neg \text{height } l \geq \text{height } r$

hence $\text{fib}(\max(\text{height } l)(\text{height } r) + 3) = \text{fib}(\text{height } r + 3)$

by(*simp add: max_absorb1*)

also have $\dots = \text{fib}(\text{height } r + 2) + \text{fib}(\text{height } r + 1)$

by (*simp add: numeral_eq_Suc*)

also have $\dots \leq \text{size1 } r + \text{fib}(\text{height } r + 1)$

using *Node* **by** (*simp*)

also have $\dots \leq \text{size1 } r + \text{size1 } l$

using *Node fib_mono[OF 2]* **by** *auto*

also have $\dots = \text{size1}(\text{Node } l(a,h) r)$

by *simp*

finally show *?thesis*

by (*simp*)

```

qed
also have ... = size1 (Node l (a,h) r)
  by simp
finally show ?case by (simp del: fib.simps add: numeral_eq_Suc)
qed auto

```

```

lemma avl_fib_bound_auto: avl t  $\implies$  fib (height t + 2)  $\leq$  size1 t
proof (induction t rule: tree2_induct)
  case Leaf thus ?case by (simp)
next
  case (Node l a h r)
  have 1: height l + 1  $\leq$  height r + 2 and 2: height r + 1  $\leq$  height l + 2
    using Node.prems by auto
  have left: height l  $\geq$  height r  $\implies$  ?case (is ?asm  $\implies$  _)
    using Node fib_mono[OF 1] by (simp add: max.absorb1)
  have right: height l  $\leq$  height r  $\implies$  ?case
    using Node fib_mono[OF 2] by (simp add: max.absorb2)
  show ?case using left right using Node.prems by simp linarith
qed

```

An exponential lower bound for *fib*:

```

lemma fib_lowerbound:
  defines  $\varphi \equiv (1 + \text{sqrt } 5) / 2$ 
  shows real (fib(n+2))  $\geq$   $\varphi^n$ 
proof (induction n rule: fib.induct)
  case 1
  then show ?case by simp
next
  case 2
  then show ?case by (simp add:  $\varphi$ _def real_le_sqrt)
next
  case (3 n)
  have  $\varphi^{\text{Suc } (Suc n)} = \varphi^2 * \varphi^n$ 
    by (simp add: field_simps power2_eq_square)
  also have ... =  $(\varphi + 1) * \varphi^n$ 
    by (simp_all add:  $\varphi$ _def power2_eq_square field_simps)
  also have ... =  $\varphi^{\text{Suc } n} + \varphi^n$ 
    by (simp add: field_simps)
  also have ...  $\leq$  real (fib (Suc n + 2)) + real (fib (n + 2))
    by (intro add_mono 3.IH)
  finally show ?case by simp
qed

```

The size of an AVL tree is (at least) exponential in its height:

```

lemma avl_size_lowerbound:
  defines  $\varphi \equiv (1 + \text{sqrt } 5) / 2$ 
  assumes avl t
  shows  $\varphi^{\text{height } t} \leq \text{size1 } t$ 
proof –
  have  $\varphi^{\text{height } t} \leq \text{fib } (\text{height } t + 2)$ 
    unfolding  $\varphi\_def$  by(rule fib_lowerbound)
  also have  $\dots \leq \text{size1 } t$ 
    using avl_fib_bound[of t] assms by simp
  finally show ?thesis .
qed

```

The height of an AVL tree is most $1 / \log 2 \varphi \approx 1.44$ times worse than $\log 2$ (*real (size1 t)*):

```

lemma avl_height_upperbound:
  defines  $\varphi \equiv (1 + \text{sqrt } 5) / 2$ 
  assumes avl t
  shows  $\text{height } t \leq (1 / \log 2 \varphi) * \log 2 (\text{size1 } t)$ 
proof –
  have  $\varphi > 0 \ \varphi > 1$  by(auto simp: \varphi\_def pos\_add\_strict)
  hence  $\text{height } t = \log \varphi (\varphi^{\text{height } t})$  by(simp add: log_nat_power)
  also have  $\dots \leq \log \varphi (\text{size1 } t)$ 
    using avl_size_lowerbound[OF assms(2), folded \varphi\_def]  $\langle 1 < \varphi \rangle$ 
    by (simp add: le_log_of_power)
  also have  $\dots = (1 / \log 2 \varphi) * \log 2 (\text{size1 } t)$ 
    by(simp add: log_base_change[of 2 \varphi])
  finally show ?thesis .
qed

```

end

15 Function *lookup* for Tree2

```

theory Lookup2
imports
  Tree2
  Cmp
  Map_Specs
begin

fun lookup :: (('a::linorder * 'b) * 'c) tree  $\Rightarrow$  'a  $\Rightarrow$  'b option where
  lookup Leaf x = None |
  lookup (Node l ((a,b), _) r) x =
    (case cmp x a of LT  $\Rightarrow$  lookup l x | GT  $\Rightarrow$  lookup r x | EQ  $\Rightarrow$  Some b)

```

```

lemma lookup_map_of:
  sorted1(inorder t)  $\implies$  lookup t x = map_of (inorder t) x
by(induction t rule: tree2_induct) (auto simp: map_of_simps split: option.split)

end

```

16 AVL Tree Implementation of Maps

```

theory AVL_Map
imports
  AVL_Set
  Lookup2
begin

fun update :: 'a::linorder  $\Rightarrow$  'b  $\Rightarrow$  ('a*'b) tree_ht  $\Rightarrow$  ('a*'b) tree_ht where
  update x y Leaf = Node Leaf ((x,y), 1) Leaf |
  update x y (Node l ((a,b), h) r) = (case cmp x a of
    EQ  $\Rightarrow$  Node l ((x,y), h) r |
    LT  $\Rightarrow$  balL (update x y l) (a,b) r |
    GT  $\Rightarrow$  balR l (a,b) (update x y r))

fun delete :: 'a::linorder  $\Rightarrow$  ('a*'b) tree_ht  $\Rightarrow$  ('a*'b) tree_ht where
  delete _ Leaf = Leaf |
  delete x (Node l ((a,b), h) r) = (case cmp x a of
    EQ  $\Rightarrow$  if l = Leaf then r
      else let (l', ab') = split_max l in balR l' ab' r |
    LT  $\Rightarrow$  balR (delete x l) (a,b) r |
    GT  $\Rightarrow$  balL l (a,b) (delete x r))

```

16.1 Functional Correctness

```

theorem inorder_update:
  sorted1(inorder t)  $\implies$  inorder(update x y t) = upd_list x y (inorder t)
by (induct t) (auto simp: upd_list_simps inorder_balL inorder_balR)

```

```

theorem inorder_delete:
  sorted1(inorder t)  $\implies$  inorder (delete x t) = del_list x (inorder t)
by(induction t)
  (auto simp: del_list_simps inorder_balL inorder_balR
    inorder_split_maxD split: prod.splits)

```

16.2 AVL invariants

16.2.1 Insertion maintains AVL balance

theorem *avl_update*:

assumes *avl t*

shows *avl(update x y t)*

$(\text{height}(\text{update } x \ y \ t) = \text{height } t \vee \text{height}(\text{update } x \ y \ t) = \text{height } t + 1)$

using *assms*

proof (*induction x y t rule: update.induct*)

case *eq2: (2 x y l a b h r)*

case *1*

show *?case*

proof(*cases x = a*)

case *True with eq2 1 show ?thesis by simp*

next

case *False*

with *eq2 1 show ?thesis*

proof(*cases x < a*)

case *True with eq2 1 show ?thesis by (auto intro!: avl_balL)*

next

case *False with eq2 1 <x≠a> show ?thesis by (auto intro!: avl_balR)*

qed

qed

case *2*

show *?case*

proof(*cases x = a*)

case *True with eq2 1 show ?thesis by simp*

next

case *False*

show *?thesis*

proof(*cases x < a*)

case *True*

show *?thesis*

proof(*cases height (update x y l) = height r + 2*)

case *False with eq2 2 <x < a> show ?thesis by (auto simp: height_balL2)*

next

case *True*

hence $(\text{height}(\text{balL}(\text{update } x \ y \ l) \ (a,b) \ r) = \text{height } r + 2) \vee$

$(\text{height}(\text{balL}(\text{update } x \ y \ l) \ (a,b) \ r) = \text{height } r + 3) \text{ (is } ?A \vee ?B)$

using *eq2 2 <x < a> height_balL[OF _ _ True] by simp*

thus *?thesis*

```

    proof
      assume ?A with 2 ⟨x < a⟩ show ?thesis by (auto)
    next
      assume ?B with True 1 eq2(2) ⟨x < a⟩ show ?thesis by (simp)
arith
    qed
  qed
next
  case False
  show ?thesis
  proof(cases height (update x y r) = height l + 2)
    case False with eq2 2 ⟨¬x < a⟩ show ?thesis by (auto simp:
height_balR2)
  next
    case True
    hence (height (balR l (a,b) (update x y r)) = height l + 2) ∨
      (height (balR l (a,b) (update x y r)) = height l + 3) (is ?A ∨ ?B)
    using eq2 2 ⟨¬x < a⟩ ⟨x ≠ a⟩ height_balR[OF _ _ True] by simp
    thus ?thesis
  proof
    assume ?A with 2 ⟨¬x < a⟩ show ?thesis by (auto)
  next
    assume ?B with True 1 eq2(4) ⟨¬x < a⟩ show ?thesis by (simp)
arith
  qed
  qed
  qed
  qed
qed simp_all

```

16.2.2 Deletion maintains AVL balance

theorem *avl_delete*:

assumes *avl t*

shows $avl(delete\ x\ t)$ **and** $height\ t = (height\ (delete\ x\ t)) \vee height\ t = height\ (delete\ x\ t) + 1$

using *assms*

proof (*induct t rule: tree2_induct*)

case (*Node l ab h r*)

obtain *a b* **where** [*simp*]: $ab = (a,b)$ **by** *fastforce*

case *1*

show *?case*

proof(*cases x = a*)

case *True* **with** *Node 1* **show** *?thesis*

```

    using avl_split_max[of l] by (auto intro!: avl_balR split: prod.split)
next
case False
show ?thesis
proof(cases x < a)
  case True with Node 1 show ?thesis by (auto intro!: avl_balR)
next
  case False with Node 1 ⟨x ≠ a⟩ show ?thesis by (auto intro!: avl_balL)
qed
qed
case 2
show ?case
proof(cases x = a)
  case True then show ?thesis using 1 avl_split_max[of l]
  by(auto simp: balR_def max_absorb2 split!: if_splits prod.split tree.split)
next
  case False
  show ?thesis
  proof(cases x < a)
    case True
    show ?thesis
    proof(cases height r = height (delete x l) + 2)
      case False with Node 1 ⟨x < a⟩ show ?thesis by(auto simp:
balR_def)
    next
      case True
      thus ?thesis using height_balR[OF __ True, of ab] 2 Node(1,2) ⟨x
< a⟩ by simp linarith
    qed
  next
    case False
    show ?thesis
    proof(cases height l = height (delete x r) + 2)
      case False with Node 1 ⟨¬x < a⟩ ⟨x ≠ a⟩ show ?thesis by(auto
simp: balL_def)
    next
      case True
      thus ?thesis
      using height_balL[OF __ True, of ab] 2 Node(3,4) ⟨¬x < a⟩ ⟨x
≠ a⟩ by auto
    qed
  qed
qed
qed simp_all

```

```

interpretation M: Map_by_Ordered
where empty = empty and lookup = lookup and update = update and
delete = delete
and inorder = inorder and inv = avl
proof (standard, goal_cases)
  case 1 show ?case by (simp add: empty_def)
next
  case 2 thus ?case by(simp add: lookup_map_of)
next
  case 3 thus ?case by(simp add: inorder_update)
next
  case 4 thus ?case by(simp add: inorder_delete)
next
  case 5 show ?case by (simp add: empty_def)
next
  case 6 thus ?case by(simp add: avl_update(1))
next
  case 7 thus ?case by(simp add: avl_delete(1))
qed

end

```

17 AVL Tree with Balance Factors (1)

```

theory AVL_Bal_Set
imports
  Cmp
  Isin2
begin

```

This version detects height increase/decrease from above via the change in balance factors.

```

datatype bal = Lh | Bal | Rh

```

```

type_synonym 'a tree_bal = ('a * bal) tree

```

Invariant:

```

fun avl :: 'a tree_bal  $\Rightarrow$  bool where
avl Leaf = True |
avl (Node l (a,b) r) =
  ((case b of
    Bal  $\Rightarrow$  height r = height l |

```

$Lh \Rightarrow \text{height } l = \text{height } r + 1 \mid$
 $Rh \Rightarrow \text{height } r = \text{height } l + 1)$
 $\wedge \text{avl } l \wedge \text{avl } r)$

17.1 Code

fun *is_bal* **where**

is_bal (Node *l* (*a*,*b*) *r*) = (*b* = *Bal*)

fun *incr* **where**

incr *t* *t'* = (*t* = *Leaf* \vee *is_bal* *t* \wedge \neg *is_bal* *t'*)

fun *rot2* **where**

rot2 *A* *a* *B* *c* *C* = (case *B* of
(Node *B*₁ (*b*, *bb*) *B*₂) \Rightarrow
let *b*₁ = if *bb* = *Rh* then *Lh* else *Bal*;
*b*₂ = if *bb* = *Lh* then *Rh* else *Bal*
in Node (Node *A* (*a*,*b*₁) *B*₁) (*b*,*Bal*) (Node *B*₂ (*c*,*b*₂) *C*))

fun *balL* :: '*a* tree_bal \Rightarrow '*a* \Rightarrow *bal* \Rightarrow '*a* tree_bal \Rightarrow '*a* tree_bal **where**

balL *AB* *c* *bc* *C* = (case *bc* of
Bal \Rightarrow Node *AB* (*c*,*Lh*) *C* |
Rh \Rightarrow Node *AB* (*c*,*Bal*) *C* |
Lh \Rightarrow (case *AB* of
Node *A* (*a*,*Lh*) *B* \Rightarrow Node *A* (*a*,*Bal*) (Node *B* (*c*,*Bal*) *C*) |
Node *A* (*a*,*Bal*) *B* \Rightarrow Node *A* (*a*,*Rh*) (Node *B* (*c*,*Lh*) *C*) |
Node *A* (*a*,*Rh*) *B* \Rightarrow *rot2* *A* *a* *B* *c* *C*))

fun *balR* :: '*a* tree_bal \Rightarrow '*a* \Rightarrow *bal* \Rightarrow '*a* tree_bal \Rightarrow '*a* tree_bal **where**

balR *A* *a* *ba* *BC* = (case *ba* of
Bal \Rightarrow Node *A* (*a*,*Rh*) *BC* |
Lh \Rightarrow Node *A* (*a*,*Bal*) *BC* |
Rh \Rightarrow (case *BC* of
Node *B* (*c*,*Rh*) *C* \Rightarrow Node (Node *A* (*a*,*Bal*) *B*) (*c*,*Bal*) *C* |
Node *B* (*c*,*Bal*) *C* \Rightarrow Node (Node *A* (*a*,*Rh*) *B*) (*c*,*Lh*) *C* |
Node *B* (*c*,*Lh*) *C* \Rightarrow *rot2* *A* *a* *B* *c* *C*))

fun *insert* :: '*a*::*linorder* \Rightarrow '*a* tree_bal \Rightarrow '*a* tree_bal **where**

insert *x* *Leaf* = Node *Leaf* (*x*, *Bal*) *Leaf* |
insert *x* (Node *l* (*a*, *b*) *r*) = (case *cmp* *x* *a* of
EQ \Rightarrow Node *l* (*a*, *b*) *r* |
LT \Rightarrow let *l'* = *insert* *x* *l* in if *incr* *l* *l'* then *balL* *l'* *a* *b* *r* else Node *l'* (*a*,*b*)
r |
GT \Rightarrow let *r'* = *insert* *x* *r* in if *incr* *r* *r'* then *balR* *l* *a* *b* *r'* else Node *l* (*a*,*b*)

$r')$

fun *decr* **where**

decr $t\ t' = (t \neq \text{Leaf} \wedge (t' = \text{Leaf} \vee \neg \text{is_bal } t \wedge \text{is_bal } t'))$

fun *split_max* :: $'a\ \text{tree_bal} \Rightarrow 'a\ \text{tree_bal} * 'a$ **where**

split_max (*Node* $l\ (a,\ ba)\ r$) =

(*if* $r = \text{Leaf}$ *then* (l,a)

else *let* $(r',a') = \text{split_max } r$;

$t' = \text{if } \text{decr } r\ r' \text{ then } \text{balL } l\ a\ ba\ r' \text{ else } \text{Node } l\ (a,ba)\ r'$

in (t', a'))

fun *delete* :: $'a::\text{linorder} \Rightarrow 'a\ \text{tree_bal} \Rightarrow 'a\ \text{tree_bal}$ **where**

delete $_ \text{Leaf} = \text{Leaf} \mid$

delete $x\ (\text{Node } l\ (a,\ ba)\ r) =$

(*case* *cmp* $x\ a$ *of*

$\text{EQ} \Rightarrow \text{if } l = \text{Leaf} \text{ then } r$

$\text{else } \text{let } (l', a') = \text{split_max } l \text{ in}$

$\text{if } \text{decr } l\ l' \text{ then } \text{balR } l'\ a'\ ba\ r \text{ else } \text{Node } l'\ (a',ba)\ r \mid$

$\text{LT} \Rightarrow \text{let } l' = \text{delete } x\ l \text{ in if } \text{decr } l\ l' \text{ then } \text{balR } l'\ a\ ba\ r \text{ else } \text{Node } l'$
 $(a,ba)\ r \mid$

$\text{GT} \Rightarrow \text{let } r' = \text{delete } x\ r \text{ in if } \text{decr } r\ r' \text{ then } \text{balL } l\ a\ ba\ r' \text{ else } \text{Node } l$
 $(a,ba)\ r')$

17.2 Proofs

lemmas *split_max_induct* = *split_max.induct*[*case_names Node Leaf*]

lemmas *splits* = *if_splits tree.splits bal.splits*

declare *Let_def* [*simp*]

17.2.1 Proofs about insertion

lemma *avl_insert*: $\text{avl } t \Longrightarrow$

$\text{avl}(\text{insert } x\ t) \wedge$

$\text{height}(\text{insert } x\ t) = \text{height } t + (\text{if } \text{incr } t\ (\text{insert } x\ t) \text{ then } 1 \text{ else } 0)$

apply(*induction* $x\ t$ *rule*: *insert.induct*)

apply(*auto split!*: *splits*)

done

The following two auxiliary lemma merely simplify the proof of *in-order_insert*.

lemma [*simp*]: $[\] \neq \text{ins_list } x\ xs$

by(*cases* xs) *auto*

lemma [simp]: $avl\ t \implies insert\ x\ t \neq \langle l, (a, Rh), \langle \rangle \rangle \wedge insert\ x\ t \neq \langle \langle \rangle, (a, Lh), r \rangle$
by(drule avl_insert[of _ x]) (auto split: splits)

theorem inorder_insert:
 $\llbracket avl\ t; sorted(inorder\ t) \rrbracket \implies inorder(insert\ x\ t) = ins_list\ x\ (inorder\ t)$
apply(induction t)
apply (auto simp: ins_list_simps split!: splits)
done

17.2.2 Proofs about deletion

lemma inorder_balR:
 $\llbracket ba = Rh \longrightarrow r \neq Leaf; avl\ r \rrbracket$
 $\implies inorder\ (balR\ l\ a\ ba\ r) = inorder\ l\ @\ a\ \# inorder\ r$
by (auto split: splits)

lemma inorder_balL:
 $\llbracket ba = Lh \longrightarrow l \neq Leaf; avl\ l \rrbracket$
 $\implies inorder\ (balL\ l\ a\ ba\ r) = inorder\ l\ @\ a\ \# inorder\ r$
by (auto split: splits)

lemma height_1_iff: $avl\ t \implies height\ t = Suc\ 0 \iff (\exists x. t = Node\ Leaf\ (x, Bal)\ Leaf)$
by(cases t) (auto split: splits prod.splits)

lemma avl_split_max:
 $\llbracket split_max\ t = (t', a); avl\ t; t \neq Leaf \rrbracket \implies$
 $avl\ t' \wedge height\ t = height\ t' + (if\ decr\ t\ t'\ then\ 1\ else\ 0)$
apply(induction t arbitrary: t' a rule: split_max_induct)
apply(auto simp: max_absorb1 max_absorb2 height_1_iff split!: splits prod.splits)
done

lemma avl_delete: $avl\ t \implies$
 $avl\ (delete\ x\ t) \wedge$
 $height\ t = height\ (delete\ x\ t) + (if\ decr\ t\ (delete\ x\ t)\ then\ 1\ else\ 0)$
apply(induction x t rule: delete_induct)
apply(auto simp: max_absorb1 max_absorb2 height_1_iff dest: avl_split_max split!: splits prod.splits)
done

lemma *inorder_split_maxD*:
 $\llbracket \text{split_max } t = (t', a); t \neq \text{Leaf}; \text{avl } t \rrbracket \implies$
 $\text{inorder } t' @ [a] = \text{inorder } t$
apply(*induction t arbitrary: t' rule: split_max.induct*)
apply(*fastforce split!: splits prod.splits*)
apply *simp*
done

lemma *neq_Leaf_if_height_neq_0*: $\text{height } t \neq 0 \implies t \neq \text{Leaf}$
by *auto*

lemma *split_max_Leaf*: $\llbracket t \neq \text{Leaf}; \text{avl } t \rrbracket \implies \text{split_max } t = (\langle \rangle, x) \longleftrightarrow$
 $t = \text{Node Leaf } (x, \text{Bal}) \text{ Leaf}$
by(*cases t*) (*auto split: splits prod.splits*)

theorem *inorder_delete*:
 $\llbracket \text{avl } t; \text{sorted}(\text{inorder } t) \rrbracket \implies \text{inorder } (\text{delete } x \ t) = \text{del_list } x \ (\text{inorder } t)$
apply(*induction t rule: tree2_induct*)
apply(*auto simp: del_list_simps inorder_balR inorder_balL avl_delete_inorder_split_maxD*
 $\text{split_max_Leaf neq_Leaf_if_height_neq_0}$
 $\text{simp del: balL.simps balR.simps split!: splits prod.splits}$)
done

17.2.3 Set Implementation

interpretation *S*: *Set_by_Ordered*
where *empty* = *Leaf* **and** *isin* = *isin*
and *insert* = *insert*
and *delete* = *delete*
and *inorder* = *inorder* **and** *inv* = *avl*
proof (*standard, goal_cases*)
case 1 **show** ?*case* **by** (*simp*)
next
case 2 **thus** ?*case* **by**(*simp add: isin_set_inorder*)
next
case 3 **thus** ?*case* **by**(*simp add: inorder_insert*)
next
case 4 **thus** ?*case* **by**(*simp add: inorder_delete*)
next
case 5 **thus** ?*case* **by** (*simp*)
next
case 6 **thus** ?*case* **by** (*simp add: avl_insert*)

```

next
  case  $\gamma$  thus ?case by (simp add: avl_delete)
qed

end

```

18 AVL Tree with Balance Factors (2)

```

theory AVL_Bal2_Set

```

```

imports

```

```

  Cmp

```

```

  Isin2

```

```

begin

```

This version passes a flag (*Same/Diff*) back up to signal if the height changed.

```

datatype bal = Lh | Bal | Rh

```

```

type_synonym 'a tree_bal = ('a * bal) tree

```

Invariant:

```

fun avl :: 'a tree_bal  $\Rightarrow$  bool where

```

```

  avl Leaf = True |

```

```

  avl (Node l (a,b) r) =

```

```

    ((case b of
```

```

      Bal  $\Rightarrow$  height r = height l |
```

```

      Lh  $\Rightarrow$  height l = height r + 1 |
```

```

      Rh  $\Rightarrow$  height r = height l + 1)
```

```

     $\wedge$  avl l  $\wedge$  avl r)

```

18.1 Code

```

datatype 'a alt = Same 'a | Diff 'a

```

```

type_synonym 'a tree_bal2 = 'a tree_bal alt

```

```

fun tree :: 'a alt  $\Rightarrow$  'a where

```

```

  tree(Same t) = t |

```

```

  tree(Diff t) = t

```

```

fun rot2 where

```

```

  rot2 A a B c C = (case B of
```

```

    (Node B1 (b, bb) B2)  $\Rightarrow$ 
```

```

    let b1 = if bb = Rh then Lh else Bal;
```

$b_2 = \text{if } bb = Lh \text{ then } Rh \text{ else } Bal$
in Node (Node A (a,b₁) B₁) (b,Bal) (Node B₂ (c,b₂) C))

fun *balL* :: 'a tree_bal2 ⇒ 'a ⇒ bal ⇒ 'a tree_bal ⇒ 'a tree_bal2 **where**
balL AB' c bc C = (case AB' of
Same AB ⇒ Same (Node AB (c,bc) C) |
Diff AB ⇒ (case bc of
Bal ⇒ Diff (Node AB (c,Lh) C) |
Rh ⇒ Same (Node AB (c,Bal) C) |
Lh ⇒ (case AB of
Node A (a,Lh) B ⇒ Same(Node A (a,Bal) (Node B (c,Bal) C)) |
Node A (a,Rh) B ⇒ Same(rot2 A a B c C))))

fun *balR* :: 'a tree_bal ⇒ 'a ⇒ bal ⇒ 'a tree_bal2 ⇒ 'a tree_bal2 **where**
balR A a ba BC' = (case BC' of
Same BC ⇒ Same (Node A (a,ba) BC) |
Diff BC ⇒ (case ba of
Bal ⇒ Diff (Node A (a,Rh) BC) |
Lh ⇒ Same (Node A (a,Bal) BC) |
Rh ⇒ (case BC of
Node B (c,Rh) C ⇒ Same(Node (Node A (a,Bal) B) (c,Bal) C) |
Node B (c,Lh) C ⇒ Same(rot2 A a B c C))))

fun *ins* :: 'a::linorder ⇒ 'a tree_bal ⇒ 'a tree_bal2 **where**
ins x Leaf = Diff(Node Leaf (x, Bal) Leaf) |
ins x (Node l (a, b) r) = (case cmp x a of
EQ ⇒ Same(Node l (a, b) r) |
LT ⇒ balL (ins x l) a b r |
GT ⇒ balR l a b (ins x r))

definition *insert* :: 'a::linorder ⇒ 'a tree_bal ⇒ 'a tree_bal **where**
insert x t = tree(ins x t)

fun *baldR* :: 'a tree_bal ⇒ 'a ⇒ bal ⇒ 'a tree_bal2 ⇒ 'a tree_bal2 **where**
baldR AB c bc C' = (case C' of
Same C ⇒ Same (Node AB (c,bc) C) |
Diff C ⇒ (case bc of
Bal ⇒ Same (Node AB (c,Lh) C) |
Rh ⇒ Diff (Node AB (c,Bal) C) |
Lh ⇒ (case AB of
Node A (a,Lh) B ⇒ Diff(Node A (a,Bal) (Node B (c,Bal) C)) |
Node A (a,Bal) B ⇒ Same(Node A (a,Rh) (Node B (c,Lh) C)) |
Node A (a,Rh) B ⇒ Diff(rot2 A a B c C))))

```

fun baldL :: 'a tree_bal2 ⇒ 'a ⇒ bal ⇒ 'a tree_bal ⇒ 'a tree_bal2 where
baldL A' a ba BC = (case A' of
  Same A ⇒ Same (Node A (a,ba) BC) |
  Diff A ⇒ (case ba of
    Bal ⇒ Same (Node A (a,Rh) BC) |
    Lh ⇒ Diff (Node A (a,Bal) BC) |
    Rh ⇒ (case BC of
      Node B (c,Rh) C ⇒ Diff(Node (Node A (a,Bal) B) (c,Bal) C) |
      Node B (c,Bal) C ⇒ Same(Node (Node A (a,Rh) B) (c,Lh) C) |
      Node B (c,Lh) C ⇒ Diff(rot2 A a B c C))))

```

```

fun split_max :: 'a tree_bal ⇒ 'a tree_bal2 * 'a where
split_max (Node l (a, ba) r) =
  (if r = Leaf then (Diff l,a) else let (r',a') = split_max r in (baldR l a ba
  r', a'))

```

```

fun del :: 'a::linorder ⇒ 'a tree_bal ⇒ 'a tree_bal2 where
del _ Leaf = Same Leaf |
del x (Node l (a, ba) r) =
  (case cmp x a of
    EQ ⇒ if l = Leaf then Diff r
          else let (l', a') = split_max l in baldL l' a' ba r |
    LT ⇒ baldL (del x l) a ba r |
    GT ⇒ baldR l a ba (del x r))

```

```

definition delete :: 'a::linorder ⇒ 'a tree_bal ⇒ 'a tree_bal where
delete x t = tree(del x t)

```

```

lemmas split_max_induct = split_max.induct[case_names Node Leaf]

```

```

lemmas splits = if_splits tree.splits alt.splits bal.splits

```

18.2 Proofs

18.2.1 Proofs about insertion

```

lemma avl_ins_case: avl t ⇒ case ins x t of
  Same t' ⇒ avl t' ∧ height t' = height t |
  Diff t' ⇒ avl t' ∧ height t' = height t + 1 ∧
  (∀ l a r. t' = Node l (a,Bal) r → a = x ∧ l = Leaf ∧ r = Leaf)
apply(induction x t rule: ins.induct)
apply(auto simp: max_absorb1 split!: splits)
done

```

corollary *avl_insert*: $avl\ t \implies avl(insert\ x\ t)$
using *avl_ins_case*[of *t x*] **by** (*simp add: insert_def split: splits*)

lemma *ins_Diff*[*simp*]: $avl\ t \implies$
 $ins\ x\ t \neq Diff\ Leaf \wedge$
 $(ins\ x\ t = Diff\ (Node\ l\ (a, Bal)\ r) \iff t = Leaf \wedge a = x \wedge l = Leaf \wedge$
 $r = Leaf) \wedge$
 $ins\ x\ t \neq Diff\ (Node\ l\ (a, Rh)\ Leaf) \wedge$
 $ins\ x\ t \neq Diff\ (Node\ Leaf\ (a, Lh)\ r)$
by(*drule avl_ins_case*[of $_ x$]) (*auto split: splits*)

theorem *inorder_ins*:
 $\llbracket avl\ t; sorted(inorder\ t) \rrbracket \implies inorder(tree(ins\ x\ t)) = ins_list\ x\ (inorder\ t)$
apply(*induction t*)
apply (*auto simp: ins_list_simps split!: splits*)
done

18.2.2 Proofs about deletion

lemma *inorder_balDL*:
 $\llbracket ba = Rh \implies r \neq Leaf; avl\ r \rrbracket$
 $\implies inorder\ (tree(balDL\ l\ a\ ba\ r)) = inorder\ (tree\ l) @ a \# inorder\ r$
by (*auto split: splits*)

lemma *inorder_balDR*:
 $\llbracket ba = Lh \implies l \neq Leaf; avl\ l \rrbracket$
 $\implies inorder\ (tree(balDR\ l\ a\ ba\ r)) = inorder\ l @ a \# inorder\ (tree\ r)$
by (*auto split: splits*)

lemma *avl_split_max*:
 $\llbracket split_max\ t = (t', a); avl\ t; t \neq Leaf \rrbracket \implies case\ t'\ of$
 $Same\ t' \Rightarrow avl\ t' \wedge height\ t = height\ t' \mid$
 $Diff\ t' \Rightarrow avl\ t' \wedge height\ t = height\ t' + 1$
apply(*induction t arbitrary: t' a rule: split_max_induct*)
apply(*fastforce simp: max_absorb1 max_absorb2 split!: splits prod.splits*)
apply *simp*
done

lemma *avl_del_case*: $avl\ t \implies case\ del\ x\ t\ of$
 $Same\ t' \Rightarrow avl\ t' \wedge height\ t = height\ t' \mid$
 $Diff\ t' \Rightarrow avl\ t' \wedge height\ t = height\ t' + 1$

```

apply(induction x t rule: del.induct)
  apply(auto simp: max_absorb1 max_absorb2 dest: avl_split_max split!:
splits prod.splits)
done

```

```

corollary avl_delete: avl t  $\implies$  avl(delete x t)
using avl_del_case[of t x] by(simp add: delete_def split: splits)

```

```

lemma inorder_split_maxD:
   $\llbracket \text{split\_max } t = (t', a); t \neq \text{Leaf}; \text{avl } t \rrbracket \implies$ 
   $\text{inorder } (\text{tree } t') @ [a] = \text{inorder } t$ 
apply(induction t arbitrary: t' rule: split_max.induct)
  apply(fastforce split!: splits prod.splits)
apply simp
done

```

```

lemma neq_Leaf_if_height_neq_0[simp]: height t  $\neq$  0  $\implies$  t  $\neq$  Leaf
by auto

```

```

theorem inorder_del:
   $\llbracket \text{avl } t; \text{sorted}(\text{inorder } t) \rrbracket \implies \text{inorder } (\text{tree}(\text{del } x t)) = \text{del\_list } x (\text{inorder } t)$ 
apply(induction t rule: tree2_induct)
apply(auto simp: del_list_simps inorder_balD inorder_balR avl_delete
inorder_split_maxD)
  simp del: balR.simps balL.simps split!: splits prod.splits)
done

```

18.2.3 Set Implementation

```

interpretation S: Set_by_Ordered
where empty = Leaf and isin = isin
  and insert = insert
  and delete = delete
  and inorder = inorder and inv = avl
proof (standard, goal_cases)
  case 1 show ?case by (simp)
next
  case 2 thus ?case by(simp add: isin_set_inorder)
next
  case 3 thus ?case by(simp add: inorder_ins insert_def)
next
  case 4 thus ?case by(simp add: inorder_del delete_def)
next

```

```

    case 5 thus ?case by (simp)
next
  case 6 thus ?case by (simp add: avl_insert)
next
  case 7 thus ?case by (simp add: avl_delete)
qed

end

```

19 Height-Balanced Trees

```

theory Height_Balanced_Tree
imports
  Cmp
  Isin2
begin

```

Height-balanced trees (HBTs) can be seen as a generalization of AVL trees. The code and the proofs were obtained by small modifications of the AVL theories. This is an implementation of sets via HBTs.

```

type_synonym 'a tree_ht = ('a*nat) tree

```

```

definition empty :: 'a tree_ht where
empty = Leaf

```

The maximal amount by which the height of two siblings may differ:

```

locale HBT =
fixes m :: nat
assumes [arith]: m > 0
begin

```

Invariant:

```

fun hbt :: 'a tree_ht ⇒ bool where
hbt Leaf = True |
hbt (Node l (a,n) r) =
  (abs(int(height l) - int(height r)) ≤ int(m) ∧
   n = max (height l) (height r) + 1 ∧ hbt l ∧ hbt r)

```

```

fun ht :: 'a tree_ht ⇒ nat where
ht Leaf = 0 |
ht (Node l (a,n) r) = n

```

```

definition node :: 'a tree_ht ⇒ 'a ⇒ 'a tree_ht ⇒ 'a tree_ht where
node l a r = Node l (a, max (ht l) (ht r) + 1) r

```

definition *balL* :: 'a tree_ht ⇒ 'a ⇒ 'a tree_ht ⇒ 'a tree_ht **where**
balL AB b C =
 (if ht AB = ht C + m + 1 then
 case AB of
 Node A (a, _) B ⇒
 if ht A ≥ ht B then node A a (node B b C)
 else
 case B of
 Node B₁ (ab, _) B₂ ⇒ node (node A a B₁) ab (node B₂ b C)
 else node AB b C)

definition *balR* :: 'a tree_ht ⇒ 'a ⇒ 'a tree_ht ⇒ 'a tree_ht **where**
balR A a BC =
 (if ht BC = ht A + m + 1 then
 case BC of
 Node B (b, _) C ⇒
 if ht B ≤ ht C then node (node A a B) b C
 else
 case B of
 Node B₁ (ab, _) B₂ ⇒ node (node A a B₁) ab (node B₂ b C)
 else node A a BC)

fun *insert* :: 'a::linorder ⇒ 'a tree_ht ⇒ 'a tree_ht **where**
insert x Leaf = Node Leaf (x, 1) Leaf |
insert x (Node l (a, n) r) = (case cmp x a of
 EQ ⇒ Node l (a, n) r |
 LT ⇒ *balL* (insert x l) a r |
 GT ⇒ *balR* l a (insert x r))

fun *split_max* :: 'a tree_ht ⇒ 'a tree_ht * 'a **where**
split_max (Node l (a, _) r) =
 (if r = Leaf then (l,a) else let (r',a') = *split_max* r in (*balL* l a r', a'))

lemmas *split_max_induct* = *split_max.induct*[*case_names* Node Leaf]

fun *delete* :: 'a::linorder ⇒ 'a tree_ht ⇒ 'a tree_ht **where**
delete _ Leaf = Leaf |
delete x (Node l (a, n) r) =
 (case cmp x a of
 EQ ⇒ if l = Leaf then r
 else let (l', a') = *split_max* l in *balR* l' a' r |
 LT ⇒ *balR* (delete x l) a r |
 GT ⇒ *balL* l a (delete x r))

19.1 Functional Correctness Proofs

19.1.1 Proofs for insert

lemma *inorder_balL*:

$inorder (balL\ l\ a\ r) = inorder\ l\ @\ a\ \# \infty\ inorder\ r$
by (*auto simp: node_def balL_def split:tree.splits*)

lemma *inorder_balR*:

$inorder (balR\ l\ a\ r) = inorder\ l\ @\ a\ \# \infty\ inorder\ r$
by (*auto simp: node_def balR_def split:tree.splits*)

theorem *inorder_insert*:

$sorted(inorder\ t) \implies inorder(insert\ x\ t) = ins_list\ x\ (inorder\ t)$
by (*induct t*)
(*auto simp: ins_list_simps inorder_balL inorder_balR*)

19.1.2 Proofs for delete

lemma *inorder_split_maxD*:

$\llbracket split_max\ t = (t',a); t \neq Leaf \rrbracket \implies$
 $inorder\ t' @ [a] = inorder\ t$
by(*induction t arbitrary: t' rule: split_max.induct*)
(*auto simp: inorder_balL split: if_splits prod.splits tree.split*)

theorem *inorder_delete*:

$sorted(inorder\ t) \implies inorder\ (delete\ x\ t) = del_list\ x\ (inorder\ t)$
by(*induction t*)
(*auto simp: del_list_simps inorder_balL inorder_balR inorder_split_maxD split: prod.splits*)

19.2 Invariant preservation

19.2.1 Insertion maintains balance

declare *Let_def [simp]*

lemma *ht_height[simp]*: $hbt\ t \implies ht\ t = height\ t$

by (*cases t rule: tree2_cases*) *simp_all*

First, a fast but relatively manual proof with many lemmas:

lemma *height_balL*:

$\llbracket hbt\ l; hbt\ r; height\ l = height\ r + m + 1 \rrbracket \implies$
 $height\ (balL\ l\ a\ r) \in \{height\ r + m + 1, height\ r + m + 2\}$
by (*auto simp: node_def balL_def split:tree.split*)

lemma *height_balR*:

$\llbracket \text{hbt } l; \text{hbt } r; \text{height } r = \text{height } l + m + 1 \rrbracket \implies$
 $\text{height } (\text{balR } l \ a \ r) \in \{\text{height } l + m + 1, \text{height } l + m + 2\}$
by(*auto simp add: node_def balR_def split: tree.split*)

lemma *height_node[simp]*: $\text{height}(\text{node } l \ a \ r) = \max(\text{height } l) (\text{height } r) + 1$
by (*simp add: node_def*)

lemma *height_balL2*:

$\llbracket \text{hbt } l; \text{hbt } r; \text{height } l \neq \text{height } r + m + 1 \rrbracket \implies$
 $\text{height } (\text{balL } l \ a \ r) = 1 + \max(\text{height } l) (\text{height } r)$
by (*simp_all add: balL_def*)

lemma *height_balR2*:

$\llbracket \text{hbt } l; \text{hbt } r; \text{height } r \neq \text{height } l + m + 1 \rrbracket \implies$
 $\text{height } (\text{balR } l \ a \ r) = 1 + \max(\text{height } l) (\text{height } r)$
by (*simp_all add: balR_def*)

lemma *hbt_balL*:

$\llbracket \text{hbt } l; \text{hbt } r; \text{height } r - m \leq \text{height } l \wedge \text{height } l \leq \text{height } r + m + 1 \rrbracket$
 $\implies \text{hbt}(\text{balL } l \ a \ r)$
by(*auto simp: balL_def node_def max_def split!: if_splits tree.split*)

lemma *hbt_balR*:

$\llbracket \text{hbt } l; \text{hbt } r; \text{height } l - m \leq \text{height } r \wedge \text{height } r \leq \text{height } l + m + 1 \rrbracket$
 $\implies \text{hbt}(\text{balR } l \ a \ r)$
by(*auto simp: balR_def node_def max_def split!: if_splits tree.split*)

Insertion maintains *hbt*. Requires simultaneous proof.

theorem *hbt_insert*:

$\text{hbt } t \implies \text{hbt}(\text{insert } x \ t)$
 $\text{hbt } t \implies \text{height } (\text{insert } x \ t) \in \{\text{height } t, \text{height } t + 1\}$

proof (*induction t rule: tree2_induct*)

case (*Node l a _ r*)

case 1

show *?case*

proof(*cases x = a*)

case True with Node 1 show ?thesis by simp

next

case False

show *?thesis*

proof(*cases x < a*)

case True with 1 Node(1,2) show ?thesis by (auto intro!: hbt_balL)

```

    next
      case False with 1 Node(3,4) ⟨x≠a⟩ show ?thesis by (auto intro!:
hbt_balR)
      qed
      qed
      case 2
      show ?case
      proof(cases x = a)
        case True with 2 show ?thesis by simp
      next
        case False
        show ?thesis
        proof(cases x < a)
          case True
          show ?thesis
          proof(cases height (insert x l) = height r + m + 1)
            case False with 2 Node(1,2) ⟨x < a⟩ show ?thesis by (auto simp:
height_balL2)
          next
            case True
            hence (height (balL (insert x l) a r) = height r + m + 1) ∨
              (height (balL (insert x l) a r) = height r + m + 2) (is ?A ∨ ?B)
              using 2 Node(1,2) height_balL[OF __ True] by simp
            thus ?thesis
            proof
              assume ?A with 2 Node(2) True ⟨x < a⟩ show ?thesis by (auto)
            next
              assume ?B with 2 Node(2) True ⟨x < a⟩ show ?thesis by (simp)
            arith
          qed
          qed
        next
          case False
          show ?thesis
          proof(cases height (insert x r) = height l + m + 1)
            case False with 2 Node(3,4) ⟨¬x < a⟩ show ?thesis by (auto simp:
height_balR2)
          next
            case True
            hence (height (balR l a (insert x r)) = height l + m + 1) ∨
              (height (balR l a (insert x r)) = height l + m + 2) (is ?A ∨ ?B)
              using Node 2 height_balR[OF __ True] by simp
            thus ?thesis
            proof

```

```

      assume ?A with 2 Node(4) True <¬x < a> show ?thesis by (auto)
    next
      assume ?B with 2 Node(4) True <¬x < a> show ?thesis by (simp)
arith
  qed
  qed
  qed
  qed
qed simp_all

```

Now an automatic proof without lemmas:

```

theorem hbt_insert_auto: hbt t  $\implies$ 
  hbt(insert x t)  $\wedge$  height (insert x t)  $\in$  {height t, height t + 1}
apply (induction t rule: tree2_induct)

apply (auto simp: balL_def balR_def node_def max_absorb1 max_absorb2
split!: if_split tree.split)
done

```

19.2.2 Deletion maintains balance

```

lemma hbt_split_max:
   $\llbracket$  hbt t; t  $\neq$  Leaf  $\rrbracket \implies$ 
  hbt (fst (split_max t))  $\wedge$ 
  height t  $\in$  {height(fst (split_max t)), height(fst (split_max t)) + 1}
by(induct t rule: split_max_induct)
  (auto simp: balL_def node_def max_absorb2 split!: prod.split if_split
tree.split)

```

Deletion maintains hbt:

```

theorem hbt_delete:
  hbt t  $\implies$  hbt(delete x t)
  hbt t  $\implies$  height t  $\in$  {height (delete x t), height (delete x t) + 1}
proof (induct t rule: tree2_induct)
  case (Node l a n r)
  case 1
  thus ?case
    using Node hbt_split_max[of l] by (auto intro!: hbt_balL hbt_balR split:
prod.split)
  case 2
  show ?case
  proof(cases x = a)
    case True then show ?thesis using 1 hbt_split_max[of l]
    by(auto simp: balR_def max_absorb2 split!: if_splits prod.split tree.split)

```

```

next
  case False
  show ?thesis
  proof(cases x < a)
    case True
    show ?thesis
    proof(cases height r = height (delete x l) + m + 1)
      case False with Node 1 ⟨x < a⟩ show ?thesis by(auto simp:
balR_def)
      next
      case True
      hence (height (balR (delete x l) a r) = height (delete x l) + m + 1)
∨
      height (balR (delete x l) a r) = height (delete x l) + m + 2 (is ?A
∨ ?B)
      using Node 2 height_balR[OF __ True] by simp
      thus ?thesis
      proof
        assume ?A with ⟨x < a⟩ Node 2 show ?thesis by(auto simp:
balR_def split!: if_splits)
        next
        assume ?B with ⟨x < a⟩ Node 2 show ?thesis by(auto simp:
balR_def split!: if_splits)
      qed
    qed
  next
  case False
  show ?thesis
  proof(cases height l = height (delete x r) + m + 1)
    case False with Node 1 ⟨¬x < a⟩ ⟨x ≠ a⟩ show ?thesis by(auto
simp: balL_def)
    next
    case True
    hence (height (balL l a (delete x r)) = height (delete x r) + m + 1)
∨
    height (balL l a (delete x r)) = height (delete x r) + m + 2 (is ?A
∨ ?B)
    using Node 2 height_balL[OF __ True] by simp
    thus ?thesis
    proof
      assume ?A with ⟨¬x < a⟩ ⟨x ≠ a⟩ Node 2 show ?thesis by(auto
simp: balL_def split!: if_splits)
      next
      assume ?B with ⟨¬x < a⟩ ⟨x ≠ a⟩ Node 2 show ?thesis by(auto

```

```

simp: balL_def split: if_splits)
  qed
  qed
  qed
  qed
qed simp_all

```

A more automatic proof. Complete automation as for insertion seems hard due to resource requirements.

```

theorem hbt_delete_auto:
  hbt t  $\implies$  hbt(delete x t)
  hbt t  $\implies$  height t  $\in$  {height (delete x t), height (delete x t) + 1}
proof (induct t rule: tree2_induct)
  case (Node l a n r)
  case 1
  thus ?case
    using Node hbt_split_max[of l] by (auto intro!: hbt_balL hbt_balR split:
prod.split)
  case 2
  show ?case
  proof(cases x = a)
    case True thus ?thesis
      using 2 hbt_split_max[of l]
      by(auto simp: balR_def max_absorb2 split!: if_splits prod.split tree.split)
  next
    case False thus ?thesis
      using height_balL[of l delete x r a] height_balR[of delete x l r a] 2
Node
      by(auto simp: balL_def balR_def split!: if_split)
  qed
qed simp_all

```

19.3 Overall correctness

```

interpretation S: Set_by_Ordered
where empty = empty and isin = isin and insert = insert and delete =
delete
and inorder = inorder and inv = hbt
proof (standard, goal_cases)
  case 1 show ?case by (simp add: empty_def)
next
  case 2 thus ?case by(simp add: isin_set_inorder)
next
  case 3 thus ?case by(simp add: inorder_insert)

```

```

next
  case 4 thus ?case by (simp add: inorder_delete)
next
  case 5 thus ?case by (simp add: empty_def)
next
  case 6 thus ?case by (simp add: hbt_insert(1))
next
  case 7 thus ?case by (simp add: hbt_delete(1))
qed

end

end

```

20 Red-Black Trees

```

theory RBT
imports Tree2
begin

```

```

datatype color = Red | Black

```

```

type_synonym 'a rbt = ('a*color)tree

```

```

abbreviation R where R l a r ≡ Node l (a, Red) r

```

```

abbreviation B where B l a r ≡ Node l (a, Black) r

```

```

fun baliL :: 'a rbt ⇒ 'a ⇒ 'a rbt ⇒ 'a rbt where
baliL (R (R t1 a t2) b t3) c t4 = R (B t1 a t2) b (B t3 c t4) |
baliL (R t1 a (R t2 b t3)) c t4 = R (B t1 a t2) b (B t3 c t4) |
baliL t1 a t2 = B t1 a t2

```

```

fun baliR :: 'a rbt ⇒ 'a ⇒ 'a rbt ⇒ 'a rbt where
baliR t1 a (R t2 b (R t3 c t4)) = R (B t1 a t2) b (B t3 c t4) |
baliR t1 a (R (R t2 b t3) c t4) = R (B t1 a t2) b (B t3 c t4) |
baliR t1 a t2 = B t1 a t2

```

```

fun paint :: color ⇒ 'a rbt ⇒ 'a rbt where
paint c Leaf = Leaf |
paint c (Node l (a,_) r) = Node l (a,c) r

```

```

fun baldL :: 'a rbt ⇒ 'a ⇒ 'a rbt ⇒ 'a rbt where
baldL (R t1 a t2) b t3 = R (B t1 a t2) b t3 |

```

```

baldL t1 a (B t2 b t3) = baliR t1 a (R t2 b t3) |
baldL t1 a (R (B t2 b t3) c t4) = R (B t1 a t2) b (baliR t3 c (paint Red
t4)) |
baldL t1 a t2 = R t1 a t2

```

```

fun baldR :: 'a rbt ⇒ 'a ⇒ 'a rbt ⇒ 'a rbt where
baldR t1 a (R t2 b t3) = R t1 a (B t2 b t3) |
baldR (B t1 a t2) b t3 = baliL (R t1 a t2) b t3 |
baldR (R t1 a (B t2 b t3)) c t4 = R (baliL (paint Red t1) a t2) b (B t3 c
t4) |
baldR t1 a t2 = R t1 a t2

```

```

fun join :: 'a rbt ⇒ 'a rbt ⇒ 'a rbt where
join Leaf t = t |
join t Leaf = t |
join (R t1 a t2) (R t3 c t4) =
  (case join t2 t3 of
    R u2 b u3 ⇒ (R (R t1 a u2) b (R u3 c t4)) |
    t23 ⇒ R t1 a (R t23 c t4)) |
join (B t1 a t2) (B t3 c t4) =
  (case join t2 t3 of
    R u2 b u3 ⇒ R (B t1 a u2) b (B u3 c t4) |
    t23 ⇒ baldL t1 a (B t23 c t4)) |
join t1 (R t2 a t3) = R (join t1 t2) a t3 |
join (R t1 a t2) t3 = R t1 a (join t2 t3)

```

end

21 Red-Black Tree Implementation of Sets

```

theory RBT_Set

```

```

imports

```

```

  Complex_Main

```

```

  RBT

```

```

  Cmp

```

```

  Isin2

```

```

begin

```

```

definition empty :: 'a rbt where

```

```

empty = Leaf

```

```

fun ins :: 'a::linorder ⇒ 'a rbt ⇒ 'a rbt where

```

```

ins x Leaf = R Leaf x Leaf |

```

```

ins x (B l a r) =
  (case cmp x a of
    LT => baliL (ins x l) a r |
    GT => baliR l a (ins x r) |
    EQ => B l a r) |
ins x (R l a r) =
  (case cmp x a of
    LT => R (ins x l) a r |
    GT => R l a (ins x r) |
    EQ => R l a r)

```

definition *insert* :: 'a::linorder => 'a rbt => 'a rbt **where**
insert x t = paint Black (ins x t)

fun *color* :: 'a rbt => color **where**
color Leaf = Black |
color (Node _ (_, c) _) = c

fun *del* :: 'a::linorder => 'a rbt => 'a rbt **where**
del x Leaf = Leaf |
del x (Node l (a, _) r) =
 (case cmp x a of
 LT => if l ≠ Leaf ∧ color l = Black
 then baldL (del x l) a r else R (del x l) a r |
 GT => if r ≠ Leaf ∧ color r = Black
 then baldR l a (del x r) else R l a (del x r) |
 EQ => join l r)

definition *delete* :: 'a::linorder => 'a rbt => 'a rbt **where**
delete x t = paint Black (del x t)

21.1 Functional Correctness Proofs

lemma *inorder_paint*: *inorder*(paint c t) = *inorder* t
by(cases t) (auto)

lemma *inorder_baliL*:
inorder(baliL l a r) = *inorder* l @ a # *inorder* r
by(cases (l,a,r) rule: baliL.cases) (auto)

lemma *inorder_baliR*:
inorder(baliR l a r) = *inorder* l @ a # *inorder* r
by(cases (l,a,r) rule: baliR.cases) (auto)

lemma *inorder_ins*:
 $sorted(inorder\ t) \implies inorder(ins\ x\ t) = ins_list\ x\ (inorder\ t)$
by(*induction* $x\ t$ *rule*: *ins.induct*)
(auto *simp*: *ins_list_simps* *inorder_baliL* *inorder_baliR*)

lemma *inorder_insert*:
 $sorted(inorder\ t) \implies inorder(insert\ x\ t) = ins_list\ x\ (inorder\ t)$
by (*simp* *add*: *insert_def* *inorder_ins* *inorder_paint*)

lemma *inorder_baldL*:
 $inorder(baldL\ l\ a\ r) = inorder\ l\ @\ a\ \#\ inorder\ r$
by(*cases* (l, a, r) *rule*: *baldL.cases*)
(auto *simp*: *inorder_baliL* *inorder_baliR* *inorder_paint*)

lemma *inorder_baldR*:
 $inorder(baldR\ l\ a\ r) = inorder\ l\ @\ a\ \#\ inorder\ r$
by(*cases* (l, a, r) *rule*: *baldR.cases*)
(auto *simp*: *inorder_baliL* *inorder_baliR* *inorder_paint*)

lemma *inorder_join*:
 $inorder(join\ l\ r) = inorder\ l\ @\ inorder\ r$
by(*induction* $l\ r$ *rule*: *join.induct*)
(auto *simp*: *inorder_baldL* *inorder_baldR* *split*: *tree.split* *color.split*)

lemma *inorder_del*:
 $sorted(inorder\ t) \implies inorder(del\ x\ t) = del_list\ x\ (inorder\ t)$
by(*induction* $x\ t$ *rule*: *del.induct*)
(auto *simp*: *del_list_simps* *inorder_join* *inorder_baldL* *inorder_baldR*)

lemma *inorder_delete*:
 $sorted(inorder\ t) \implies inorder(delete\ x\ t) = del_list\ x\ (inorder\ t)$
by (auto *simp*: *delete_def* *inorder_del* *inorder_paint*)

21.2 Structural invariants

lemma *neq_Black[simp]*: $(c \neq Black) = (c = Red)$
by (*cases* c) *auto*

The proofs are due to Markus Reiter and Alexander Krauss.

fun *bheight* :: 'a rbt \Rightarrow nat **where**
bheight Leaf = 0 |
bheight (Node $l\ (x, c)\ r$) = (if $c = Black$ then *bheight* $l + 1$ else *bheight* l)

fun *invc* :: 'a rbt \Rightarrow bool **where**

$invc\ Leaf = True \mid$
 $invc\ (Node\ l\ (a,c)\ r) =$
 $((c = Red \longrightarrow color\ l = Black \wedge color\ r = Black) \wedge invc\ l \wedge invc\ r)$

Weaker version:

abbreviation $invc2 :: 'a\ rbt \Rightarrow bool$ **where**
 $invc2\ t \equiv invc(\text{paint}\ Black\ t)$

fun $invh :: 'a\ rbt \Rightarrow bool$ **where**
 $invh\ Leaf = True \mid$
 $invh\ (Node\ l\ (x, c)\ r) = (bheight\ l = bheight\ r \wedge invh\ l \wedge invh\ r)$

lemma $invc2I: invc\ t \Longrightarrow invc2\ t$
by ($cases\ t\ rule: tree2_cases$) $simp+$

definition $rbt :: 'a\ rbt \Rightarrow bool$ **where**
 $rbt\ t = (invc\ t \wedge invh\ t \wedge color\ t = Black)$

lemma $color_paint_Black: color\ (\text{paint}\ Black\ t) = Black$
by ($cases\ t$) $auto$

lemma $paint2: \text{paint}\ c2\ (\text{paint}\ c1\ t) = \text{paint}\ c2\ t$
by ($cases\ t$) $auto$

lemma $invh_paint: invh\ t \Longrightarrow invh\ (\text{paint}\ c\ t)$
by ($cases\ t$) $auto$

lemma $invc_baliL:$
 $\llbracket invc2\ l; invc\ r \rrbracket \Longrightarrow invc\ (baliL\ l\ a\ r)$
by ($induct\ l\ a\ r\ rule: baliL.induct$) $auto$

lemma $invc_baliR:$
 $\llbracket invc\ l; invc2\ r \rrbracket \Longrightarrow invc\ (baliR\ l\ a\ r)$
by ($induct\ l\ a\ r\ rule: baliR.induct$) $auto$

lemma $bheight_baliL:$
 $bheight\ l = bheight\ r \Longrightarrow bheight\ (baliL\ l\ a\ r) = Suc\ (bheight\ l)$
by ($induct\ l\ a\ r\ rule: baliL.induct$) $auto$

lemma $bheight_baliR:$
 $bheight\ l = bheight\ r \Longrightarrow bheight\ (baliR\ l\ a\ r) = Suc\ (bheight\ l)$
by ($induct\ l\ a\ r\ rule: baliR.induct$) $auto$

lemma $invh_baliL:$

$\llbracket \text{invh } l; \text{invh } r; \text{bheight } l = \text{bheight } r \rrbracket \implies \text{invh } (\text{baliL } l \ a \ r)$
by (*induct l a r rule: baliL.induct*) *auto*

lemma *invh_baliR*:

$\llbracket \text{invh } l; \text{invh } r; \text{bheight } l = \text{bheight } r \rrbracket \implies \text{invh } (\text{baliR } l \ a \ r)$
by (*induct l a r rule: baliR.induct*) *auto*

All in one:

lemma *inv_baliR*: $\llbracket \text{invh } l; \text{invh } r; \text{invc } l; \text{invc2 } r; \text{bheight } l = \text{bheight } r \rrbracket$
 $\implies \text{invc } (\text{baliR } l \ a \ r) \wedge \text{invh } (\text{baliR } l \ a \ r) \wedge \text{bheight } (\text{baliR } l \ a \ r) = \text{Suc}$
 $(\text{bheight } l)$
by (*induct l a r rule: baliR.induct*) *auto*

lemma *inv_baliL*: $\llbracket \text{invh } l; \text{invh } r; \text{invc2 } l; \text{invc } r; \text{bheight } l = \text{bheight } r \rrbracket$
 $\implies \text{invc } (\text{baliL } l \ a \ r) \wedge \text{invh } (\text{baliL } l \ a \ r) \wedge \text{bheight } (\text{baliL } l \ a \ r) = \text{Suc}$
 $(\text{bheight } l)$
by (*induct l a r rule: baliL.induct*) *auto*

21.2.1 Insertion

lemma *invc_ins*: $\text{invc } t \longrightarrow \text{invc2 } (\text{ins } x \ t) \wedge (\text{color } t = \text{Black} \longrightarrow \text{invc}$
 $(\text{ins } x \ t))$
by (*induct x t rule: ins.induct*) (*auto simp: invc_baliL invc_baliR invc2I*)

lemma *invh_ins*: $\text{invh } t \implies \text{invh } (\text{ins } x \ t) \wedge \text{bheight } (\text{ins } x \ t) = \text{bheight } t$
by(*induct x t rule: ins.induct*)
(auto simp: invh_baliL invh_baliR bheight_baliL bheight_baliR)

theorem *rbt_insert*: $\text{rbt } t \implies \text{rbt } (\text{insert } x \ t)$

by (*simp add: invc_ins invh_ins color_paint_Black invh_paint rbt_def insert_def*)

All in one:

lemma *inv_ins*: $\llbracket \text{invc } t; \text{invh } t \rrbracket \implies$
 $\text{invc2 } (\text{ins } x \ t) \wedge (\text{color } t = \text{Black} \longrightarrow \text{invc } (\text{ins } x \ t)) \wedge$
 $\text{invh}(\text{ins } x \ t) \wedge \text{bheight } (\text{ins } x \ t) = \text{bheight } t$
by (*induct x t rule: ins.induct*) (*auto simp: inv_baliL inv_baliR invc2I*)

theorem *rbt_insert2*: $\text{rbt } t \implies \text{rbt } (\text{insert } x \ t)$

by (*simp add: inv_ins color_paint_Black invh_paint rbt_def insert_def*)

21.2.2 Deletion

lemma *bheight_paint_Red*:

$\text{color } t = \text{Black} \implies \text{bheight } (\text{paint Red } t) = \text{bheight } t - 1$

by (*cases t*) *auto*

lemma *invh_baldL_invc*:

$\llbracket \text{invh } l; \text{invh } r; \text{bheight } l + 1 = \text{bheight } r; \text{invc } r \rrbracket$
 $\implies \text{invh } (\text{baldL } l \ a \ r) \wedge \text{bheight } (\text{baldL } l \ a \ r) = \text{bheight } r$

by (*induct l a r rule: baldL.induct*)

(*auto simp: invh_baliR invh_paint bheight_baliR bheight_paint_Red*)

lemma *invh_baldL_Black*:

$\llbracket \text{invh } l; \text{invh } r; \text{bheight } l + 1 = \text{bheight } r; \text{color } r = \text{Black} \rrbracket$
 $\implies \text{invh } (\text{baldL } l \ a \ r) \wedge \text{bheight } (\text{baldL } l \ a \ r) = \text{bheight } r$

by (*induct l a r rule: baldL.induct*) (*auto simp add: invh_baliR bheight_baliR*)

lemma *invc_baldL*: $\llbracket \text{invc2 } l; \text{invc } r; \text{color } r = \text{Black} \rrbracket \implies \text{invc } (\text{baldL } l \ a \ r)$

by (*induct l a r rule: baldL.induct*) (*simp_all add: invc_baliR*)

lemma *invc2_baldL*: $\llbracket \text{invc2 } l; \text{invc } r \rrbracket \implies \text{invc2 } (\text{baldL } l \ a \ r)$

by (*induct l a r rule: baldL.induct*) (*auto simp: invc_baliR paint2 invc2I*)

lemma *invh_baldR_invc*:

$\llbracket \text{invh } l; \text{invh } r; \text{bheight } l = \text{bheight } r + 1; \text{invc } l \rrbracket$
 $\implies \text{invh } (\text{baldR } l \ a \ r) \wedge \text{bheight } (\text{baldR } l \ a \ r) = \text{bheight } l$

by(*induct l a r rule: baldR.induct*)

(*auto simp: invh_baliL bheight_baliL invh_paint bheight_paint_Red*)

lemma *invc_baldR*: $\llbracket \text{invc } l; \text{invc2 } r; \text{color } l = \text{Black} \rrbracket \implies \text{invc } (\text{baldR } l \ a \ r)$

by (*induct l a r rule: baldR.induct*) (*simp_all add: invc_baliL*)

lemma *invc2_baldR*: $\llbracket \text{invc } l; \text{invc2 } r \rrbracket \implies \text{invc2 } (\text{baldR } l \ a \ r)$

by (*induct l a r rule: baldR.induct*) (*auto simp: invc_baliL paint2 invc2I*)

lemma *invh_join*:

$\llbracket \text{invh } l; \text{invh } r; \text{bheight } l = \text{bheight } r \rrbracket$
 $\implies \text{invh } (\text{join } l \ r) \wedge \text{bheight } (\text{join } l \ r) = \text{bheight } l$

by (*induct l r rule: join.induct*)

(*auto simp: invh_baldL_Black split: tree.splits color.splits*)

lemma *invc_join*:

$\llbracket \text{invc } l; \text{invc } r \rrbracket \implies$

$(\text{color } l = \text{Black} \wedge \text{color } r = \text{Black} \longrightarrow \text{invc } (\text{join } l \ r)) \wedge \text{invc2 } (\text{join } l \ r)$

by (*induct l r rule: join.induct*)

(*auto simp: invc_baldL invc2I split: tree.splits color.splits*)

All in one:

lemma *inv_baldL*:

$\llbracket \text{invh } l; \text{invh } r; \text{bheight } l + 1 = \text{bheight } r; \text{invc2 } l; \text{invc } r \rrbracket$
 $\implies \text{invh } (\text{baldL } l \ a \ r) \wedge \text{bheight } (\text{baldL } l \ a \ r) = \text{bheight } r$
 $\wedge \text{invc2 } (\text{baldL } l \ a \ r) \wedge (\text{color } r = \text{Black} \longrightarrow \text{invc } (\text{baldL } l \ a \ r))$

by (*induct l a r rule: baldL.induct*)

(*auto simp: inv_baliR invh_paint bheight_baliR bheight_paint_Red paint2 invc2I*)

lemma *inv_baldR*:

$\llbracket \text{invh } l; \text{invh } r; \text{bheight } l = \text{bheight } r + 1; \text{invc } l; \text{invc2 } r \rrbracket$
 $\implies \text{invh } (\text{baldR } l \ a \ r) \wedge \text{bheight } (\text{baldR } l \ a \ r) = \text{bheight } l$
 $\wedge \text{invc2 } (\text{baldR } l \ a \ r) \wedge (\text{color } l = \text{Black} \longrightarrow \text{invc } (\text{baldR } l \ a \ r))$

by (*induct l a r rule: baldR.induct*)

(*auto simp: inv_baliL invh_paint bheight_baliL bheight_paint_Red paint2 invc2I*)

lemma *inv_join*:

$\llbracket \text{invh } l; \text{invh } r; \text{bheight } l = \text{bheight } r; \text{invc } l; \text{invc } r \rrbracket$
 $\implies \text{invh } (\text{join } l \ r) \wedge \text{bheight } (\text{join } l \ r) = \text{bheight } l$
 $\wedge \text{invc2 } (\text{join } l \ r) \wedge (\text{color } l = \text{Black} \wedge \text{color } r = \text{Black} \longrightarrow \text{invc } (\text{join } l \ r))$

by (*induct l r rule: join.induct*)

(*auto simp: invh_baldL_Black inv_baldL invc2I split: tree.splits color.splits*)

lemma *neq_LeafD*: $t \neq \text{Leaf} \implies \exists l \ x \ c \ r. t = \text{Node } l \ (x, c) \ r$

by(*cases t rule: tree2_cases*) *auto*

lemma *inv_del*: $\llbracket \text{invh } t; \text{invc } t \rrbracket \implies$

$\text{invh } (\text{del } x \ t) \wedge$
 $(\text{color } t = \text{Red} \longrightarrow \text{bheight } (\text{del } x \ t) = \text{bheight } t \wedge \text{invc } (\text{del } x \ t)) \wedge$
 $(\text{color } t = \text{Black} \longrightarrow \text{bheight } (\text{del } x \ t) = \text{bheight } t - 1 \wedge \text{invc2 } (\text{del } x \ t))$

by(*induct x t rule: del.induct*)

(*auto simp: inv_baldL inv_baldR inv_join dest!: neq_LeafD*)

theorem *rbt_delete*: $\text{rbt } t \implies \text{rbt } (\text{delete } x \ t)$

by (*metis delete_def rbt_def color_paint_Black inv_del invh_paint*)

Overall correctness:

interpretation *S*: *Set_by_Ordered*

where *empty* = *empty* **and** *isin* = *isin* **and** *insert* = *insert* **and** *delete* = *delete*

```

and inorder = inorder and inv = rbt
proof (standard, goal_cases)
  case 1 show ?case by (simp add: empty_def)
next
  case 2 thus ?case by(simp add: isin_set_inorder)
next
  case 3 thus ?case by(simp add: inorder_insert)
next
  case 4 thus ?case by(simp add: inorder_delete)
next
  case 5 thus ?case by (simp add: rbt_def empty_def)
next
  case 6 thus ?case by (simp add: rbt_insert)
next
  case 7 thus ?case by (simp add: rbt_delete)
qed

```

21.3 Height-Size Relation

```

lemma rbt_height_bheight_if: invc t  $\implies$  invh t  $\implies$ 
  height t  $\leq$  2 * bheight t + (if color t = Black then 0 else 1)
by(induction t) (auto split: if_split_asm)

```

```

lemma rbt_height_bheight: rbt t  $\implies$  height t / 2  $\leq$  bheight t
by(auto simp: rbt_def dest: rbt_height_bheight_if)

```

```

lemma bheight_size_bound: invc t  $\implies$  invh t  $\implies$  2 t  $\leq$  size1 t
by (induction t) auto

```

```

lemma rbt_height_le: assumes rbt t shows height t  $\leq$  2 * log 2 (size1 t)
proof –

```

```

  have 2 height t / 2  $\leq$  2 bheight t
    using rbt_height_bheight[OF assms] by (simp)
  also have ...  $\leq$  size1 t using assms
    by (simp add: powr_realpow bheight_size_bound rbt_def)
  finally have 2 height t / 2  $\leq$  size1 t .
  hence height t / 2  $\leq$  log 2 (size1 t)
    by (simp add: le_log_iff size1_size del: divide_le_eq_numeral1(1))
  thus ?thesis by simp

```

qed

end

22 Alternative Deletion in Red-Black Trees

```
theory RBT_Set2
imports RBT_Set
begin
```

This is a conceptually simpler version of deletion. Instead of the tricky *join* function this version follows the standard approach of replacing the deleted element (in function *del*) by the minimal element in its right subtree.

```
fun split_min :: 'a rbt  $\Rightarrow$  'a  $\times$  'a rbt where
split_min (Node l (a, _) r) =
  (if l = Leaf then (a,r)
   else let (x,l') = split_min l
          in (x, if color l = Black then baldL l' a r else R l' a r))
```

```
fun del :: 'a::linorder  $\Rightarrow$  'a rbt  $\Rightarrow$  'a rbt where
del x Leaf = Leaf |
del x (Node l (a, _) r) =
  (case cmp x a of
   LT  $\Rightarrow$  let l' = del x l in if l  $\neq$  Leaf  $\wedge$  color l = Black
             then baldL l' a r else R l' a r |
   GT  $\Rightarrow$  let r' = del x r in if r  $\neq$  Leaf  $\wedge$  color r = Black
             then baldR l a r' else R l a r' |
   EQ  $\Rightarrow$  if r = Leaf then l else let (a',r') = split_min r in
             if color r = Black then baldR l a' r' else R l a' r')
```

The first two *lets* speed up the automatic proof of *inv_del* below.

```
definition delete :: 'a::linorder  $\Rightarrow$  'a rbt  $\Rightarrow$  'a rbt where
delete x t = paint Black (del x t)
```

22.1 Functional Correctness Proofs

```
declare Let_def[simp]
```

```
lemma split_minD:
```

```
split_min t = (x,t')  $\Longrightarrow$  t  $\neq$  Leaf  $\Longrightarrow$  x  $\#$  inorder t' = inorder t
by(induction t arbitrary: t' rule: split_min.induct)
(auto simp: inorder_baldL sorted_lems split: prod.splits if_splits)
```

```
lemma inorder_del:
```

```
sorted(inorder t)  $\Longrightarrow$  inorder(del x t) = del_list x (inorder t)
by(induction x t rule: del.induct)
(auto simp: del_list_simps inorder_baldL inorder_baldR split_minD split:
prod.splits)
```

lemma *inorder_delete*:
 $sorted(inorder\ t) \implies inorder(delete\ x\ t) = del_list\ x\ (inorder\ t)$
by (*auto simp: delete_def inorder_del inorder_paint*)

22.2 Structural invariants

lemma *neq_Red[simp]*: $(c \neq Red) = (c = Black)$
by (*cases c auto*)

22.2.1 Deletion

lemma *inv_split_min*: $\llbracket split_min\ t = (x, t');\ t \neq Leaf;\ invh\ t;\ invc\ t \rrbracket$
 \implies
 $invh\ t' \wedge$
 $(color\ t = Red \longrightarrow bheight\ t' = bheight\ t \wedge invc\ t') \wedge$
 $(color\ t = Black \longrightarrow bheight\ t' = bheight\ t - 1 \wedge invc2\ t')$
apply(*induction t arbitrary: x t' rule: split_min.induct*)
apply(*auto simp: inv_baldR inv_baldL invc2I dest!: neq_LeafD*
 $split: if_splits\ prod.splits$)
done

An automatic proof. It is quite brittle, e.g. inlining the *lets* in *RBT_Set2.del* breaks it.

lemma *inv_del*: $\llbracket invh\ t;\ invc\ t \rrbracket \implies$
 $invh\ (del\ x\ t) \wedge$
 $(color\ t = Red \longrightarrow bheight\ (del\ x\ t) = bheight\ t \wedge invc\ (del\ x\ t)) \wedge$
 $(color\ t = Black \longrightarrow bheight\ (del\ x\ t) = bheight\ t - 1 \wedge invc2\ (del\ x\ t))$
apply(*induction x t rule: del.induct*)
apply(*auto simp: inv_baldR inv_baldL invc2I dest!: inv_split_min dest:*
 neq_LeafD
 $split!: prod.splits\ if_splits$)
done

A structured proof where one can see what is used in each case.

lemma *inv_del2*: $\llbracket invh\ t;\ invc\ t \rrbracket \implies$
 $invh\ (del\ x\ t) \wedge$
 $(color\ t = Red \longrightarrow bheight\ (del\ x\ t) = bheight\ t \wedge invc\ (del\ x\ t)) \wedge$
 $(color\ t = Black \longrightarrow bheight\ (del\ x\ t) = bheight\ t - 1 \wedge invc2\ (del\ x\ t))$
proof(*induction x t rule: del.induct*)
case (1 *x*)
then show *?case by simp*
next
case (2 *x l a c r*)
note *if_split[split del]*
show *?case*

```

proof cases
  assume  $x < a$ 
  show ?thesis
proof cases
  assume  $l = \text{Leaf}$  thus ?thesis using  $\langle x < a \rangle$  2.prem1 by(auto)
next
  assume  $l \neq \text{Leaf}$ 
  show ?thesis
  proof (cases color l)
    assume *: color l = Black
    hence bheight l > 0 using  $l \text{ neq\_LeafD[of } l]$  by auto
    thus ?thesis using  $\langle x < a \rangle$  2.IH(1) 2.prem1 inv_baldL[of del x l] *
  l by(auto)
  next
    assume color l = Red
    thus ?thesis using  $\langle x < a \rangle$  2.prem1 2.IH(1) by(auto)
  qed
qed
next
  assume  $\neg x < a$ 
  show ?thesis
proof cases
  assume  $x > a$ 
  show ?thesis using  $\langle a < x \rangle$  2.IH(2) 2.prem1 neq_LeafD[of r] inv_baldR[of
  __ del x r]
    by(auto split: if_split)

  next
    assume  $\neg x > a$ 
    show ?thesis using 2.prem1  $\langle \neg x < a \rangle \langle \neg x > a \rangle$ 
    by(auto simp: inv_baldR invc2I dest!: inv_split_min dest: neq_LeafD
split: prod.split if_split)
  qed
qed
qed

```

theorem *rbt_delete*: $\text{rbt } t \implies \text{rbt } (\text{delete } x \ t)$
by (*metis delete_def rbt_def color_paint_Black inv_del invh_paint*)

Overall correctness:

interpretation *S*: *Set_by_Ordered*
where *empty* = *empty* **and** *isin* = *isin* **and** *insert* = *insert* **and** *delete* =
delete
and *inorder* = *inorder* **and** *inv* = *rbt*

```

proof (standard, goal_cases)
  case 1 show ?case by (simp add: empty_def)
next
  case 2 thus ?case by(simp add: isin_set_inorder)
next
  case 3 thus ?case by(simp add: inorder_insert)
next
  case 4 thus ?case by(simp add: inorder_delete)
next
  case 5 thus ?case by (simp add: rbt_def empty_def)
next
  case 6 thus ?case by (simp add: rbt_insert)
next
  case 7 thus ?case by (simp add: rbt_delete)
qed

end

```

23 Red-Black Tree Implementation of Maps

```

theory RBT_Map

```

```

imports

```

```

  RBT_Set

```

```

  Lookup2

```

```

begin

```

```

fun upd :: 'a::linorder  $\Rightarrow$  'b  $\Rightarrow$  ('a*'b) rbt  $\Rightarrow$  ('a*'b) rbt where

```

```

upd x y Leaf = R Leaf (x,y) Leaf |

```

```

upd x y (B l (a,b) r) = (case cmp x a of

```

```

  LT  $\Rightarrow$  baliL (upd x y l) (a,b) r |

```

```

  GT  $\Rightarrow$  baliR l (a,b) (upd x y r) |

```

```

  EQ  $\Rightarrow$  B l (x,y) r |

```

```

upd x y (R l (a,b) r) = (case cmp x a of

```

```

  LT  $\Rightarrow$  R (upd x y l) (a,b) r |

```

```

  GT  $\Rightarrow$  R l (a,b) (upd x y r) |

```

```

  EQ  $\Rightarrow$  R l (x,y) r)

```

```

definition update :: 'a::linorder  $\Rightarrow$  'b  $\Rightarrow$  ('a*'b) rbt  $\Rightarrow$  ('a*'b) rbt where

```

```

update x y t = paint Black (upd x y t)

```

```

fun del :: 'a::linorder  $\Rightarrow$  ('a*'b)rbt  $\Rightarrow$  ('a*'b)rbt where

```

```

del x Leaf = Leaf |

```

$del\ x\ (Node\ l\ ((a,b),\ c)\ r) = (case\ cmp\ x\ a\ of$
 $LT \Rightarrow if\ l \neq Leaf \wedge color\ l = Black$
 $\quad then\ baldL\ (del\ x\ l)\ (a,b)\ r\ else\ R\ (del\ x\ l)\ (a,b)\ r\ |$
 $GT \Rightarrow if\ r \neq Leaf \wedge color\ r = Black$
 $\quad then\ baldR\ l\ (a,b)\ (del\ x\ r)\ else\ R\ l\ (a,b)\ (del\ x\ r)\ |$
 $EQ \Rightarrow join\ l\ r)$

definition $delete :: 'a::linorder \Rightarrow ('a*'b)\ rbt \Rightarrow ('a*'b)\ rbt$ **where**
 $delete\ x\ t = paint\ Black\ (del\ x\ t)$

23.1 Functional Correctness Proofs

lemma $inorder_upd$:
 $sorted1(inorder\ t) \Longrightarrow inorder(upd\ x\ y\ t) = upd_list\ x\ y\ (inorder\ t)$
by($induction\ x\ y\ t\ rule: upd.induct$)
 $(auto\ simp: upd_list_simps\ inorder_baliL\ inorder_baliR)$

lemma $inorder_update$:
 $sorted1(inorder\ t) \Longrightarrow inorder(update\ x\ y\ t) = upd_list\ x\ y\ (inorder\ t)$
by($simp\ add: update_def\ inorder_upd\ inorder_paint$)

lemma $inorder_del$:
 $sorted1(inorder\ t) \Longrightarrow inorder(del\ x\ t) = del_list\ x\ (inorder\ t)$
by($induction\ x\ t\ rule: del.induct$)
 $(auto\ simp: del_list_simps\ inorder_join\ inorder_baldL\ inorder_baldR)$

lemma $inorder_delete$:
 $sorted1(inorder\ t) \Longrightarrow inorder(delete\ x\ t) = del_list\ x\ (inorder\ t)$
by($simp\ add: delete_def\ inorder_del\ inorder_paint$)

23.2 Structural invariants

23.2.1 Update

lemma $invc_upd$: **assumes** $invc\ t$
shows $color\ t = Black \Longrightarrow invc\ (upd\ x\ y\ t)\ invc2\ (upd\ x\ y\ t)$
using $assms$
by ($induct\ x\ y\ t\ rule: upd.induct$) ($auto\ simp: invc_baliL\ invc_baliR\ invc2I$)

lemma $invh_upd$: **assumes** $invh\ t$
shows $invh\ (upd\ x\ y\ t)\ bheight\ (upd\ x\ y\ t) = bheight\ t$
using $assms$
by($induct\ x\ y\ t\ rule: upd.induct$)
 $(auto\ simp: invh_baliL\ invh_baliR\ bheight_baliL\ bheight_baliR)$

theorem *rbt_update*: $rbt\ t \implies rbt\ (update\ x\ y\ t)$
by (*simp add: invc_upd(2) invh_upd(1) color_paint_Black invh_paint*
rbt_def update_def)

23.2.2 Deletion

lemma *del_invc_invh*: $invh\ t \implies invc\ t \implies invh\ (del\ x\ t) \wedge$
 $(color\ t = Red \wedge bheight\ (del\ x\ t) = bheight\ t \wedge invc\ (del\ x\ t) \vee$
 $color\ t = Black \wedge bheight\ (del\ x\ t) = bheight\ t - 1 \wedge invc2\ (del\ x\ t))$

proof (*induct x t rule: del.induct*)

case (*2 x _ y _ c*)

have $x = y \vee x < y \vee x > y$ **by** *auto*

thus *?case proof* (*elim disjE*)

assume $x = y$

with *2 show ?thesis*

by (*cases c*) (*simp_all add: invh_join invc_join*)

next

assume $x < y$

with *2 show ?thesis*

by(*cases c*)

(*auto simp: invh_baldL_invc invc_baldL invc2_baldL dest: neq_LeafD*)

next

assume $y < x$

with *2 show ?thesis*

by(*cases c*)

(*auto simp: invh_baldR_invc invc_baldR invc2_baldR dest: neq_LeafD*)

qed

qed *auto*

theorem *rbt_delete*: $rbt\ t \implies rbt\ (delete\ k\ t)$

by (*metis delete_def rbt_def color_paint_Black del_invc_invh invc2I invh_paint*)

interpretation *M*: *Map_by_Ordered*

where *empty* = *empty* **and** *lookup* = *lookup* **and** *update* = *update* **and**
delete = *delete*

and *inorder* = *inorder* **and** *inv* = *rbt*

proof (*standard, goal_cases*)

case *1 show ?case* **by** (*simp add: empty_def*)

next

case *2 thus ?case* **by**(*simp add: lookup_map_of*)

next

case *3 thus ?case* **by**(*simp add: inorder_update*)

next

case *4 thus ?case* **by**(*simp add: inorder_delete*)

```

next
  case 5 thus ?case by (simp add: rbt_def empty_def)
next
  case 6 thus ?case by (simp add: rbt_update)
next
  case 7 thus ?case by (simp add: rbt_delete)
qed

end

```

24 2-3 Trees

```

theory Tree23
imports Main
begin

```

```

class height =
fixes height :: 'a ⇒ nat

```

```

datatype 'a tree23 =
  Leaf (⟨⟩) |
  Node2 'a tree23 'a 'a tree23 (⟨_, _, _⟩) |
  Node3 'a tree23 'a 'a tree23 'a 'a tree23 (⟨_, _, _, _⟩)

```

```

fun inorder :: 'a tree23 ⇒ 'a list where
inorder Leaf = [] |
inorder(Node2 l a r) = inorder l @ a # inorder r |
inorder(Node3 l a m b r) = inorder l @ a # inorder m @ b # inorder r

```

```

instantiation tree23 :: (type)height
begin

```

```

fun height_tree23 :: 'a tree23 ⇒ nat where
height Leaf = 0 |
height (Node2 l _ r) = Suc(max (height l) (height r)) |
height (Node3 l _ m _ r) = Suc(max (height l) (max (height m) (height
r)))

```

```

instance ..

```

```

end

```

Completeness:

```

fun complete :: 'a tree23  $\Rightarrow$  bool where
  complete Leaf = True |
  complete (Node2 l _ r) = (height l = height r  $\wedge$  complete l & complete r) |
  complete (Node3 l _ m _ r) =
    (height l = height m & height m = height r & complete l & complete m
    & complete r)

```

```

lemma ht_sz_if_complete: complete t  $\Longrightarrow$  2 ^ height t  $\leq$  size t + 1
by (induction t) auto

```

```

end

```

25 2-3 Tree Implementation of Sets

```

theory Tree23_Set

```

```

imports

```

```

  Tree23

```

```

  Cmp

```

```

  Set_Specs

```

```

begin

```

```

declare sorted_wrt.simps(2)[simp del]

```

```

definition empty :: 'a tree23 where

```

```

  empty = Leaf

```

```

fun isin :: 'a::linorder tree23  $\Rightarrow$  'a  $\Rightarrow$  bool where

```

```

  isin Leaf x = False |

```

```

  isin (Node2 l a r) x =

```

```

    (case cmp x a of

```

```

      LT  $\Rightarrow$  isin l x |

```

```

      EQ  $\Rightarrow$  True |

```

```

      GT  $\Rightarrow$  isin r x) |

```

```

  isin (Node3 l a m b r) x =

```

```

    (case cmp x a of

```

```

      LT  $\Rightarrow$  isin l x |

```

```

      EQ  $\Rightarrow$  True |

```

```

      GT  $\Rightarrow$ 

```

```

        (case cmp x b of

```

```

          LT  $\Rightarrow$  isin m x |

```

```

          EQ  $\Rightarrow$  True |

```

```

          GT  $\Rightarrow$  isin r x))

```

datatype 'a upI = TI 'a tree23 | OF 'a tree23 'a 'a tree23

fun treeI :: 'a upI ⇒ 'a tree23 **where**

treeI (TI t) = t |

treeI (OF l a r) = Node2 l a r

fun ins :: 'a::linorder ⇒ 'a tree23 ⇒ 'a upI **where**

ins x Leaf = OF Leaf x Leaf |

ins x (Node2 l a r) =

(case cmp x a of

LT ⇒

(case ins x l of

TI l' => TI (Node2 l' a r) |

OF l1 b l2 => TI (Node3 l1 b l2 a r)) |

EQ ⇒ TI (Node2 l a r) |

GT ⇒

(case ins x r of

TI r' => TI (Node2 l a r') |

OF r1 b r2 => TI (Node3 l a r1 b r2))) |

ins x (Node3 l a m b r) =

(case cmp x a of

LT ⇒

(case ins x l of

TI l' => TI (Node3 l' a m b r) |

OF l1 c l2 => OF (Node2 l1 c l2) a (Node2 m b r)) |

EQ ⇒ TI (Node3 l a m b r) |

GT ⇒

(case cmp x b of

GT ⇒

(case ins x r of

TI r' => TI (Node3 l a m b r') |

OF r1 c r2 => OF (Node2 l a m) b (Node2 r1 c r2)) |

EQ ⇒ TI (Node3 l a m b r) |

LT ⇒

(case ins x m of

TI m' => TI (Node3 l a m' b r) |

OF m1 c m2 => OF (Node2 l a m1) c (Node2 m2 b r))))

hide_const insert

definition insert :: 'a::linorder ⇒ 'a tree23 ⇒ 'a tree23 **where**

insert x t = treeI(ins x t)

datatype 'a upD = TD 'a tree23 | UF 'a tree23

```

fun treeD :: 'a upD ⇒ 'a tree23 where
treeD (TD t) = t |
treeD (UF t) = t

```

```

fun node21 :: 'a upD ⇒ 'a ⇒ 'a tree23 ⇒ 'a upD where
node21 (TD t1) a t2 = TD(Node2 t1 a t2) |
node21 (UF t1) a (Node2 t2 b t3) = UF(Node3 t1 a t2 b t3) |
node21 (UF t1) a (Node3 t2 b t3 c t4) = TD(Node2 (Node2 t1 a t2) b
(Node2 t3 c t4))

```

```

fun node22 :: 'a tree23 ⇒ 'a ⇒ 'a upD ⇒ 'a upD where
node22 t1 a (TD t2) = TD(Node2 t1 a t2) |
node22 (Node2 t1 b t2) a (UF t3) = UF(Node3 t1 b t2 a t3) |
node22 (Node3 t1 b t2 c t3) a (UF t4) = TD(Node2 (Node2 t1 b t2) c
(Node2 t3 a t4))

```

```

fun node31 :: 'a upD ⇒ 'a ⇒ 'a tree23 ⇒ 'a ⇒ 'a tree23 ⇒ 'a upD where
node31 (TD t1) a t2 b t3 = TD(Node3 t1 a t2 b t3) |
node31 (UF t1) a (Node2 t2 b t3) c t4 = TD(Node2 (Node3 t1 a t2 b t3)
c t4) |
node31 (UF t1) a (Node3 t2 b t3 c t4) d t5 = TD(Node3 (Node2 t1 a t2)
b (Node2 t3 c t4) d t5)

```

```

fun node32 :: 'a tree23 ⇒ 'a ⇒ 'a upD ⇒ 'a ⇒ 'a tree23 ⇒ 'a upD where
node32 t1 a (TD t2) b t3 = TD(Node3 t1 a t2 b t3) |
node32 t1 a (UF t2) b (Node2 t3 c t4) = TD(Node2 t1 a (Node3 t2 b t3 c
t4)) |
node32 t1 a (UF t2) b (Node3 t3 c t4 d t5) = TD(Node3 t1 a (Node2 t2 b
t3) c (Node2 t4 d t5))

```

```

fun node33 :: 'a tree23 ⇒ 'a ⇒ 'a tree23 ⇒ 'a ⇒ 'a upD ⇒ 'a upD where
node33 l a m b (TD r) = TD(Node3 l a m b r) |
node33 t1 a (Node2 t2 b t3) c (UF t4) = TD(Node2 t1 a (Node3 t2 b t3 c
t4)) |
node33 t1 a (Node3 t2 b t3 c t4) d (UF t5) = TD(Node3 t1 a (Node2 t2 b
t3) c (Node2 t4 d t5))

```

```

fun split_min :: 'a tree23 ⇒ 'a * 'a upD where
split_min (Node2 Leaf a Leaf) = (a, UF Leaf) |
split_min (Node3 Leaf a Leaf b Leaf) = (a, TD(Node2 Leaf b Leaf)) |
split_min (Node2 l a r) = (let (x,l') = split_min l in (x, node21 l' a r)) |

```

$split_min (Node3\ l\ a\ m\ b\ r) = (let\ (x,l') = split_min\ l\ in\ (x,\ node31\ l'\ a\ m\ b\ r))$

In the base cases of $split_min$ and del it is enough to check if one subtree is a *Leaf*, in which case completeness implies that so are the others. Exercise.

fun $del :: 'a::linorder \Rightarrow 'a\ tree23 \Rightarrow 'a\ upD$ **where**

$del\ x\ Leaf = TD\ Leaf\ |$

$del\ x\ (Node2\ Leaf\ a\ Leaf) =$

$(if\ x = a\ then\ UF\ Leaf\ else\ TD(Node2\ Leaf\ a\ Leaf))\ |$

$del\ x\ (Node3\ Leaf\ a\ Leaf\ b\ Leaf) =$

$TD(if\ x = a\ then\ Node2\ Leaf\ b\ Leaf\ else$

$if\ x = b\ then\ Node2\ Leaf\ a\ Leaf$

$else\ Node3\ Leaf\ a\ Leaf\ b\ Leaf)\ |$

$del\ x\ (Node2\ l\ a\ r) =$

$(case\ cmp\ x\ a\ of$

$LT \Rightarrow node21\ (del\ x\ l)\ a\ r\ |$

$GT \Rightarrow node22\ l\ a\ (del\ x\ r)\ |$

$EQ \Rightarrow let\ (a',r') = split_min\ r\ in\ node22\ l\ a'\ r')\ |$

$del\ x\ (Node3\ l\ a\ m\ b\ r) =$

$(case\ cmp\ x\ a\ of$

$LT \Rightarrow node31\ (del\ x\ l)\ a\ m\ b\ r\ |$

$EQ \Rightarrow let\ (a',m') = split_min\ m\ in\ node32\ l\ a'\ m'\ b\ r\ |$

$GT \Rightarrow$

$(case\ cmp\ x\ b\ of$

$LT \Rightarrow node32\ l\ a\ (del\ x\ m)\ b\ r\ |$

$EQ \Rightarrow let\ (b',r') = split_min\ r\ in\ node33\ l\ a\ m\ b'\ r'\ |$

$GT \Rightarrow node33\ l\ a\ m\ b\ (del\ x\ r))$

definition $delete :: 'a::linorder \Rightarrow 'a\ tree23 \Rightarrow 'a\ tree23$ **where**

$delete\ x\ t = treeD(del\ x\ t)$

25.1 Functional Correctness

25.1.1 Proofs for isin

lemma $isin_set: sorted(inorder\ t) \Longrightarrow isin\ t\ x = (x \in set\ (inorder\ t))$

by $(induction\ t)\ (auto\ simp: isin_simps)$

25.1.2 Proofs for insert

lemma $inorder_ins:$

$sorted(inorder\ t) \Longrightarrow inorder(treeI(ins\ x\ t)) = ins_list\ x\ (inorder\ t)$

by $(induction\ t)\ (auto\ simp: ins_list_simps\ split: upI.splits)$

lemma $inorder_insert:$

$sorted(inorder\ t) \implies inorder(insert\ a\ t) = ins_list\ a\ (inorder\ t)$
by(*simp add: insert_def inorder_ins*)

25.1.3 Proofs for delete

lemma *inorder_node21: height r > 0 \implies*
 $inorder\ (treeD\ (node21\ l'\ a\ r)) = inorder\ (treeD\ l')\ @\ a\ \# \text{inorder}\ r$
by(*induct l' a r rule: node21.induct*) *auto*

lemma *inorder_node22: height l > 0 \implies*
 $inorder\ (treeD\ (node22\ l\ a\ r')) = inorder\ l\ @\ a\ \# \text{inorder}\ (treeD\ r')$
by(*induct l a r' rule: node22.induct*) *auto*

lemma *inorder_node31: height m > 0 \implies*
 $inorder\ (treeD\ (node31\ l'\ a\ m\ b\ r)) = inorder\ (treeD\ l')\ @\ a\ \# \text{inorder}\ m$
 $@\ b\ \# \text{inorder}\ r$
by(*induct l' a m b r rule: node31.induct*) *auto*

lemma *inorder_node32: height r > 0 \implies*
 $inorder\ (treeD\ (node32\ l\ a\ m'\ b\ r)) = inorder\ l\ @\ a\ \# \text{inorder}\ (treeD\ m')$
 $@\ b\ \# \text{inorder}\ r$
by(*induct l a m' b r rule: node32.induct*) *auto*

lemma *inorder_node33: height m > 0 \implies*
 $inorder\ (treeD\ (node33\ l\ a\ m\ b\ r')) = inorder\ l\ @\ a\ \# \text{inorder}\ m\ @\ b\ \#$
 $inorder\ (treeD\ r')$
by(*induct l a m b r' rule: node33.induct*) *auto*

lemmas *inorder_nodes = inorder_node21 inorder_node22*
inorder_node31 inorder_node32 inorder_node33

lemma *split_minD:*
 $split_min\ t = (x, t') \implies complete\ t \implies height\ t > 0 \implies$
 $x\ \# \text{inorder}\ (treeD\ t') = \text{inorder}\ t$
by(*induction t arbitrary: t' rule: split_min.induct*)
(auto simp: inorder_nodes split: prod.splits)

lemma *inorder_del: $\llbracket complete\ t ; sorted(inorder\ t) \rrbracket \implies$*
 $inorder(treeD\ (del\ x\ t)) = del_list\ x\ (inorder\ t)$
by(*induction t rule: del.induct*)
(auto simp: del_list_simps inorder_nodes split_minD split!: if_split prod.splits)

lemma *inorder_delete: $\llbracket complete\ t ; sorted(inorder\ t) \rrbracket \implies$*
 $inorder(delete\ x\ t) = del_list\ x\ (inorder\ t)$

by(*simp add: delete_def inorder_del*)

25.2 Completeness

25.2.1 Proofs for insert

First a standard proof that *ins* preserves *complete*.

fun *hI* :: 'a *upI* \Rightarrow *nat* **where**
hI (*TI* *t*) = *height* *t* |
hI (*OF* *l a r*) = *height* *l*

lemma *complete_ins*: *complete* *t* \Longrightarrow *complete* (*treeI*(*ins a t*)) \wedge *hI*(*ins a t*) = *height* *t*

by (*induct* *t*) (*auto split!*: *if_split upI.split*)

Now an alternative proof (by Brian Huffman) that runs faster because two properties (completeness and height) are combined in one predicate.

inductive *full* :: *nat* \Rightarrow 'a *tree23* \Rightarrow *bool* **where**

full 0 *Leaf* |
[[*full* *n l*; *full* *n r*] \Longrightarrow *full* (*Suc* *n*) (*Node2* *l p r*) |
[[*full* *n l*; *full* *n m*; *full* *n r*] \Longrightarrow *full* (*Suc* *n*) (*Node3* *l p m q r*)

inductive_cases *full_elims*:

full *n Leaf*
full *n* (*Node2* *l p r*)
full *n* (*Node3* *l p m q r*)

inductive_cases *full_0_elim*: *full* 0 *t*

inductive_cases *full_Suc_elim*: *full* (*Suc* *n*) *t*

lemma *full_0_iff* [*simp*]: *full* 0 *t* \longleftrightarrow *t* = *Leaf*

by (*auto elim: full_0_elim intro: full.intros*)

lemma *full_Leaf_iff* [*simp*]: *full* *n Leaf* \longleftrightarrow *n* = 0

by (*auto elim: full_elims intro: full.intros*)

lemma *full_Suc_Node2_iff* [*simp*]:

full (*Suc* *n*) (*Node2* *l p r*) \longleftrightarrow *full* *n l* \wedge *full* *n r*
by (*auto elim: full_elims intro: full.intros*)

lemma *full_Suc_Node3_iff* [*simp*]:

full (*Suc* *n*) (*Node3* *l p m q r*) \longleftrightarrow *full* *n l* \wedge *full* *n m* \wedge *full* *n r*
by (*auto elim: full_elims intro: full.intros*)

lemma *full_imp_height*: $full\ n\ t \implies height\ t = n$
by (*induct set: full, simp_all*)

lemma *full_imp_complete*: $full\ n\ t \implies complete\ t$
by (*induct set: full, auto dest: full_imp_height*)

lemma *complete_imp_full*: $complete\ t \implies full\ (height\ t)\ t$
by (*induct t, simp_all*)

lemma *complete_iff_full*: $complete\ t \longleftrightarrow (\exists n. full\ n\ t)$
by (*auto elim!: complete_imp_full full_imp_complete*)

The *insert* function either preserves the height of the tree, or increases it by one. The constructor returned by the *insert* function determines which: A return value of the form *TI t* indicates that the height will be the same. A value of the form *OF l p r* indicates an increase in height.

fun *full_i* :: $nat \Rightarrow 'a\ upI \Rightarrow bool$ **where**
full_i *n* (*TI t*) $\longleftrightarrow full\ n\ t$ |
full_i *n* (*OF l p r*) $\longleftrightarrow full\ n\ l \wedge full\ n\ r$

lemma *full_i_ins*: $full\ n\ t \implies full_i\ n\ (ins\ a\ t)$
by (*induct rule: full.induct*) (*auto split: upI.split*)

The *insert* operation preserves completeance.

lemma *complete_insert*: $complete\ t \implies complete\ (insert\ a\ t)$
unfolding *complete_iff_full insert_def*
apply (*erule exE*)
apply (*drule full_i_ins [of _ _ a]*)
apply (*cases ins a t*)
apply (*auto intro: full.intros*)
done

25.3 Proofs for delete

fun *hD* :: $'a\ upD \Rightarrow nat$ **where**
hD (*TD t*) = *height t* |
hD (*UF t*) = *height t + 1*

lemma *complete_treeD_node21*:
 $\llbracket complete\ r; complete\ (treeD\ l'); height\ r = hD\ l' \rrbracket \implies complete\ (treeD\ (node21\ l'\ a\ r))$
by(*induct l' a r rule: node21.induct*) *auto*

lemma *complete_treeD_node22*:

$\llbracket \text{complete}(\text{treeD } r'); \text{ complete } l; \text{ hD } r' = \text{height } l \rrbracket \implies \text{complete}(\text{treeD } (\text{node22 } l \ a \ r'))$
by(*induct l a r' rule: node22.induct*) *auto*

lemma *complete_treeD_node31:*

$\llbracket \text{complete}(\text{treeD } l'); \text{ complete } m; \text{ complete } r; \text{ hD } l' = \text{height } r; \text{ height } m = \text{height } r \rrbracket$
 $\implies \text{complete}(\text{treeD } (\text{node31 } l' \ a \ m \ b \ r))$
by(*induct l' a m b r rule: node31.induct*) *auto*

lemma *complete_treeD_node32:*

$\llbracket \text{complete } l; \text{ complete}(\text{treeD } m'); \text{ complete } r; \text{ height } l = \text{height } r; \text{ hD } m' = \text{height } r \rrbracket$
 $\implies \text{complete}(\text{treeD } (\text{node32 } l \ a \ m' \ b \ r))$
by(*induct l a m' b r rule: node32.induct*) *auto*

lemma *complete_treeD_node33:*

$\llbracket \text{complete } l; \text{ complete } m; \text{ complete}(\text{treeD } r'); \text{ height } l = \text{hD } r'; \text{ height } m = \text{hD } r' \rrbracket$
 $\implies \text{complete}(\text{treeD } (\text{node33 } l \ a \ m \ b \ r'))$
by(*induct l a m b r' rule: node33.induct*) *auto*

lemmas *completes = complete_treeD_node21 complete_treeD_node22 complete_treeD_node31 complete_treeD_node32 complete_treeD_node33*

lemma *height'_node21:*

$\text{height } r > 0 \implies \text{hD}(\text{node21 } l' \ a \ r) = \max(\text{hD } l') (\text{height } r) + 1$
by(*induct l' a r rule: node21.induct*)(*simp_all*)

lemma *height'_node22:*

$\text{height } l > 0 \implies \text{hD}(\text{node22 } l \ a \ r') = \max(\text{height } l) (\text{hD } r') + 1$
by(*induct l a r' rule: node22.induct*)(*simp_all*)

lemma *height'_node31:*

$\text{height } m > 0 \implies \text{hD}(\text{node31 } l \ a \ m \ b \ r) =$
 $\max(\text{hD } l) (\max(\text{height } m) (\text{height } r)) + 1$
by(*induct l a m b r rule: node31.induct*)(*simp_all add: max_def*)

lemma *height'_node32:*

$\text{height } r > 0 \implies \text{hD}(\text{node32 } l \ a \ m \ b \ r) =$
 $\max(\text{height } l) (\max(\text{hD } m) (\text{height } r)) + 1$
by(*induct l a m b r rule: node32.induct*)(*simp_all add: max_def*)

lemma *height'_node33:*

$height\ m > 0 \implies hD(node33\ l\ a\ m\ b\ r) =$
 $max\ (height\ l)\ (max\ (height\ m)\ (hD\ r)) + 1$
by(*induct l a m b r rule: node33.induct*)(*simp_all add: max_def*)

lemmas *heights = height'_node21 height'_node22*
height'_node31 height'_node32 height'_node33

lemma *height_split_min:*

$split_min\ t = (x, t') \implies height\ t > 0 \implies complete\ t \implies hD\ t' = height\ t$

by(*induct t arbitrary: x t' rule: split_min.induct*)
(auto simp: heights split: prod.splits)

lemma *height_del: complete t \implies hD(del x t) = height t*

by(*induction x t rule: del.induct*)
(auto simp: heights max_def height_split_min split: prod.splits)

lemma *complete_split_min:*

$\llbracket split_min\ t = (x, t'); complete\ t; height\ t > 0 \rrbracket \implies complete\ (treeD\ t')$

by(*induct t arbitrary: x t' rule: split_min.induct*)
(auto simp: heights height_split_min completes split: prod.splits)

lemma *complete_treeD_del: complete t \implies complete(treeD(del x t))*

by(*induction x t rule: del.induct*)
(auto simp: completes complete_split_min height_del height_split_min split: prod.splits)

corollary *complete_delete: complete t \implies complete(delete x t)*

by(*simp add: delete_def complete_treeD_del*)

25.4 Overall Correctness

interpretation *S: Set_by_Ordered*

where *empty = empty and isin = isin and insert = insert and delete = delete*

and *inorder = inorder and inv = complete*

proof (*standard, goal_cases*)

case 2 **thus** *?case by*(*simp add: isin_set*)

next

case 3 **thus** *?case by*(*simp add: inorder_insert*)

next

case 4 **thus** *?case by*(*simp add: inorder_delete*)

next

case 6 **thus** *?case by*(*simp add: complete_insert*)

```

next
  case 7 thus ?case by(simp add: complete_delete)
qed (simp add: empty_def)+

end

```

26 2-3 Tree Implementation of Maps

```

theory Tree23_Map

```

```

imports

```

```

  Tree23_Set

```

```

  Map_Specs

```

```

begin

```

```

fun lookup :: ('a::linorder * 'b) tree23 ⇒ 'a ⇒ 'b option where
lookup Leaf x = None |
lookup (Node2 l (a,b) r) x = (case cmp x a of
  LT ⇒ lookup l x |
  GT ⇒ lookup r x |
  EQ ⇒ Some b) |
lookup (Node3 l (a1,b1) m (a2,b2) r) x = (case cmp x a1 of
  LT ⇒ lookup l x |
  EQ ⇒ Some b1 |
  GT ⇒ (case cmp x a2 of
    LT ⇒ lookup m x |
    EQ ⇒ Some b2 |
    GT ⇒ lookup r x))

```

```

fun upd :: 'a::linorder ⇒ 'b ⇒ ('a*'b) tree23 ⇒ ('a*'b) upI where
upd x y Leaf = OF Leaf (x,y) Leaf |
upd x y (Node2 l ab r) = (case cmp x (fst ab) of
  LT ⇒ (case upd x y l of
    TI l' => TI (Node2 l' ab r)
    | OF l1 ab' l2 => TI (Node3 l1 ab' l2 ab r)) |
  EQ ⇒ TI (Node2 l (x,y) r) |
  GT ⇒ (case upd x y r of
    TI r' => TI (Node2 l ab r')
    | OF r1 ab' r2 => TI (Node3 l ab r1 ab' r2))) |
upd x y (Node3 l ab1 m ab2 r) = (case cmp x (fst ab1) of
  LT ⇒ (case upd x y l of
    TI l' => TI (Node3 l' ab1 m ab2 r)
    | OF l1 ab' l2 => OF (Node2 l1 ab' l2) ab1 (Node2 m ab2 r)) |
  EQ ⇒ TI (Node3 l (x,y) m ab2 r) |

```

$$\begin{aligned}
GT &\Rightarrow (\text{case cmp } x \text{ (fst ab2) of} \\
&\quad LT \Rightarrow (\text{case upd } x \text{ y m of} \\
&\quad\quad TI \text{ m}' \Rightarrow TI \text{ (Node3 l ab1 m' ab2 r)} \\
&\quad\quad | OF \text{ m1 ab' m2} \Rightarrow OF \text{ (Node2 l ab1 m1) ab' (Node2 m2} \\
&\text{ab2 r)}) | \\
&\quad EQ \Rightarrow TI \text{ (Node3 l ab1 m (x,y) r)} | \\
&\quad GT \Rightarrow (\text{case upd } x \text{ y r of} \\
&\quad\quad TI \text{ r}' \Rightarrow TI \text{ (Node3 l ab1 m ab2 r')} \\
&\quad\quad | OF \text{ r1 ab' r2} \Rightarrow OF \text{ (Node2 l ab1 m) ab2 (Node2 r1 ab' r2)}))
\end{aligned}$$

definition *update* :: 'a::linorder \Rightarrow 'b \Rightarrow ('a*'b) tree23 \Rightarrow ('a*'b) tree23
where

update a b t = treeI(upd a b t)

fun *del* :: 'a::linorder \Rightarrow ('a*'b) tree23 \Rightarrow ('a*'b) upD **where**

del x Leaf = TD Leaf |

del x (Node2 Leaf ab1 Leaf) = (if x=fst ab1 then UF Leaf else TD(Node2 Leaf ab1 Leaf)) |

del x (Node3 Leaf ab1 Leaf ab2 Leaf) = TD(if x=fst ab1 then Node2 Leaf ab2 Leaf

else if x=fst ab2 then Node2 Leaf ab1 Leaf else Node3 Leaf ab1 Leaf ab2 Leaf) |

del x (Node2 l ab1 r) = (case cmp x (fst ab1) of

LT \Rightarrow node21 (del x l) ab1 r |

GT \Rightarrow node22 l ab1 (del x r) |

EQ \Rightarrow let (ab1',t) = split_min r in node22 l ab1' t) |

del x (Node3 l ab1 m ab2 r) = (case cmp x (fst ab1) of

LT \Rightarrow node31 (del x l) ab1 m ab2 r |

EQ \Rightarrow let (ab1',m') = split_min m in node32 l ab1' m' ab2 r |

GT \Rightarrow (case cmp x (fst ab2) of

LT \Rightarrow node32 l ab1 (del x m) ab2 r |

EQ \Rightarrow let (ab2',r') = split_min r in node33 l ab1 m ab2' r' |

GT \Rightarrow node33 l ab1 m ab2 (del x r))

definition *delete* :: 'a::linorder \Rightarrow ('a*'b) tree23 \Rightarrow ('a*'b) tree23 **where**
delete x t = treeD(del x t)

26.1 Functional Correctness

lemma *lookup_map_of*:

sorted1(inorder t) \Longrightarrow lookup t x = map_of (inorder t) x

by (*induction t*) (*auto simp: map_of_simps split: option.split*)

lemma *inorder_upd*:

$sorted1(inorder\ t) \implies inorder(treeI(upd\ x\ y\ t)) = upd_list\ x\ y\ (inorder\ t)$

by(*induction* *t*) (*auto simp: upd_list_simps split: upI.splits*)

corollary *inorder_update*:

$sorted1(inorder\ t) \implies inorder(update\ x\ y\ t) = upd_list\ x\ y\ (inorder\ t)$

by(*simp add: update_def inorder_upd*)

lemma *inorder_del*: $\llbracket complete\ t ; sorted1(inorder\ t) \rrbracket \implies$

$inorder(treeD\ (del\ x\ t)) = del_list\ x\ (inorder\ t)$

by(*induction* *t* *rule: del.induct*)

(*auto simp: del_list_simps inorder_nodes split_minD split!: if_split prod.splits*)

corollary *inorder_delete*: $\llbracket complete\ t ; sorted1(inorder\ t) \rrbracket \implies$

$inorder(delete\ x\ t) = del_list\ x\ (inorder\ t)$

by(*simp add: delete_def inorder_del*)

26.2 Balancedness

lemma *complete_upd*: $complete\ t \implies complete\ (treeI(upd\ x\ y\ t)) \wedge hI(upd\ x\ y\ t) = height\ t$

by (*induct* *t*) (*auto split!: if_split upI.split*)

corollary *complete_update*: $complete\ t \implies complete\ (update\ x\ y\ t)$

by (*simp add: update_def complete_upd*)

lemma *height_del*: $complete\ t \implies hD(del\ x\ t) = height\ t$

by(*induction* *x t* *rule: del.induct*)

(*auto simp add: heights_max_def height_split_min split: prod.split*)

lemma *complete_treeD_del*: $complete\ t \implies complete(treeD(del\ x\ t))$

by(*induction* *x t* *rule: del.induct*)

(*auto simp: completes_complete_split_min height_del height_split_min split: prod.split*)

corollary *complete_delete*: $complete\ t \implies complete(delete\ x\ t)$

by(*simp add: delete_def complete_treeD_del*)

26.3 Overall Correctness

```
interpretation M: Map_by_Ordered
where empty = empty and lookup = lookup and update = update and
delete = delete
and inorder = inorder and inv = complete
proof (standard, goal_cases)
  case 1 thus ?case by(simp add: empty_def)
next
  case 2 thus ?case by(simp add: lookup_map_of)
next
  case 3 thus ?case by(simp add: inorder_update)
next
  case 4 thus ?case by(simp add: inorder_delete)
next
  case 5 thus ?case by(simp add: empty_def)
next
  case 6 thus ?case by(simp add: complete_update)
next
  case 7 thus ?case by(simp add: complete_delete)
qed

end
```

27 2-3 Tree from List

```
theory Tree23_of_List
imports Tree23
begin
```

Linear-time bottom up conversion of a list of items into a complete 2-3 tree whose inorder traversal yields the list of items.

27.1 Code

Nonempty lists of 2-3 trees alternating with items, starting and ending with a 2-3 tree:

```
datatype 'a tree23s = T 'a tree23 | TTs 'a tree23 'a 'a tree23s
```

```
abbreviation not_T ts == ( $\forall t. ts \neq T t$ )
```

```
fun len :: 'a tree23s  $\Rightarrow$  nat where
  len (T _) = 1 |
  len (TTs _ _ ts) = len ts + 1
```

fun *trees* :: 'a tree23s \Rightarrow 'a tree23 set **where**
trees (T t) = {t} |
trees (TTs t a ts) = {t} \cup *trees* ts

Join pairs of adjacent trees:

fun *join_adj* :: 'a tree23s \Rightarrow 'a tree23s **where**
join_adj (TTs t1 a (T t2)) = T(Node2 t1 a t2) |
join_adj (TTs t1 a (TTs t2 b (T t3))) = T(Node3 t1 a t2 b t3) |
join_adj (TTs t1 a (TTs t2 b ts)) = TTs (Node2 t1 a t2) b (*join_adj* ts)

Towards termination of *join_all*:

lemma *len_ge2*:
 $not_T\ ts \Longrightarrow len\ ts \geq 2$
by(cases ts rule: *join_adj.cases*) auto

lemma [*measure_function*]: *is_measure len*
by(rule *is_measure_trivial*)

lemma *len_join_adj_div2*:
 $not_T\ ts \Longrightarrow len(join_adj\ ts) \leq len\ ts\ div\ 2$
by(induction ts rule: *join_adj.induct*) auto

lemma *len_join_adj1*: $not_T\ ts \Longrightarrow len(join_adj\ ts) < len\ ts$
using *len_join_adj_div2*[of ts] *len_ge2*[of ts] **by** simp

corollary *len_join_adj2*[*termination_simp*]: $len(join_adj\ (TTs\ t\ a\ ts)) \leq len\ ts$
using *len_join_adj1*[of TTs t a ts] **by** simp

fun *join_all* :: 'a tree23s \Rightarrow 'a tree23 **where**
join_all (T t) = t |
join_all ts = *join_all* (*join_adj* ts)

fun *leaves* :: 'a list \Rightarrow 'a tree23s **where**
leaves [] = T Leaf |
leaves (a # as) = TTs Leaf a (*leaves* as)

definition *tree23_of_list* :: 'a list \Rightarrow 'a tree23 **where**
tree23_of_list as = *join_all*(*leaves* as)

27.2 Functional correctness

27.2.1 *inorder*:

fun *inorder2* :: 'a tree23s \Rightarrow 'a list **where**
inorder2 (T t) = *inorder* t |
inorder2 (TTs t a ts) = *inorder* t @ a # *inorder2* ts

lemma *inorder2_join_adj*: *not_T* ts \Longrightarrow *inorder2*(*join_adj* ts) = *inorder2* ts
by (*induction* ts *rule*: *join_adj.induct*) *auto*

lemma *inorder_join_all*: *inorder* (*join_all* ts) = *inorder2* ts
proof (*induction* ts *rule*: *join_all.induct*)
 case 1 **thus** ?*case* **by** *simp*
next
 case (2 t a ts)
 thus ?*case* **using** *inorder2_join_adj*[*of* TTs t a ts]
 by (*simp* *add*: *le_imp_less_Suc*)
qed

lemma *inorder2_leaves*: *inorder2*(*leaves* as) = as
by(*induction* as) *auto*

lemma *inorder*: *inorder*(*tree23_of_list* as) = as
by(*simp* *add*: *tree23_of_list_def* *inorder_join_all* *inorder2_leaves*)

27.2.2 Completeness:

lemma *complete_join_adj*:
 $\forall t \in \text{trees } ts. \text{complete } t \wedge \text{height } t = n \Longrightarrow \text{not_T } ts \Longrightarrow$
 $\forall t \in \text{trees } (\text{join_adj } ts). \text{complete } t \wedge \text{height } t = \text{Suc } n$
by (*induction* ts *rule*: *join_adj.induct*) *auto*

lemma *complete_join_all*:
 $\forall t \in \text{trees } ts. \text{complete } t \wedge \text{height } t = n \Longrightarrow \text{complete } (\text{join_all } ts)$
proof (*induction* ts *arbitrary*: n *rule*: *join_all.induct*)
 case 1 **thus** ?*case* **by** *simp*
next
 case (2 t a ts)
 thus ?*case*
 apply *simp* **using** *complete_join_adj*[*of* TTs t a ts n, *simplified*] **by**
blast
qed

lemma *complete_leaves*: $t \in \text{trees } (\text{leaves } as) \implies \text{complete } t \wedge \text{height } t = 0$
by (*induction as*) *auto*

corollary *complete*: $\text{complete}(\text{tree23_of_list } as)$
by(*simp add: tree23_of_list_def complete_leaves complete_join_all[of _ 0]*)

27.3 Linear running time

fun *T_join_adj* :: 'a tree23s \Rightarrow nat **where**
T_join_adj (TTs t1 a (T t2)) = 1 |
T_join_adj (TTs t1 a (TTs t2 b (T t3))) = 1 |
T_join_adj (TTs t1 a (TTs t2 b ts)) = *T_join_adj* ts + 1

fun *T_join_all* :: 'a tree23s \Rightarrow nat **where**
T_join_all (T t) = 1 |
T_join_all ts = *T_join_adj* ts + *T_join_all* (join_adj ts) + 1

fun *T_leaves* :: 'a list \Rightarrow nat **where**
T_leaves [] = 1 |
T_leaves (a # as) = *T_leaves* as + 1

definition *T_tree23_of_list* :: 'a list \Rightarrow nat **where**
T_tree23_of_list as = *T_leaves* as + *T_join_all*(leaves as) + 1

lemma *T_join_adj*: $\text{not_T } ts \implies \text{T_join_adj } ts \leq \text{len } ts \text{ div } 2$
by(*induction ts rule: T_join_adj.induct*) *auto*

lemma *len_ge_1*: $\text{len } ts \geq 1$
by(*cases ts*) *auto*

lemma *T_join_all*: $\text{T_join_all } ts \leq 2 * \text{len } ts$
proof(*induction ts rule: join_all.induct*)

case 1 **thus** ?*case* **by** *simp*
next
 case (2 t a ts)
 let ?ts = TTs t a ts
 have $\text{T_join_all } ?ts = \text{T_join_adj } ?ts + \text{T_join_all } (\text{join_adj } ?ts) + 1$
 by *simp*
 also have $\dots \leq \text{len } ?ts \text{ div } 2 + \text{T_join_all } (\text{join_adj } ?ts) + 1$
 using *T_join_adj[of ?ts]* **by** *simp*
 also have $\dots \leq \text{len } ?ts \text{ div } 2 + 2 * \text{len } (\text{join_adj } ?ts) + 1$

```

    using 2.IH by simp
    also have ... ≤ len ?ts div 2 + 2 * (len ?ts div 2) + 1
      using len_join_adj_div2[of ?ts] by simp
    also have ... ≤ 2 * len ?ts using len_ge_1[of ?ts] by linarith
    finally show ?case .
qed

```

```

lemma T_leaves: T_leaves as = length as + 1
by(induction as) auto

```

```

lemma len_leaves: len(leaves as) = length as + 1
by(induction as) auto

```

```

lemma T_tree23_of_list: T_tree23_of_list as ≤ 3*(length as) + 4
using T_join_all[of leaves as] by(simp add: T_tree23_of_list_def T_leaves
len_leaves)

```

end

28 2-3-4 Trees

```

theory Tree234
imports Main
begin

```

```

class height =
fixes height :: 'a ⇒ nat

```

```

datatype 'a tree234 =
  Leaf (⟨⟩) |
  Node2 'a tree234 'a 'a tree234 (⟨_, _, _⟩) |
  Node3 'a tree234 'a 'a tree234 'a 'a tree234 (⟨_, _, _, _, _⟩) |
  Node4 'a tree234 'a 'a tree234 'a 'a tree234 'a 'a tree234
  (⟨_, _, _, _, _, _⟩)

```

```

fun inorder :: 'a tree234 ⇒ 'a list where
inorder Leaf = [] |
inorder(Node2 l a r) = inorder l @ a # inorder r |
inorder(Node3 l a m b r) = inorder l @ a # inorder m @ b # inorder r |
inorder(Node4 l a m b n c r) = inorder l @ a # inorder m @ b # inorder
n @ c # inorder r

```

```

instantiation tree234 :: (type)height
begin

fun height_tree234 :: 'a tree234 ⇒ nat where
height Leaf = 0 |
height (Node2 l _ r) = Suc(max (height l) (height r)) |
height (Node3 l _ m _ r) = Suc(max (height l) (max (height m) (height
r))) |
height (Node4 l _ m _ n _ r) = Suc(max (height l) (max (height m) (max
(height n) (height r))))

instance ..

end

    Balanced:

fun bal :: 'a tree234 ⇒ bool where
bal Leaf = True |
bal (Node2 l _ r) = (bal l & bal r & height l = height r) |
bal (Node3 l _ m _ r) = (bal l & bal m & bal r & height l = height m &
height m = height r) |
bal (Node4 l _ m _ n _ r) = (bal l & bal m & bal n & bal r & height l =
height m & height m = height n & height n = height r)

end

```

29 2-3-4 Tree Implementation of Sets

```

theory Tree234_Set

```

```

imports

```

```

    Tree234

```

```

    Cmp

```

```

    Set_Specs

```

```

begin

```

```

declare sorted_wrt.simps(2)[simp del]

```

29.1 Set operations on 2-3-4 trees

```

definition empty :: 'a tree234 where

```

```

empty = Leaf

```

```

fun isin :: 'a::linorder tree234 ⇒ 'a ⇒ bool where

```

```

isin Leaf x = False |

```

```

isin (Node2 l a r) x =
  (case cmp x a of LT => isin l x | EQ => True | GT => isin r x) |
isin (Node3 l a m b r) x =
  (case cmp x a of LT => isin l x | EQ => True | GT => (case cmp x b of
    LT => isin m x | EQ => True | GT => isin r x)) |
isin (Node4 t1 a t2 b t3 c t4) x =
  (case cmp x b of
    LT =>
      (case cmp x a of
        LT => isin t1 x |
        EQ => True |
        GT => isin t2 x) |
    EQ => True |
    GT =>
      (case cmp x c of
        LT => isin t3 x |
        EQ => True |
        GT => isin t4 x))

```

datatype 'a up_i = T_i 'a tree₂₃₄ | Up_i 'a tree₂₃₄ 'a 'a tree₂₃₄

```

fun treei :: 'a upi => 'a tree234 where
treei (Ti t) = t |
treei (Upi l a r) = Node2 l a r

```

```

fun ins :: 'a::linorder => 'a tree234 => 'a upi where
ins x Leaf = Upi Leaf x Leaf |
ins x (Node2 l a r) =
  (case cmp x a of
    LT => (case ins x l of
      Ti l' => Ti (Node2 l' a r)
      | Upi l1 b l2 => Ti (Node3 l1 b l2 a r)) |
    EQ => Ti (Node2 l x r) |
    GT => (case ins x r of
      Ti r' => Ti (Node2 l a r')
      | Upi r1 b r2 => Ti (Node3 l a r1 b r2))) |
ins x (Node3 l a m b r) =
  (case cmp x a of
    LT => (case ins x l of
      Ti l' => Ti (Node3 l' a m b r)
      | Upi l1 c l2 => Upi (Node2 l1 c l2) a (Node2 m b r)) |
    EQ => Ti (Node3 l a m b r) |
    GT => (case cmp x b of
      GT => (case ins x r of

```

$$\begin{aligned}
& T_i r' \Rightarrow T_i (\text{Node3 } l \ a \ m \ b \ r') \\
& \quad | \text{Up}_i \ r1 \ c \ r2 \Rightarrow \text{Up}_i (\text{Node2 } l \ a \ m) \ b \ (\text{Node2 } r1 \ c \ r2)) \ | \\
& EQ \Rightarrow T_i (\text{Node3 } l \ a \ m \ b \ r) \ | \\
& LT \Rightarrow (\text{case ins } x \ m \ \text{of} \\
& \quad T_i \ m' \Rightarrow T_i (\text{Node3 } l \ a \ m' \ b \ r) \\
& \quad | \text{Up}_i \ m1 \ c \ m2 \Rightarrow \text{Up}_i (\text{Node2 } l \ a \ m1) \ c \ (\text{Node2 } m2 \ b \\
& r)))) \ | \\
& \text{ins } x \ (\text{Node4 } t1 \ a \ t2 \ b \ t3 \ c \ t4) = \\
& \quad (\text{case cmp } x \ b \ \text{of} \\
& \quad \quad LT \Rightarrow \\
& \quad \quad \quad (\text{case cmp } x \ a \ \text{of} \\
& \quad \quad \quad \quad LT \Rightarrow \\
& \quad \quad \quad \quad \quad (\text{case ins } x \ t1 \ \text{of} \\
& \quad \quad \quad \quad \quad \quad T_i \ t \Rightarrow T_i (\text{Node4 } t \ a \ t2 \ b \ t3 \ c \ t4) \ | \\
& \quad \quad \quad \quad \quad \quad \text{Up}_i \ l \ y \ r \Rightarrow \text{Up}_i (\text{Node2 } l \ y \ r) \ a \ (\text{Node3 } t2 \ b \ t3 \ c \ t4)) \ | \\
& \quad \quad \quad \quad EQ \Rightarrow T_i (\text{Node4 } t1 \ a \ t2 \ b \ t3 \ c \ t4) \ | \\
& \quad \quad \quad \quad GT \Rightarrow \\
& \quad \quad \quad \quad \quad (\text{case ins } x \ t2 \ \text{of} \\
& \quad \quad \quad \quad \quad \quad T_i \ t \Rightarrow T_i (\text{Node4 } t1 \ a \ t \ b \ t3 \ c \ t4) \ | \\
& \quad \quad \quad \quad \quad \quad \text{Up}_i \ l \ y \ r \Rightarrow \text{Up}_i (\text{Node2 } t1 \ a \ l) \ y \ (\text{Node3 } r \ b \ t3 \ c \ t4))) \ | \\
& \quad \quad \quad EQ \Rightarrow T_i (\text{Node4 } t1 \ a \ t2 \ b \ t3 \ c \ t4) \ | \\
& \quad \quad \quad GT \Rightarrow \\
& \quad \quad \quad \quad (\text{case cmp } x \ c \ \text{of} \\
& \quad \quad \quad \quad \quad LT \Rightarrow \\
& \quad \quad \quad \quad \quad \quad (\text{case ins } x \ t3 \ \text{of} \\
& \quad \quad \quad \quad \quad \quad \quad T_i \ t \Rightarrow T_i (\text{Node4 } t1 \ a \ t2 \ b \ t \ c \ t4) \ | \\
& \quad \quad \quad \quad \quad \quad \quad \text{Up}_i \ l \ y \ r \Rightarrow \text{Up}_i (\text{Node2 } t1 \ a \ t2) \ b \ (\text{Node3 } l \ y \ r \ c \ t4)) \ | \\
& \quad \quad \quad \quad \quad EQ \Rightarrow T_i (\text{Node4 } t1 \ a \ t2 \ b \ t3 \ c \ t4) \ | \\
& \quad \quad \quad \quad \quad GT \Rightarrow \\
& \quad \quad \quad \quad \quad \quad (\text{case ins } x \ t4 \ \text{of} \\
& \quad \quad \quad \quad \quad \quad \quad T_i \ t \Rightarrow T_i (\text{Node4 } t1 \ a \ t2 \ b \ t3 \ c \ t) \ | \\
& \quad \quad \quad \quad \quad \quad \quad \text{Up}_i \ l \ y \ r \Rightarrow \text{Up}_i (\text{Node2 } t1 \ a \ t2) \ b \ (\text{Node3 } t3 \ c \ l \ y \ r))))
\end{aligned}$$

hide_const *insert*

definition *insert* :: 'a::linorder \Rightarrow 'a tree234 \Rightarrow 'a tree234 **where**
insert *x* *t* = tree_i(*ins* *x* *t*)

datatype 'a up_d = T_d 'a tree234 | Up_d 'a tree234

fun tree_d :: 'a up_d \Rightarrow 'a tree234 **where**
tree_d (T_d *t*) = *t* |
tree_d (Up_d *t*) = *t*

```

fun node21 :: 'a upd ⇒ 'a ⇒ 'a tree234 ⇒ 'a upd where
node21 (Td l) a r = Td(Node2 l a r) |
node21 (Upd l) a (Node2 lr b rr) = Upd(Node3 l a lr b rr) |
node21 (Upd l) a (Node3 lr b mr c rr) = Td(Node2 (Node2 l a lr) b (Node2
mr c rr)) |
node21 (Upd t1) a (Node4 t2 b t3 c t4 d t5) = Td(Node2 (Node2 t1 a t2)
b (Node3 t3 c t4 d t5))

```

```

fun node22 :: 'a tree234 ⇒ 'a ⇒ 'a upd ⇒ 'a upd where
node22 l a (Td r) = Td(Node2 l a r) |
node22 (Node2 ll b rl) a (Upd r) = Upd(Node3 ll b rl a r) |
node22 (Node3 ll b ml c rl) a (Upd r) = Td(Node2 (Node2 ll b ml) c (Node2
rl a r)) |
node22 (Node4 t1 a t2 b t3 c t4) d (Upd t5) = Td(Node2 (Node2 t1 a t2)
b (Node3 t3 c t4 d t5))

```

```

fun node31 :: 'a upd ⇒ 'a ⇒ 'a tree234 ⇒ 'a ⇒ 'a tree234 ⇒ 'a upd where
node31 (Td t1) a t2 b t3 = Td(Node3 t1 a t2 b t3) |
node31 (Upd t1) a (Node2 t2 b t3) c t4 = Td(Node2 (Node3 t1 a t2 b t3)
c t4) |
node31 (Upd t1) a (Node3 t2 b t3 c t4) d t5 = Td(Node3 (Node2 t1 a t2)
b (Node2 t3 c t4) d t5) |
node31 (Upd t1) a (Node4 t2 b t3 c t4 d t5) e t6 = Td(Node3 (Node2 t1 a
t2) b (Node3 t3 c t4 d t5) e t6)

```

```

fun node32 :: 'a tree234 ⇒ 'a ⇒ 'a upd ⇒ 'a ⇒ 'a tree234 ⇒ 'a upd where
node32 t1 a (Td t2) b t3 = Td(Node3 t1 a t2 b t3) |
node32 t1 a (Upd t2) b (Node2 t3 c t4) = Td(Node2 t1 a (Node3 t2 b t3 c
t4)) |
node32 t1 a (Upd t2) b (Node3 t3 c t4 d t5) = Td(Node3 t1 a (Node2 t2 b
t3) c (Node2 t4 d t5)) |
node32 t1 a (Upd t2) b (Node4 t3 c t4 d t5 e t6) = Td(Node3 t1 a (Node2
t2 b t3) c (Node3 t4 d t5 e t6))

```

```

fun node33 :: 'a tree234 ⇒ 'a ⇒ 'a tree234 ⇒ 'a ⇒ 'a upd ⇒ 'a upd where
node33 l a m b (Td r) = Td(Node3 l a m b r) |
node33 t1 a (Node2 t2 b t3) c (Upd t4) = Td(Node2 t1 a (Node3 t2 b t3 c
t4)) |
node33 t1 a (Node3 t2 b t3 c t4) d (Upd t5) = Td(Node3 t1 a (Node2 t2 b
t3) c (Node2 t4 d t5)) |
node33 t1 a (Node4 t2 b t3 c t4 d t5) e (Upd t6) = Td(Node3 t1 a (Node2
t2 b t3) c (Node3 t4 d t5 e t6))

```

```

fun node41 :: 'a upd ⇒ 'a ⇒ 'a tree234 ⇒ 'a ⇒ 'a tree234 ⇒ 'a ⇒ 'a

```

```

tree234 ⇒ 'a up_d where
node41 (T_d t1) a t2 b t3 c t4 = T_d(Node4 t1 a t2 b t3 c t4) |
node41 (Up_d t1) a (Node2 t2 b t3) c t4 d t5 = T_d(Node3 (Node3 t1 a t2 b
t3) c t4 d t5) |
node41 (Up_d t1) a (Node3 t2 b t3 c t4) d t5 e t6 = T_d(Node4 (Node2 t1 a
t2) b (Node2 t3 c t4) d t5 e t6) |
node41 (Up_d t1) a (Node4 t2 b t3 c t4 d t5) e t6 f t7 = T_d(Node4 (Node2
t1 a t2) b (Node3 t3 c t4 d t5) e t6 f t7)

```

```

fun node42 :: 'a tree234 ⇒ 'a ⇒ 'a up_d ⇒ 'a ⇒ 'a tree234 ⇒ 'a ⇒ 'a
tree234 ⇒ 'a up_d where
node42 t1 a (T_d t2) b t3 c t4 = T_d(Node4 t1 a t2 b t3 c t4) |
node42 (Node2 t1 a t2) b (Up_d t3) c t4 d t5 = T_d(Node3 (Node3 t1 a t2 b
t3) c t4 d t5) |
node42 (Node3 t1 a t2 b t3) c (Up_d t4) d t5 e t6 = T_d(Node4 (Node2 t1 a
t2) b (Node2 t3 c t4) d t5 e t6) |
node42 (Node4 t1 a t2 b t3 c t4) d (Up_d t5) e t6 f t7 = T_d(Node4 (Node2
t1 a t2) b (Node3 t3 c t4 d t5) e t6 f t7)

```

```

fun node43 :: 'a tree234 ⇒ 'a ⇒ 'a tree234 ⇒ 'a ⇒ 'a up_d ⇒ 'a ⇒ 'a
tree234 ⇒ 'a up_d where
node43 t1 a t2 b (T_d t3) c t4 = T_d(Node4 t1 a t2 b t3 c t4) |
node43 t1 a (Node2 t2 b t3) c (Up_d t4) d t5 = T_d(Node3 t1 a (Node3 t2 b
t3 c t4) d t5) |
node43 t1 a (Node3 t2 b t3 c t4) d (Up_d t5) e t6 = T_d(Node4 t1 a (Node2
t2 b t3) c (Node2 t4 d t5) e t6) |
node43 t1 a (Node4 t2 b t3 c t4 d t5) e (Up_d t6) f t7 = T_d(Node4 t1 a
(Node2 t2 b t3) c (Node3 t4 d t5 e t6) f t7)

```

```

fun node44 :: 'a tree234 ⇒ 'a ⇒ 'a tree234 ⇒ 'a ⇒ 'a tree234 ⇒ 'a ⇒ 'a
up_d ⇒ 'a up_d where
node44 t1 a t2 b t3 c (T_d t4) = T_d(Node4 t1 a t2 b t3 c t4) |
node44 t1 a t2 b (Node2 t3 c t4) d (Up_d t5) = T_d(Node3 t1 a t2 b (Node3
t3 c t4 d t5)) |
node44 t1 a t2 b (Node3 t3 c t4 d t5) e (Up_d t6) = T_d(Node4 t1 a t2 b
(Node2 t3 c t4) d (Node2 t5 e t6)) |
node44 t1 a t2 b (Node4 t3 c t4 d t5 e t6) f (Up_d t7) = T_d(Node4 t1 a t2
b (Node2 t3 c t4) d (Node3 t5 e t6 f t7))

```

```

fun split_min :: 'a tree234 ⇒ 'a * 'a up_d where
split_min (Node2 Leaf a Leaf) = (a, Up_d Leaf) |
split_min (Node3 Leaf a Leaf b Leaf) = (a, T_d(Node2 Leaf b Leaf)) |
split_min (Node4 Leaf a Leaf b Leaf c Leaf) = (a, T_d(Node3 Leaf b Leaf c
Leaf)) |

```

$split_min (Node2\ l\ a\ r) = (let\ (x,l') = split_min\ l\ in\ (x,\ node21\ l'\ a\ r))\ |\$
 $split_min (Node3\ l\ a\ m\ b\ r) = (let\ (x,l') = split_min\ l\ in\ (x,\ node31\ l'\ a\ m\ b\ r))\ |\$
 $split_min (Node4\ l\ a\ m\ b\ n\ c\ r) = (let\ (x,l') = split_min\ l\ in\ (x,\ node41\ l'\ a\ m\ b\ n\ c\ r))$

fun $del :: 'a::linorder \Rightarrow 'a\ tree234 \Rightarrow 'a\ up_d$ **where**

$del\ k\ Leaf = T_d\ Leaf\ |\$

$del\ k\ (Node2\ Leaf\ p\ Leaf) = (if\ k=p\ then\ Up_d\ Leaf\ else\ T_d(Node2\ Leaf\ p\ Leaf))\ |\$

$del\ k\ (Node3\ Leaf\ p\ Leaf\ q\ Leaf) = T_d(if\ k=p\ then\ Node2\ Leaf\ q\ Leaf\ else\ if\ k=q\ then\ Node2\ Leaf\ p\ Leaf\ else\ Node3\ Leaf\ p\ Leaf\ q\ Leaf)\ |\$

$del\ k\ (Node4\ Leaf\ a\ Leaf\ b\ Leaf\ c\ Leaf) =$
 $T_d(if\ k=a\ then\ Node3\ Leaf\ b\ Leaf\ c\ Leaf\ else$
 $\quad if\ k=b\ then\ Node3\ Leaf\ a\ Leaf\ c\ Leaf\ else$
 $\quad if\ k=c\ then\ Node3\ Leaf\ a\ Leaf\ b\ Leaf$
 $\quad else\ Node4\ Leaf\ a\ Leaf\ b\ Leaf\ c\ Leaf)\ |\$

$del\ k\ (Node2\ l\ a\ r) = (case\ cmp\ k\ a\ of$
 $\quad LT \Rightarrow node21\ (del\ k\ l)\ a\ r\ |\$
 $\quad GT \Rightarrow node22\ l\ a\ (del\ k\ r)\ |\$
 $\quad EQ \Rightarrow let\ (a',t) = split_min\ r\ in\ node22\ l\ a'\ t)\ |\$

$del\ k\ (Node3\ l\ a\ m\ b\ r) = (case\ cmp\ k\ a\ of$
 $\quad LT \Rightarrow node31\ (del\ k\ l)\ a\ m\ b\ r\ |\$
 $\quad EQ \Rightarrow let\ (a',m') = split_min\ m\ in\ node32\ l\ a'\ m'\ b\ r\ |\$
 $\quad GT \Rightarrow (case\ cmp\ k\ b\ of$
 $\quad\quad LT \Rightarrow node32\ l\ a\ (del\ k\ m)\ b\ r\ |\$
 $\quad\quad EQ \Rightarrow let\ (b',r') = split_min\ r\ in\ node33\ l\ a\ m\ b'\ r'\ |\$
 $\quad\quad GT \Rightarrow node33\ l\ a\ m\ b\ (del\ k\ r))\ |\$

$del\ k\ (Node4\ l\ a\ m\ b\ n\ c\ r) = (case\ cmp\ k\ b\ of$
 $\quad LT \Rightarrow (case\ cmp\ k\ a\ of$
 $\quad\quad LT \Rightarrow node41\ (del\ k\ l)\ a\ m\ b\ n\ c\ r\ |\$
 $\quad\quad EQ \Rightarrow let\ (a',m') = split_min\ m\ in\ node42\ l\ a'\ m'\ b\ n\ c\ r\ |\$
 $\quad\quad GT \Rightarrow node42\ l\ a\ (del\ k\ m)\ b\ n\ c\ r)\ |\$
 $\quad EQ \Rightarrow let\ (b',n') = split_min\ n\ in\ node43\ l\ a\ m\ b'\ n'\ c\ r\ |\$
 $\quad GT \Rightarrow (case\ cmp\ k\ c\ of$
 $\quad\quad LT \Rightarrow node43\ l\ a\ m\ b\ (del\ k\ n)\ c\ r\ |\$
 $\quad\quad EQ \Rightarrow let\ (c',r') = split_min\ r\ in\ node44\ l\ a\ m\ b\ n\ c'\ r'\ |\$
 $\quad\quad GT \Rightarrow node44\ l\ a\ m\ b\ n\ c\ (del\ k\ r))$

definition $delete :: 'a::linorder \Rightarrow 'a\ tree234 \Rightarrow 'a\ tree234$ **where**

$delete\ x\ t = tree_d(del\ x\ t)$

29.2 Functional correctness

29.2.1 Functional correctness of `isin`:

lemma *isin_set*: $\text{sorted}(\text{inorder } t) \implies \text{isin } t \ x = (x \in \text{set } (\text{inorder } t))$
by (*induction t*) (*auto simp: isin_simps*)

29.2.2 Functional correctness of `insert`:

lemma *inorder_ins*:
 $\text{sorted}(\text{inorder } t) \implies \text{inorder}(\text{tree}_i(\text{ins } x \ t)) = \text{ins_list } x \ (\text{inorder } t)$
by(*induction t*) (*auto, auto simp: ins_list_simps split!: if_splits up_i.splits*)

lemma *inorder_insert*:
 $\text{sorted}(\text{inorder } t) \implies \text{inorder}(\text{insert } a \ t) = \text{ins_list } a \ (\text{inorder } t)$
by(*simp add: insert_def inorder_ins*)

29.2.3 Functional correctness of `delete`

lemma *inorder_node21*: $\text{height } r > 0 \implies$
 $\text{inorder } (\text{tree}_d(\text{node21 } l' \ a \ r)) = \text{inorder } (\text{tree}_d \ l') \ @ \ a \ \# \ \text{inorder } r$
by(*induct l' a r rule: node21.induct*) *auto*

lemma *inorder_node22*: $\text{height } l > 0 \implies$
 $\text{inorder } (\text{tree}_d(\text{node22 } l \ a \ r')) = \text{inorder } l \ @ \ a \ \# \ \text{inorder } (\text{tree}_d \ r')$
by(*induct l a r' rule: node22.induct*) *auto*

lemma *inorder_node31*: $\text{height } m > 0 \implies$
 $\text{inorder } (\text{tree}_d(\text{node31 } l' \ a \ m \ b \ r)) = \text{inorder } (\text{tree}_d \ l') \ @ \ a \ \# \ \text{inorder } m$
 $\ @ \ b \ \# \ \text{inorder } r$
by(*induct l' a m b r rule: node31.induct*) *auto*

lemma *inorder_node32*: $\text{height } r > 0 \implies$
 $\text{inorder } (\text{tree}_d(\text{node32 } l \ a \ m' \ b \ r)) = \text{inorder } l \ @ \ a \ \# \ \text{inorder } (\text{tree}_d \ m')$
 $\ @ \ b \ \# \ \text{inorder } r$
by(*induct l a m' b r rule: node32.induct*) *auto*

lemma *inorder_node33*: $\text{height } m > 0 \implies$
 $\text{inorder } (\text{tree}_d(\text{node33 } l \ a \ m \ b \ r')) = \text{inorder } l \ @ \ a \ \# \ \text{inorder } m \ @ \ b \ \#$
 $\ \text{inorder } (\text{tree}_d \ r')$
by(*induct l a m b r' rule: node33.induct*) *auto*

lemma *inorder_node41*: $\text{height } m > 0 \implies$
 $\text{inorder } (\text{tree}_d(\text{node41 } l' \ a \ m \ b \ n \ c \ r)) = \text{inorder } (\text{tree}_d \ l') \ @ \ a \ \# \ \text{inorder}$
 $\ m \ @ \ b \ \# \ \text{inorder } n \ @ \ c \ \# \ \text{inorder } r$

by(*induct l' a m b n c r rule: node41.induct*) *auto*

lemma *inorder_node42: height l > 0 \implies*

inorder (tree_d (node42 l a m b n c r)) = inorder l @ a # inorder (tree_d m) @ b # inorder n @ c # inorder r

by(*induct l a m b n c r rule: node42.induct*) *auto*

lemma *inorder_node43: height m > 0 \implies*

inorder (tree_d (node43 l a m b n c r)) = inorder l @ a # inorder m @ b # inorder (tree_d n) @ c # inorder r

by(*induct l a m b n c r rule: node43.induct*) *auto*

lemma *inorder_node44: height n > 0 \implies*

inorder (tree_d (node44 l a m b n c r)) = inorder l @ a # inorder m @ b # inorder n @ c # inorder (tree_d r)

by(*induct l a m b n c r rule: node44.induct*) *auto*

lemmas *inorder_nodes = inorder_node21 inorder_node22*

inorder_node31 inorder_node32 inorder_node33

inorder_node41 inorder_node42 inorder_node43 inorder_node44

lemma *split_minD:*

split_min t = (x,t') \implies bal t \implies height t > 0 \implies

x # inorder (tree_d t') = inorder t

by(*induction t arbitrary: t' rule: split_min.induct*)

(*auto simp: inorder_nodes split: prod.splits*)

lemma *inorder_del: \llbracket bal t ; sorted(inorder t) $\rrbracket \implies$*

inorder (tree_d (del x t)) = del_list x (inorder t)

by(*induction t rule: del.induct*)

(*auto simp: inorder_nodes del_list_simps split_minD split!: if_split prod.splits*)

lemma *inorder_delete: \llbracket bal t ; sorted(inorder t) $\rrbracket \implies$*

inorder (delete x t) = del_list x (inorder t)

by(*simp add: delete_def inorder_del*)

29.3 Balancedness

29.3.1 Proofs for insert

First a standard proof that *ins* preserves *bal*.

instantiation *up_i :: (type)height*

begin

```

fun height_upi :: 'a up_i  $\Rightarrow$  nat where
height (T_i t) = height t |
height (Up_i l a r) = height l

```

```

instance ..

```

```

end

```

```

lemma bal_ins: bal t  $\Longrightarrow$  bal (tree_i(ins a t))  $\wedge$  height(ins a t) = height t
by (induct t) (auto split!: if_split up_i.split)

```

Now an alternative proof (by Brian Huffman) that runs faster because two properties (balance and height) are combined in one predicate.

```

inductive full :: nat  $\Rightarrow$  'a tree234  $\Rightarrow$  bool where
full 0 Leaf |
[[full n l; full n r]]  $\Longrightarrow$  full (Suc n) (Node2 l p r) |
[[full n l; full n m; full n r]]  $\Longrightarrow$  full (Suc n) (Node3 l p m q r) |
[[full n l; full n m; full n m'; full n r]]  $\Longrightarrow$  full (Suc n) (Node4 l p m q m' q' r)

```

```

inductive_cases full_elims:

```

```

full n Leaf
full n (Node2 l p r)
full n (Node3 l p m q r)
full n (Node4 l p m q m' q' r)

```

```

inductive_cases full_0_elim: full 0 t

```

```

inductive_cases full_Suc_elim: full (Suc n) t

```

```

lemma full_0_iff [simp]: full 0 t  $\longleftrightarrow$  t = Leaf
by (auto elim: full_0_elim intro: full.intros)

```

```

lemma full_Leaf_iff [simp]: full n Leaf  $\longleftrightarrow$  n = 0
by (auto elim: full_elims intro: full.intros)

```

```

lemma full_Suc_Node2_iff [simp]:
full (Suc n) (Node2 l p r)  $\longleftrightarrow$  full n l  $\wedge$  full n r
by (auto elim: full_elims intro: full.intros)

```

```

lemma full_Suc_Node3_iff [simp]:
full (Suc n) (Node3 l p m q r)  $\longleftrightarrow$  full n l  $\wedge$  full n m  $\wedge$  full n r
by (auto elim: full_elims intro: full.intros)

```

lemma *full_Suc_Node4_iff* [*simp*]:
 $full (Suc\ n) (Node4\ l\ p\ m\ q\ m'\ q'\ r) \longleftrightarrow full\ n\ l \wedge full\ n\ m \wedge full\ n\ m' \wedge full\ n\ r$
by (*auto elim: full_elims intro: full.intros*)

lemma *full_imp_height*: $full\ n\ t \implies height\ t = n$
by (*induct set: full, simp_all*)

lemma *full_imp_bal*: $full\ n\ t \implies bal\ t$
by (*induct set: full, auto dest: full_imp_height*)

lemma *bal_imp_full*: $bal\ t \implies full\ (height\ t)\ t$
by (*induct t, simp_all*)

lemma *bal_iff_full*: $bal\ t \longleftrightarrow (\exists n. full\ n\ t)$
by (*auto elim!: bal_imp_full full_imp_bal*)

The *insert* function either preserves the height of the tree, or increases it by one. The constructor returned by the *insert* function determines which: A return value of the form $T_i\ t$ indicates that the height will be the same. A value of the form $Up_i\ l\ p\ r$ indicates an increase in height.

primrec *full_i* :: $nat \Rightarrow 'a\ up_i \Rightarrow bool$ **where**
 $full_i\ n\ (T_i\ t) \longleftrightarrow full\ n\ t \mid$
 $full_i\ n\ (Up_i\ l\ p\ r) \longleftrightarrow full\ n\ l \wedge full\ n\ r$

lemma *full_i_ins*: $full\ n\ t \implies full_i\ n\ (ins\ a\ t)$
by (*induct rule: full.induct*) (*auto, auto split: up_i.split*)

The *insert* operation preserves balance.

lemma *bal_insert*: $bal\ t \implies bal\ (insert\ a\ t)$
unfolding *bal_iff_full insert_def*
apply (*erule exE*)
apply (*drule full_i_ins [of _ _ a]*)
apply (*cases ins a t*)
apply (*auto intro: full.intros*)
done

29.3.2 Proofs for delete

instantiation *up_d* :: $(type)height$
begin

fun *height_{up_d}* :: $'a\ up_d \Rightarrow nat$ **where**
 $height\ (T_d\ t) = height\ t \mid$

$height (Up_d t) = height t + 1$

instance ..

end

lemma *bal_tree_d_node21*:

$\llbracket bal\ r; bal\ (tree_d\ l); height\ r = height\ l \rrbracket \implies bal\ (tree_d\ (node21\ l\ a\ r))$
by(*induct l a r rule: node21.induct*) *auto*

lemma *bal_tree_d_node22*:

$\llbracket bal\ (tree_d\ r); bal\ l; height\ r = height\ l \rrbracket \implies bal\ (tree_d\ (node22\ l\ a\ r))$
by(*induct l a r rule: node22.induct*) *auto*

lemma *bal_tree_d_node31*:

$\llbracket bal\ (tree_d\ l); bal\ m; bal\ r; height\ l = height\ r; height\ m = height\ r \rrbracket$
 $\implies bal\ (tree_d\ (node31\ l\ a\ m\ b\ r))$
by(*induct l a m b r rule: node31.induct*) *auto*

lemma *bal_tree_d_node32*:

$\llbracket bal\ l; bal\ (tree_d\ m); bal\ r; height\ l = height\ r; height\ m = height\ r \rrbracket$
 $\implies bal\ (tree_d\ (node32\ l\ a\ m\ b\ r))$
by(*induct l a m b r rule: node32.induct*) *auto*

lemma *bal_tree_d_node33*:

$\llbracket bal\ l; bal\ m; bal\ (tree_d\ r); height\ l = height\ r; height\ m = height\ r \rrbracket$
 $\implies bal\ (tree_d\ (node33\ l\ a\ m\ b\ r))$
by(*induct l a m b r rule: node33.induct*) *auto*

lemma *bal_tree_d_node41*:

$\llbracket bal\ (tree_d\ l); bal\ m; bal\ n; bal\ r; height\ l = height\ r; height\ m = height\ r; height\ n = height\ r \rrbracket$
 $\implies bal\ (tree_d\ (node41\ l\ a\ m\ b\ n\ c\ r))$
by(*induct l a m b n c r rule: node41.induct*) *auto*

lemma *bal_tree_d_node42*:

$\llbracket bal\ l; bal\ (tree_d\ m); bal\ n; bal\ r; height\ l = height\ r; height\ m = height\ r; height\ n = height\ r \rrbracket$
 $\implies bal\ (tree_d\ (node42\ l\ a\ m\ b\ n\ c\ r))$
by(*induct l a m b n c r rule: node42.induct*) *auto*

lemma *bal_tree_d_node43*:

$\llbracket bal\ l; bal\ m; bal\ (tree_d\ n); bal\ r; height\ l = height\ r; height\ m = height\ r; height\ n = height\ r \rrbracket$

$\implies \text{bal } (\text{tree}_d (\text{node43 } l \ a \ m \ b \ n \ c \ r))$
by(*induct l a m b n c r rule: node43.induct*) *auto*

lemma *bal_tree_d_node44*:
 $\llbracket \text{bal } l; \text{bal } m; \text{bal } n; \text{bal } (\text{tree}_d \ r); \text{height } l = \text{height } r; \text{height } m = \text{height } r; \text{height } n = \text{height } r \rrbracket$
 $\implies \text{bal } (\text{tree}_d (\text{node44 } l \ a \ m \ b \ n \ c \ r))$
by(*induct l a m b n c r rule: node44.induct*) *auto*

lemmas *bals = bal_tree_d_node21 bal_tree_d_node22*
bal_tree_d_node31 bal_tree_d_node32 bal_tree_d_node33
bal_tree_d_node41 bal_tree_d_node42 bal_tree_d_node43 bal_tree_d_node44

lemma *height_node21*:
 $\text{height } r > 0 \implies \text{height}(\text{node21 } l \ a \ r) = \max (\text{height } l) (\text{height } r) + 1$
by(*induct l a r rule: node21.induct*)(*simp_all add: max.assoc*)

lemma *height_node22*:
 $\text{height } l > 0 \implies \text{height}(\text{node22 } l \ a \ r) = \max (\text{height } l) (\text{height } r) + 1$
by(*induct l a r rule: node22.induct*)(*simp_all add: max.assoc*)

lemma *height_node31*:
 $\text{height } m > 0 \implies \text{height}(\text{node31 } l \ a \ m \ b \ r) =$
 $\max (\text{height } l) (\max (\text{height } m) (\text{height } r)) + 1$
by(*induct l a m b r rule: node31.induct*)(*simp_all add: max_def*)

lemma *height_node32*:
 $\text{height } r > 0 \implies \text{height}(\text{node32 } l \ a \ m \ b \ r) =$
 $\max (\text{height } l) (\max (\text{height } m) (\text{height } r)) + 1$
by(*induct l a m b r rule: node32.induct*)(*simp_all add: max_def*)

lemma *height_node33*:
 $\text{height } m > 0 \implies \text{height}(\text{node33 } l \ a \ m \ b \ r) =$
 $\max (\text{height } l) (\max (\text{height } m) (\text{height } r)) + 1$
by(*induct l a m b r rule: node33.induct*)(*simp_all add: max_def*)

lemma *height_node41*:
 $\text{height } m > 0 \implies \text{height}(\text{node41 } l \ a \ m \ b \ n \ c \ r) =$
 $\max (\text{height } l) (\max (\text{height } m) (\max (\text{height } n) (\text{height } r))) + 1$
by(*induct l a m b n c r rule: node41.induct*)(*simp_all add: max_def*)

lemma *height_node42*:
 $\text{height } l > 0 \implies \text{height}(\text{node42 } l \ a \ m \ b \ n \ c \ r) =$
 $\max (\text{height } l) (\max (\text{height } m) (\max (\text{height } n) (\text{height } r))) + 1$

by(*induct l a m b n c r rule: node42.induct*)(*simp_all add: max_def*)

lemma *height_node43*:

height m > 0 \implies *height*(*node43 l a m b n c r*) =
max (*height l*) (*max* (*height m*) (*max* (*height n*) (*height r*))) + 1

by(*induct l a m b n c r rule: node43.induct*)(*simp_all add: max_def*)

lemma *height_node44*:

height n > 0 \implies *height*(*node44 l a m b n c r*) =
max (*height l*) (*max* (*height m*) (*max* (*height n*) (*height r*))) + 1

by(*induct l a m b n c r rule: node44.induct*)(*simp_all add: max_def*)

lemmas *heights = height_node21 height_node22*

height_node31 height_node32 height_node33

height_node41 height_node42 height_node43 height_node44

lemma *height_split_min*:

split_min t = (x, t') \implies *height t > 0* \implies *bal t* \implies *height t' = height t*

by(*induct t arbitrary: x t' rule: split_min.induct*)

(*auto simp: heights split: prod.splits*)

lemma *height_del: bal t* \implies *height*(*del x t*) = *height t*

by(*induction x t rule: del.induct*)

(*auto simp add: heights height_split_min split!: if_split prod.split*)

lemma *bal_split_min*:

\llbracket *split_min t = (x, t'); bal t; height t > 0* $\rrbracket \implies$ *bal* (*tree_d t*)

by(*induct t arbitrary: x t' rule: split_min.induct*)

(*auto simp: heights height_split_min bals split: prod.splits*)

lemma *bal_tree_d_del: bal t* \implies *bal*(*tree_d*(*del x t*))

by(*induction x t rule: del.induct*)

(*auto simp: bals bal_split_min height_del height_split_min split!: if_split prod.split*)

corollary *bal_delete: bal t* \implies *bal*(*delete x t*)

by(*simp add: delete_def bal_tree_d_del*)

29.4 Overall Correctness

interpretation *S: Set_by_Ordered*

where *empty = empty* **and** *isin = isin* **and** *insert = insert* **and** *delete = delete*

and *inorder = inorder* **and** *inv = bal*

```

proof (standard, goal_cases)
  case 2 thus ?case by(simp add: isin_set)
next
  case 3 thus ?case by(simp add: inorder_insert)
next
  case 4 thus ?case by(simp add: inorder_delete)
next
  case 6 thus ?case by(simp add: bal_insert)
next
  case 7 thus ?case by(simp add: bal_delete)
qed (simp add: empty_def)+

end

```

30 2-3-4 Tree Implementation of Maps

```

theory Tree234_Map
imports
  Tree234_Set
  Map_Specs
begin

```

30.1 Map operations on 2-3-4 trees

```

fun lookup :: ('a::linorder * 'b) tree234  $\Rightarrow$  'a  $\Rightarrow$  'b option where
  lookup Leaf x = None |
  lookup (Node2 l (a,b) r) x = (case cmp x a of
    LT  $\Rightarrow$  lookup l x |
    GT  $\Rightarrow$  lookup r x |
    EQ  $\Rightarrow$  Some b) |
  lookup (Node3 l (a1,b1) m (a2,b2) r) x = (case cmp x a1 of
    LT  $\Rightarrow$  lookup l x |
    EQ  $\Rightarrow$  Some b1 |
    GT  $\Rightarrow$  (case cmp x a2 of
      LT  $\Rightarrow$  lookup m x |
      EQ  $\Rightarrow$  Some b2 |
      GT  $\Rightarrow$  lookup r x)) |
  lookup (Node4 t1 (a1,b1) t2 (a2,b2) t3 (a3,b3) t4) x = (case cmp x a2 of
    LT  $\Rightarrow$  (case cmp x a1 of
      LT  $\Rightarrow$  lookup t1 x | EQ  $\Rightarrow$  Some b1 | GT  $\Rightarrow$  lookup t2 x) |
    EQ  $\Rightarrow$  Some b2 |
    GT  $\Rightarrow$  (case cmp x a3 of
      LT  $\Rightarrow$  lookup t3 x | EQ  $\Rightarrow$  Some b3 | GT  $\Rightarrow$  lookup t4 x))

```

```

fun upd :: 'a::linorder => 'b => ('a*'b) tree234 => ('a*'b) upi where
upd x y Leaf = Upi Leaf (x,y) Leaf |
upd x y (Node2 l ab r) = (case cmp x (fst ab) of
  LT => (case upd x y l of
    Ti l' => Ti (Node2 l' ab r)
    | Upi l1 ab' l2 => Ti (Node3 l1 ab' l2 ab r)) |
  EQ => Ti (Node2 l (x,y) r) |
  GT => (case upd x y r of
    Ti r' => Ti (Node2 l ab r')
    | Upi r1 ab' r2 => Ti (Node3 l ab r1 ab' r2))) |
upd x y (Node3 l ab1 m ab2 r) = (case cmp x (fst ab1) of
  LT => (case upd x y l of
    Ti l' => Ti (Node3 l' ab1 m ab2 r)
    | Upi l1 ab' l2 => Upi (Node2 l1 ab' l2) ab1 (Node2 m ab2 r)) |
  EQ => Ti (Node3 l (x,y) m ab2 r) |
  GT => (case cmp x (fst ab2) of
    LT => (case upd x y m of
      Ti m' => Ti (Node3 l ab1 m' ab2 r)
      | Upi m1 ab' m2 => Upi (Node2 l ab1 m1) ab' (Node2 m2
ab2 r)) |
    EQ => Ti (Node3 l ab1 m (x,y) r) |
    GT => (case upd x y r of
      Ti r' => Ti (Node3 l ab1 m ab2 r')
      | Upi r1 ab' r2 => Upi (Node2 l ab1 m) ab2 (Node2 r1 ab'
r2)))) |
upd x y (Node4 t1 ab1 t2 ab2 t3 ab3 t4) = (case cmp x (fst ab2) of
  LT => (case cmp x (fst ab1) of
    LT => (case upd x y t1 of
      Ti t1' => Ti (Node4 t1' ab1 t2 ab2 t3 ab3 t4)
      | Upi t11 q t12 => Upi (Node2 t11 q t12) ab1 (Node3 t2 ab2
t3 ab3 t4)) |
    EQ => Ti (Node4 t1 (x,y) t2 ab2 t3 ab3 t4) |
    GT => (case upd x y t2 of
      Ti t2' => Ti (Node4 t1 ab1 t2' ab2 t3 ab3 t4)
      | Upi t21 q t22 => Upi (Node2 t1 ab1 t21) q (Node3 t22 ab2
t3 ab3 t4)) |
    EQ => Ti (Node4 t1 ab1 t2 (x,y) t3 ab3 t4) |
    GT => (case cmp x (fst ab3) of
      LT => (case upd x y t3 of
        Ti t3' => Ti (Node4 t1 ab1 t2 ab2 t3' ab3 t4)
        | Upi t31 q t32 => Upi (Node2 t1 ab1 t2) ab2 q (Node3 t31
q t32 ab3 t4)) |
      EQ => Ti (Node4 t1 ab1 t2 ab2 t3 (x,y) t4) |
      GT => (case upd x y t4 of

```

$$T_i t4' \Rightarrow T_i (\text{Node4 } t1 \text{ ab1 } t2 \text{ ab2 } t3 \text{ ab3 } t4') \\ | \text{Up}_i t41 \text{ q } t42 \Rightarrow \text{Up}_i (\text{Node2 } t1 \text{ ab1 } t2) \text{ ab2 } (\text{Node3 } t3 \text{ ab3 } \\ t41 \text{ q } t42))))$$

definition *update* :: 'a::linorder \Rightarrow 'b \Rightarrow ('a*'b) tree234 \Rightarrow ('a*'b) tree234

where

update x y t = tree_i(upd x y t)

fun *del* :: 'a::linorder \Rightarrow ('a*'b) tree234 \Rightarrow ('a*'b) up_d **where**

del x Leaf = T_d Leaf |

del x (Node2 Leaf ab1 Leaf) = (if x=fst ab1 then Up_d Leaf else T_d(Node2 Leaf ab1 Leaf)) |

del x (Node3 Leaf ab1 Leaf ab2 Leaf) = T_d(if x=fst ab1 then Node2 Leaf ab2 Leaf

else if x=fst ab2 then Node2 Leaf ab1 Leaf else Node3 Leaf ab1 Leaf ab2 Leaf) |

del x (Node4 Leaf ab1 Leaf ab2 Leaf ab3 Leaf) =

T_d(if x = fst ab1 then Node3 Leaf ab2 Leaf ab3 Leaf else

if x = fst ab2 then Node3 Leaf ab1 Leaf ab3 Leaf else

if x = fst ab3 then Node3 Leaf ab1 Leaf ab2 Leaf

else Node4 Leaf ab1 Leaf ab2 Leaf ab3 Leaf) |

del x (Node2 l ab1 r) = (case cmp x (fst ab1) of

LT \Rightarrow node21 (del x l) ab1 r |

GT \Rightarrow node22 l ab1 (del x r) |

EQ \Rightarrow let (ab1',t) = split_min r in node22 l ab1' t) |

del x (Node3 l ab1 m ab2 r) = (case cmp x (fst ab1) of

LT \Rightarrow node31 (del x l) ab1 m ab2 r |

EQ \Rightarrow let (ab1',m') = split_min m in node32 l ab1' m' ab2 r |

GT \Rightarrow (case cmp x (fst ab2) of

LT \Rightarrow node32 l ab1 (del x m) ab2 r |

EQ \Rightarrow let (ab2',r') = split_min r in node33 l ab1 m ab2' r' |

GT \Rightarrow node33 l ab1 m ab2 (del x r))) |

del x (Node4 t1 ab1 t2 ab2 t3 ab3 t4) = (case cmp x (fst ab2) of

LT \Rightarrow (case cmp x (fst ab1) of

LT \Rightarrow node41 (del x t1) ab1 t2 ab2 t3 ab3 t4 |

EQ \Rightarrow let (ab',t2') = split_min t2 in node42 t1 ab' t2' ab2 t3 ab3

t4 |

GT \Rightarrow node42 t1 ab1 (del x t2) ab2 t3 ab3 t4) |

EQ \Rightarrow let (ab',t3') = split_min t3 in node43 t1 ab1 t2 ab' t3' ab3 t4 |

GT \Rightarrow (case cmp x (fst ab3) of

LT \Rightarrow node43 t1 ab1 t2 ab2 (del x t3) ab3 t4 |

EQ \Rightarrow let (ab',t4') = split_min t4 in node44 t1 ab1 t2 ab2 t3 ab'

t4' |

GT \Rightarrow node44 t1 ab1 t2 ab2 t3 ab3 (del x t4)))

definition *delete* :: 'a::linorder \Rightarrow ('a*'b) tree234 \Rightarrow ('a*'b) tree234 **where**
delete x t = tree_d(del x t)

30.2 Functional correctness

lemma *lookup_map_of*:
 sorted1(*inorder* t) \Longrightarrow lookup t x = map_of (*inorder* t) x
by (*induction* t) (*auto simp: map_of_simps split: option.split*)

lemma *inorder_upd*:
 sorted1(*inorder* t) \Longrightarrow *inorder*(tree_i(upd a b t)) = upd_list a b (*inorder* t)
by(*induction* t)
 (*auto simp: upd_list_simps, auto simp: upd_list_simps split: up_i.splits*)

lemma *inorder_update*:
 sorted1(*inorder* t) \Longrightarrow *inorder*(update a b t) = upd_list a b (*inorder* t)
by(*simp add: update_def inorder_upd*)

lemma *inorder_del*: \llbracket bal t ; sorted1(*inorder* t) $\rrbracket \Longrightarrow$
inorder(tree_d (del x t)) = del_list x (*inorder* t)
by(*induction* t *rule: del.induct*)
 (*auto simp: del_list_simps inorder_nodes split_minD split!: if_splits prod.splits*)

lemma *inorder_delete*: \llbracket bal t ; sorted1(*inorder* t) $\rrbracket \Longrightarrow$
inorder(delete x t) = del_list x (*inorder* t)
by(*simp add: delete_def inorder_del*)

30.3 Balancedness

lemma *bal_upd*: bal t \Longrightarrow bal (tree_i(upd x y t)) \wedge height(upd x y t) = height t
by (*induct* t) (*auto, auto split!: if_split up_i.split*)

lemma *bal_update*: bal t \Longrightarrow bal (update x y t)
by (*simp add: update_def bal_upd*)

lemma *height_del*: bal t \Longrightarrow height(del x t) = height t
by(*induction* x t *rule: del.induct*)
 (*auto simp add: heights height_split_min split!: if_split prod.split*)

lemma *bal_tree_d_del*: bal t \Longrightarrow bal(tree_d(del x t))

```

by(induction x t rule: del.induct)
  (auto simp: bals bal_split_min height_del height_split_min split!: if_split
prod.split)

```

```

corollary bal_delete: bal t  $\implies$  bal(delete x t)
by(simp add: delete_def bal_tree_d_del)

```

30.4 Overall Correctness

```

interpretation M: Map_by_Ordered
where empty = empty and lookup = lookup and update = update and
delete = delete
and inorder = inorder and inv = bal
proof (standard, goal_cases)
  case 2 thus ?case by(simp add: lookup_map_of)
next
  case 3 thus ?case by(simp add: inorder_update)
next
  case 4 thus ?case by(simp add: inorder_delete)
next
  case 6 thus ?case by(simp add: bal_update)
next
  case 7 thus ?case by(simp add: bal_delete)
qed (simp add: empty_def)+

end

```

31 1-2 Brother Tree Implementation of Sets

```

theory Brother12_Set
imports
  Cmp
  Set_Specs
  HOL-Number_Theory.Fib
begin

```

31.1 Data Type and Operations

```

datatype 'a bro =
  N0 |
  N1 'a bro |
  N2 'a bro 'a 'a bro |

  L2 'a |

```

$N3 \ 'a \ bro \ 'a \ 'a \ bro \ 'a \ 'a \ bro$

definition $empty :: 'a \ bro \ where$
 $empty = N0$

fun $inorder :: 'a \ bro \Rightarrow 'a \ list \ where$
 $inorder \ N0 = [] \ |$
 $inorder \ (N1 \ t) = inorder \ t \ |$
 $inorder \ (N2 \ l \ a \ r) = inorder \ l \ @ \ a \ \# \ inorder \ r \ |$
 $inorder \ (L2 \ a) = [a] \ |$
 $inorder \ (N3 \ t1 \ a1 \ t2 \ a2 \ t3) = inorder \ t1 \ @ \ a1 \ \# \ inorder \ t2 \ @ \ a2 \ \# \ inorder \ t3$

fun $isin :: 'a \ bro \Rightarrow 'a::linorder \Rightarrow bool \ where$
 $isin \ N0 \ x = False \ |$
 $isin \ (N1 \ t) \ x = isin \ t \ x \ |$
 $isin \ (N2 \ l \ a \ r) \ x =$
 $(case \ cmp \ x \ a \ of$
 $LT \Rightarrow isin \ l \ x \ |$
 $EQ \Rightarrow True \ |$
 $GT \Rightarrow isin \ r \ x)$

fun $n1 :: 'a \ bro \Rightarrow 'a \ bro \ where$
 $n1 \ (L2 \ a) = N2 \ N0 \ a \ N0 \ |$
 $n1 \ (N3 \ t1 \ a1 \ t2 \ a2 \ t3) = N2 \ (N2 \ t1 \ a1 \ t2) \ a2 \ (N1 \ t3) \ |$
 $n1 \ t = N1 \ t$

hide_const (open) $insert$

locale $insert$
begin

fun $n2 :: 'a \ bro \Rightarrow 'a \Rightarrow 'a \ bro \Rightarrow 'a \ bro \ where$
 $n2 \ (L2 \ a1) \ a2 \ t = N3 \ N0 \ a1 \ N0 \ a2 \ t \ |$
 $n2 \ (N3 \ t1 \ a1 \ t2 \ a2 \ t3) \ a3 \ (N1 \ t4) = N2 \ (N2 \ t1 \ a1 \ t2) \ a2 \ (N2 \ t3 \ a3 \ t4) \ |$
 $n2 \ (N3 \ t1 \ a1 \ t2 \ a2 \ t3) \ a3 \ t4 = N3 \ (N2 \ t1 \ a1 \ t2) \ a2 \ (N1 \ t3) \ a3 \ t4 \ |$
 $n2 \ t1 \ a1 \ (L2 \ a2) = N3 \ t1 \ a1 \ N0 \ a2 \ N0 \ |$
 $n2 \ (N1 \ t1) \ a1 \ (N3 \ t2 \ a2 \ t3 \ a3 \ t4) = N2 \ (N2 \ t1 \ a1 \ t2) \ a2 \ (N2 \ t3 \ a3 \ t4) \ |$
 $n2 \ t1 \ a1 \ (N3 \ t2 \ a2 \ t3 \ a3 \ t4) = N3 \ t1 \ a1 \ (N1 \ t2) \ a2 \ (N2 \ t3 \ a3 \ t4) \ |$
 $n2 \ t1 \ a \ t2 = N2 \ t1 \ a \ t2$

fun $ins :: 'a::linorder \Rightarrow 'a \ bro \Rightarrow 'a \ bro \ where$
 $ins \ x \ N0 = L2 \ x \ |$
 $ins \ x \ (N1 \ t) = n1 \ (ins \ x \ t) \ |$

```

ins x (N2 l a r) =
  (case cmp x a of
    LT => n2 (ins x l) a r |
    EQ => N2 l a r |
    GT => n2 l a (ins x r))

```

```

fun tree :: 'a bro => 'a bro where
tree (L2 a) = N2 N0 a N0 |
tree (N3 t1 a1 t2 a2 t3) = N2 (N2 t1 a1 t2) a2 (N1 t3) |
tree t = t

```

```

definition insert :: 'a::linorder => 'a bro => 'a bro where
insert x t = tree(ins x t)

```

end

locale delete
begin

```

fun n2 :: 'a bro => 'a => 'a bro => 'a bro where
n2 (N1 t1) a1 (N1 t2) = N1 (N2 t1 a1 t2) |
n2 (N1 (N1 t1)) a1 (N2 (N1 t2) a2 (N2 t3 a3 t4)) =
  N1 (N2 (N2 t1 a1 t2) a2 (N2 t3 a3 t4)) |
n2 (N1 (N1 t1)) a1 (N2 (N2 t2 a2 t3) a3 (N1 t4)) =
  N1 (N2 (N2 t1 a1 t2) a2 (N2 t3 a3 t4)) |
n2 (N1 (N1 t1)) a1 (N2 (N2 t2 a2 t3) a3 (N2 t4 a4 t5)) =
  N2 (N2 (N1 t1) a1 (N2 t2 a2 t3)) a3 (N1 (N2 t4 a4 t5)) |
n2 (N2 (N1 t1) a1 (N2 t2 a2 t3)) a3 (N1 (N1 t4)) =
  N1 (N2 (N2 t1 a1 t2) a2 (N2 t3 a3 t4)) |
n2 (N2 (N2 t1 a1 t2) a2 (N1 t3)) a3 (N1 (N1 t4)) =
  N1 (N2 (N2 t1 a1 t2) a2 (N2 t3 a3 t4)) |
n2 (N2 (N2 t1 a1 t2) a2 (N2 t3 a3 t4)) a5 (N1 (N1 t5)) =
  N2 (N1 (N2 t1 a1 t2)) a2 (N2 (N2 t3 a3 t4) a5 (N1 t5)) |
n2 t1 a1 t2 = N2 t1 a1 t2

```

```

fun split_min :: 'a bro => ('a × 'a bro) option where
split_min N0 = None |
split_min (N1 t) =
  (case split_min t of
    None => None |
    Some (a, t') => Some (a, N1 t')) |
split_min (N2 t1 a t2) =
  (case split_min t1 of
    None => Some (a, N1 t2) |

```

$Some (b, t1') \Rightarrow Some (b, n2 t1' a t2))$

```
fun del :: 'a::linorder  $\Rightarrow$  'a bro  $\Rightarrow$  'a bro where
del _ N0      = N0 |
del x (N1 t)  = N1 (del x t) |
del x (N2 l a r) =
  (case cmp x a of
    LT  $\Rightarrow$  n2 (del x l) a r |
    GT  $\Rightarrow$  n2 l a (del x r) |
    EQ  $\Rightarrow$  (case split_min r of
      None  $\Rightarrow$  N1 l |
      Some (b, r')  $\Rightarrow$  n2 l b r'))
```

```
fun tree :: 'a bro  $\Rightarrow$  'a bro where
tree (N1 t) = t |
tree t = t
```

```
definition delete :: 'a::linorder  $\Rightarrow$  'a bro  $\Rightarrow$  'a bro where
delete a t = tree (del a t)
```

end

31.2 Invariants

```
fun B :: nat  $\Rightarrow$  'a bro set
and U :: nat  $\Rightarrow$  'a bro set where
B 0 = {N0} |
B (Suc h) = { N2 t1 a t2 | t1 a t2.
  t1  $\in$  B h  $\cup$  U h  $\wedge$  t2  $\in$  B h  $\vee$  t1  $\in$  B h  $\wedge$  t2  $\in$  B h  $\cup$  U h } |
U 0 = {} |
U (Suc h) = N1 ' B h
```

abbreviation $T h \equiv B h \cup U h$

```
fun Bp :: nat  $\Rightarrow$  'a bro set where
Bp 0 = B 0  $\cup$  L2 ' UNIV |
Bp (Suc 0) = B (Suc 0)  $\cup$  {N3 N0 a N0 b N0 | a b. True} |
Bp (Suc(Suc h)) = B (Suc(Suc h))  $\cup$ 
  {N3 t1 a t2 b t3 | t1 a t2 b t3. t1  $\in$  B (Suc h)  $\wedge$  t2  $\in$  U (Suc h)  $\wedge$  t3  $\in$ 
  B (Suc h)}
```

```
fun Um :: nat  $\Rightarrow$  'a bro set where
Um 0 = {} |
Um (Suc h) = N1 ' T h
```

31.3 Functional Correctness Proofs

31.3.1 Proofs for `isin`

lemma *isin_set*:

$t \in T h \implies \text{sorted}(\text{inorder } t) \implies \text{isin } t x = (x \in \text{set}(\text{inorder } t))$
by(*induction h arbitrary: t*) (*fastforce simp: isin_simps split: if_splits*)**+**

31.3.2 Proofs for `insertion`

lemma *inorder_n1*: $\text{inorder}(n1\ t) = \text{inorder } t$

by(*cases t rule: n1.cases*) (*auto simp: sorted_lems*)

context *insert*

begin

lemma *inorder_n2*: $\text{inorder}(n2\ l\ a\ r) = \text{inorder } l @ a \# \text{inorder } r$

by(*cases (l,a,r) rule: n2.cases*) (*auto simp: sorted_lems*)

lemma *inorder_tree*: $\text{inorder}(\text{tree } t) = \text{inorder } t$

by(*cases t*) *auto*

lemma *inorder_ins*: $t \in T h \implies$

$\text{sorted}(\text{inorder } t) \implies \text{inorder}(\text{ins } a\ t) = \text{ins_list } a\ (\text{inorder } t)$

by(*induction h arbitrary: t*) (*auto simp: ins_list_simps inorder_n1 inorder_n2*)

lemma *inorder_insert*: $t \in T h \implies$

$\text{sorted}(\text{inorder } t) \implies \text{inorder}(\text{insert } a\ t) = \text{ins_list } a\ (\text{inorder } t)$

by(*simp add: insert_def inorder_ins inorder_tree*)

end

31.3.3 Proofs for `deletion`

context *delete*

begin

lemma *inorder_tree*: $\text{inorder}(\text{tree } t) = \text{inorder } t$

by(*cases t*) *auto*

lemma *inorder_n2*: $\text{inorder}(n2\ l\ a\ r) = \text{inorder } l @ a \# \text{inorder } r$

by(*cases (l,a,r) rule: n2.cases*) (*auto*)

lemma *inorder_split_min*:

$t \in T h \implies (\text{split_min } t = \text{None} \longleftrightarrow \text{inorder } t = []) \wedge$
 $(\text{split_min } t = \text{Some}(a, t') \longrightarrow \text{inorder } t = a \# \text{inorder } t')$
by(*induction h arbitrary: t a t'*) (*auto simp: inorder_n2 split: option.splits*)

lemma *inorder_del*:

$t \in T h \implies \text{sorted}(\text{inorder } t) \implies \text{inorder}(\text{del } x \ t) = \text{del_list } x \ (\text{inorder } t)$

apply (*induction h arbitrary: t*)
apply (*auto simp: del_list_simps inorder_n2 split: option.splits*)
apply (*auto simp: del_list_simps inorder_n2*
inorder_split_min[OF UnI1] inorder_split_min[OF UnI2] split: option.splits)
done

lemma *inorder_delete*:

$t \in T h \implies \text{sorted}(\text{inorder } t) \implies \text{inorder}(\text{delete } x \ t) = \text{del_list } x \ (\text{inorder } t)$

by(*simp add: delete_def inorder_del inorder_tree*)

end

31.4 Invariant Proofs

31.4.1 Proofs for insertion

lemma *n1_type*: $t \in Bp \ h \implies n1 \ t \in T \ (Suc \ h)$

by(*cases h rule: Bp.cases*) *auto*

context *insert*

begin

lemma *tree_type*: $t \in Bp \ h \implies \text{tree } t \in B \ h \cup B \ (Suc \ h)$

by(*cases h rule: Bp.cases*) *auto*

lemma *n2_type*:

$(t1 \in Bp \ h \wedge t2 \in T \ h \longrightarrow n2 \ t1 \ a \ t2 \in Bp \ (Suc \ h)) \wedge$

$(t1 \in T \ h \wedge t2 \in Bp \ h \longrightarrow n2 \ t1 \ a \ t2 \in Bp \ (Suc \ h))$

apply(*cases h rule: Bp.cases*)

apply (*auto*)[2]

apply(*rule conjI impI | erule conjE exE imageE | simp | erule disjE*)+

done

lemma *Bp_if_B*: $t \in B \ h \implies t \in Bp \ h$

by (*cases h rule: Bp.cases*) *simp_all*

An automatic proof:

lemma

$(t \in B h \longrightarrow \text{ins } x t \in Bp h) \wedge (t \in U h \longrightarrow \text{ins } x t \in T h)$

apply(*induction h arbitrary: t*)

apply (*simp*)

apply (*fastforce simp: Bp_if_B n2_type dest: n1_type*)

done

A detailed proof:

lemma *ins_type*:

shows $t \in B h \implies \text{ins } x t \in Bp h$ **and** $t \in U h \implies \text{ins } x t \in T h$

proof(*induction h arbitrary: t*)

case 0

{ **case** 1 **thus** ?*case* **by** *simp*

next

case 2 **thus** ?*case* **by** *simp* }

next

case (*Suc h*)

{ **case** 1

then obtain *t1 a t2* **where** [*simp*]: $t = N2 t1 a t2$ **and**

t1: $t1 \in T h$ **and** *t2*: $t2 \in T h$ **and** *t12*: $t1 \in B h \vee t2 \in B h$

by *auto*

have ?*case* **if** $x < a$

proof –

have *n2* ($\text{ins } x t1$) *a t2* $\in Bp (Suc h)$

proof *cases*

assume $t1 \in B h$

with *t2* **show** ?*thesis* **by** (*simp add: Suc.IH(1) n2_type*)

next

assume $t1 \notin B h$

hence 1: $t1 \in U h$ **and** 2: $t2 \in B h$ **using** *t1 t12* **by** *auto*

show ?*thesis* **by** (*metis Suc.IH(2)[OF 1] Bp_if_B[OF 2] n2_type*)

qed

with $\langle x < a \rangle$ **show** ?*case* **by** *simp*

qed

moreover

have ?*case* **if** $a < x$

proof –

have *n2* *t1 a* ($\text{ins } x t2$) $\in Bp (Suc h)$

proof *cases*

assume $t2 \in B h$

with *t1* **show** ?*thesis* **by** (*simp add: Suc.IH(1) n2_type*)

next

assume $t2 \notin B h$

```

    hence 1:  $t1 \in B h$  and 2:  $t2 \in U h$  using  $t2 t12$  by auto
    show ?thesis by (metis Bp_if_B[OF 1] Suc.IH(2)[OF 2] n2_type)
qed
with  $\langle a < x \rangle$  show ?case by simp
qed
moreover
have ?case if  $x = a$ 
proof -
  from 1 have  $t \in Bp (Suc h)$  by(rule Bp_if_B)
  thus ?case using  $\langle x = a \rangle$  by simp
qed
ultimately show ?case by auto
next
case 2 thus ?case using Suc(1) n1_type by fastforce }
qed

```

```

lemma insert_type:
   $t \in B h \implies insert\ x\ t \in B h \cup B (Suc\ h)$ 
unfolding insert_def by (metis ins_type(1) tree_type)

```

end

31.4.2 Proofs for deletion

```

lemma B_simps[simp]:
   $N1\ t \in B\ h = False$ 
   $L2\ y \in B\ h = False$ 
   $(N3\ t1\ a1\ t2\ a2\ t3) \in B\ h = False$ 
   $N0 \in B\ h \longleftrightarrow h = 0$ 
by (cases h, auto)+

```

```

context delete
begin

```

```

lemma n2_type1:
   $\llbracket t1 \in Um\ h; t2 \in B\ h \rrbracket \implies n2\ t1\ a\ t2 \in T (Suc\ h)$ 
apply(cases h rule: Bp.cases)
apply auto[2]
apply(erule exE bexE conjE imageE | simp | erule disjE)+
done

```

```

lemma n2_type2:
   $\llbracket t1 \in B\ h; t2 \in Um\ h \rrbracket \implies n2\ t1\ a\ t2 \in T (Suc\ h)$ 
apply(cases h rule: Bp.cases)

```

```

apply auto[2]
apply(erule exE bexE conjE imageE | simp | erule disjE)+
done

```

```

lemma n2_type3:
   $\llbracket t1 \in T h ; t2 \in T h \rrbracket \implies n2\ t1\ a\ t2 \in T\ (Suc\ h)$ 
apply(cases h rule: Bp.cases)
apply auto[2]
apply(erule exE bexE conjE imageE | simp | erule disjE)+
done

```

```

lemma split_minNoneN0:  $\llbracket t \in B\ h ; split\_min\ t = None \rrbracket \implies t = N0$ 
by (cases t) (auto split: option.splits)

```

```

lemma split_minNoneN1 :  $\llbracket t \in U\ h ; split\_min\ t = None \rrbracket \implies t = N1\ N0$ 
by (cases h) (auto simp: split_minNoneN0 split: option.splits)

```

```

lemma split_min_type:
   $t \in B\ h \implies split\_min\ t = Some\ (a, t') \implies t' \in T\ h$ 
   $t \in U\ h \implies split\_min\ t = Some\ (a, t') \implies t' \in Um\ h$ 
proof (induction h arbitrary: t a t')
  case (Suc h)
  { case 1
    then obtain t1 a t2 where [simp]:  $t = N2\ t1\ a\ t2$  and
       $t12: t1 \in T\ h\ t2 \in T\ h\ t1 \in B\ h \vee t2 \in B\ h$ 
      by auto
    show ?case
    proof (cases split_min t1)
      case None
      show ?thesis
      proof cases
        assume  $t1 \in B\ h$ 
        with split_minNoneN0[OF this None] 1 show ?thesis by(auto)
      next
        assume  $t1 \notin B\ h$ 
        thus ?thesis using 1 None by (auto)
      qed
    next
      case [simp]: (Some bt')
      obtain b t1' where [simp]:  $bt' = (b, t1')$  by fastforce
      show ?thesis
      proof cases
        assume  $t1 \in B\ h$ 
        from Suc.IH(1)[OF this] 1 have  $t1' \in T\ h$  by simp

```

```

    from n2_type3[OF this t12(2)] 1 show ?thesis by auto
  next
    assume t1  $\notin$  B h
    hence t1: t1  $\in$  U h and t2: t2  $\in$  B h using t12 by auto
    from Suc.IH(2)[OF t1] have t1'  $\in$  Um h by simp
    from n2_type1[OF this t2] 1 show ?thesis by auto
  qed
}
{ case 2
  then obtain t1 where [simp]: t = N1 t1 and t1: t1  $\in$  B h by auto
  show ?case
  proof (cases split_min t1)
    case None
      with split_minNoneN0[OF t1 None] 2 show ?thesis by(auto)
    next
      case [simp]: (Some bt')
        obtain b t1' where [simp]: bt' = (b,t1') by fastforce
        from Suc.IH(1)[OF t1] have t1'  $\in$  T h by simp
        thus ?thesis using 2 by auto
      qed
    }
  qed auto
}

```

lemma *del_type*:

$t \in B h \implies del\ x\ t \in T h$

$t \in U h \implies del\ x\ t \in Um\ h$

proof (*induction h arbitrary: x t*)

case (Suc h)

{ case 1

then obtain l a r where [simp]: t = N2 l a r and

lr: l \in T h r \in T h l \in B h \vee r \in B h by auto

have ?case if $x < a$

proof *cases*

assume l \in B h

from n2_type3[OF Suc.IH(1)[OF this] lr(2)]

show ?thesis using $\langle x < a \rangle$ by(simp)

next

assume l \notin B h

hence l \in U h r \in B h using lr by auto

from n2_type1[OF Suc.IH(2)[OF this(1)] this(2)]

show ?thesis using $\langle x < a \rangle$ by(simp)

qed

moreover

```

have ?case if  $x > a$ 
proof cases
  assume  $r \in B h$ 
  from  $n2\_type3[OF\ lr(1)\ Suc.IH(1)[OF\ this]]$ 
  show ?thesis using  $\langle x > a \rangle$  by(simp)
next
  assume  $r \notin B h$ 
  hence  $l \in B h\ r \in U h$  using  $lr$  by auto
  from  $n2\_type2[OF\ this(1)\ Suc.IH(2)[OF\ this(2)]]$ 
  show ?thesis using  $\langle x > a \rangle$  by(simp)
qed
moreover
have ?case if [simp]:  $x = a$ 
proof (cases split_min r)
  case None
  show ?thesis
  proof cases
    assume  $r \in B h$ 
    with split_minNoneN0[OF this None]  $lr$  show ?thesis by(simp)
  next
    assume  $r \notin B h$ 
    hence  $r \in U h$  using  $lr$  by auto
    with split_minNoneN1[OF this None]  $lr(3)$  show ?thesis by (simp)
  qed
next
  case [simp]: (Some br')
  obtain  $b\ r'$  where [simp]:  $br' = (b, r')$  by fastforce
  show ?thesis
  proof cases
    assume  $r \in B h$ 
    from split_min_type(1)[OF this]  $n2\_type3[OF\ lr(1)]$ 
    show ?thesis by simp
  next
    assume  $r \notin B h$ 
    hence  $l \in B h$  and  $r \in U h$  using  $lr$  by auto
    from split_min_type(2)[OF this(2)]  $n2\_type2[OF\ this(1)]$ 
    show ?thesis by simp
  qed
qed
ultimately show ?case by auto
}
{ case 2 with  $Suc.IH(1)$  show ?case by auto }
qed auto

```

lemma *tree_type*: $t \in T (h+1) \implies tree\ t \in B (h+1) \cup B\ h$
by(*auto*)

lemma *delete_type*: $t \in B\ h \implies delete\ x\ t \in B\ h \cup B(h-1)$
unfolding *delete_def*

by (*cases h*) (*simp, metis del_type(1) tree_type Suc_eq_plus1 diff_Suc_1*)

end

31.5 Overall correctness

interpretation *Set_by_Ordered*

where *empty* = *empty* **and** *isin* = *isin* **and** *insert* = *insert.insert*

and *delete* = *delete.delete* **and** *inorder* = *inorder* **and** *inv* = $\lambda t. \exists h. t \in B\ h$

proof (*standard, goal_cases*)

case 2 **thus** ?*case* **by**(*auto intro!*: *isin_set*)

next

case 3 **thus** ?*case* **by**(*auto intro!*: *insert.inorder_insert*)

next

case 4 **thus** ?*case* **by**(*auto intro!*: *delete.inorder_delete*)

next

case 6 **thus** ?*case* **using** *insert.insert_type* **by** *blast*

next

case 7 **thus** ?*case* **using** *delete.delete_type* **by** *blast*

qed (*auto simp: empty_def*)

31.6 Height-Size Relation

By Daniel Stüwe

fun *fib_tree* :: *nat* \Rightarrow *unit bro* **where**

fib_tree 0 = *N0*

| *fib_tree* (*Suc* 0) = *N2 N0 () N0*

| *fib_tree* (*Suc*(*Suc* *h*)) = *N2 (fib_tree (h+1)) () (N1 (fib_tree h))*

fun *fib'* :: *nat* \Rightarrow *nat* **where**

fib' 0 = 0

| *fib'* (*Suc* 0) = 1

| *fib'* (*Suc*(*Suc* *h*)) = 1 + *fib'* (*Suc* *h*) + *fib'* *h*

fun *size* :: '*a bro* \Rightarrow *nat* **where**

size *N0* = 0

| *size* (*N1* *t*) = *size* *t*

| *size* (*N2* *t1* _ *t2*) = 1 + *size* *t1* + *size* *t2*

```

lemma fib_tree_B: fib_tree h ∈ B h
by (induction h rule: fib_tree.induct) auto

declare [[names_short]]

lemma size_fib': size (fib_tree h) = fib' h
by (induction h rule: fib_tree.induct) auto

lemma fibfib: fib' h + 1 = fib (Suc(Suc h))
by (induction h rule: fib_tree.induct) auto

lemma B_N2_cases[consumes 1]:
assumes N2 t1 a t2 ∈ B (Suc n)
obtains
  (BB) t1 ∈ B n and t2 ∈ B n |
  (UB) t1 ∈ U n and t2 ∈ B n |
  (BU) t1 ∈ B n and t2 ∈ U n
using assms by auto

lemma size_bounded: t ∈ B h ⇒ size t ≥ size (fib_tree h)
unfolding size_fib' proof (induction h arbitrary: t rule: fib'.induct)
case (∃ h t')
  note main = ∫
  then obtain t1 a t2 where t': t' = N2 t1 a t2 by auto
  with main have N2 t1 a t2 ∈ B (Suc (Suc h)) by auto
  thus ?case proof (cases rule: B_N2_cases)
    case BB
      then obtain x y z where t2: t2 = N2 x y z ∨ t2 = N2 z y x x ∈ B h
    by auto
      show ?thesis unfolding t' using main(1)[OF BB(1)] main(2)[OF
t2(2)] t2(1) by auto
    next
      case UB
        then obtain t11 where t1: t1 = N1 t11 t11 ∈ B h by auto
        show ?thesis unfolding t' t1(1) using main(2)[OF t1(2)] main(1)[OF
UB(2)] by simp
      next
        case BU
          then obtain t22 where t2: t2 = N1 t22 t22 ∈ B h by auto
          show ?thesis unfolding t' t2(1) using main(2)[OF t2(2)] main(1)[OF
BU(1)] by simp
    qed
qed auto

```

```

theorem  $t \in B \ h \implies \text{fib } (h + 2) \leq \text{size } t + 1$ 
using size_bounded
by (simp add: size_fib' fibfib[symmetric] del: fib.simps)

end

```

32 1-2 Brother Tree Implementation of Maps

```

theory Brother12_Map
imports
  Brother12_Set
  Map_Specs
begin

```

```

fun lookup :: ('a × 'b) bro  $\Rightarrow$  'a::linorder  $\Rightarrow$  'b option where
lookup N0 x = None |
lookup (N1 t) x = lookup t x |
lookup (N2 l (a,b) r) x =
  (case cmp x a of
    LT  $\Rightarrow$  lookup l x |
    EQ  $\Rightarrow$  Some b |
    GT  $\Rightarrow$  lookup r x)

```

```

locale update = insert
begin

```

```

fun upd :: 'a::linorder  $\Rightarrow$  'b  $\Rightarrow$  ('a×'b) bro  $\Rightarrow$  ('a×'b) bro where
upd x y N0 = L2 (x,y) |
upd x y (N1 t) = n1 (upd x y t) |
upd x y (N2 l (a,b) r) =
  (case cmp x a of
    LT  $\Rightarrow$  n2 (upd x y l) (a,b) r |
    EQ  $\Rightarrow$  N2 l (a,y) r |
    GT  $\Rightarrow$  n2 l (a,b) (upd x y r))

```

```

definition update :: 'a::linorder  $\Rightarrow$  'b  $\Rightarrow$  ('a×'b) bro  $\Rightarrow$  ('a×'b) bro where
update x y t = tree(upd x y t)

```

```

end

```

```

context delete
begin

```

```

fun del :: 'a::linorder => ('a×'b) bro => ('a×'b) bro where
  del _ N0      = N0 |
  del x (N1 t)  = N1 (del x t) |
  del x (N2 l (a,b) r) =
    (case cmp x a of
      LT => n2 (del x l) (a,b) r |
      GT => n2 l (a,b) (del x r) |
      EQ => (case split_min r of
        None => N1 l |
        Some (ab, r') => n2 l ab r'))

```

```

definition delete :: 'a::linorder => ('a×'b) bro => ('a×'b) bro where
  delete a t = tree (del a t)

```

end

32.1 Functional Correctness Proofs

32.1.1 Proofs for lookup

```

lemma lookup_map_of: t ∈ T h =>
  sorted1(inorder t) => lookup t x = map_of (inorder t) x
by(induction h arbitrary: t) (auto simp: map_of_simps split: option.splits)

```

32.1.2 Proofs for update

context update

begin

```

lemma inorder_upd: t ∈ T h =>
  sorted1(inorder t) => inorder(upd x y t) = upd_list x y (inorder t)
by(induction h arbitrary: t) (auto simp: upd_list_simps inorder_n1 inorder_n2)

```

```

lemma inorder_update: t ∈ T h =>
  sorted1(inorder t) => inorder(update x y t) = upd_list x y (inorder t)
by(simp add: update_def inorder_upd inorder_tree)

```

end

32.1.3 Proofs for deletion

context delete

begin

```

lemma inorder_del:
   $t \in T h \implies \text{sorted1}(\text{inorder } t) \implies \text{inorder}(\text{del } x \ t) = \text{del\_list } x \ (\text{inorder } t)$ 
  apply (induction h arbitrary: t)
  apply (auto simp: del_list_simps inorder_n2)
  apply (auto simp: del_list_simps inorder_n2
    inorder_split_min[OF UnI1] inorder_split_min[OF UnI2] split: option.splits)
  done

```

```

lemma inorder_delete:
   $t \in T h \implies \text{sorted1}(\text{inorder } t) \implies \text{inorder}(\text{delete } x \ t) = \text{del\_list } x \ (\text{inorder } t)$ 
by(simp add: delete_def inorder_del inorder_tree)

```

end

32.2 Invariant Proofs

32.2.1 Proofs for update

```

context update
begin

```

```

lemma upd_type:
   $(t \in B h \longrightarrow \text{upd } x \ y \ t \in Bp \ h) \wedge (t \in U h \longrightarrow \text{upd } x \ y \ t \in T h)$ 
apply(induction h arbitrary: t)
  apply (simp)
apply (fastforce simp: Bp_if_B n2_type dest: n1_type)
done

```

```

lemma update_type:
   $t \in B h \implies \text{update } x \ y \ t \in B h \cup B (\text{Suc } h)$ 
unfolding update_def by (metis upd_type tree_type)

```

end

32.2.2 Proofs for deletion

```

context delete
begin

```

```

lemma del_type:
   $t \in B h \implies \text{del } x \ t \in T h$ 

```

$t \in U h \implies del\ x\ t \in U m\ h$
proof (*induction h arbitrary: x t*)
case (*Suc h*)
{ case 1
then obtain $l\ a\ b\ r$ **where** $[simp]: t = N2\ l\ (a,b)\ r$ **and**
 $lr: l \in T\ h\ r \in T\ h\ l \in B\ h \vee r \in B\ h$ **by** *auto*
have $?case$ **if** $x < a$
proof *cases*
assume $l \in B\ h$
from $n2_type3[OF\ Suc.IH(1)[OF\ this]\ lr(2)]$
show $?thesis$ **using** $\langle x < a \rangle$ **by**(*simp*)
next
assume $l \notin B\ h$
hence $l \in U\ h\ r \in B\ h$ **using** lr **by** *auto*
from $n2_type1[OF\ Suc.IH(2)[OF\ this(1)]\ this(2)]$
show $?thesis$ **using** $\langle x < a \rangle$ **by**(*simp*)
qed
moreover
have $?case$ **if** $x > a$
proof *cases*
assume $r \in B\ h$
from $n2_type3[OF\ lr(1)\ Suc.IH(1)[OF\ this]]$
show $?thesis$ **using** $\langle x > a \rangle$ **by**(*simp*)
next
assume $r \notin B\ h$
hence $l \in B\ h\ r \in U\ h$ **using** lr **by** *auto*
from $n2_type2[OF\ this(1)\ Suc.IH(2)[OF\ this(2)]]$
show $?thesis$ **using** $\langle x > a \rangle$ **by**(*simp*)
qed
moreover
have $?case$ **if** $[simp]: x = a$
proof (*cases split_min r*)
case *None*
show $?thesis$
proof *cases*
assume $r \in B\ h$
with $split_minNoneN0[OF\ this\ None]\ lr$ **show** $?thesis$ **by**(*simp*)
next
assume $r \notin B\ h$
hence $r \in U\ h$ **using** lr **by** *auto*
with $split_minNoneN1[OF\ this\ None]\ lr(3)$ **show** $?thesis$ **by** (*simp*)
qed
next
case $[simp]: (Some\ br')$

```

obtain  $b\ r'$  where [simp]:  $br' = (b,r')$  by fastforce
show ?thesis
proof cases
  assume  $r \in B\ h$ 
  from split_min_type(1)[OF this] n2_type3[OF lr(1)]
  show ?thesis by simp
next
  assume  $r \notin B\ h$ 
  hence  $l \in B\ h$  and  $r \in U\ h$  using lr by auto
  from split_min_type(2)[OF this(2)] n2_type2[OF this(1)]
  show ?thesis by simp
qed
qed
ultimately show ?case by auto
}
{ case 2 with Suc.IH(1) show ?case by auto }
qed auto

```

lemma *delete_type*:

$t \in B\ h \implies \text{delete } x\ t \in B\ h \cup B(h-1)$

unfolding *delete_def*

by (*cases h*) (*simp*, *metis del_type(1)* *tree_type Suc_eq_plus1 diff_Suc_1*)

end

32.3 Overall correctness

interpretation *Map_by_Ordered*

where *empty* = *empty* **and** *lookup* = *lookup* **and** *update* = *update.update*
and *delete* = *delete.delete* **and** *inorder* = *inorder* **and** *inv* = $\lambda t. \exists h. t \in B\ h$

proof (*standard*, *goal_cases*)

case 2 **thus** *?case* **by**(*auto intro!*: *lookup_map_of*)

next

case 3 **thus** *?case* **by**(*auto intro!*: *update.inorder_update*)

next

case 4 **thus** *?case* **by**(*auto intro!*: *delete.inorder_delete*)

next

case 6 **thus** *?case* **using** *update.update_type* **by** (*metis Un_iff*)

next

case 7 **thus** *?case* **using** *delete.delete_type* **by** *blast*

qed (*auto simp*: *empty_def*)

end

33 AA Tree Implementation of Sets

theory *AA_Set*

imports

Isin2

Cmp

begin

type_synonym *'a aa_tree* = (*'a*nat*) *tree*

definition *empty* :: *'a aa_tree* **where**
empty = *Leaf*

fun *lvl* :: *'a aa_tree* \Rightarrow *nat* **where**
lvl Leaf = 0 |
lvl (Node _ (_, lv) _) = *lv*

fun *invar* :: *'a aa_tree* \Rightarrow *bool* **where**
invar Leaf = *True* |
invar (Node l (a, h) r) =
 (*invar l* \wedge *invar r* \wedge
h = *lvl l* + 1 \wedge (*h* = *lvl r* + 1 \vee ($\exists lr b rr. r = \text{Node } lr (b, h) rr \wedge h = \text{lvl } rr + 1$)))

fun *skew* :: *'a aa_tree* \Rightarrow *'a aa_tree* **where**
skew (Node (Node t1 (b, lvb) t2) (a, lva) t3) =
 (*if lva* = *lvb* *then Node t1 (b, lvb) (Node t2 (a, lva) t3)* *else Node (Node*
t1 (b, lvb) t2) (a, lva) t3) |
skew t = *t*

fun *split* :: *'a aa_tree* \Rightarrow *'a aa_tree* **where**
split (Node t1 (a, lva) (Node t2 (b, lvb) (Node t3 (c, lvc) t4))) =
 (*if lva* = *lvb* \wedge *lvb* = *lvc* — *lva* = *lvc* *suffices*
then Node (Node t1 (a, lva) t2) (b, lva+1) (Node t3 (c, lva) t4)
else Node t1 (a, lva) (Node t2 (b, lvb) (Node t3 (c, lvc) t4)))) |
split t = *t*

hide_const (**open**) *insert*

fun *insert* :: *'a::linorder* \Rightarrow *'a aa_tree* \Rightarrow *'a aa_tree* **where**
insert x Leaf = *Node Leaf (x, 1) Leaf* |
insert x (Node t1 (a, lv) t2) =
 (*case cmp x a of*
LT \Rightarrow *split (skew (Node (insert x t1) (a, lv) t2))*) |

$GT \Rightarrow \text{split} (\text{skew} (\text{Node } t1 (a,lv) (\text{insert } x t2))) \mid$
 $EQ \Rightarrow \text{Node } t1 (x, lv) t2)$

fun *sngl* :: 'a aa_tree \Rightarrow bool **where**
sngl Leaf = False |
sngl (Node _ _ Leaf) = True |
sngl (Node _ (_, lva) (Node _ (_, lvb) _)) = (lva > lvb)

definition *adjust* :: 'a aa_tree \Rightarrow 'a aa_tree **where**
adjust t =
(case t of
Node l (x,lv) r \Rightarrow
(if lvl l \geq lv-1 \wedge lvl r \geq lv-1 then t else
if lvl r < lv-1 \wedge sngl l then skew (Node l (x,lv-1) r) else
if lvl r < lv-1
then case l of
Node t1 (a,lva) (Node t2 (b,lvb) t3)
 \Rightarrow Node (Node t1 (a,lva) t2) (b,lvb+1) (Node t3 (x,lv-1) r)
else
if lvl r < lv then split (Node l (x,lv-1) r)
else
case r of
Node t1 (b,lvb) t4 \Rightarrow
(case t1 of
Node t2 (a,lva) t3
 \Rightarrow Node (Node l (x,lv-1) t2) (a,lva+1)
(split (Node t3 (b, if sngl t1 then lva else lva+1) t4))))))

In the paper, the last case of *adjust* is expressed with the help of an incorrect auxiliary function `nlvl`.

Function *split_max* below is called `dellrg` in the paper. The latter is incorrect for two reasons: `dellrg` is meant to delete the largest element but recurses on the left instead of the right subtree; the invariant is not restored.

fun *split_max* :: 'a aa_tree \Rightarrow 'a aa_tree * 'a **where**
split_max (Node l (a,lv) Leaf) = (l,a) |
split_max (Node l (a,lv) r) = (let (r',b) = *split_max* r in (*adjust*(Node l (a,lv) r'), b))

fun *delete* :: 'a::linorder \Rightarrow 'a aa_tree \Rightarrow 'a aa_tree **where**
delete _ Leaf = Leaf |
delete x (Node l (a,lv) r) =
(case cmp x a of
LT \Rightarrow *adjust* (Node (delete x l) (a,lv) r) |
GT \Rightarrow *adjust* (Node l (a,lv) (delete x r)) |

$EQ \Rightarrow$ (if $l = \text{Leaf}$ then r
 else let $(l',b) = \text{split_max } l$ in $\text{adjust } (\text{Node } l' (b,lw) r)$)

fun *pre_adjust* **where**

pre_adjust (*Node* $l (a,lw) r$) = (*invar* $l \wedge$ *invar* $r \wedge$
 ($lw = lvl\ l + 1 \wedge (lw = lvl\ r + 1 \vee lw = lvl\ r + 2 \vee lw = lvl\ r \wedge \text{sngl } r)$) \vee
 ($lw = lvl\ l + 2 \wedge (lw = lvl\ r + 1 \vee lw = lvl\ r \wedge \text{sngl } r)$))

declare *pre_adjust.simps* [*simp del*]

33.1 Auxiliary Proofs

lemma *split_case*: *split* $t =$ (case t of

Node $t1 (x,lwx) (\text{Node } t2 (y,lvy) (\text{Node } t3 (z,lvz) t4)) \Rightarrow$

(if $lwx = lvy \wedge lvy = lvz$

then *Node* (*Node* $t1 (x,lwx) t2) (y,lwx+1) (\text{Node } t3 (z,lvx) t4)$

else t)

| $t \Rightarrow t$)

by(*auto split: tree.split*)

lemma *skew_case*: *skew* $t =$ (case t of

Node (*Node* $t1 (y,lvy) t2) (x,lwx) t3 \Rightarrow$

(if $lwx = lvy$ then *Node* $t1 (y, lwx) (\text{Node } t2 (x,lwx) t3)$ else t)

| $t \Rightarrow t$)

by(*auto split: tree.split*)

lemma *lvl_0_iff*: *invar* $t \Longrightarrow lvl\ t = 0 \longleftrightarrow t = \text{Leaf}$

by(*cases t auto*)

lemma *lvl_Suc_iff*: $lvl\ t = \text{Suc } n \longleftrightarrow (\exists l\ a\ r. t = \text{Node } l (a,\text{Suc } n) r)$

by(*cases t auto*)

lemma *lvl_skew*: $lvl\ (\text{skew } t) = lvl\ t$

by(*cases t rule: skew.cases auto*)

lemma *lvl_split*: $lvl\ (\text{split } t) = lvl\ t \vee lvl\ (\text{split } t) = lvl\ t + 1 \wedge \text{sngl } (\text{split } t)$

by(*cases t rule: split.cases auto*)

lemma *invar_2Nodes*:*invar* (*Node* $l (x,lw) (\text{Node } rl (rx, rlw) rr)$) =

(*invar* $l \wedge$ *invar* $\langle rl, (rx, rlw), rr \rangle \wedge lw = \text{Suc } (lvl\ l) \wedge$

$(lw = \text{Suc } rlw \vee rlw = lw \wedge lw = \text{Suc } (lvl\ rr))$)

by *simp*

lemma *invar_NodeLeaf*[*simp*]:
 $invar (Node\ l\ (x,lv)\ Leaf) = (invar\ l \wedge lv = Suc\ (lvl\ l) \wedge lv = Suc\ 0)$
by *simp*

lemma *sngl_if_invar*: $invar (Node\ l\ (a, n)\ r) \implies n = lvl\ r \implies sngl\ r$
by(*cases r rule: sngl.cases*) *clarsimp*+

33.2 Invariance

33.2.1 Proofs for insert

lemma *lvl_insert_aux*:
 $lvl (insert\ x\ t) = lvl\ t \vee lvl (insert\ x\ t) = lvl\ t + 1 \wedge sngl (insert\ x\ t)$
apply(*induction t*)
apply (*auto simp: lvl_skew*)
apply (*metis Suc_eq_plus1 lvl.simps(2) lvl_split lvl_skew*)
done

lemma *lvl_insert*: **obtains**
(Same) $lvl (insert\ x\ t) = lvl\ t \mid$
(Incr) $lvl (insert\ x\ t) = lvl\ t + 1 \wedge sngl (insert\ x\ t)$
using *lvl_insert_aux* **by** *blast*

lemma *lvl_insert_sngl*: $invar\ t \implies sngl\ t \implies lvl(insert\ x\ t) = lvl\ t$
proof (*induction t rule: insert.induct*)
case ($2\ x\ t1\ a\ lv\ t2$)
consider (*LT*) $x < a \mid$ (*GT*) $x > a \mid$ (*EQ*) $x = a$
using *less_linear* **by** *blast*
thus *?case proof cases*
case *LT*
thus *?thesis using 2* **by** (*auto simp add: skew_case split_case split: tree.splits*)
next
case *GT*
thus *?thesis using 2*
proof (*cases t1 rule: tree2_cases*)
case *Node*
thus *?thesis using 2 GT*
apply (*auto simp add: skew_case split_case split: tree.splits*)
by (*metis less_not_refl2 lvl.simps(2) lvl_insert_aux n_not_Suc_n sngl.simps(3)*)
qed (*auto simp add: lvl_0_iff*)
qed *simp*

qed *simp*

lemma *skew_invar*: $invar\ t \implies skew\ t = t$
by(*cases t rule: skew.cases*) *auto*

lemma *split_invar*: $invar\ t \implies split\ t = t$
by(*cases t rule: split.cases*) *clarsimp+*

lemma *invar_NodeL*:
 $\llbracket invar(Node\ l\ (x,\ n)\ r); invar\ l'; lvl\ l' = lvl\ l \rrbracket \implies invar(Node\ l'\ (x,\ n)\ r)$
by(*auto*)

lemma *invar_NodeR*:
 $\llbracket invar(Node\ l\ (x,\ n)\ r); n = lvl\ r + 1; invar\ r'; lvl\ r' = lvl\ r \rrbracket \implies invar(Node\ l\ (x,\ n)\ r')$
by(*auto*)

lemma *invar_NodeR2*:
 $\llbracket invar(Node\ l\ (x,\ n)\ r); snl\ r'; n = lvl\ r + 1; invar\ r'; lvl\ r' = n \rrbracket \implies invar(Node\ l\ (x,\ n)\ r')$
by(*cases r' rule: snl.cases*) *clarsimp+*

lemma *lvl_insert_incr_iff*: $(lvl(insert\ a\ t) = lvl\ t + 1) \longleftrightarrow (\exists l\ x\ r.\ insert\ a\ t = Node\ l\ (x,\ lvl\ t + 1)\ r \wedge lvl\ l = lvl\ r)$
apply(*cases t rule: tree2_cases*)
apply(*auto simp add: skew_case split_case split: if_splits*)
apply(*auto split: tree_splits if_splits*)
done

lemma *invar_insert*: $invar\ t \implies invar(insert\ a\ t)$
proof(*induction t rule: tree2_induct*)
 case *N*: (*Node l x n r*)
 hence *il*: $invar\ l$ **and** *ir*: $invar\ r$ **by** *auto*
 note *iil* = *N.IH(1)[OF il]*
 note *iir* = *N.IH(2)[OF ir]*
 let *?t* = *Node l (x, n) r*
 have $a < x \vee a = x \vee x < a$ **by** *auto*
 moreover
 have *?case if a < x*
 proof (*cases rule: lvl_insert[of a l]*)
 case (*Same*) **thus** *?thesis*
 using $\langle a < x \rangle invar_NodeL[OF\ N.prem\ iil\ Same]$

```

    by (simp add: skew_invar split_invar del: invar.simps)
next
case (Incr)
then obtain t1 w t2 where ial[simp]: insert a l = Node t1 (w, n) t2
  using N.premis by (auto simp: lvl_Suc_iff)
have l12: lvl t1 = lvl t2
  by (metis Incr(1) ial lvl_insert_incr_iff tree.inject)
have insert a ?t = split(skew(Node (insert a l) (x,n) r))
  by(simp add: <a<x>)
also have skew(Node (insert a l) (x,n) r) = Node t1 (w,n) (Node t2
(x,n) r)
  by(simp)
also have invar(split ...)
proof (cases r rule: tree2_cases)
case Leaf
hence l = Leaf using N.premis by(auto simp: lvl_0_iff)
thus ?thesis using Leaf ial by simp
next
case [simp]: (Node t3 y m t4)
show ?thesis
proof cases
assume m = n thus ?thesis using N(3) iil by(auto)
next
assume m ≠ n thus ?thesis using N(3) iil l12 by(auto)
qed
qed
finally show ?thesis .
qed
moreover
have ?case if x < a
proof -
from <invar ?t> have n = lvl r ∨ n = lvl r + 1 by auto
thus ?case
proof
assume 0: n = lvl r
have insert a ?t = split(skew(Node l (x, n) (insert a r)))
  using <a>x> by(auto)
also have skew(Node l (x,n) (insert a r)) = Node l (x,n) (insert a r)
  using N.premis by(simp add: skew_case split: tree.split)
also have invar(split ...)
proof -
from lvl_insert_sngl[OF ir sngl_if_invar[OF <invar ?t> 0], of a]
obtain t1 y t2 where iar: insert a r = Node t1 (y,n) t2
  using N.premis 0 by (auto simp: lvl_Suc_iff)

```

```

    from N.prems iar 0 iir
    show ?thesis by (auto simp: split_case split: tree_splits)
qed
finally show ?thesis .
next
assume 1:  $n = \text{lvl } r + 1$ 
hence sngl ?t by (cases r) auto
show ?thesis
proof (cases rule: lvl_insert[of a r])
  case (Same)
  show ?thesis using  $\langle x < a \rangle$  il ir invar_NodeR[OF N.prems 1 iir Same]
    by (auto simp add: skew_invar split_invar)
  next
  case (Incr)
  thus ?thesis using invar_NodeR2[OF  $\langle \text{invar } ?t \rangle$  Incr(2) 1 iir] 1  $\langle x$ 
    < a  $\rangle$ 
    by (auto simp add: skew_invar split_invar split: if_splits)
qed
qed
qed
moreover
have  $a = x \implies ?\text{case}$  using N.prems by auto
ultimately show ?case by blast
qed simp

```

33.2.2 Proofs for delete

lemma *invarL*: $\text{ASSUMPTION}(\text{invar } \langle l, (a, \text{lv}), r \rangle) \implies \text{invar } l$
 by (simp add: ASSUMPTION_def)

lemma *invarR*: $\text{ASSUMPTION}(\text{invar } \langle l, (a, \text{lv}), r \rangle) \implies \text{invar } r$
 by (simp add: ASSUMPTION_def)

lemma *sngl_NodeI*:
 $\text{sngl } (\text{Node } l (a, \text{lv}) r) \implies \text{sngl } (\text{Node } l' (a', \text{lv}) r)$
 by (cases *r* rule: tree2_cases) (simp_all)

declare *invarL*[simp] *invarR*[simp]

lemma *pre_cases*:
assumes *pre_adjust* (*Node* *l* (*x*, *lv*) *r*)
obtains
 (*tSngl*) $\text{invar } l \wedge \text{invar } r \wedge$

```

     $lv = \text{Suc } (lvl \ r) \wedge lvl \ l = lvl \ r \mid$ 
    (tDouble) invar l  $\wedge$  invar r  $\wedge$ 
     $lv = lvl \ r \wedge \text{Suc } (lvl \ l) = lvl \ r \wedge \text{sngl } r \mid$ 
    (rDown) invar l  $\wedge$  invar r  $\wedge$ 
     $lv = \text{Suc } (\text{Suc } (lvl \ r)) \wedge lv = \text{Suc } (lvl \ l) \mid$ 
    (lDown_tSngl) invar l  $\wedge$  invar r  $\wedge$ 
     $lv = \text{Suc } (lvl \ r) \wedge lv = \text{Suc } (\text{Suc } (lvl \ l)) \mid$ 
    (lDown_tDouble) invar l  $\wedge$  invar r  $\wedge$ 
     $lv = lvl \ r \wedge lv = \text{Suc } (\text{Suc } (lvl \ l)) \wedge \text{sngl } r$ 
using assms unfolding pre_adjust.simps
by auto

declare invar.simps(2)[simp del] invar_2Nodes[simp add]

lemma invar_adjust:
  assumes pre: pre_adjust (Node l (a,lv) r)
  shows invar(adjust (Node l (a,lv) r))
using pre proof (cases rule: pre_cases)
  case (tDouble) thus ?thesis unfolding adjust_def by (cases r) (auto simp: invar.simps(2))
next
  case (rDown)
  from rDown obtain llv ll la lr where l: l = Node ll (la, llv) lr by (cases l) auto
  from rDown show ?thesis unfolding adjust_def by (auto simp: l invar.simps(2) split: tree.splits)
next
  case (lDown_tDouble)
  from lDown_tDouble obtain rlv rr ra rl where r: r = Node rl (ra, rlv)
  rr by (cases r) auto
  from lDown_tDouble and r obtain rrlv rrr rra rrl where
    rr : rr = Node rrr (rra, rrlv) rrl by (cases rr) auto
  from lDown_tDouble show ?thesis unfolding adjust_def r rr
  apply (cases rl rule: tree2_cases) apply (auto simp add: invar.simps(2) split!: if_split)
  using lDown_tDouble by (auto simp: split_case lvl_0_iff elim:lvl.elims split: tree.split)
qed (auto simp: split_case invar.simps(2) adjust_def split: tree.splits)

lemma lvl_adjust:
  assumes pre_adjust (Node l (a,lv) r)
  shows  $lv = lvl \ (\text{adjust}(\text{Node } l \ (a,lv) \ r)) \vee lv = lvl \ (\text{adjust}(\text{Node } l \ (a,lv) \ r)) + 1$ 
using assms(1)

```

```

proof(cases rule: pre_cases)
  case lDown_tSngl thus ?thesis
    using lvl_split[of ⟨l, (a, lvl r), r⟩] by (auto simp: adjust_def)
next
  case lDown_tDouble thus ?thesis
    by (auto simp: adjust_def invar.simps(2) split: tree.split)
qed (auto simp: adjust_def split: tree.splits)

```

```

lemma sngl_adjust: assumes pre_adjust (Node l (a,lv) r)
  sngl ⟨l, (a, lv), r⟩ lv = lvl (adjust ⟨l, (a, lv), r⟩)
  shows sngl (adjust ⟨l, (a, lv), r⟩)
using assms proof (cases rule: pre_cases)
  case rDown
  thus ?thesis using assms(2,3) unfolding adjust_def
    by (auto simp add: skew_case) (auto split: tree.split)
qed (auto simp: adjust_def skew_case split_case split: tree.split)

```

```

definition post_del t t' ==
  invar t' ∧
  (lvl t' = lvl t ∨ lvl t' + 1 = lvl t) ∧
  (lvl t' = lvl t ∧ sngl t → sngl t')

```

```

lemma pre_adj_if_postR:
  invar⟨lv, (l, a), r⟩ ⇒ post_del r r' ⇒ pre_adjust ⟨lv, (l, a), r'⟩
by(cases sngl r)
  (auto simp: pre_adjust.simps post_del_def invar.simps(2) elim: sngl.elims)

```

```

lemma pre_adj_if_postL:
  invar⟨l, (a, lv), r⟩ ⇒ post_del l l' ⇒ pre_adjust ⟨l', (b, lv), r'⟩
by(cases sngl r)
  (auto simp: pre_adjust.simps post_del_def invar.simps(2) elim: sngl.elims)

```

```

lemma post_del_adjL:
  [ invar⟨l, (a, lv), r⟩; pre_adjust ⟨l', (b, lv), r'⟩ ]
  ⇒ post_del ⟨l, (a, lv), r⟩ (adjust ⟨l', (b, lv), r'⟩)
unfolding post_del_def
by (metis invar_adjust lvl_adjust sngl_NodeI sngl_adjust lvl.simps(2))

```

```

lemma post_del_adjR:
assumes invar⟨l, (a,lv), r⟩ pre_adjust ⟨l, (a,lv), r'⟩ post_del r r'
shows post_del ⟨l, (a,lv), r⟩ (adjust ⟨l, (a,lv), r'⟩)
proof(unfold post_del_def, safe del: disjCI)
  let ?t = ⟨l, (a,lv), r⟩
  let ?t' = adjust ⟨l, (a,lv), r'⟩

```

```

show invar ?t' by(rule invar_adjust[OF assms(2)])
show lvl ?t' = lvl ?t ∨ lvl ?t' + 1 = lvl ?t
  using lvl_adjust[OF assms(2)] by auto
show sngl ?t' if as: lvl ?t' = lvl ?t sngl ?t
proof –
  have s: sngl ⟨l, (a,lv), r⟩
  proof(cases r' rule: tree2_cases)
    case Leaf thus ?thesis by simp
  next
    case Node thus ?thesis using as(2) assms(1,3)
    by (cases r rule: tree2_cases) (auto simp: post_del_def)
  qed
  show ?thesis using as(1) sngl_adjust[OF assms(2) s] by simp
qed
qed

declare prod.splits[split]

theorem post_split_max:
  [[ invar t; (t', x) = split_max t; t ≠ Leaf ]] ⇒ post_del t t'
proof (induction t arbitrary: t' rule: split_max.induct)
  case (2 l a lv rl bl rr)
  let ?r = ⟨rl, bl, rr⟩
  let ?t = ⟨l, (a, lv), ?r⟩
  from 2.prems(2) obtain r' where r': (r', x) = split_max ?r
  and [simp]: t = adjust ⟨l, (a, lv), r⟩ by auto
  from 2.IH[OF _ r'] ⟨invar ?t⟩ have post: post_del ?r r' by simp
  note preR = pre_adj_if_postR[OF ⟨invar ?t⟩ post]
  show ?case by (simp add: post_del_adjR[OF 2.prems(1) preR post])
qed (auto simp: post_del_def)

theorem post_delete: invar t ⇒ post_del t (delete x t)
proof (induction t rule: tree2_induct)
  case (Node l a lv r)

  let ?l' = delete x l and ?r' = delete x r
  let ?t = Node l (a,lv) r let ?t' = delete x ?t

  from Node.prems have inv_l: invar l and inv_r: invar r by (auto)

  note post_l' = Node.IH(1)[OF inv_l]
  note preL = pre_adj_if_postL[OF Node.prems post_l']

  note post_r' = Node.IH(2)[OF inv_r]

```

```

note preR = pre_adj_if_postR[OF Node.premis post_r']

show ?case
proof (cases rule: linorder_cases[of x a])
  case less
  thus ?thesis using Node.premis by (simp add: post_del_adjL preL)
next
  case greater
  thus ?thesis using Node.premis by (simp add: post_del_adjR preR
post_r')
next
  case equal
  show ?thesis
  proof cases
    assume l = Leaf thus ?thesis using equal Node.premis
    by(auto simp: post_del_def invar.simps(2))
  next
    assume l ≠ Leaf thus ?thesis using equal
    by simp (metis Node.premis inv_l post_del_adjL post_split_max
pre_adj_if_postL)
  qed
  qed
qed (simp add: post_del_def)

declare invar_2Nodes[simp del]

```

33.3 Functional Correctness

33.3.1 Proofs for insert

```

lemma inorder_split: inorder(split t) = inorder t
by(cases t rule: split.cases) (auto)

```

```

lemma inorder_skew: inorder(skew t) = inorder t
by(cases t rule: skew.cases) (auto)

```

```

lemma inorder_insert:
  sorted(inorder t) ⇒ inorder(insert x t) = ins_list x (inorder t)
by(induction t) (auto simp: ins_list_simps inorder_split inorder_skew)

```

33.3.2 Proofs for delete

```

lemma inorder_adjust: t ≠ Leaf ⇒ pre_adjust t ⇒ inorder(adjust t)
= inorder t
by(cases t)

```

```
(auto simp: adjust_def inorder_skew inorder_split invar.simps(2) pre_adjust.simps
  split: tree.splits)
```

lemma *split_maxD*:

```
[[ split_max t = (t',x); t ≠ Leaf; invar t ]] ⇒ inorder t' @ [x] = inorder
t
```

```
by(induction t arbitrary: t' rule: split_max.induct)
```

```
(auto simp: sorted_lems inorder_adjust pre_adj_if_postR post_split_max
  split: prod.splits)
```

lemma *inorder_delete*:

```
invar t ⇒ sorted(inorder t) ⇒ inorder(delete x t) = del_list x (inorder
t)
```

```
by(induction t)
```

```
(auto simp: del_list_simps inorder_adjust pre_adj_if_postL pre_adj_if_postR
```

```
  post_split_max post_delete split_maxD split: prod.splits)
```

interpretation *S*: *Set_by_Ordered*

```
where empty = empty and isin = isin and insert = insert and delete =
delete
```

```
and inorder = inorder and inv = invar
```

```
proof (standard, goal_cases)
```

```
  case 1 show ?case by (simp add: empty_def)
```

```
next
```

```
  case 2 thus ?case by (simp add: isin_set_inorder)
```

```
next
```

```
  case 3 thus ?case by (simp add: inorder_insert)
```

```
next
```

```
  case 4 thus ?case by (simp add: inorder_delete)
```

```
next
```

```
  case 5 thus ?case by (simp add: empty_def)
```

```
next
```

```
  case 6 thus ?case by (simp add: invar_insert)
```

```
next
```

```
  case 7 thus ?case using post_delete by (auto simp: post_del_def)
```

```
qed
```

```
end
```

34 AA Tree Implementation of Maps

```
theory AA_Map
```

```

imports
  AA_Set
  Lookup2
begin

fun update :: 'a::linorder  $\Rightarrow$  'b  $\Rightarrow$  ('a*'b) aa_tree  $\Rightarrow$  ('a*'b) aa_tree where
update x y Leaf = Node Leaf ((x,y), 1) Leaf |
update x y (Node t1 ((a,b), lv) t2) =
  (case cmp x a of
    LT  $\Rightarrow$  split (skew (Node (update x y t1) ((a,b), lv) t2)) |
    GT  $\Rightarrow$  split (skew (Node t1 ((a,b), lv) (update x y t2))) |
    EQ  $\Rightarrow$  Node t1 ((x,y), lv) t2)

fun delete :: 'a::linorder  $\Rightarrow$  ('a*'b) aa_tree  $\Rightarrow$  ('a*'b) aa_tree where
delete _ Leaf = Leaf |
delete x (Node l ((a,b), lv) r) =
  (case cmp x a of
    LT  $\Rightarrow$  adjust (Node (delete x l) ((a,b), lv) r) |
    GT  $\Rightarrow$  adjust (Node l ((a,b), lv) (delete x r)) |
    EQ  $\Rightarrow$  (if l = Leaf then r
      else let (l',ab') = split_max l in adjust (Node l' (ab', lv) r)))

```

34.1 Invariance

34.1.1 Proofs for insert

```

lemma lvl_update_aux:
  lvl (update x y t) = lvl t  $\vee$  lvl (update x y t) = lvl t + 1  $\wedge$  sngl (update x
y t)
apply(induction t)
apply (auto simp: lvl_skew)
apply (metis Suc_eq_plus1 lvl_simps(2) lvl_split lvl_skew)+
done

```

```

lemma lvl_update: obtains
  (Same) lvl (update x y t) = lvl t |
  (Incr) lvl (update x y t) = lvl t + 1 sngl (update x y t)
using lvl_update_aux by fastforce

```

```

declare invar_simps(2)[simp]

```

```

lemma lvl_update_sngl: invar t  $\Longrightarrow$  sngl t  $\Longrightarrow$  lvl(update x y t) = lvl t
proof (induction t rule: update.induct)
  case (2 x y t1 a b lv t2)

```

```

consider (LT)  $x < a$  | (GT)  $x > a$  | (EQ)  $x = a$ 
  using less_linear by blast
thus ?case proof cases
  case LT
    thus ?thesis using 2 by (auto simp add: skew_case split_case split:
tree.splits)
  next
    case GT
      thus ?thesis using 2 proof (cases t1)
        case Node
          thus ?thesis using 2 GT
            apply (auto simp add: skew_case split_case split: tree.splits)
            by (metis less_not_refl2 lvl.simps(2) lvl_update_aux n_not_Suc_n
sngl.simps(3))+
              qed (auto simp add: lvl_0_iff)
            qed simp
          qed simp
        qed simp

```

```

lemma lvl_update_incr_iff: ( $lvl(\text{update } a \ b \ t) = lvl \ t + 1$ )  $\longleftrightarrow$ 
  ( $\exists l \ x \ r. \text{update } a \ b \ t = \text{Node } l \ (x, lvl \ t + 1) \ r \wedge lvl \ l = lvl \ r$ )
apply(cases t)
apply(auto simp add: skew_case split_case split: if_splits)
apply(auto split: tree.splits if_splits)
done

```

```

lemma invar_update:  $invar \ t \implies invar(\text{update } a \ b \ t)$ 
proof(induction t rule: tree2_induct)
  case N: (Node  $l \ xy \ n \ r$ )
    hence il:  $invar \ l$  and ir:  $invar \ r$  by auto
    note iil =  $N.IH(1)[OF \ il]$ 
    note iir =  $N.IH(2)[OF \ ir]$ 
    obtain  $x \ y$  where [simp]:  $xy = (x, y)$  by fastforce
    let ? $t$  =  $\text{Node } l \ (xy, n) \ r$ 
    have  $a < x \vee a = x \vee x < a$  by auto
    moreover
      have ?case if  $a < x$ 
      proof (cases rule: lvl_update[of a b l])
        case (Same) thus ?thesis
          using  $\langle a < x \rangle invar\_NodeL[OF \ N.prem \ iil \ Same]$ 
          by (simp add: skew_invar split_invar del: invar.simps)
        next
          case (Incr)
            then obtain  $t1 \ w \ t2$  where ial[simp]:  $\text{update } a \ b \ l = \text{Node } t1 \ (w, n) \ t2$ 
              using  $N.prem$  by (auto simp: lvl_Suc_iff)

```

```

have l12: lvl t1 = lvl t2
  by (metis Incr(1) ial lvl_update_incr_iff tree.inject)
have update a b ?t = split(skew(Node (update a b l) (xy, n) r))
  by(simp add: ‹a<x›)
also have skew(Node (update a b l) (xy, n) r) = Node t1 (w, n) (Node
t2 (xy, n) r)
  by(simp)
also have invar(split ...)
proof (cases r rule: tree2_cases)
  case Leaf
  hence l = Leaf using N.prem1 by(auto simp: lvl_0_iff)
  thus ?thesis using Leaf ial by simp
next
  case [simp]: (Node t3 y m t4)
  show ?thesis
  proof cases
    assume m = n thus ?thesis using N(3) iil by(auto)
  next
    assume m ≠ n thus ?thesis using N(3) iil l12 by(auto)
  qed
qed
finally show ?thesis .
qed
moreover
have ?case if x < a
proof –
  from ‹invar ?t› have n = lvl r ∨ n = lvl r + 1 by auto
  thus ?case
proof
  assume 0: n = lvl r
  have update a b ?t = split(skew(Node l (xy, n) (update a b r)))
    using ‹a>x› by(auto)
  also have skew(Node l (xy, n) (update a b r)) = Node l (xy, n) (update
a b r)
    using N.prem1 by(simp add: skew_case split: tree.split)
  also have invar(split ...)
proof –
  from lvl_update_sngl[OF ir sngl_if_invar[OF ‹invar ?t› 0], of a b]
  obtain t1 p t2 where iar: update a b r = Node t1 (p, n) t2
    using N.prem1 0 by (auto simp: lvl_Suc_iff)
  from N.prem1 iar 0 iir
  show ?thesis by (auto simp: split_case split: tree.splits)
qed
finally show ?thesis .

```

```

next
  assume 1: n = lvl r + 1
  hence sngl ?t by(cases r) auto
  show ?thesis
  proof (cases rule: lvl_update[of a b r])
    case (Same)
    show ?thesis using ⟨x < a⟩ il ir invar_NodeR[OF N.prem1 iir Same]
      by (auto simp add: skew_invar split_invar)
    next
    case (Incr)
    thus ?thesis using invar_NodeR2[OF ⟨invar ?t⟩ Incr(2) 1 iir] 1 ⟨x
< a⟩
      by (auto simp add: skew_invar split_invar split: if_splits)
  qed
  qed
  moreover
  have a = x ⟹ ?case using N.prem1 by auto
  ultimately show ?case by blast
qed simp

```

34.1.2 Proofs for delete

```

declare invar.simps(2)[simp del]

theorem post_delete: invar t ⟹ post_del t (delete x t)
proof (induction t rule: tree2_induct)
  case (Node l ab lv r)

  obtain a b where [simp]: ab = (a,b) by fastforce

  let ?l' = delete x l and ?r' = delete x r
  let ?t = Node l (ab, lv) r let ?t' = delete x ?t

  from Node.prem1 have inv_l: invar l and inv_r: invar r by (auto)

  note post_l' = Node.IH(1)[OF inv_l]
  note preL = pre_adj_if_postL[OF Node.prem1 post_l']

  note post_r' = Node.IH(2)[OF inv_r]
  note preR = pre_adj_if_postR[OF Node.prem1 post_r']

  show ?case
  proof (cases rule: linorder_cases[of x a])

```

```

    case less
    thus ?thesis using Node.prems by (simp add: post_del_adjL preL)
next
    case greater
    thus ?thesis using Node.prems preR by (simp add: post_del_adjR
post_r')
next
    case equal
    show ?thesis
    proof cases
    assume l = Leaf thus ?thesis using equal Node.prems
    by (auto simp: post_del_def invar.simps(2))
next
    assume l ≠ Leaf thus ?thesis using equal Node.prems
    by simp (metis inv_l post_del_adjL post_split_max pre_adj_if_postL)
qed
qed
qed (simp add: post_del_def)

```

34.2 Functional Correctness Proofs

theorem *inorder_update*:

$sorted1(inorder\ t) \implies inorder(update\ x\ y\ t) = upd_list\ x\ y\ (inorder\ t)$
by (*induct* *t*) (*auto simp: upd_list_simps inorder_split inorder_skew*)

theorem *inorder_delete*:

$\llbracket invar\ t; sorted1(inorder\ t) \rrbracket \implies$
 $inorder(delete\ x\ t) = del_list\ x\ (inorder\ t)$
by(*induction* *t*)
(*auto simp: del_list_simps inorder_adjust pre_adj_if_postL pre_adj_if_postR*
post_split_max post_delete split_maxD split: prod.splits)

interpretation *I*: *Map_by_Ordered*

where *empty* = *empty* **and** *lookup* = *lookup* **and** *update* = *update* **and**
delete = *delete*

and *inorder* = *inorder* **and** *inv* = *invar*

proof (*standard, goal_cases*)

case 1 **show** ?*case* **by** (*simp add: empty_def*)

next

case 2 **thus** ?*case* **by**(*simp add: lookup_map_of*)

next

case 3 **thus** ?*case* **by**(*simp add: inorder_update*)

next

```

    case 4 thus ?case by(simp add: inorder_delete)
next
    case 5 thus ?case by(simp add: empty_def)
next
    case 6 thus ?case by(simp add: invar_update)
next
    case 7 thus ?case using post_delete by(auto simp: post_del_def)
qed

end

```

35 Join-Based Implementation of Sets

```

theory Set2_Join
imports
  Isin2
begin

```

This theory implements the set operations *insert*, *delete*, *union*, *intersection* and *difference*. The implementation is based on binary search trees. All operations are reduced to a single operation *join l x r* that joins two BSTs *l* and *r* and an element *x* such that $l < x < r$.

The theory is based on theory *HOL-Data_Structures.Tree2* where nodes have an additional field. This field is ignored here but it means that this theory can be instantiated with red-black trees (see theory *Set2_Join_RBT.thy*) and other balanced trees. This approach is very concrete and fixes the type of trees. Alternatively, one could assume some abstract type *t* of trees with suitable decomposition and recursion operators on it.

```

locale Set2_Join =
fixes join :: ('a::linorder*'b) tree  $\Rightarrow$  'a  $\Rightarrow$  ('a*'b) tree  $\Rightarrow$  ('a*'b) tree
fixes inv :: ('a*'b) tree  $\Rightarrow$  bool
assumes set_join: set_tree (join l a r) = set_tree l  $\cup$  {a}  $\cup$  set_tree r
assumes bst_join: bst (Node l (a, b) r)  $\Longrightarrow$  bst (join l a r)
assumes inv_Leaf: inv  $\langle \rangle$ 
assumes inv_join:  $\llbracket$  inv l; inv r  $\rrbracket \Longrightarrow$  inv (join l a r)
assumes inv_Node:  $\llbracket$  inv (Node l (a,b) r)  $\rrbracket \Longrightarrow$  inv l  $\wedge$  inv r
begin

```

```

declare set_join [simp] Let_def[simp]

```

35.1 split_min

```

fun split_min :: ('a*'b) tree  $\Rightarrow$  'a  $\times$  ('a*'b) tree where
split_min (Node l (a, _) r) =

```

(if $l = \text{Leaf}$ then (a,r) else let $(m,l') = \text{split_min } l$ in $(m, \text{join } l' a r)$)

lemma *split_min_set*:

$\llbracket \text{split_min } t = (m,t'); t \neq \text{Leaf} \rrbracket \implies m \in \text{set_tree } t \wedge \text{set_tree } t = \{m\} \cup \text{set_tree } t'$

proof(*induction t arbitrary: t' rule: tree2_induct*)

case *Node* **thus** ?*case* **by**(*auto split: prod.splits if_splits dest: inv_Node*)

next

case *Leaf* **thus** ?*case* **by** *simp*

qed

lemma *split_min_bst*:

$\llbracket \text{split_min } t = (m,t'); \text{bst } t; t \neq \text{Leaf} \rrbracket \implies \text{bst } t' \wedge (\forall x \in \text{set_tree } t'. m < x)$

proof(*induction t arbitrary: t' rule: tree2_induct*)

case *Node* **thus** ?*case* **by**(*fastforce simp: split_min_set bst_join split: prod.splits if_splits*)

next

case *Leaf* **thus** ?*case* **by** *simp*

qed

lemma *split_min_inv*:

$\llbracket \text{split_min } t = (m,t'); \text{inv } t; t \neq \text{Leaf} \rrbracket \implies \text{inv } t'$

proof(*induction t arbitrary: t' rule: tree2_induct*)

case *Node* **thus** ?*case* **by**(*auto simp: inv_join split: prod.splits if_splits dest: inv_Node*)

next

case *Leaf* **thus** ?*case* **by** *simp*

qed

35.2 join2

definition *join2* :: $('a*'b)$ tree \Rightarrow $('a*'b)$ tree \Rightarrow $('a*'b)$ tree **where**

join2 l $r = (\text{if } r = \text{Leaf} \text{ then } l \text{ else let } (m,r') = \text{split_min } r \text{ in } \text{join } l m r')$

lemma *set_join2[simp]*: $\text{set_tree } (\text{join2 } l r) = \text{set_tree } l \cup \text{set_tree } r$

by(*simp add: join2_def split_min_set split: prod.split*)

lemma *bst_join2*: $\llbracket \text{bst } l; \text{bst } r; \forall x \in \text{set_tree } l. \forall y \in \text{set_tree } r. x < y \rrbracket$

$\implies \text{bst } (\text{join2 } l r)$

by(*simp add: join2_def bst_join split_min_set split_min_bst split: prod.split*)

lemma *inv_join2*: $\llbracket \text{inv } l; \text{inv } r \rrbracket \implies \text{inv } (\text{join2 } l r)$

by(*simp add: join2_def inv_join split_min_set split_min_inv split: prod.split*)

35.3 split

```

fun split :: ('a*'b)tree  $\Rightarrow$  'a  $\Rightarrow$  ('a*'b)tree  $\times$  bool  $\times$  ('a*'b)tree where
  split Leaf k = (Leaf, False, Leaf) |
  split (Node l (a, _) r) x =
    (case cmp x a of
      LT  $\Rightarrow$  let (l1,b,l2) = split l x in (l1, b, join l2 a r) |
      GT  $\Rightarrow$  let (r1,b,r2) = split r x in (join l a r1, b, r2) |
      EQ  $\Rightarrow$  (l, True, r))

```

```

lemma split: split t x = (l,b,r)  $\Longrightarrow$  bst t  $\Longrightarrow$ 
  set_tree l = {a  $\in$  set_tree t. a < x}  $\wedge$  set_tree r = {a  $\in$  set_tree t. x < a}

```

```

 $\wedge$  (b = (x  $\in$  set_tree t))  $\wedge$  bst l  $\wedge$  bst r

```

```

proof(induction t arbitrary: l b r rule: tree2_induct)

```

```

  case Leaf thus ?case by simp

```

```

next

```

```

  case (Node y a b z l c r)

```

```

  consider (LT) l1 xin l2 where (l1,xin,l2) = split y x

```

```

    and split <y, (a, b), z> x = (l1, xin, join l2 a z) and cmp x a = LT

```

```

  | (GT) r1 xin r2 where (r1,xin,r2) = split z x

```

```

    and split <y, (a, b), z> x = (join y a r1, xin, r2) and cmp x a = GT

```

```

  | (EQ) split <y, (a, b), z> x = (y, True, z) and cmp x a = EQ

```

```

    by (force split: cmp_val.splits prod.splits if_splits)

```

```

thus ?case

```

```

proof cases

```

```

  case (LT l1 xin l2)

```

```

    with Node.IH(1)[OF <(l1,xin,l2) = split y x>[symmetric]] Node.prem

```

```

    show ?thesis by (force intro!: bst_join)

```

```

next

```

```

  case (GT r1 xin r2)

```

```

    with Node.IH(2)[OF <(r1,xin,r2) = split z x>[symmetric]] Node.prem

```

```

    show ?thesis by (force intro!: bst_join)

```

```

next

```

```

  case EQ

```

```

    with Node.prem show ?thesis by auto

```

```

qed

```

```

qed

```

```

lemma split_inv: split t x = (l,b,r)  $\Longrightarrow$  inv t  $\Longrightarrow$  inv l  $\wedge$  inv r

```

```

proof(induction t arbitrary: l b r rule: tree2_induct)

```

```

  case Leaf thus ?case by simp

```

```

next

```

case *Node*
thus *?case by(force simp: inv_join split!: prod.splits if_splits dest!: inv_Node)*
qed

declare *split.simps[simp del]*

35.4 *insert*

definition *insert :: 'a \Rightarrow ('a*'b) tree \Rightarrow ('a*'b) tree where*
insert x t = (let (l,_,r) = split t x in join l x r)

lemma *set_tree_insert: bst t \Longrightarrow set_tree (insert x t) = {x} \cup set_tree t*
by(*auto simp add: insert_def split split: prod.split*)

lemma *bst_insert: bst t \Longrightarrow bst (insert x t)*
by(*auto simp add: insert_def bst_join dest: split split: prod.split*)

lemma *inv_insert: inv t \Longrightarrow inv (insert x t)*
by(*force simp: insert_def inv_join dest: split_inv split: prod.split*)

35.5 *delete*

definition *delete :: 'a \Rightarrow ('a*'b) tree \Rightarrow ('a*'b) tree where*
delete x t = (let (l,_,r) = split t x in join2 l r)

lemma *set_tree_delete: bst t \Longrightarrow set_tree (delete x t) = set_tree t - {x}*
by(*auto simp: delete_def split split: prod.split*)

lemma *bst_delete: bst t \Longrightarrow bst (delete x t)*
by(*force simp add: delete_def intro: bst_join2 dest: split split: prod.split*)

lemma *inv_delete: inv t \Longrightarrow inv (delete x t)*
by(*force simp: delete_def inv_join2 dest: split_inv split: prod.split*)

35.6 *union*

fun *union :: ('a*'b)tree \Rightarrow ('a*'b)tree \Rightarrow ('a*'b)tree where*
union t1 t2 =

(if t1 = Leaf then t2 else
if t2 = Leaf then t1 else
case t1 of Node l1 (a, _) r1 \Rightarrow
let (l2,_,r2) = split t2 a;
l' = union l1 l2; r' = union r1 r2
in join l' a r')

declare *union.simps* [*simp del*]

lemma *set_tree_union*: $bst\ t2 \implies set_tree\ (union\ t1\ t2) = set_tree\ t1 \cup set_tree\ t2$

proof(*induction t1 t2 rule: union.induct*)

case (1 *t1 t2*)

then show *?case*

by (*auto simp: union.simps[of t1 t2] split split: tree.split prod.split*)

qed

lemma *bst_union*: $\llbracket bst\ t1; bst\ t2 \rrbracket \implies bst\ (union\ t1\ t2)$

proof(*induction t1 t2 rule: union.induct*)

case (1 *t1 t2*)

thus *?case*

by(*fastforce simp: union.simps[of t1 t2] set_tree_union split intro!: bst_join*

split: tree.split prod.split)

qed

lemma *inv_union*: $\llbracket inv\ t1; inv\ t2 \rrbracket \implies inv\ (union\ t1\ t2)$

proof(*induction t1 t2 rule: union.induct*)

case (1 *t1 t2*)

thus *?case*

by(*auto simp: union.simps[of t1 t2] inv_join split_inv split!: tree.split prod.split dest: inv_Node*)

qed

35.7 *inter*

fun *inter* :: $('a*'b)tree \Rightarrow ('a*'b)tree \Rightarrow ('a*'b)tree$ **where**
inter t1 t2 =

(if t1 = Leaf then Leaf else

if t2 = Leaf then Leaf else

case t1 of Node l1 (a, _) r1 \Rightarrow

let (l2,b,r2) = split t2 a;

l' = inter l1 l2; r' = inter r1 r2

in if b then join l' a r' else join2 l' r')

declare *inter.simps* [*simp del*]

lemma *set_tree_inter*:

$\llbracket bst\ t1; bst\ t2 \rrbracket \implies set_tree\ (inter\ t1\ t2) = set_tree\ t1 \cap set_tree\ t2$

proof(*induction t1 t2 rule: inter.induct*)

case (1 *t1 t2*)

```

show ?case
proof (cases t1 rule: tree2_cases)
  case Leaf thus ?thesis by (simp add: inter.simps)
next
  case [simp]: (Node l1 a _ r1)
  show ?thesis
  proof (cases t2 = Leaf)
    case True thus ?thesis by (simp add: inter.simps)
  next
    case False
    let ?L1 = set_tree l1 let ?R1 = set_tree r1
    have *: a ∉ ?L1 ∪ ?R1 using ⟨bst t1⟩ by (fastforce)
    obtain l2 b r2 where sp: split t2 a = (l2,b,r2) using prod_cases3 by
blast
    let ?L2 = set_tree l2 let ?R2 = set_tree r2 let ?A = if b then {a}
else {}
    have t2: set_tree t2 = ?L2 ∪ ?R2 ∪ ?A and
      **: ?L2 ∩ ?R2 = {} a ∉ ?L2 ∪ ?R2 ?L1 ∩ ?R2 = {} ?L2 ∩ ?R1
= {}
    using split[OF sp] ⟨bst t1⟩ ⟨bst t2⟩ by (force, force, force, force,
force)
    have IHl: set_tree (inter l1 l2) = set_tree l1 ∩ set_tree l2
    using 1.IH(1)[OF _ False _ _ sp[symmetric]] 1.prem1(1,2) split[OF
sp] by simp
    have IHR: set_tree (inter r1 r2) = set_tree r1 ∩ set_tree r2
    using 1.IH(2)[OF _ False _ _ sp[symmetric]] 1.prem1(1,2) split[OF
sp] by simp
    have set_tree t1 ∩ set_tree t2 = (?L1 ∪ ?R1 ∪ {a}) ∩ (?L2 ∪ ?R2
∪ ?A)
    by(simp add: t2)
    also have ... = (?L1 ∩ ?L2) ∪ (?R1 ∩ ?R2) ∪ ?A
    using * ** by auto
    also have ... = set_tree (inter t1 t2)
    using IHl IHR sp inter.simps[of t1 t2] False by(simp)
    finally show ?thesis by simp
  qed
qed
qed

```

```

lemma bst_inter: [ [ bst t1; bst t2 ] ] ⇒ bst (inter t1 t2)
proof(induction t1 t2 rule: inter.induct)
  case (1 t1 t2)
  thus ?case
  by(fastforce simp: inter.simps[of t1 t2] set_tree_inter split

```

intro!: bst_join bst_join2 split: tree.split prod.split)
qed

lemma *inv_inter*: $\llbracket \text{inv } t1; \text{inv } t2 \rrbracket \implies \text{inv } (\text{inter } t1 \ t2)$
proof(*induction* *t1 t2* *rule*: *inter.induct*)

case (1 *t1 t2*)

thus ?*case*

by(*auto simp*: *inter.simps*[*of t1 t2*] *inv_join inv_join2 split_inv*
split!: *tree.split prod.split dest*: *inv_Node*)

qed

35.8 *diff*

fun *diff* :: ('*a**'*b*)*tree* \Rightarrow ('*a**'*b*)*tree* \Rightarrow ('*a**'*b*)*tree* **where**
diff t1 t2 =

(*if t1* = *Leaf* *then Leaf* *else*
if t2 = *Leaf* *then t1* *else*
case t2 of Node l2 (a, _) r2 \Rightarrow
let (l1,_,r1) = split t1 a;
l' = diff l1 l2; r' = diff r1 r2
in join2 l' r')

declare *diff.simps* [*simp del*]

lemma *set_tree_diff*:

$\llbracket \text{bst } t1; \text{bst } t2 \rrbracket \implies \text{set_tree } (\text{diff } t1 \ t2) = \text{set_tree } t1 - \text{set_tree } t2$

proof(*induction* *t1 t2* *rule*: *diff.induct*)

case (1 *t1 t2*)

show ?*case*

proof (*cases t2 rule*: *tree2_cases*)

case *Leaf* **thus** ?*thesis* **by** (*simp add*: *diff.simps*)

next

case [*simp*]: (*Node l2 a _ r2*)

show ?*thesis*

proof (*cases t1* = *Leaf*)

case *True* **thus** ?*thesis* **by** (*simp add*: *diff.simps*)

next

case *False*

let ?*L2* = *set_tree l2* **let** ?*R2* = *set_tree r2*

obtain *l1 b r1* **where** *sp*: *split t1 a* = (*l1,b,r1*) **using** *prod_cases3* **by**

blast

let ?*L1* = *set_tree l1* **let** ?*R1* = *set_tree r1* **let** ?*A* = *if b then* {*a*}

else {}

have *t1*: *set_tree t1* = ?*L1* \cup ?*R1* \cup ?*A* **and**

```

      **:  $a \notin ?L1 \cup ?R1$   $?L1 \cap ?R2 = \{\}$   $?L2 \cap ?R1 = \{\}$ 
      using split[OF sp] <bst t1> <bst t2> by (force, force, force, force)
      have IHl: set_tree (diff l1 l2) = set_tree l1 - set_tree l2
      using 1.IH(1)[OF False _ _ _ sp[symmetric]] 1.prem(1,2) split[OF
sp] by simp
      have IHr: set_tree (diff r1 r2) = set_tree r1 - set_tree r2
      using 1.IH(2)[OF False _ _ _ sp[symmetric]] 1.prem(1,2) split[OF
sp] by simp
      have set_tree t1 - set_tree t2 = (?L1  $\cup$  ?R1) - (?L2  $\cup$  ?R2  $\cup$  {a})
      by(simp add: t1)
      also have ... = (?L1 - ?L2)  $\cup$  (?R1 - ?R2)
      using ** by auto
      also have ... = set_tree (diff t1 t2)
      using IHl IHr sp diff.simps[of t1 t2] False by(simp)
      finally show ?thesis by simp
    qed
  qed
qed

```

```

lemma bst_diff: [ bst t1; bst t2 ]  $\implies$  bst (diff t1 t2)
proof(induction t1 t2 rule: diff.induct)
  case (1 t1 t2)
  thus ?case
  by(fastforce simp: diff.simps[of t1 t2] set_tree_diff split
  intro!: bst_join bst_join2 split: tree.split prod.split)
qed

```

```

lemma inv_diff: [ inv t1; inv t2 ]  $\implies$  inv (diff t1 t2)
proof(induction t1 t2 rule: diff.induct)
  case (1 t1 t2)
  thus ?case
  by(auto simp: diff.simps[of t1 t2] inv_join inv_join2 split_inv
  split!: tree.split prod.split dest: inv_Node)
qed

```

Locale *Set2_Join* implements locale *Set2*:

```

sublocale Set2
where empty = Leaf and insert = insert and delete = delete and isin =
isin
and union = union and inter = inter and diff = diff
and set = set_tree and invar =  $\lambda t. inv t \wedge bst t$ 
proof (standard, goal_cases)
  case 1 show ?case by (simp)
next

```

```

    case 2 thus ?case by(simp add: isin_set_tree)
next
    case 3 thus ?case by (simp add: set_tree_insert)
next
    case 4 thus ?case by (simp add: set_tree_delete)
next
    case 5 thus ?case by (simp add: inv_Leaf)
next
    case 6 thus ?case by (simp add: bst_insert inv_insert)
next
    case 7 thus ?case by (simp add: bst_delete inv_delete)
next
    case 8 thus ?case by(simp add: set_tree_union)
next
    case 9 thus ?case by(simp add: set_tree_inter)
next
    case 10 thus ?case by(simp add: set_tree_diff)
next
    case 11 thus ?case by (simp add: bst_union inv_union)
next
    case 12 thus ?case by (simp add: bst_inter inv_inter)
next
    case 13 thus ?case by (simp add: bst_diff inv_diff)
qed

```

end

interpretation *unbal: Set2_Join*

where *join* = $\lambda l x r. \text{Node } l (x, ()) r$ and *inv* = $\lambda t. \text{True}$

proof (*standard, goal_cases*)

```

    case 1 show ?case by simp
next
    case 2 thus ?case by simp
next
    case 3 thus ?case by simp
next
    case 4 thus ?case by simp
next
    case 5 thus ?case by simp
qed

```

end

36 Join-Based Implementation of Sets via RBTs

```
theory Set2_Join_RBT
imports
  Set2_Join
  RBT_Set
begin
```

36.1 Code

Function *joinL* joins two trees (and an element). Precondition: *bheight l* \leq *bheight r*. Method: Descend along the left spine of *r* until you find a subtree with the same *bheight* as *l*, then combine them into a new red node.

```
fun joinL :: 'a rbt  $\Rightarrow$  'a  $\Rightarrow$  'a rbt  $\Rightarrow$  'a rbt where
joinL l x r =
  (if bheight l  $\geq$  bheight r then R l x r
   else case r of
     B l' x' r'  $\Rightarrow$  baliL (joinL l x l') x' r' |
     R l' x' r'  $\Rightarrow$  R (joinL l x l') x' r')
```

```
fun joinR :: 'a rbt  $\Rightarrow$  'a  $\Rightarrow$  'a rbt  $\Rightarrow$  'a rbt where
joinR l x r =
  (if bheight l  $\leq$  bheight r then R l x r
   else case l of
     B l' x' r'  $\Rightarrow$  baliR l' x' (joinR r' x r) |
     R l' x' r'  $\Rightarrow$  R l' x' (joinR r' x r))
```

```
definition join :: 'a rbt  $\Rightarrow$  'a  $\Rightarrow$  'a rbt  $\Rightarrow$  'a rbt where
join l x r =
  (if bheight l  $>$  bheight r
   then paint Black (joinR l x r)
   else if bheight l  $<$  bheight r
   then paint Black (joinL l x r)
   else B l x r)
```

```
declare joinL.simps[simp del]
declare joinR.simps[simp del]
```

36.2 Properties

36.2.1 Color and height invariants

```
lemma invc2_joinL:
   $\llbracket$  invc l; invc r; bheight l  $\leq$  bheight r  $\rrbracket \Longrightarrow$ 
  invc2 (joinL l x r)
```

$\wedge (\text{bheight } l \neq \text{bheight } r \wedge \text{color } r = \text{Black} \longrightarrow \text{invc}(\text{joinL } l \ x \ r))$
proof (*induct l x r rule: joinL.induct*)
case (*1 l x r*) **thus** ?case
by(*auto simp: invc_baliL invc2I joinL.simps[of l x r] split!: tree.splits if_splits*)
qed

lemma *invc2_joinR*:
 $\llbracket \text{invc } l; \text{invh } l; \text{invc } r; \text{invh } r; \text{bheight } l \geq \text{bheight } r \rrbracket \Longrightarrow$
 $\text{invc2 } (\text{joinR } l \ x \ r)$
 $\wedge (\text{bheight } l \neq \text{bheight } r \wedge \text{color } l = \text{Black} \longrightarrow \text{invc}(\text{joinR } l \ x \ r))$
proof (*induct l x r rule: joinR.induct*)
case (*1 l x r*) **thus** ?case
by(*fastforce simp: invc_baliR invc2I joinR.simps[of l x r] split!: tree.splits if_splits*)
qed

lemma *bheight_joinL*:
 $\llbracket \text{invh } l; \text{invh } r; \text{bheight } l \leq \text{bheight } r \rrbracket \Longrightarrow \text{bheight } (\text{joinL } l \ x \ r) = \text{bheight } r$
proof (*induct l x r rule: joinL.induct*)
case (*1 l x r*) **thus** ?case
by(*auto simp: bheight_baliL joinL.simps[of l x r] split!: tree.split*)
qed

lemma *invh_joinL*:
 $\llbracket \text{invh } l; \text{invh } r; \text{bheight } l \leq \text{bheight } r \rrbracket \Longrightarrow \text{invh } (\text{joinL } l \ x \ r)$
proof (*induct l x r rule: joinL.induct*)
case (*1 l x r*) **thus** ?case
by(*auto simp: invh_baliL bheight_joinL joinL.simps[of l x r] split!: tree.split color.split*)
qed

lemma *bheight_joinR*:
 $\llbracket \text{invh } l; \text{invh } r; \text{bheight } l \geq \text{bheight } r \rrbracket \Longrightarrow \text{bheight } (\text{joinR } l \ x \ r) = \text{bheight } l$
proof (*induct l x r rule: joinR.induct*)
case (*1 l x r*) **thus** ?case
by(*fastforce simp: bheight_baliR joinR.simps[of l x r] split!: tree.split*)
qed

lemma *invh_joinR*:
 $\llbracket \text{invh } l; \text{invh } r; \text{bheight } l \geq \text{bheight } r \rrbracket \Longrightarrow \text{invh } (\text{joinR } l \ x \ r)$
proof (*induct l x r rule: joinR.induct*)

case (1 l x r) **thus** ?case
by(fastforce simp: invh_baliR bheight_joinR joinR.simps[of l x r]
split!: tree.split color.split)
qed

All invariants in one:

lemma inv_joinL: $\llbracket \text{invc } l; \text{invc } r; \text{invh } l; \text{invh } r; \text{bheight } l \leq \text{bheight } r \rrbracket$
 $\implies \text{invc2 } (\text{joinL } l \ x \ r) \wedge (\text{bheight } l \neq \text{bheight } r \wedge \text{color } r = \text{Black} \longrightarrow$
 $\text{invc } (\text{joinL } l \ x \ r))$
 $\wedge \text{invh } (\text{joinL } l \ x \ r) \wedge \text{bheight } (\text{joinL } l \ x \ r) = \text{bheight } r$
proof (induct l x r rule: joinL.induct)
case (1 l x r) **thus** ?case
by(auto simp: inv_baliL invc2I joinL.simps[of l x r] split!: tree.splits
if_splits)
qed

lemma inv_joinR: $\llbracket \text{invc } l; \text{invc } r; \text{invh } l; \text{invh } r; \text{bheight } l \geq \text{bheight } r \rrbracket$
 $\implies \text{invc2 } (\text{joinR } l \ x \ r) \wedge (\text{bheight } l \neq \text{bheight } r \wedge \text{color } l = \text{Black} \longrightarrow$
 $\text{invc } (\text{joinR } l \ x \ r))$
 $\wedge \text{invh } (\text{joinR } l \ x \ r) \wedge \text{bheight } (\text{joinR } l \ x \ r) = \text{bheight } l$
proof (induct l x r rule: joinR.induct)
case (1 l x r) **thus** ?case
by(auto simp: inv_baliR invc2I joinR.simps[of l x r] split!: tree.splits
if_splits)
qed

lemma rbt_join: $\llbracket \text{invc } l; \text{invh } l; \text{invc } r; \text{invh } r \rrbracket \implies \text{rbt}(\text{join } l \ x \ r)$
by(simp add: inv_joinL inv_joinR invh_paint rbt_def color_paint_Black
join_def)

To make sure the the black height is not increased unnecessarily:

lemma bheight_paint_Black: $\text{bheight}(\text{paint } \text{Black } t) \leq \text{bheight } t + 1$
by(cases t) auto

lemma $\llbracket \text{rbt } l; \text{rbt } r \rrbracket \implies \text{bheight}(\text{join } l \ x \ r) \leq \max (\text{bheight } l) (\text{bheight } r)$
 $+ 1$
using bheight_paint_Black[of joinL l x r] bheight_paint_Black[of joinR l
x r]
bheight_joinL[of l r x] bheight_joinR[of l r x]
by(auto simp: max_def rbt_def join_def)

36.2.2 Inorder properties

Currently unused. Instead *Tree2.set_tree* and *Tree2.bst* properties are proved directly.

lemma *inorder_joinL*: $bheight\ l \leq bheight\ r \implies inorder(joinL\ l\ x\ r) = inorder\ l\ @\ x\ \# \inorder\ r$

proof(*induction l x r rule: joinL.induct*)

case (*1 l x r*)

thus ?*case by*(*auto simp: inorder_baliL joinL.simps[of l x r] split!: tree.splits color.splits*)

qed

lemma *inorder_joinR*:

$inorder(joinR\ l\ x\ r) = inorder\ l\ @\ x\ \# \inorder\ r$

proof(*induction l x r rule: joinR.induct*)

case (*1 l x r*)

thus ?*case by* (*force simp: inorder_baliR joinR.simps[of l x r] split!: tree.splits color.splits*)

qed

lemma $inorder(join\ l\ x\ r) = inorder\ l\ @\ x\ \# \inorder\ r$

by(*auto simp: inorder_joinL inorder_joinR inorder_paint join_def split!: tree.splits color.splits if_splits dest!: arg_cong[where f = inorder]*)

36.2.3 Set and bst properties

lemma *set_baliL*:

$set_tree(baliL\ l\ a\ r) = set_tree\ l\ \cup\ \{a\} \cup set_tree\ r$

by(*cases (l,a,r) rule: baliL.cases*) (*auto*)

lemma *set_joinL*:

$bheight\ l \leq bheight\ r \implies set_tree\ (joinL\ l\ x\ r) = set_tree\ l\ \cup\ \{x\} \cup set_tree\ r$

proof(*induction l x r rule: joinL.induct*)

case (*1 l x r*)

thus ?*case by*(*auto simp: set_baliL joinL.simps[of l x r] split!: tree.splits color.splits*)

qed

lemma *set_baliR*:

$set_tree(baliR\ l\ a\ r) = set_tree\ l\ \cup\ \{a\} \cup set_tree\ r$

by(*cases (l,a,r) rule: baliR.cases*) (*auto*)

lemma *set_joinR*:

$set_tree (joinR\ l\ x\ r) = set_tree\ l \cup \{x\} \cup set_tree\ r$

proof(*induction l x r rule: joinR.induct*)

case (*1 l x r*)

thus *?case* **by**(*force simp: set_baliR joinR.simps[of l x r] split!: tree.splits color.splits*)

qed

lemma *set_paint*: $set_tree (paint\ c\ t) = set_tree\ t$

by (*cases t*) *auto*

lemma *set_join*: $set_tree (join\ l\ x\ r) = set_tree\ l \cup \{x\} \cup set_tree\ r$

by(*simp add: set_joinL set_joinR set_paint join_def*)

lemma *bst_baliL*:

$\llbracket bst\ l; bst\ r; \forall x \in set_tree\ l. x < a; \forall x \in set_tree\ r. a < x \rrbracket$

$\implies bst\ (baliL\ l\ a\ r)$

by(*cases (l,a,r) rule: baliL.cases*) (*auto simp: ball_Un*)

lemma *bst_baliR*:

$\llbracket bst\ l; bst\ r; \forall x \in set_tree\ l. x < a; \forall x \in set_tree\ r. a < x \rrbracket$

$\implies bst\ (baliR\ l\ a\ r)$

by(*cases (l,a,r) rule: baliR.cases*) (*auto simp: ball_Un*)

lemma *bst_joinL*:

$\llbracket bst\ (Node\ l\ (a,\ n)\ r); bheight\ l \leq bheight\ r \rrbracket$

$\implies bst\ (joinL\ l\ a\ r)$

proof(*induction l a r rule: joinL.induct*)

case (*1 l a r*)

thus *?case*

by(*auto simp: set_baliL joinL.simps[of l a r] set_joinL ball_Un intro!*)
bst_baliL

split!: tree.splits color.splits)

qed

lemma *bst_joinR*:

$\llbracket bst\ l; bst\ r; \forall x \in set_tree\ l. x < a; \forall y \in set_tree\ r. a < y \rrbracket$

$\implies bst\ (joinR\ l\ a\ r)$

proof(*induction l a r rule: joinR.induct*)

case (*1 l a r*)

thus *?case*

by(*auto simp: set_baliR joinR.simps[of l a r] set_joinR ball_Un intro!*)
bst_baliR

split!: tree.splits color.splits)

qed

lemma *bst_paint*: $bst (paint\ c\ t) = bst\ t$
by(*cases t*) *auto*

lemma *bst_join*:
 $bst (Node\ l\ (a,\ n)\ r) \implies bst (join\ l\ a\ r)$
by(*auto simp: bst_paint bst_joinL bst_joinR join_def*)

lemma *inv_join*: $\llbracket invc\ l; invh\ l; invc\ r; invh\ r \rrbracket \implies invc(join\ l\ x\ r) \wedge invh(join\ l\ x\ r)$
by (*simp add: inv_joinL inv_joinR invh_paint join_def*)

36.2.4 Interpretation of *Set2_Join* with Red-Black Tree

global_interpretation *RBT*: *Set2_Join*
where *join* = *join* **and** *inv* = $\lambda t. invc\ t \wedge invh\ t$
defines *insert_rbt* = *RBT.insert* **and** *delete_rbt* = *RBT.delete* **and** *split_rbt*
= *RBT.split*
and *join2_rbt* = *RBT.join2* **and** *split_min_rbt* = *RBT.split_min*
proof (*standard, goal_cases*)
 case 1 show ?case by (*rule set_join*)
next
 case 2 thus ?case by (*simp add: bst_join*)
next
 case 3 show ?case by *simp*
next
 case 4 thus ?case by (*simp add: inv_join*)
next
 case 5 thus ?case by *simp*
qed

The invariant does not guarantee that the root node is black. This is not required to guarantee that the height is logarithmic in the size — Exercise.

end
theory *Array_Specs*
imports *Main*
begin

 Array Specifications

locale *Array* =
fixes *lookup* :: $'ar \Rightarrow nat \Rightarrow 'a$
fixes *update* :: $nat \Rightarrow 'a \Rightarrow 'ar \Rightarrow 'ar$
fixes *len* :: $'ar \Rightarrow nat$

fixes *array* :: 'a list \Rightarrow 'ar

fixes *list* :: 'ar \Rightarrow 'a list

fixes *invar* :: 'ar \Rightarrow bool

assumes *lookup*: *invar ar* \Longrightarrow $n < \text{len } ar \Longrightarrow \text{lookup } ar \ n = \text{list } ar \ ! \ n$

assumes *update*: *invar ar* \Longrightarrow $n < \text{len } ar \Longrightarrow \text{list}(\text{update } n \ x \ ar) = (\text{list } ar)[n:=x]$

assumes *len_array*: *invar ar* \Longrightarrow $\text{len } ar = \text{length } (\text{list } ar)$

assumes *array*: *list (array xs)* = *xs*

assumes *invar_update*: *invar ar* \Longrightarrow $n < \text{len } ar \Longrightarrow \text{invar}(\text{update } n \ x \ ar)$

assumes *invar_array*: *invar(array xs)*

locale *Array_Flex* = *Array* +

fixes *add_lo* :: 'a \Rightarrow 'ar \Rightarrow 'ar

fixes *del_lo* :: 'ar \Rightarrow 'ar

fixes *add_hi* :: 'a \Rightarrow 'ar \Rightarrow 'ar

fixes *del_hi* :: 'ar \Rightarrow 'ar

assumes *add_lo*: *invar ar* \Longrightarrow $\text{list}(\text{add_lo } a \ ar) = a \ \# \ \text{list } ar$

assumes *del_lo*: *invar ar* \Longrightarrow $\text{list}(\text{del_lo } ar) = \text{tl } (\text{list } ar)$

assumes *add_hi*: *invar ar* \Longrightarrow $\text{list}(\text{add_hi } a \ ar) = \text{list } ar \ @ \ [a]$

assumes *del_hi*: *invar ar* \Longrightarrow $\text{list}(\text{del_hi } ar) = \text{butlast } (\text{list } ar)$

assumes *invar_add_lo*: *invar ar* \Longrightarrow *invar (add_lo a ar)*

assumes *invar_del_lo*: *invar ar* \Longrightarrow *invar (del_lo ar)*

assumes *invar_add_hi*: *invar ar* \Longrightarrow *invar (add_hi a ar)*

assumes *invar_del_hi*: *invar ar* \Longrightarrow *invar (del_hi ar)*

end

37 Braun Trees

theory *Braun_Tree*

imports *HOL-Library.Tree_Real*

begin

Braun Trees were studied by Braun and Rem [5] and later Hoogerwoord [10].

fun *braun* :: 'a tree \Rightarrow bool **where**

braun Leaf = *True* |

braun (Node l x r) = $((\text{size } l = \text{size } r \vee \text{size } l = \text{size } r + 1) \wedge \text{braun } l \wedge \text{braun } r)$

lemma *braun_Node'*:
 $\text{braun } (\text{Node } l \ x \ r) = (\text{size } r \leq \text{size } l \wedge \text{size } l \leq \text{size } r + 1 \wedge \text{braun } l \wedge \text{braun } r)$
by *auto*

The shape of a Braun-tree is uniquely determined by its size:

lemma *braun_unique*: $\llbracket \text{braun } (t1::\text{unit tree}); \text{braun } t2; \text{size } t1 = \text{size } t2 \rrbracket \implies t1 = t2$
proof (*induction t1 arbitrary: t2*)
case *Leaf* **thus** *?case* **by** *simp*
next
case (*Node l1 _ r1*)
from *Node.prem*s(3) **have** $t2 \neq \text{Leaf}$ **by** *auto*
then obtain *l2 x2 r2* **where** [*simp*]: $t2 = \text{Node } l2 \ x2 \ r2$ **by** (*meson neq_Leaf_iff*)
with *Node.prem*s **have** $\text{size } l1 = \text{size } l2 \wedge \text{size } r1 = \text{size } r2$ **by** *auto*
thus *?case* **using** *Node.prem*s(1,2) *Node.IH* **by** *auto*
qed

Braun trees are almost complete:

lemma *acomplete_if_braun*: $\text{braun } t \implies \text{acomplete } t$
proof(*induction t*)
case *Leaf* **show** *?case* **by** (*simp add: acomplete_def*)
next
case (*Node l x r*) **thus** *?case* **using** *acomplete_Node_if_wbal2* **by** *force*
qed

37.1 Numbering Nodes

We show that a tree is a Braun tree iff a parity-based numbering (*braun_indices*) of nodes yields an interval of numbers.

fun *braun_indices* :: $'a \text{ tree} \Rightarrow \text{nat set}$ **where**
 $\text{braun_indices } \text{Leaf} = \{ \} \mid$
 $\text{braun_indices } (\text{Node } l \ _ \ r) = \{1\} \cup (*) \ 2 \ ' \ \text{braun_indices } l \cup \text{Suc } ' (*) \ 2 \ ' \ \text{braun_indices } r$

lemma *braun_indices1*: $0 \notin \text{braun_indices } t$
by (*induction t*) *auto*

lemma *finite_braun_indices*: $\text{finite}(\text{braun_indices } t)$
by (*induction t*) *auto*

One direction:

```

lemma braun_indices_if_braun: braun t  $\implies$  braun_indices t = {1..size
t}
proof(induction t)
  case Leaf thus ?case by simp
next
  have *: (*) 2 ‘ {a..b}  $\cup$  Suc ‘ (*) 2 ‘ {a..b} = {2*a..2*b+1} (is ?l = ?r)
for a b
  proof
    show ?l  $\subseteq$  ?r by auto
  next
    have  $\exists x2 \in \{a..b\}. x \in \{Suc (2*x2), 2*x2\}$  if *:  $x \in \{2*a .. 2*b+1\}$ 
for x
    proof –
      have  $x \text{ div } 2 \in \{a..b\}$  using * by auto
      moreover have  $x \in \{2 * (x \text{ div } 2), Suc(2 * (x \text{ div } 2))\}$  by auto
      ultimately show ?thesis by blast
    qed
    thus ?r  $\subseteq$  ?l by fastforce
  qed
case (Node l x r)
hence size l = size r  $\vee$  size l = size r + 1 (is ?A  $\vee$  ?B) by auto
thus ?case
proof
  assume ?A
  with Node show ?thesis by (auto simp: *)
next
  assume ?B
  with Node show ?thesis by (auto simp: * atLeastAtMostSuc_conv)
qed
qed

```

The other direction is more complicated. The following proof is due to Thomas Sewell.

```

lemma disj_evens_odds: (*) 2 ‘ A  $\cap$  Suc ‘ (*) 2 ‘ B = {}
using double_not_eq_Suc_double by auto

```

```

lemma card_braun_indices: card (braun_indices t) = size t

```

```

proof (induction t)

```

```

  case Leaf thus ?case by simp

```

```

next

```

```

  case Node

```

```

  thus ?case

```

```

  by(auto simp: UNION_singleton_eq_range finite_braun_indices card_Un_disjoint
card_insert_if_disj_evens_odds card_image inj_on_def

```

braun_indices1)
qed

lemma *braun_indices_intvl_base_1*:
assumes *bi*: *braun_indices t = {m..n}*
shows $\{m..n\} = \{1..size\ t\}$
proof (*cases t = Leaf*)
case *True* **then show** *?thesis* **using** *bi* **by** *simp*
next
case *False*
note *eqs = eqset_imp_iff[OF bi]*
from *eqs[of 0]* **have** *0: 0 < m*
by (*simp add: braun_indices1*)
from *eqs[of 1]* **have** *1: m ≤ 1*
by (*cases t; simp add: False*)
from *0 1* **have** *eq1: m = 1* **by** *simp*
from *card_braun_indices[of t]* **show** *?thesis*
by (*simp add: bi eq1*)
qed

lemma *even_of_intvl_intvl*:
fixes *S :: nat set*
assumes $S = \{m..n\} \cap \{i. \text{even } i\}$
shows $\exists m' n'. S = (\lambda i. i * 2) \text{ ` } \{m'..n'\}$
apply (*rule exI[where x=Suc m div 2], rule exI[where x=n div 2]*)
apply (*fastforce simp add: assms mult.commute*)
done

lemma *odd_of_intvl_intvl*:
fixes *S :: nat set*
assumes $S = \{m..n\} \cap \{i. \text{odd } i\}$
shows $\exists m' n'. S = \text{Suc } \text{ ` } (\lambda i. i * 2) \text{ ` } \{m'..n'\}$
proof –
have *step1: $\exists m'. S = \text{Suc } \text{ ` } (\{m'..n - 1\} \cap \{i. \text{even } i\})$*
apply (*rule_tac x=if n = 0 then 1 else m - 1 in exI*)
apply (*auto simp: assms image_def elim!: oddE*)
done
thus *?thesis*
by (*metis even_of_intvl_intvl*)
qed

lemma *image_int_eq_image*:
 $(\forall i \in S. f\ i \in T) \implies (f \text{ ` } S) \cap T = f \text{ ` } S$
 $(\forall i \in S. f\ i \notin T) \implies (f \text{ ` } S) \cap T = \{\}$

by auto

lemma braun_indices1_le:

$i \in \text{braun_indices } t \implies \text{Suc } 0 \leq i$

using braun_indices1_not_less_eq_eq **by** blast

lemma braun_if_braun_indices: $\text{braun_indices } t = \{1..size\ t\} \implies \text{braun } t$

proof(induction t)

case Leaf

then show ?case **by** simp

next

case (Node l x r)

obtain t **where** $t = \text{Node } l\ x\ r$ **by** simp

from Node.premis **have** $\text{eq}: \{2 .. size\ t\} = (\lambda i. i * 2) \text{ ` braun_indices } l \cup \text{Suc } \text{ ` } (\lambda i. i * 2) \text{ ` braun_indices } r$

(**is** ?R = ?S \cup ?T)

apply clarsimp

apply (drule_tac $f = \lambda S. S \cap \{2.. \}$ **in** arg_cong)

apply (simp add: t mult.commute Int_Un_distrib2 image_int_eq_image braun_indices1_le)

done

then have ST: ?S = ?R \cap {i. even i} ?T = ?R \cap {i. odd i}

by (simp_all add: Int_Un_distrib2 image_int_eq_image)

from ST **have** l: braun_indices l = {1 .. size l}

by (fastforce dest: braun_indices_intvl_base_1 dest!: even_of_intvl_intvl
simp: mult.commute inj_image_eq_iff[OF inj_onI])

from ST **have** r: braun_indices r = {1 .. size r}

by (fastforce dest: braun_indices_intvl_base_1 dest!: odd_of_intvl_intvl
simp: mult.commute inj_image_eq_iff[OF inj_onI])

note STa = ST[THEN eqset_imp_iff, THEN iffD2]

note STb = STa[of size t] STa[of size t - 1]

then have sizes: size l = size r \vee size l = size r + 1

apply (clarsimp simp: t l r inj_image_mem_iff[OF inj_onI])

apply (cases even (size l); cases even (size r); clarsimp elim!: oddE;
fastforce)

done

from l r sizes **show** ?case

by (clarsimp simp: Node.IH)

qed

lemma braun_iff_braun_indices: $\text{braun } t \longleftrightarrow \text{braun_indices } t = \{1..size\ t\}$

using braun_if_braun_indices braun_indices_if_braun **by** blast

end

38 Arrays via Braun Trees

theory *Array_Braun*

imports

Array_Specs

Braun_Tree

begin

38.1 Array

fun *lookup1* :: 'a tree \Rightarrow nat \Rightarrow 'a **where**

lookup1 (Node l x r) n = (if n=1 then x else *lookup1* (if even n then l else r) (n div 2))

fun *update1* :: nat \Rightarrow 'a \Rightarrow 'a tree \Rightarrow 'a tree **where**

update1 n x Leaf = Node Leaf x Leaf |

update1 n x (Node l a r) =

(if n=1 then Node l x r else

if even n then Node (*update1* (n div 2) x l) a r

else Node l a (*update1* (n div 2) x r))

fun *adds* :: 'a list \Rightarrow nat \Rightarrow 'a tree \Rightarrow 'a tree **where**

adds [] n t = t |

adds (x#xs) n t = *adds* xs (n+1) (*update1* (n+1) x t)

fun *list* :: 'a tree \Rightarrow 'a list **where**

list Leaf = [] |

list (Node l x r) = x # *splice* (*list* l) (*list* r)

38.1.1 Functional Correctness

lemma *size_list*: *size*(*list* t) = *size* t

by(*induction* t)(*auto*)

lemma *minus1_div2*: (n - Suc 0) div 2 = (if odd n then n div 2 else n div 2 - 1)

by *auto arith*

```

lemma nth_splice:  $\llbracket n < \text{size } xs + \text{size } ys; \text{size } ys \leq \text{size } xs; \text{size } xs \leq \text{size } ys + 1 \rrbracket$ 
   $\implies \text{splice } xs \text{ } ys \ ! \ n = (\text{if even } n \text{ then } xs \text{ else } ys) \ ! \ (n \ \text{div} \ 2)$ 
apply(induction xs ys arbitrary: n rule: splice.induct)
apply (auto simp: nth_Cons' minus1_div2)
done

```

```

lemma div2_in_bounds:
   $\llbracket \text{braun } (Node \ l \ x \ r); n \in \{1.. \text{size}(Node \ l \ x \ r)\}; n > 1 \rrbracket \implies$ 
   $(\text{odd } n \longrightarrow n \ \text{div} \ 2 \in \{1.. \text{size } r\}) \wedge (\text{even } n \longrightarrow n \ \text{div} \ 2 \in \{1.. \text{size } l\})$ 
by auto arith

```

```

declare upt_Suc[simp del]

```

```

lookup1 lemma nth_list_lookup1:  $\llbracket \text{braun } t; i < \text{size } t \rrbracket \implies \text{list } t \ ! \ i =$ 
lookup1 t (i+1)
proof(induction t arbitrary: i)
  case Leaf thus ?case by simp
next
  case Node
  thus ?case using div2_in_bounds[OF Node.prem1], of i+1]
  by (auto simp: nth_splice minus1_div2 size_list)
qed

```

```

lemma list_eq_map_lookup1:  $\text{braun } t \implies \text{list } t = \text{map } (\text{lookup1 } t) \ [1.. < \text{size } t + 1]$ 
by(auto simp add: list_eq_iff_nth_eq size_list nth_list_lookup1)

```

```

update1 lemma size_update1:  $\llbracket \text{braun } t; n \in \{1.. \text{size } t\} \rrbracket \implies \text{size}(\text{update1 } n \ x \ t) = \text{size } t$ 
proof(induction t arbitrary: n)
  case Leaf thus ?case by simp
next
  case Node thus ?case using div2_in_bounds[OF Node.prem1] by simp
qed

```

```

lemma braun_update1:  $\llbracket \text{braun } t; n \in \{1.. \text{size } t\} \rrbracket \implies \text{braun}(\text{update1 } n \ x \ t)$ 
proof(induction t arbitrary: n)
  case Leaf thus ?case by simp
next
  case Node thus ?case
  using div2_in_bounds[OF Node.prem1] by (simp add: size_update1)

```

qed

lemma *lookup1_update1*: $\llbracket \text{braun } t; n \in \{1.. \text{size } t\} \rrbracket \implies$
lookup1 (*update1* *n x t*) *m* = (if *n=m* then *x* else *lookup1 t m*)

proof(*induction t arbitrary: m n*)

case *Leaf*

then show *?case* **by** *simp*

next

have *aux*: $\llbracket \text{odd } n; \text{odd } m \rrbracket \implies n \text{ div } 2 = (m::\text{nat}) \text{ div } 2 \longleftrightarrow m=n$ **for**
m n

using *odd_two_times_div_two_succ* **by** *fastforce*

case *Node*

thus *?case* **using** *div2_in_bounds[OF Node.prem1]* **by** (*auto simp: aux*)

qed

lemma *list_update1*: $\llbracket \text{braun } t; n \in \{1.. \text{size } t\} \rrbracket \implies \text{list}(\text{update1 } n x t)$
= (*list t*)[*n-1* := *x*]

by(*auto simp add: list_eq_map_lookup1 list_eq_iff_nth_eq lookup1_update1*
size_update1 braun_update1)

 A second proof of $\llbracket \text{braun } ?t; ?n \in \{1.. \text{size } ?t\} \rrbracket \implies \text{list}(\text{update1 } ?n ?x$
?t) = (*list ?t*)[*?n - 1* := *?x*]:

lemma *diff1_eq_iff*: $n > 0 \implies n - \text{Suc } 0 = m \longleftrightarrow n = m+1$

by *arith*

lemma *list_update_splice*:

$\llbracket n < \text{size } xs + \text{size } ys; \text{size } ys \leq \text{size } xs; \text{size } xs \leq \text{size } ys + 1 \rrbracket \implies$

(*splice xs ys*) [*n* := *x*] =

(if even *n* then *splice* (*xs*[*n* div 2 := *x*]) *ys* else *splice xs* (*ys*[*n* div 2 := *x*]))

by(*induction xs ys arbitrary: n rule: splice.induct*) (*auto split: nat.split*)

lemma *list_update2*: $\llbracket \text{braun } t; n \in \{1.. \text{size } t\} \rrbracket \implies \text{list}(\text{update1 } n x t)$
= (*list t*)[*n-1* := *x*]

proof(*induction t arbitrary: n*)

case *Leaf* **thus** *?case* **by** *simp*

next

case (*Node l a r*) **thus** *?case* **using** *div2_in_bounds[OF Node.prem1]*

by(*auto simp: list_update_splice diff1_eq_iff size_list split: nat.split*)

qed

adds **lemma** *splice_last*: **shows**

$\text{size } ys \leq \text{size } xs \implies \text{splice } (xs @ [x]) ys = \text{splice } xs ys @ [x]$

and $\text{size } ys+1 \geq \text{size } xs \implies \text{splice } xs (ys @ [y]) = \text{splice } xs ys @ [y]$

by(*induction xs ys arbitrary: x y rule: splice.induct*) (*auto*)

lemma *list_add_hi*: $\text{braun } t \implies \text{list}(\text{update1 } (\text{Suc}(\text{size } t)) \ x \ t) = \text{list } t \ @ \ [x]$
by(*induction t*)(*auto simp: splice_last size_list*)

lemma *size_add_hi*: $\text{braun } t \implies m = \text{size } t \implies \text{size}(\text{update1 } (\text{Suc } m) \ x \ t) = \text{size } t + 1$
by(*induction t arbitrary: m*)(*auto*)

lemma *braun_add_hi*: $\text{braun } t \implies \text{braun}(\text{update1 } (\text{Suc}(\text{size } t)) \ x \ t)$
by(*induction t*)(*auto simp: size_add_hi*)

lemma *size_braun_adds*:
 $\llbracket \text{braun } t; \text{size } t = n \rrbracket \implies \text{size}(\text{adds } xs \ n \ t) = \text{size } t + \text{length } xs \wedge \text{braun } (\text{adds } xs \ n \ t)$
by(*induction xs arbitrary: t n*)(*auto simp: braun_add_hi size_add_hi*)

lemma *list_adds*: $\llbracket \text{braun } t; \text{size } t = n \rrbracket \implies \text{list}(\text{adds } xs \ n \ t) = \text{list } t \ @ \ xs$
by(*induction xs arbitrary: t n*)(*auto simp: size_braun_adds list_add_hi size_add_hi braun_add_hi*)

38.1.2 Array Implementation

interpretation *A*: *Array*

where *lookup* = $\lambda(t,l) \ n. \text{lookup1 } t \ (n+1)$
and *update* = $\lambda n \ x \ (t,l). (\text{update1 } (n+1) \ x \ t, l)$
and *len* = $\lambda(t,l). \ l$
and *array* = $\lambda xs. (\text{adds } xs \ 0 \ \text{Leaf}, \text{length } xs)$
and *invar* = $\lambda(t,l). \text{braun } t \wedge l = \text{size } t$
and *list* = $\lambda(t,l). \text{list } t$

proof (*standard, goal_cases*)

case 1 thus ?case by (*simp add: nth_list_lookup1 split: prod.splits*)
next

case 2 thus ?case by (*simp add: list_update1 split: prod.splits*)
next

case 3 thus ?case by (*simp add: size_list split: prod.splits*)
next

case 4 thus ?case by (*simp add: list_adds*)
next

case 5 thus ?case by (*simp add: braun_update1 size_update1 split: prod.splits*)
next

case 6 thus ?case by (*simp add: size_braun_adds split: prod.splits*)
qed

38.2 Flexible Array

```
fun add_lo where  
add_lo x Leaf = Node Leaf x Leaf |  
add_lo x (Node l a r) = Node (add_lo a r) x l
```

```
fun merge where  
merge Leaf r = r |  
merge (Node l a r) rr = Node rr a (merge l r)
```

```
fun del_lo where  
del_lo Leaf = Leaf |  
del_lo (Node l a r) = merge l r
```

```
fun del_hi :: nat ⇒ 'a tree ⇒ 'a tree where  
del_hi n Leaf = Leaf |  
del_hi n (Node l x r) =  
  (if n = 1 then Leaf  
   else if even n  
     then Node (del_hi (n div 2) l) x r  
     else Node l x (del_hi (n div 2) r))
```

38.2.1 Functional Correctness

```
add_lo lemma list_add_lo: braun t ⇒ list (add_lo a t) = a # list t  
by(induction t arbitrary: a) auto
```

```
lemma braun_add_lo: braun t ⇒ braun(add_lo x t)  
by(induction t arbitrary: x) (auto simp add: list_add_lo simp flip: size_list)
```

```
del_lo lemma list_merge: braun (Node l x r) ⇒ list(merge l r) = splice  
(list l) (list r)  
by (induction l r rule: merge.induct) auto
```

```
lemma braun_merge: braun (Node l x r) ⇒ braun(merge l r)  
by (induction l r rule: merge.induct)(auto simp add: list_merge simp flip: size_list)
```

```
lemma list_del_lo: braun t ⇒ list(del_lo t) = tl (list t)  
by (cases t) (simp_all add: list_merge)
```

```
lemma braun_del_lo: braun t ⇒ braun(del_lo t)  
by (cases t) (simp_all add: braun_merge)
```

del_hi **lemma** *list_Nil_iff*: $list\ t = [] \longleftrightarrow t = Leaf$
by(*cases t*) *simp_all*

lemma *butlast_splice*: $butlast\ (splice\ xs\ ys) =$
(if size xs > size ys then splice (butlast xs) ys else splice xs (butlast ys))
by(*induction xs ys rule: splice.induct*) (*auto*)

lemma *list_del_hi*: $braun\ t \implies size\ t = st \implies list(del_hi\ st\ t) = butlast(list\ t)$
apply(*induction t arbitrary: st*)
by(*auto simp: list_Nil_iff size_list butlast_splice*)

lemma *braun_del_hi*: $braun\ t \implies size\ t = st \implies braun(del_hi\ st\ t)$
apply(*induction t arbitrary: st*)
by(*auto simp: list_del_hi simp flip: size_list*)

38.2.2 Flexible Array Implementation

interpretation *AF*: *Array_Flex*
where $lookup = \lambda(t,l)\ n.\ lookup1\ t\ (n+1)$
and $update = \lambda n\ x\ (t,l).\ (update1\ (n+1)\ x\ t,\ l)$
and $len = \lambda(t,l).\ l$
and $array = \lambda xs.\ (adds\ xs\ 0\ Leaf,\ length\ xs)$
and $invar = \lambda(t,l).\ braun\ t \wedge l = size\ t$
and $list = \lambda(t,l).\ list\ t$
and $add_lo = \lambda x\ (t,l).\ (add_lo\ x\ t,\ l+1)$
and $del_lo = \lambda(t,l).\ (del_lo\ t,\ l-1)$
and $add_hi = \lambda x\ (t,l).\ (update1\ (Suc\ l)\ x\ t,\ l+1)$
and $del_hi = \lambda(t,l).\ (del_hi\ l\ t,\ l-1)$
proof (*standard, goal_cases*)
case 1 thus ?case by (*simp add: list_add_lo split: prod.splits*)
next
case 2 thus ?case by (*simp add: list_del_lo split: prod.splits*)
next
case 3 thus ?case by (*simp add: list_add_hi braun_add_hi split: prod.splits*)
next
case 4 thus ?case by (*simp add: list_del_hi split: prod.splits*)
next
case 5 thus ?case by (*simp add: braun_add_lo list_add_lo flip: size_list split: prod.splits*)
next
case 6 thus ?case by (*simp add: braun_del_lo list_del_lo flip: size_list split: prod.splits*)
next

```

  case 7 thus ?case by (simp add: size_add_hi braun_add_hi split: prod.splits)
next
  case 8 thus ?case by (simp add: braun_del_hi list_del_hi flip: size_list
split: prod.splits)
qed

```

38.3 Faster

38.3.1 Size

```

fun diff :: 'a tree  $\Rightarrow$  nat  $\Rightarrow$  nat where
diff Leaf _ = 0 |
diff (Node l x r) n = (if n=0 then 1 else if even n then diff r (n div 2 -
1) else diff l (n div 2))

```

```

fun size_fast :: 'a tree  $\Rightarrow$  nat where
size_fast Leaf = 0 |
size_fast (Node l x r) = (let n = size_fast r in 1 + 2*n + diff l n)

```

```

declare Let_def[simp]

```

```

lemma diff: braun t  $\Longrightarrow$  size t : {n, n + 1}  $\Longrightarrow$  diff t n = size t - n
by(induction t arbitrary: n) auto

```

```

lemma size_fast: braun t  $\Longrightarrow$  size_fast t = size t
by(induction t) (auto simp add: diff)

```

38.3.2 Initialization with 1 element

```

fun braun_of_naive :: 'a  $\Rightarrow$  nat  $\Rightarrow$  'a tree where
braun_of_naive x n = (if n=0 then Leaf
  else let m = (n-1) div 2
    in if odd n then Node (braun_of_naive x m) x (braun_of_naive x m)
    else Node (braun_of_naive x (m + 1)) x (braun_of_naive x m))

```

```

fun braun2_of :: 'a  $\Rightarrow$  nat  $\Rightarrow$  'a tree * 'a tree where
braun2_of x n = (if n = 0 then (Leaf, Node Leaf x Leaf)
  else let (s,t) = braun2_of x ((n-1) div 2)
    in if odd n then (Node s x s, Node t x s) else (Node t x s, Node t x t))

```

```

definition braun_of :: 'a  $\Rightarrow$  nat  $\Rightarrow$  'a tree where
braun_of x n = fst (braun2_of x n)

```

```

declare braun2_of.simps [simp del]

```

lemma *braun2_of_size_braun*: $\text{braun2_of } x \ n = (s,t) \implies \text{size } s = n \wedge \text{size } t = n+1 \wedge \text{braun } s \wedge \text{braun } t$
proof(*induction* $x \ n$ *arbitrary*: $s \ t$ *rule*: *braun2_of.induct*)
 case (1 $x \ n$)
 then show ?*case*
 by (*auto simp*: *braun2_of.simps*[*of* $x \ n$] *split*: *prod.splits if_splits*) *presburger+*
qed

lemma *braun2_of_replicate*:
 $\text{braun2_of } x \ n = (s,t) \implies \text{list } s = \text{replicate } n \ x \wedge \text{list } t = \text{replicate } (n+1) \ x$
proof(*induction* $x \ n$ *arbitrary*: $s \ t$ *rule*: *braun2_of.induct*)
 case (1 $x \ n$)
 have $x \# \text{replicate } m \ x = \text{replicate } (m+1) \ x$ **for** m **by** *simp*
 with 1 **show** ?*case*
 apply (*auto simp*: *braun2_of.simps*[*of* $x \ n$] *replicate.simps*(2)[*of* 0 x]
 simp del: *replicate.simps*(2) *split*: *prod.splits if_splits*)
 by *presburger+*
qed

corollary *braun_braun_of*: $\text{braun}(\text{braun_of } x \ n)$
unfolding *braun_of_def* **by** (*metis eq_fst_iff braun2_of_size_braun*)

corollary *list_braun_of*: $\text{list}(\text{braun_of } x \ n) = \text{replicate } n \ x$
unfolding *braun_of_def* **by** (*metis eq_fst_iff braun2_of_replicate*)

38.3.3 Proof Infrastructure

Originally due to Thomas Sewell.

take_nth **fun** *take_nth* :: $\text{nat} \Rightarrow \text{nat} \Rightarrow 'a \ \text{list} \Rightarrow 'a \ \text{list}$ **where**
take_nth $i \ k \ [] = [] \mid$
take_nth $i \ k \ (x \# xs) = (\text{if } i = 0 \ \text{then } x \# \text{take_nth } (2^k - 1) \ k \ xs$
 else take_nth $(i - 1) \ k \ xs)$

This is the more concise definition but seems to complicate the proofs:

lemma *take_nth_eq_nth*: $\text{take_nth } i \ k \ xs = \text{nth } xs \ (\bigcup n. \{n * 2^k + i\})$
proof(*induction* xs *arbitrary*: i)
 case *Nil*
 then show ?*case* **by** *simp*
next
 case (*Cons* $x \ xs$)
 show ?*case*

proof cases
assume [*simp*]: $i = 0$
have $(\bigcup n. \{(n+1) * 2^k - 1\}) = \{m. \exists n. \text{Suc } m = n * 2^k\}$
apply (*auto simp del: mult_Suc*)
by (*metis diff_Suc_Suc diff_zero mult_eq_0_iff not0_implies_Suc*)
thus ?thesis **by** (*simp add: Cons.IH ac_simps nthns_Cons*)
next
assume [*arith*]: $i \neq 0$
have $(\bigcup n. \{n * 2^k + i - 1\}) = \{m. \exists n. \text{Suc } m = n * 2^k + i\}$
apply *auto*
by (*metis diff_Suc_Suc diff_zero*)
thus ?thesis **by** (*simp add: Cons.IH nthns_Cons*)
qed
qed

lemma *take_nthns_drop*:
 $\text{take_nthns } i \ k \ (\text{drop } j \ xs) = \text{take_nthns } (i + j) \ k \ xs$
by (*induct xs arbitrary: i j; simp add: drop_Cons split: nat.split*)

lemma *take_nthns_00*:
 $\text{take_nthns } 0 \ 0 \ xs = xs$
by (*induct xs; simp*)

lemma *splice_take_nthns*:
 $\text{splice } (\text{take_nthns } 0 \ (\text{Suc } 0) \ xs) \ (\text{take_nthns } (\text{Suc } 0) \ (\text{Suc } 0) \ xs) = xs$
by (*induct xs; simp*)

lemma *take_nthns_take_nthns*:
 $\text{take_nthns } i \ m \ (\text{take_nthns } j \ n \ xs) = \text{take_nthns } ((i * 2^n) + j) \ (m + n) \ xs$
by (*induct xs arbitrary: i j; simp add: algebra_simps power_add*)

lemma *take_nthns_empty*:
 $(\text{take_nthns } i \ k \ xs = []) = (\text{length } xs \leq i)$
by (*induction xs arbitrary: i k auto*)

lemma *hd_take_nthns*:
 $i < \text{length } xs \implies \text{hd}(\text{take_nthns } i \ k \ xs) = xs ! i$
by (*induction xs arbitrary: i k auto*)

lemma *take_nthns_01_splice*:
 $\llbracket \text{length } xs = \text{length } ys \vee \text{length } xs = \text{length } ys + 1 \rrbracket \implies$
 $\text{take_nthns } 0 \ (\text{Suc } 0) \ (\text{splice } xs \ ys) = xs \wedge$
 $\text{take_nthns } (\text{Suc } 0) \ (\text{Suc } 0) \ (\text{splice } xs \ ys) = ys$
by (*induct xs arbitrary: ys; case_tac ys; simp*)

```

lemma length_take_nths_00:
  length (take_nths 0 (Suc 0) xs) = length (take_nths (Suc 0) (Suc 0) xs)
  ∨
  length (take_nths 0 (Suc 0) xs) = length (take_nths (Suc 0) (Suc 0) xs)
  + 1
by (induct xs) auto

```

```

braun_list fun braun_list :: 'a tree ⇒ 'a list ⇒ bool where
braun_list Leaf xs = (xs = []) |
braun_list (Node l x r) xs = (xs ≠ [] ∧ x = hd xs ∧
  braun_list l (take_nths 1 1 xs) ∧
  braun_list r (take_nths 2 1 xs))

```

```

lemma braun_list_eq:
  braun_list t xs = (braun t ∧ xs = list t)
proof (induct t arbitrary: xs)
  case Leaf
  show ?case by simp
next
  case Node
  show ?case
    using length_take_nths_00[of xs] splice_take_nths[of xs]
    by (auto simp: neq_Nil_conv Node.hyps size_list[symmetric] take_nths_01_splice)
qed

```

38.3.4 Converting a list of elements into a Braun tree

```

fun nodes :: 'a tree list ⇒ 'a list ⇒ 'a tree list ⇒ 'a tree list where
nodes (l#ls) (x#xs) (r#rs) = Node l x r # nodes ls xs rs |
nodes (l#ls) (x#xs) [] = Node l x Leaf # nodes ls xs [] |
nodes [] (x#xs) (r#rs) = Node Leaf x r # nodes [] xs rs |
nodes [] (x#xs) [] = Node Leaf x Leaf # nodes [] xs [] |
nodes ls [] rs = []

```

```

fun brauns :: nat ⇒ 'a list ⇒ 'a tree list where
brauns k xs = (if xs = [] then [] else
  let ys = take (2k) xs;
      zs = drop (2k) xs;
      ts = brauns (k+1) zs
  in nodes ts ys (drop (2k) ts))

```

```

declare brauns.simps[simp del]

```

definition *brauns1* :: 'a list \Rightarrow 'a tree **where**
brauns1 *xs* = (if *xs* = [] then Leaf else *brauns* 0 *xs* ! 0)

fun *T_brauns* :: nat \Rightarrow 'a list \Rightarrow nat **where**
T_brauns *k xs* = (if *xs* = [] then 0 else
 let *ys* = take (2^{*k*}) *xs*;
 zs = drop (2^{*k*}) *xs*;
 ts = *brauns* (*k*+1) *zs*
 in 4 * min (2^{*k*}) (length *xs*) + *T_brauns* (*k*+1) *zs*)

Functional correctness The proof is originally due to Thomas Sewell.

lemma *length_nodes*:
 length (nodes *ls xs rs*) = length *xs*
by (induct *ls xs rs* rule: nodes.induct; simp)

lemma *nth_nodes*:
i < length *xs* \implies nodes *ls xs rs* ! *i* =
 Node (if *i* < length *ls* then *ls* ! *i* else Leaf) (*xs* ! *i*)
 (if *i* < length *rs* then *rs* ! *i* else Leaf)
by (induct *ls xs rs* arbitrary: *i* rule: nodes.induct;
 simp add: nth_Cons split: nat.split)

theorem *length_brauns*:
 length (brauns *k xs*) = min (length *xs*) (2^{*k*})
proof (induct *xs* arbitrary: *k* rule: measure_induct_rule[where f=length])
 case (less *xs*) **thus** ?case **by** (simp add: brauns.simps[of *k xs*] length_nodes)
qed

theorem *brauns_correct*:
i < min (length *xs*) (2^{*k*}) \implies braun_list (brauns *k xs* ! *i*) (take_nth*s* *i*
k xs)
proof (induct *xs* arbitrary: *i k* rule: measure_induct_rule[where f=length])
 case (less *xs*)
 have *xs* \neq [] **using** less.prem*s* **by** auto
 let ?*zs* = drop (2^{*k*}) *xs*
 let ?*ts* = brauns (Suc *k*) ?*zs*
 from less.hyps[of ?*zs* _ Suc *k*]
 have IH: $\llbracket j = i + 2^k; i < \min (\text{length } ?zs) (2^{k+1}) \rrbracket \implies$
 braun_list (?*ts* ! *i*) (take_nth*s* *j* (Suc *k*) *xs*) **for** *i j*
 using $\langle xs \neq [] \rangle$ **by** (simp add: take_nth_drop)
show ?case
using less.prem*s*
by (auto simp: brauns.simps[of *k xs*] nth_nodes take_nth_take_nth)

IH take_nth_empty hd_take_nth length_brauns)

qed

corollary *brauns1_correct*:

braun (brauns1 xs) ∧ list (brauns1 xs) = xs

using *brauns_correct*[of 0 xs 0]

by (*simp add: brauns1_def braun_list_eq take_nth_00*)

Running Time Analysis theorem *T_brauns*:

*T_brauns k xs = 4 * length xs*

proof (*induction xs arbitrary: k rule: measure_induct_rule*[**where** *f = length*])

case (*less xs*)

show *?case*

proof *cases*

assume *xs = []*

thus *?thesis* **by**(*simp*)

next

assume *xs ≠ []*

let *?zs = drop (2^k) xs*

have *T_brauns k xs = T_brauns (k+1) ?zs + 4 * min (2^k) (length xs)*

using *⟨xs ≠ []⟩* **by**(*simp*)

also have *... = 4 * length ?zs + 4 * min (2^k) (length xs)*

using *less*[of *?zs k+1*] *⟨xs ≠ []⟩*

by (*simp*)

also have *... = 4 * length xs*

by(*simp*)

finally show *?case* .

qed

qed

38.3.5 Converting a Braun Tree into a List of Elements

The code and the proof are originally due to Thomas Sewell (except running time).

function *list_fast_rec* :: 'a tree list ⇒ 'a list **where**

list_fast_rec ts = (let us = filter (λt. t ≠ Leaf) ts in

if us = [] then [] else

map value us @ list_fast_rec (map left us @ map right us))

by (*pat_completeness, auto*)

lemma *list_fast_rec_term1*: *ts ≠ [] ⇒ Leaf ∉ set ts ⇒*

```

    sum_list (map (size o left) ts) + sum_list (map (size o right) ts) <
sum_list (map size ts)
apply (clarsimp simp: sum_list_addf[symmetric] sum_list_map_filter')
apply (rule sum_list_strict_mono;clarsimp?)
apply (case_tac x; simp)
done

```

```

lemma list_fast_rec_term: us ≠ [] ⇒ us = filter (λt. t ≠ ⟨⟩) ts ⇒
    sum_list (map (size o left) us) + sum_list (map (size o right) us) <
sum_list (map size ts)
apply (rule order_less_le_trans, rule list_fast_rec_term1, simp_all)
apply (rule sum_list_filter_le_nat)
done

```

termination

```

apply (relation measure (sum_list o map size))
apply simp
apply (simp add: list_fast_rec_term)
done

```

```

declare list_fast_rec.simps[simp del]

```

```

definition list_fast :: 'a tree ⇒ 'a list where
list_fast t = list_fast_rec [t]

```

```

function T_list_fast_rec :: 'a tree list ⇒ nat where
T_list_fast_rec ts = (let us = filter (λt. t ≠ Leaf) ts
    in length ts + (if us = [] then 0 else
    5 * length us + T_list_fast_rec (map left us @ map right us)))
by (pat_completeness, auto)

```

termination

```

apply (relation measure (sum_list o map size))
apply simp
apply (simp add: list_fast_rec_term)
done

```

```

declare T_list_fast_rec.simps[simp del]

```

```

Functional Correctness lemma list_fast_rec_all_Leaf:
    ∀ t ∈ set ts. t = Leaf ⇒ list_fast_rec ts = []
by (simp add: filter_empty_conv list_fast_rec.simps)

```

lemma *take_nth_eq_single*:

$length\ xs - i < 2^n \implies take_nth\ i\ n\ xs = take\ 1\ (drop\ i\ xs)$

by (*induction xs arbitrary: i n; simp add: drop_Cons'*)

lemma *braun_list_Nil*:

$braun_list\ t\ [] = (t = Leaf)$

by (*cases t; simp*)

lemma *braun_list_not_Nil*: $xs \neq [] \implies$

$braun_list\ t\ xs =$

$(\exists l\ x\ r. t = Node\ l\ x\ r \wedge x = hd\ xs \wedge$

$braun_list\ l\ (take_nth\ 1\ 1\ xs) \wedge$

$braun_list\ r\ (take_nth\ 2\ 1\ xs))$

by(*cases t; simp*)

theorem *list_fast_rec_correct*:

$\llbracket length\ ts = 2^k; \forall i < 2^k. braun_list\ (ts\ !\ i)\ (take_nth\ i\ k\ xs) \rrbracket$

$\implies list_fast_rec\ ts = xs$

proof (*induct xs arbitrary: k ts rule: measure_induct_rule[where f=length]*)

case (*less xs*)

show *?case*

proof (*cases length xs < 2^k*)

case *True*

from *less.prem*s *True* **have** *filter*:

$\exists n. ts = map\ (\lambda x. Node\ Leaf\ x\ Leaf)\ xs\ @\ replicate\ n\ Leaf$

apply (*rule_tac x=length ts - length xs in exI*)

apply (*clarsimp simp: list_eq_iff_nth_eq*)

apply(*auto simp: nth_append braun_list_not_Nil take_nth_eq_single braun_list_Nil hd_drop_conv_nth*)

done

thus *?thesis*

by (*clarsimp simp: list_fast_rec.simps[of ts] o_def list_fast_rec_all_Leaf*)

next

case *False*

with *less.prem*s(2) **have** *:

$\forall i < 2^k. ts\ !\ i \neq Leaf$

$\wedge value\ (ts\ !\ i) = xs\ !\ i$

$\wedge braun_list\ (left\ (ts\ !\ i))\ (take_nth\ (i + 2^k)\ (Suc\ k)\ xs)$

$\wedge (\forall ys. ys = take_nth\ (i + 2 * 2^k)\ (Suc\ k)\ xs$

$\longrightarrow braun_list\ (right\ (ts\ !\ i))\ ys)$

by (*auto simp: take_nth_empty hd_take_nth braun_list_not_Nil take_nth_take_nth*)

algebra_simps)

have 1: $map\ value\ ts = take\ (2^k)\ xs$

```

    using less.premis(1) False by (simp add: list_eq_iff_nth_eq *)
  have 2: list_fast_rec (map left ts @ map right ts) = drop (2 ^ k) xs
    using less.premis(1) False
    by (auto intro!: Nat.diff_less less.hyps[where k= Suc k]
        simp: nth_append * take_nths_drop algebra_simps)
  from less.premis(1) False show ?thesis
    by (auto simp: list_fast_rec.simps[of ts] 1 2 * all_set_conv_all_nth)
qed

```

corollary *list_fast_correct*:

```

  braun t  $\implies$  list_fast t = list t
by (simp add: list_fast_def take_nths_00 braun_list_eq list_fast_rec_correct[where
k=0])

```

Running Time Analysis lemma *sum_tree_list_children*: $\forall t \in \text{set } ts.$
 $t \neq \text{Leaf} \implies$

```

  (∑ t ← ts. k * size t) = (∑ t ← map left ts @ map right ts. k * size t) +
  k * length ts
by(induction ts)(auto simp add: neq_Leaf_iff algebra_simps)

```

theorem *T_list_fast_rec_ub*:

```

  T_list_fast_rec ts ≤ sum_list (map (λt. 7*size t + 1) ts)
proof (induction ts rule: measure_induct_rule[where f=sum_list o map
size])
  case (less ts)
  let ?us = filter (λt. t ≠ Leaf) ts
  show ?case
  proof cases
    assume ?us = []
    thus ?thesis using T_list_fast_rec.simps[of ts]
      by(simp add: sum_list_Suc)
    next
    assume ?us ≠ []
    let ?children = map left ?us @ map right ?us
    have T_list_fast_rec ts = T_list_fast_rec ?children + 5 * length ?us
    + length ts
    using ⟨?us ≠ []⟩ T_list_fast_rec.simps[of ts] by(simp)
    also have ... ≤ (∑ t ← ?children. 7 * size t + 1) + 5 * length ?us +
    length ts
    using less[of ?children] list_fast_rec_term[of ?us] ⟨?us ≠ []⟩
    by (simp)
    also have ... = (∑ t ← ?children. 7*size t) + 7 * length ?us + length

```

```

ts
  by(simp add: sum_list_Suc o_def)
also have ... = ( $\sum t \leftarrow ?us. 7 * \text{size } t$ ) + length ts
  by(simp add: sum_tree_list_children)
also have ...  $\leq$  ( $\sum t \leftarrow ts. 7 * \text{size } t$ ) + length ts
  by(simp add: sum_list_filter_le_nat)
also have ... = ( $\sum t \leftarrow ts. 7 * \text{size } t + 1$ )
  by(simp add: sum_list_Suc)
finally show ?case .
qed
qed
end

```

39 Tries via Functions

```

theory Trie_Fun
imports
  Set_Specs
begin

```

A trie where each node maps a key to sub-tries via a function. Nice abstract model. Not efficient because of the function space.

```

datatype 'a trie = Nd bool 'a  $\Rightarrow$  'a trie option

```

```

definition empty :: 'a trie where
[simp]: empty = Nd False ( $\lambda\_.$  None)

```

```

fun isin :: 'a trie  $\Rightarrow$  'a list  $\Rightarrow$  bool where
isin (Nd b m) [] = b |
isin (Nd b m) (k # xs) = (case m k of None  $\Rightarrow$  False | Some t  $\Rightarrow$  isin t xs)

```

```

fun insert :: 'a list  $\Rightarrow$  'a trie  $\Rightarrow$  'a trie where
insert [] (Nd b m) = Nd True m |
insert (x#xs) (Nd b m) =
  (let s = (case m x of None  $\Rightarrow$  empty | Some t  $\Rightarrow$  t) in Nd b (m(x :=
Some(insert xs s))))

```

```

fun delete :: 'a list  $\Rightarrow$  'a trie  $\Rightarrow$  'a trie where
delete [] (Nd b m) = Nd False m |
delete (x#xs) (Nd b m) = Nd b
  (case m x of
    None  $\Rightarrow$  m |
    Some t  $\Rightarrow$  m(x := Some(delete xs t)))

```

Use (a tuned version of) *isin* as an abstraction function:

```
lemma isin_case: isin (Nd b m) xs =
  (case xs of
    []  $\Rightarrow$  b |
    x # ys  $\Rightarrow$  (case m x of None  $\Rightarrow$  False | Some t  $\Rightarrow$  isin t ys))
by(cases xs)auto
```

```
definition set :: 'a trie  $\Rightarrow$  'a list set where
[simp]: set t = {xs. isin t xs}
```

```
lemma isin_set: isin t xs = (xs  $\in$  set t)
by simp
```

```
lemma set_insert: set (insert xs t) = set t  $\cup$  {xs}
by (induction xs t rule: insert.induct)
  (auto simp: isin_case split!: if_splits option.splits list.splits)
```

```
lemma set_delete: set (delete xs t) = set t - {xs}
by (induction xs t rule: delete.induct)
  (auto simp: isin_case split!: if_splits option.splits list.splits)
```

```
interpretation S: Set
```

```
where empty = empty and isin = isin and insert = insert and delete =
delete
```

```
and set = set and invar =  $\lambda$ _. True
```

```
proof (standard, goal_cases)
```

```
  case 1 show ?case by (simp add: isin_case split: list.split)
```

```
next
```

```
  case 2 show ?case by(rule isin_set)
```

```
next
```

```
  case 3 show ?case by(rule set_insert)
```

```
next
```

```
  case 4 show ?case by(rule set_delete)
```

```
qed (rule TrueI)+
```

```
end
```

40 Tries via Search Trees

```
theory Trie_Map
```

```
imports
```

```
  RBT_Map
```

```
  Trie_Fun
```

begin

An implementation of tries based on maps implemented by red-black trees. Works for any kind of search tree.

Implementation of map:

```
type_synonym 'a mapi = 'a rbt
```

```
datatype 'a trie_map = Nd bool ('a * 'a trie_map) mapi
```

In principle one should be able to given an implementation of tries once and for all for any map implementation and not just for a specific one (RBT) as done here. But because the map ('a rbt) is used in a datatype, the HOL type system does not support this.

However, the development below works verbatim for any map implementation, eg *Tree_Map*, and not just *RBT_Map*, except for the termination lemma *lookup_size*.

```
term size_tree
```

```
lemma lookup_size[termination_simp]:
```

```
  fixes t :: ('a::linorder * 'a trie_map) rbt
```

```
  shows lookup t a = Some b  $\implies$  size b < Suc (size_tree (λab. Suc(Suc (size (snd(fst ab)))))) t)
```

```
apply(induction t a rule: lookup.induct)
```

```
apply(auto split: if_splits)
```

```
done
```

```
definition empty :: 'a trie_map where
```

```
[simp]: empty = Nd False Leaf
```

```
fun isin :: ('a::linorder) trie_map  $\Rightarrow$  'a list  $\Rightarrow$  bool where
```

```
isin (Nd b m) [] = b |
```

```
isin (Nd b m) (x # xs) = (case lookup m x of None  $\Rightarrow$  False | Some t  $\Rightarrow$  isin t xs)
```

```
fun insert :: ('a::linorder) list  $\Rightarrow$  'a trie_map  $\Rightarrow$  'a trie_map where
```

```
insert [] (Nd b m) = Nd True m |
```

```
insert (x#xs) (Nd b m) =
```

```
  Nd b (update x (insert xs (case lookup m x of None  $\Rightarrow$  empty | Some t  $\Rightarrow$  t)) m)
```

```
fun delete :: ('a::linorder) list  $\Rightarrow$  'a trie_map  $\Rightarrow$  'a trie_map where
```

```
delete [] (Nd b m) = Nd False m |
```

```
delete (x#xs) (Nd b m) = Nd b
```

```
  (case lookup m x of
```

$None \Rightarrow m \mid$
 $Some\ t \Rightarrow update\ x\ (delete\ xs\ t)\ m$

40.1 Correctness

Proof by stepwise refinement. First abstract to type $'a\ trie$.

fun $abs :: 'a::linorder\ trie_map \Rightarrow 'a\ trie$ **where**
 $abs\ (Nd\ b\ t) = Trie_Fun.Nd\ b\ (\lambda a. map_option\ abs\ (lookup\ t\ a))$

fun $invar :: ('a::linorder)\ trie_map \Rightarrow bool$ **where**
 $invar\ (Nd\ b\ m) = (M.invar\ m \wedge (\forall a\ t. lookup\ m\ a = Some\ t \longrightarrow invar\ t))$

lemma $isin_abs: isin\ t\ xs = Trie_Fun.isin\ (abs\ t)\ xs$
apply($induction\ t\ xs\ rule: isin.induct$)
apply($auto\ split: option.split$)
done

lemma $abs_insert: invar\ t \Longrightarrow abs(insert\ xs\ t) = Trie_Fun.insert\ xs\ (abs\ t)$
apply($induction\ xs\ t\ rule: insert.induct$)
apply($auto\ simp: M.map_specs\ RBT_Set.empty_def[symmetric]\ split: option.split$)
done

lemma $abs_delete: invar\ t \Longrightarrow abs(delete\ xs\ t) = Trie_Fun.delete\ xs\ (abs\ t)$
apply($induction\ xs\ t\ rule: delete.induct$)
apply($auto\ simp: M.map_specs\ split: option.split$)
done

lemma $invar_insert: invar\ t \Longrightarrow invar\ (insert\ xs\ t)$
apply($induction\ xs\ t\ rule: insert.induct$)
apply($auto\ simp: M.map_specs\ RBT_Set.empty_def[symmetric]\ split: option.split$)
done

lemma $invar_delete: invar\ t \Longrightarrow invar\ (delete\ xs\ t)$
apply($induction\ xs\ t\ rule: delete.induct$)
apply($auto\ simp: M.map_specs\ split: option.split$)
done

Overall correctness w.r.t. the *Set* ADT:

interpretation $S2: Set$

```

where empty = empty and isin = isin and insert = insert and delete =
delete
and set = set o abs and invar = invar
proof (standard, goal_cases)
  case 1 show ?case by (simp add: isin_case split: list.split)
next
  case 2 thus ?case by (simp add: isin_abs)
next
  case 3 thus ?case by (simp add: set_insert abs_insert del: set_def)
next
  case 4 thus ?case by (simp add: set_delete abs_delete del: set_def)
next
  case 5 thus ?case by (simp add: M.map_specs RBT_Set.empty_def[symmetric])
next
  case 6 thus ?case by (simp add: invar_insert)
next
  case 7 thus ?case by (simp add: invar_delete)
qed

end

```

41 Binary Tries and Patricia Tries

```

theory Tries_Binary
imports Set_Specs
begin

```

```

hide_const (open) insert

```

```

declare Let_def[simp]

```

```

fun sel2 :: bool  $\Rightarrow$  'a * 'a  $\Rightarrow$  'a where
sel2 b (a1,a2) = (if b then a2 else a1)

```

```

fun mod2 :: ('a  $\Rightarrow$  'a)  $\Rightarrow$  bool  $\Rightarrow$  'a * 'a  $\Rightarrow$  'a * 'a where
mod2 f b (a1,a2) = (if b then (a1,f a2) else (f a1,a2))

```

41.1 Trie

```

datatype trie = Lf | Nd bool trie * trie

```

```

definition empty :: trie where
[simp]: empty = Lf

```

```

fun isin :: trie ⇒ bool list ⇒ bool where
  isin Lf ks = False |
  isin (Nd b lr) ks =
    (case ks of
     [] ⇒ b |
     k#ks ⇒ isin (sel2 k lr) ks)

```

```

fun insert :: bool list ⇒ trie ⇒ trie where
  insert [] Lf = Nd True (Lf,Lf) |
  insert [] (Nd b lr) = Nd True lr |
  insert (k#ks) Lf = Nd False (mod2 (insert ks) k (Lf,Lf)) |
  insert (k#ks) (Nd b lr) = Nd b (mod2 (insert ks) k lr)

```

```

lemma isin_insert: isin (insert xs t) ys = (xs = ys ∨ isin t ys)
apply(induction xs t arbitrary: ys rule: insert.induct)
apply (auto split: list.splits if_splits)
done

```

A simple implementation of delete; does not shrink the trie!

```

fun delete0 :: bool list ⇒ trie ⇒ trie where
  delete0 ks Lf = Lf |
  delete0 ks (Nd b lr) =
    (case ks of
     [] ⇒ Nd False lr |
     k#ks' ⇒ Nd b (mod2 (delete0 ks') k lr))

```

```

lemma isin_delete0: isin (delete0 as t) bs = (as ≠ bs ∧ isin t bs)
apply(induction as t arbitrary: bs rule: delete0.induct)
apply (auto split: list.splits if_splits)
done

```

Now deletion with shrinking:

```

fun node :: bool ⇒ trie * trie ⇒ trie where
  node b lr = (if ¬ b ∧ lr = (Lf,Lf) then Lf else Nd b lr)

```

```

fun delete :: bool list ⇒ trie ⇒ trie where
  delete ks Lf = Lf |
  delete ks (Nd b lr) =
    (case ks of
     [] ⇒ node False lr |
     k#ks' ⇒ node b (mod2 (delete ks') k lr))

```

```

lemma isin_delete: isin (delete xs t) ys = (xs ≠ ys ∧ isin t ys)

```

```

apply(induction xs t arbitrary: ys rule: delete.induct)
  apply simp
apply (auto split: list.splits if_splits)
  apply (metis isin.simps(1))
apply (metis isin.simps(1))
done

```

```

definition set_trie :: trie  $\Rightarrow$  bool list set where
set_trie t = {xs. isin t xs}

```

```

lemma set_trie_empty: set_trie empty = {}
by(simp add: set_trie_def)

```

```

lemma set_trie_isin: isin t xs = (xs  $\in$  set_trie t)
by(simp add: set_trie_def)

```

```

lemma set_trie_insert: set_trie(insert xs t) = set_trie t  $\cup$  {xs}
by(auto simp add: isin_insert set_trie_def)

```

```

lemma set_trie_delete: set_trie(delete xs t) = set_trie t - {xs}
by(auto simp add: isin_delete set_trie_def)

```

interpretation *S*: *Set*

where *empty* = *empty* **and** *isin* = *isin* **and** *insert* = *insert* **and** *delete* = *delete*

and *set* = *set_trie* **and** *invar* = $\lambda t. True$

proof (*standard, goal_cases*)

case 1 **show** ?*case* **by** (*rule set_trie_empty*)

next

case 2 **show** ?*case* **by**(*rule set_trie_isin*)

next

case 3 **thus** ?*case* **by**(*auto simp: set_trie_insert*)

next

case 4 **show** ?*case* **by**(*rule set_trie_delete*)

qed (*rule TrueI*)**+**

41.2 Patricia Trie

```

datatype trieP = LfP | NdP bool list bool trieP * trieP

```

```

fun isinP :: trieP  $\Rightarrow$  bool list  $\Rightarrow$  bool where

```

```

isinP LfP ks = False |

```

```

isinP (NdP ps b lr) ks =

```

```

  (let n = length ps in

```

if ps = take n ks
then case drop n ks of [] ⇒ b | k#ks' ⇒ isinP (sel2 k lr) ks'
else False)

definition *emptyP :: trieP where*

[simp]: emptyP = LfP

fun *split where*

split [] ys = ([],[],ys) |

split xs [] = ([],xs,[]) |

split (x#xs) (y#ys) =

(if x≠y then ([],x#xs,y#ys)

else let (ps,xs',ys') = split xs ys in (x#ps,xs',ys'))

lemma *mod2_cong[fundef_cong]:*

[[lr = lr'; k = k'; ∧ a b. lr'=(a,b) ⇒ f (a) = f' (a) ; ∧ a b. lr'=(a,b) ⇒ f (b) = f' (b)]]

⇒ mod2 f k lr = mod2 f' k' lr'

by(*cases lr, cases lr', auto*)

fun *insertP :: bool list ⇒ trieP ⇒ trieP where*

insertP ks LfP = NdP ks True (LfP,LfP) |

insertP ks (NdP ps b lr) =

(case split ks ps of

(qs,k#ks',p#ps') ⇒

let tp = NdP ps' b lr; tk = NdP ks' True (LfP,LfP) in

NdP qs False (if k then (tp,tk) else (tk,tp)) |

(qs,k#ks',[]) ⇒

NdP ps b (mod2 (insertP ks') k lr) |

(qs,[],p#ps') ⇒

let t = NdP ps' b lr in

NdP qs True (if p then (LfP,t) else (t,LfP)) |

(qs,[],[]) ⇒ NdP ps True lr)

fun *nodeP :: bool list ⇒ bool ⇒ trieP * trieP ⇒ trieP where*

nodeP ps b lr = (if ¬ b ∧ lr = (LfP,LfP) then LfP else NdP ps b lr)

fun *deleteP :: bool list ⇒ trieP ⇒ trieP where*

deleteP ks LfP = LfP |

deleteP ks (NdP ps b lr) =

(case split ks ps of

$$\begin{aligned}
& (qs, ks', p \# ps') \Rightarrow NdP \ ps \ b \ lr \ | \\
& (qs, k \# ks', []) \Rightarrow nodeP \ ps \ b \ (mod2 \ (deleteP \ ks') \ k \ lr) \ | \\
& (qs, [], []) \Rightarrow nodeP \ ps \ False \ lr)
\end{aligned}$$

41.2.1 Functional Correctness

First step: *trieP* implements *trie* via the abstraction function *abs_trieP*:

```

fun prefix_trie :: bool list  $\Rightarrow$  trie  $\Rightarrow$  trie where
  prefix_trie [] t = t |
  prefix_trie (k#ks) t =
    (let t' = prefix_trie ks t in Nd False (if k then (Lf, t') else (t', Lf)))

fun abs_trieP :: trieP  $\Rightarrow$  trie where
  abs_trieP LfP = Lf |
  abs_trieP (NdP ps b (l,r)) = prefix_trie ps (Nd b (abs_trieP l, abs_trieP
  r))

```

Correctness of *isinP*:

```

lemma isin_prefix_trie:
  isin (prefix_trie ps t) ks
  = (ps = take (length ps) ks  $\wedge$  isin t (drop (length ps) ks))
apply(induction ps arbitrary: ks)
apply(auto split: list.split)
done

```

```

lemma abs_trieP_isinP:
  isinP t ks = isin (abs_trieP t) ks
apply(induction t arbitrary: ks rule: abs_trieP.induct)
  apply(auto simp: isin_prefix_trie split: list.split)
done

```

Correctness of *insertP*:

```

lemma prefix_trie_Lfs: prefix_trie ks (Nd True (Lf, Lf)) = insert ks Lf
apply(induction ks)
apply auto
done

```

```

lemma insert_prefix_trie_same:
  insert ps (prefix_trie ps (Nd b lr)) = prefix_trie ps (Nd True lr)
apply(induction ps)
apply auto
done

```

lemma *insert_append*: $insert (ks @ ks') (prefix_trie\ ks\ t) = prefix_trie\ ks\ (insert\ ks'\ t)$
apply(*induction ks*)
apply *auto*
done

lemma *prefix_trie_append*: $prefix_trie\ (ps @ qs)\ t = prefix_trie\ ps\ (prefix_trie\ qs\ t)$
apply(*induction ps*)
apply *auto*
done

lemma *split_if*: $split\ ks\ ps = (qs, ks', ps') \implies ks = qs @ ks' \wedge ps = qs @ ps' \wedge (ks' \neq [] \wedge ps' \neq [] \implies hd\ ks' \neq hd\ ps')$
apply(*induction ks ps arbitrary: qs ks' ps' rule: split.induct*)
apply(*auto split: prod.splits if_splits*)
done

lemma *abs_trieP_insertP*:
 $abs_trieP\ (insertP\ ks\ t) = insert\ ks\ (abs_trieP\ t)$
apply(*induction t arbitrary: ks*)
apply(*auto simp: prefix_trie_Lfs insert_prefix_trie_same insert_append prefix_trie_append dest!: split_if split: list.split prod.split if_splits*)
done

Correctness of *deleteP*:

lemma *prefix_trie_Lf*: $prefix_trie\ xs\ t = Lf \iff xs = [] \wedge t = Lf$
by(*cases xs*)(*auto*)

lemma *abs_trieP_Lf*: $abs_trieP\ t = Lf \iff t = LfP$
by(*cases t*) (*auto simp: prefix_trie_Lf*)

lemma *delete_prefix_trie*:
 $delete\ xs\ (prefix_trie\ xs\ (Nd\ b\ (l,r))) = (if\ (l,r) = (Lf,Lf)\ then\ Lf\ else\ prefix_trie\ xs\ (Nd\ False\ (l,r)))$
by(*induction xs*)(*auto simp: prefix_trie_Lf*)

lemma *delete_append_prefix_trie*:
 $delete\ (xs @ ys)\ (prefix_trie\ xs\ t) = (if\ delete\ ys\ t = Lf\ then\ Lf\ else\ prefix_trie\ xs\ (delete\ ys\ t))$
by(*induction xs*)(*auto simp: prefix_trie_Lf*)

lemma *delete_abs_trieP*:

```

    delete ks (abs_trieP t) = abs_trieP (deleteP ks t)
apply(induction t arbitrary: ks)
apply(auto simp: delete_prefix_trie delete_append_prefix_trie
        prefix_trie_append prefix_trie_Lf abs_trieP_Lf
        dest!: split_if split: if_splits list.split prod.split)
done

```

The overall correctness proof. Simply composes correctness lemmas.

```

definition set_trieP :: trieP  $\Rightarrow$  bool list set where
set_trieP = set_trie o abs_trieP

```

```

lemma isinP_set_trieP: isinP t xs = (xs  $\in$  set_trieP t)
by(simp add: abs_trieP_isinP set_trie_isin set_trieP_def)

```

```

lemma set_trieP_insertP: set_trieP (insertP xs t) = set_trieP t  $\cup$  {xs}
by(simp add: abs_trieP_insertP set_trie_insert set_trieP_def)

```

```

lemma set_trieP_deleteP: set_trieP (deleteP xs t) = set_trieP t - {xs}
by(auto simp: set_trie_delete set_trieP_def simp flip: delete_abs_trieP)

```

interpretation SP: Set

where empty = emptyP **and** isin = isinP **and** insert = insertP **and** delete = deleteP

and set = set_trieP **and** invar = $\lambda t.$ True

proof (standard, goal_cases)

case 1 **show** ?case **by** (simp add: set_trieP_def set_trie_def)

next

case 2 **show** ?case **by**(rule isinP_set_trieP)

next

case 3 **thus** ?case **by** (auto simp: set_trieP_insertP)

next

case 4 **thus** ?case **by**(auto simp: set_trieP_deleteP)

qed (rule TrueI)+

end

42 Queue Specification

theory Queue_Spec

imports Main

begin

The basic queue interface with *list*-based specification:

locale Queue =

```

fixes empty :: 'q
fixes enq :: 'a ⇒ 'q ⇒ 'q
fixes first :: 'q ⇒ 'a
fixes deq :: 'q ⇒ 'q
fixes is_empty :: 'q ⇒ bool
fixes list :: 'q ⇒ 'a list
fixes invar :: 'q ⇒ bool
assumes list_empty: list empty = []
assumes list_enq: invar q ⇒ list(enq x q) = list q @ [x]
assumes list_deq: invar q ⇒ list(deq q) = tl(list q)
assumes list_first: invar q ⇒ ¬ list q = [] ⇒ first q = hd(list q)
assumes list_is_empty: invar q ⇒ is_empty q = (list q = [])
assumes invar_empty: invar empty
assumes invar_enq: invar q ⇒ invar(enq x q)
assumes invar_deq: invar q ⇒ invar(deq q)

```

end

theory Reverse

imports Main

begin

fun T_append :: 'a list ⇒ 'a list ⇒ nat **where**

T_append [] ys = 1 |

T_append (x#xs) ys = T_append xs ys + 1

fun T_rev :: 'a list ⇒ nat **where**

T_rev [] = 1 |

T_rev (x#xs) = T_rev xs + T_append (rev xs) [x] + 1

lemma T_append: T_append xs ys = length xs + 1

by(induction xs) auto

lemma T_rev: T_rev xs ≤ (length xs + 1)²

by(induction xs) (auto simp: T_append power2_eq_square)

fun itrev :: 'a list ⇒ 'a list ⇒ 'a list **where**

itrev [] ys = ys |

itrev (x#xs) ys = itrev xs (x # ys)

lemma itrev: itrev xs ys = rev xs @ ys

by(induction xs arbitrary: ys) auto

lemma itrev_Nil: itrev xs [] = rev xs

by(simp add: itrev)

```

fun T_itrev :: 'a list  $\Rightarrow$  'a list  $\Rightarrow$  nat where
  T_itrev [] ys = 1 |
  T_itrev (x#xs) ys = T_itrev xs (x # ys) + 1

```

```

lemma T_itrev: T_itrev xs ys = length xs + 1
by(induction xs arbitrary: ys) auto

```

```

end

```

43 Queue Implementation via 2 Lists

```

theory Queue_2Lists

```

```

imports

```

```

  Queue_Spec

```

```

  Reverse

```

```

begin

```

```

  Definitions:

```

```

type_synonym 'a queue = 'a list  $\times$  'a list

```

```

fun norm :: 'a queue  $\Rightarrow$  'a queue where
  norm (fs,rs) = (if fs = [] then (itrev rs [], []) else (fs,rs))

```

```

fun enq :: 'a  $\Rightarrow$  'a queue  $\Rightarrow$  'a queue where
  enq a (fs,rs) = norm(fs, a # rs)

```

```

fun deq :: 'a queue  $\Rightarrow$  'a queue where
  deq (fs,rs) = (if fs = [] then (fs,rs) else norm(tl fs,rs))

```

```

fun first :: 'a queue  $\Rightarrow$  'a where
  first (a # fs,rs) = a

```

```

fun is_empty :: 'a queue  $\Rightarrow$  bool where
  is_empty (fs,rs) = (fs = [])

```

```

fun list :: 'a queue  $\Rightarrow$  'a list where
  list (fs,rs) = fs @ rev rs

```

```

fun invar :: 'a queue  $\Rightarrow$  bool where
  invar (fs,rs) = (fs = []  $\longrightarrow$  rs = [])

```

```

  Implementation correctness:

```

```

interpretation Queue

```

```

where empty = ([],[]) and enq = enq and deq = deq and first = first
and is_empty = is_empty and list = list and invar = invar
proof (standard, goal_cases)
  case 1 show ?case by (simp)
next
  case (2 q) thus ?case by(cases q) (simp)
next
  case (3 q) thus ?case by(cases q) (simp add: itrev_Nil)
next
  case (4 q) thus ?case by(cases q) (auto simp: neq_Nil_conv)
next
  case (5 q) thus ?case by(cases q) (auto)
next
  case 6 show ?case by(simp)
next
  case (7 q) thus ?case by(cases q) (simp)
next
  case (8 q) thus ?case by(cases q) (simp)
qed

```

Running times:

```

fun T_norm :: 'a queue  $\Rightarrow$  nat where
T_norm (fs,rs) = (if fs = [] then T_itrev rs [] else 0) + 1

```

```

fun T_enq :: 'a  $\Rightarrow$  'a queue  $\Rightarrow$  nat where
T_enq a (fs,rs) = T_norm(fs, a # rs) + 1

```

```

fun T_deq :: 'a queue  $\Rightarrow$  nat where
T_deq (fs,rs) = (if fs = [] then 0 else T_norm(tl fs,rs)) + 1

```

```

fun T_first :: 'a queue  $\Rightarrow$  nat where
T_first (a # fs,rs) = 1

```

```

fun T_is_empty :: 'a queue  $\Rightarrow$  nat where
T_is_empty (fs,rs) = 1

```

Amortized running times:

```

fun  $\Phi$  :: 'a queue  $\Rightarrow$  nat where
 $\Phi$ (fs,rs) = length rs

```

```

lemma a_enq: T_enq a (fs,rs) +  $\Phi$ (enq a (fs,rs)) -  $\Phi$ (fs,rs)  $\leq$  4
by(auto simp: T_itrev)

```

```

lemma a_deq: T_deq (fs,rs) +  $\Phi$ (deq (fs,rs)) -  $\Phi$ (fs,rs)  $\leq$  3

```

by(auto simp: T_itrev)

end

44 Priority Queue Specifications

```
theory Priority_Queue_Specs
imports HOL-Library.Multiset
begin
```

Priority queue interface + specification:

```
locale Priority_Queue =
fixes empty :: 'q
and is_empty :: 'q  $\Rightarrow$  bool
and insert :: 'a::linorder  $\Rightarrow$  'q  $\Rightarrow$  'q
and get_min :: 'q  $\Rightarrow$  'a
and del_min :: 'q  $\Rightarrow$  'q
and invar :: 'q  $\Rightarrow$  bool
and mset :: 'q  $\Rightarrow$  'a multiset
assumes mset_empty: mset empty = {#}
and is_empty: invar q  $\Longrightarrow$  is_empty q = (mset q = {#})
and mset_insert: invar q  $\Longrightarrow$  mset (insert x q) = mset q + {#x#}
and mset_del_min: invar q  $\Longrightarrow$  mset q  $\neq$  {#}  $\Longrightarrow$ 
  mset (del_min q) = mset q - {# get_min q #}
and mset_get_min: invar q  $\Longrightarrow$  mset q  $\neq$  {#}  $\Longrightarrow$  get_min q = Min_mset
(mset q)
and invar_empty: invar empty
and invar_insert: invar q  $\Longrightarrow$  invar (insert x q)
and invar_del_min: invar q  $\Longrightarrow$  mset q  $\neq$  {#}  $\Longrightarrow$  invar (del_min q)
```

Extend locale with *merge*. Need to enforce that *'q* is the same in both locales.

```
locale Priority_Queue_Merge = Priority_Queue where empty = empty
for empty :: 'q +
fixes merge :: 'q  $\Rightarrow$  'q  $\Rightarrow$  'q
assumes mset_merge:  $\llbracket$  invar q1; invar q2  $\rrbracket \Longrightarrow$  mset (merge q1 q2) =
mset q1 + mset q2
and invar_merge:  $\llbracket$  invar q1; invar q2  $\rrbracket \Longrightarrow$  invar (merge q1 q2)
```

end

45 Heaps

```
theory Heaps
```

```

imports
  HOL-Library.Tree_Multiset
  Priority_Queue_Specs
begin

  Heap = priority queue on trees:

locale Heap =
fixes insert :: ('a::linorder) ⇒ 'a tree ⇒ 'a tree
and del_min :: 'a tree ⇒ 'a tree
assumes mset_insert: heap q ⇒ mset_tree (insert x q) = {#x#} +
  mset_tree q
and mset_del_min: [| heap q; q ≠ Leaf |] ⇒ mset_tree (del_min q) =
  mset_tree q - {#value q#}
and heap_insert: heap q ⇒ heap(insert x q)
and heap_del_min: heap q ⇒ heap(del_min q)
begin

definition empty :: 'a tree where
  empty = Leaf

fun is_empty :: 'a tree ⇒ bool where
  is_empty t = (t = Leaf)

sublocale Priority_Queue where empty = empty and is_empty = is_empty
and insert = insert
and get_min = value and del_min = del_min and invar = heap and
  mset = mset_tree
proof (standard, goal_cases)
  case 1 thus ?case by (simp add: empty_def)
next
  case 2 thus ?case by(auto)
next
  case 3 thus ?case by(simp add: mset_insert)
next
  case 4 thus ?case by(simp add: mset_del_min)
next
  case 5 thus ?case by(auto simp: neq_Leaf_iff Min_insert2 simp del:
  Un_iff)
next
  case 6 thus ?case by(simp add: empty_def)
next
  case 7 thus ?case by(simp add: heap_insert)
next
  case 8 thus ?case by(simp add: heap_del_min)

```

qed

end

Once you have *merge*, *insert* and *del_min* are easy:

```
locale Heap_Merge =  
fixes merge :: 'a::linorder tree  $\Rightarrow$  'a tree  $\Rightarrow$  'a tree  
assumes mset_merge:  $\llbracket$  heap q1; heap q2  $\rrbracket \Longrightarrow$  mset_tree (merge q1 q2)  
= mset_tree q1 + mset_tree q2  
and invar_merge:  $\llbracket$  heap q1; heap q2  $\rrbracket \Longrightarrow$  heap (merge q1 q2)  
begin
```

```
fun insert :: 'a  $\Rightarrow$  'a tree  $\Rightarrow$  'a tree where  
insert x t = merge (Node Leaf x Leaf) t
```

```
fun del_min :: 'a tree  $\Rightarrow$  'a tree where  
del_min Leaf = Leaf |  
del_min (Node l x r) = merge l r
```

interpretation *Heap insert del_min*

proof(*standard, goal_cases*)

case 1 thus ?case by (simp add: mset_merge)

next

case (2 q) thus ?case by (cases q)(auto simp: mset_merge)

next

case 3 thus ?case by (simp add: invar_merge)

next

case (4 q) thus ?case by (cases q)(auto simp: invar_merge)

qed

sublocale *PQM: Priority_Queue_Merge where empty = empty and is_empty*
= is_empty and insert = insert

and *get_min = value and del_min = del_min and invar = heap and*
mset = mset_tree and merge = merge

proof(*standard, goal_cases*)

case 1 thus ?case by (simp add: mset_merge)

next

case 2 thus ?case by (simp add: invar_merge)

qed

end

end

46 Leftist Heap

theory *Leftist_Heap*

imports

HOL-Library.Pattern_Aliases

Tree2

Priority_Queue_Specs

Complex_Main

begin

fun *mset_tree* :: ('a*'b) tree \Rightarrow 'a multiset **where**
mset_tree Leaf = {#} |
mset_tree (Node l (a, _) r) = {#a#} + *mset_tree l* + *mset_tree r*

type_synonym 'a *lheap* = ('a*nat)tree

fun *mht* :: 'a *lheap* \Rightarrow nat **where**

mht Leaf = 0 |

mht (Node _ (_, n) _) = n

The invariants:

fun (**in** *linorder*) *heap* :: ('a*'b) tree \Rightarrow bool **where**

heap Leaf = True |

heap (Node l (m, _) r) =

$((\forall x \in \text{set_tree } l \cup \text{set_tree } r. m \leq x) \wedge \text{heap } l \wedge \text{heap } r)$

fun *ltree* :: 'a *lheap* \Rightarrow bool **where**

ltree Leaf = True |

ltree (Node l (a, n) r) =

$(\text{min_height } l \geq \text{min_height } r \wedge n = \text{min_height } r + 1 \wedge \text{ltree } l \ \& \ \text{ltree } r)$

definition *empty* :: 'a *lheap* **where**

empty = *Leaf*

definition *node* :: 'a *lheap* \Rightarrow 'a \Rightarrow 'a *lheap* \Rightarrow 'a *lheap* **where**

node l a r =

(let *mhl* = *mht l*; *mhr* = *mht r*

in if *mhl* \geq *mhr* then *Node l (a, mhr+1) r* else *Node r (a, mhl+1) l*)

fun *get_min* :: 'a *lheap* \Rightarrow 'a **where**

get_min (Node l (a, n) r) = a

For function *merge*:

unbundle *pattern_aliases*

```

fun merge :: 'a::ord lheap  $\Rightarrow$  'a lheap  $\Rightarrow$  'a lheap where
merge Leaf t = t |
merge t Leaf = t |
merge (Node l1 (a1, n1) r1 =: t1) (Node l2 (a2, n2) r2 =: t2) =
  (if a1  $\leq$  a2 then node l1 a1 (merge r1 t2)
   else node l2 a2 (merge t1 r2))

```

Termination of *merge*: by sum or lexicographic product of the sizes of the two arguments. Isabelle uses a lexicographic product.

```

lemma merge_code: merge t1 t2 = (case (t1,t2) of
  (Leaf, _)  $\Rightarrow$  t2 |
  (_, Leaf)  $\Rightarrow$  t1 |
  (Node l1 (a1, n1) r1, Node l2 (a2, n2) r2)  $\Rightarrow$ 
    if a1  $\leq$  a2 then node l1 a1 (merge r1 t2) else node l2 a2 (merge t1 r2))
by(induction t1 t2 rule: merge.induct) (simp_all split: tree.split)

```

```

hide_const (open) insert

```

```

definition insert :: 'a::ord  $\Rightarrow$  'a lheap  $\Rightarrow$  'a lheap where
insert x t = merge (Node Leaf (x,1) Leaf) t

```

```

fun del_min :: 'a::ord lheap  $\Rightarrow$  'a lheap where
del_min Leaf = Leaf |
del_min (Node l _ r) = merge l r

```

46.1 Lemmas

```

lemma mset_tree_empty: mset_tree t = {#}  $\longleftrightarrow$  t = Leaf
by(cases t) auto

```

```

lemma mht_eq_min_height: ltree t  $\Longrightarrow$  mht t = min_height t
by(cases t) auto

```

```

lemma ltree_node: ltree (node l a r)  $\longleftrightarrow$  ltree l  $\wedge$  ltree r
by(auto simp add: node_def mht_eq_min_height)

```

```

lemma heap_node: heap (node l a r)  $\longleftrightarrow$ 
  heap l  $\wedge$  heap r  $\wedge$  ( $\forall x \in$  set_tree l  $\cup$  set_tree r. a  $\leq$  x)
by(auto simp add: node_def)

```

```

lemma set_tree_mset: set_tree t = set_mset(mset_tree t)
by(induction t) auto

```

46.2 Functional Correctness

lemma *mset_merge*: $mset_tree (merge\ t1\ t2) = mset_tree\ t1 + mset_tree\ t2$

by (*induction* *t1 t2* *rule*: *merge.induct*) (*auto simp add*: *node_def ac_simps*)

lemma *mset_insert*: $mset_tree (insert\ x\ t) = mset_tree\ t + \{\#x\# \}$

by (*auto simp add*: *insert_def mset_merge*)

lemma *get_min*: $\llbracket heap\ t; t \neq Leaf \rrbracket \implies get_min\ t = Min(set_tree\ t)$

by (*cases* *t*) (*auto simp add*: *eq_Min_iff*)

lemma *mset_del_min*: $mset_tree (del_min\ t) = mset_tree\ t - \{\# get_min\ t\ \# \}$

by (*cases* *t*) (*auto simp*: *mset_merge*)

lemma *ltree_merge*: $\llbracket ltree\ l; ltree\ r \rrbracket \implies ltree (merge\ l\ r)$

by(*induction* *l r* *rule*: *merge.induct*)(*auto simp*: *ltree_node*)

lemma *heap_merge*: $\llbracket heap\ l; heap\ r \rrbracket \implies heap (merge\ l\ r)$

proof(*induction* *l r* *rule*: *merge.induct*)

case 3 **thus** ?*case* **by**(*auto simp*: *heap_node mset_merge ball_Un set_tree_mset*)
qed *simp_all*

lemma *ltree_insert*: $ltree\ t \implies ltree(insert\ x\ t)$

by(*simp add*: *insert_def ltree_merge del*: *merge.simps split*: *tree.split*)

lemma *heap_insert*: $heap\ t \implies heap(insert\ x\ t)$

by(*simp add*: *insert_def heap_merge del*: *merge.simps split*: *tree.split*)

lemma *ltree_del_min*: $ltree\ t \implies ltree(del_min\ t)$

by(*cases* *t*)(*auto simp add*: *ltree_merge simp del*: *merge.simps*)

lemma *heap_del_min*: $heap\ t \implies heap(del_min\ t)$

by(*cases* *t*)(*auto simp add*: *heap_merge simp del*: *merge.simps*)

Last step of functional correctness proof: combine all the above lemmas to show that leftist heaps satisfy the specification of priority queues with merge.

interpretation *lheap*: *Priority_Queue_Merge*

where *empty* = *empty* **and** *is_empty* = $\lambda t. t = Leaf$

and *insert* = *insert* **and** *del_min* = *del_min*

and *get_min* = *get_min* **and** *merge* = *merge*

and *invar* = $\lambda t. heap\ t \wedge ltree\ t$ **and** *mset* = *mset_tree*

```

proof(standard, goal_cases)
  case 1 show ?case by (simp add: empty_def)
next
  case (2 q) show ?case by (cases q auto)
next
  case 3 show ?case by(rule mset_insert)
next
  case 4 show ?case by(rule mset_del_min)
next
  case 5 thus ?case by(simp add: get_min mset_tree_empty set_tree_mset)
next
  case 6 thus ?case by(simp add: empty_def)
next
  case 7 thus ?case by(simp add: heap_insert ltree_insert)
next
  case 8 thus ?case by(simp add: heap_del_min ltree_del_min)
next
  case 9 thus ?case by (simp add: mset_merge)
next
  case 10 thus ?case by (simp add: heap_merge ltree_merge)
qed

```

46.3 Complexity

Explicit termination argument: sum of sizes

```

fun T_merge :: 'a::ord lheap  $\Rightarrow$  'a lheap  $\Rightarrow$  nat where
  T_merge Leaf t = 1 |
  T_merge t Leaf = 1 |
  T_merge (Node l1 (a1, n1) r1 =: t1) (Node l2 (a2, n2) r2 =: t2) =
    (if  $a1 \leq a2$  then T_merge r1 t2
     else T_merge t1 r2) + 1

```

```

definition T_insert :: 'a::ord  $\Rightarrow$  'a lheap  $\Rightarrow$  nat where
  T_insert x t = T_merge (Node Leaf (x, 1) Leaf) t + 1

```

```

fun T_del_min :: 'a::ord lheap  $\Rightarrow$  nat where
  T_del_min Leaf = 1 |
  T_del_min (Node l _ r) = T_merge l r + 1

```

lemma *T_merge_min_height*: *ltree l* \Longrightarrow *ltree r* \Longrightarrow *T_merge l r* \leq *min_height l* + *min_height r* + 1

```

proof(induction l r rule: merge.induct)
  case 3 thus ?case by(auto)
qed simp_all

```

corollary *T_merge_log*: **assumes** *ltree l ltree r*
shows $T_merge\ l\ r \leq \log 2\ (size1\ l) + \log 2\ (size1\ r) + 1$
using *le_log2_of_power*[*OF min_height_size1*[*of l*]]
le_log2_of_power[*OF min_height_size1*[*of r*]] *T_merge_min_height*[*of*
l r] *assms*
by *linarith*

corollary *T_insert_log*: *ltree t* \implies $T_insert\ x\ t \leq \log 2\ (size1\ t) + 3$
using *T_merge_log*[*of Node Leaf (x, 1) Leaf t*]
by(*simp add: T_insert_def split: tree.split*)

lemma *ld_ld_1_less*:

assumes $x > 0\ y > 0$ **shows** $\log 2\ x + \log 2\ y + 1 < 2 * \log 2\ (x+y)$
proof –
have $2\ powr\ (\log 2\ x + \log 2\ y + 1) = 2*x*y$
using *assms* **by**(*simp add: powr_add*)
also have $\dots < (x+y)^2$ **using** *assms*
by(*simp add: numeral_eq_Suc algebra_simps add_pos_pos*)
also have $\dots = 2\ powr\ (2 * \log 2\ (x+y))$
using *assms* **by**(*simp add: powr_add log_powr[symmetric]*)
finally show *?thesis* **by** *simp*
qed

corollary *T_del_min_log*: **assumes** *ltree t*
shows $T_del_min\ t \leq 2 * \log 2\ (size1\ t) + 1$
proof(*cases t rule: tree2_cases*)
case *Leaf* **thus** *?thesis* **using** *assms* **by** *simp*
next
case [*simp*]: (*Node l _ _ r*)
have $T_del_min\ t = T_merge\ l\ r + 1$ **by** *simp*
also have $\dots \leq \log 2\ (size1\ l) + \log 2\ (size1\ r) + 2$
using $\langle ltree\ t \rangle$ *T_merge_log*[*of l r*] **by** (*auto simp del: T_merge.simps*)
also have $\dots \leq 2 * \log 2\ (size1\ t) + 1$
using *ld_ld_1_less*[*of size1 l size1 r*] **by** (*simp*)
finally show *?thesis* .
qed

end

47 Binomial Heap

```
theory Binomial_Heap
imports
  HOL-Library.Pattern_Aliases
  Complex_Main
  Priority_Queue_Specs
begin
```

We formalize the binomial heap presentation from Okasaki's book. We show the functional correctness and complexity of all operations.

The presentation is engineered for simplicity, and most proofs are straightforward and automatic.

47.1 Binomial Tree and Heap Datatype

```
datatype 'a tree = Node (rank: nat) (root: 'a) (children: 'a tree list)
```

```
type_synonym 'a heap = 'a tree list
```

47.1.1 Multiset of elements

```
fun mset_tree :: 'a::linorder tree  $\Rightarrow$  'a multiset where
  mset_tree (Node _ a ts) = {#a#} + ( $\sum$  t $\in$ #mset ts. mset_tree t)
```

```
definition mset_heap :: 'a::linorder heap  $\Rightarrow$  'a multiset where
  mset_heap ts = ( $\sum$  t $\in$ #mset ts. mset_tree t)
```

```
lemma mset_tree_simp_alt[simp]:
  mset_tree (Node r a ts) = {#a#} + mset_heap ts
unfolding mset_heap_def by auto
declare mset_tree.simps[simp del]
```

```
lemma mset_tree_nonempty[simp]: mset_tree t  $\neq$  {#}
by (cases t) auto
```

```
lemma mset_heap_Nil[simp]:
  mset_heap [] = {#}
by (auto simp: mset_heap_def)
```

```
lemma mset_heap_Cons[simp]: mset_heap (t#ts) = mset_tree t + mset_heap
ts
by (auto simp: mset_heap_def)
```

```
lemma mset_heap_empty_iff[simp]: mset_heap ts = {#}  $\longleftrightarrow$  ts=[]
```

by (*auto simp: mset_heap_def*)

lemma *root_in_mset*[*simp*]: *root t ∈# mset_tree t*
by (*cases t*) *auto*

lemma *mset_heap_rev_eq*[*simp*]: *mset_heap (rev ts) = mset_heap ts*
by (*auto simp: mset_heap_def*)

47.1.2 Invariants

Binomial tree

fun *invar_btree* :: '*a*::*linorder* tree ⇒ *bool* **where**
invar_btree (*Node r x ts*) ↔
 (∀ *t* ∈ *set ts*. *invar_btree t*) ∧ *map rank ts = rev [0..*r*]*
 Ordering (heap) invariant

fun *invar_otree* :: '*a*::*linorder* tree ⇒ *bool* **where**
invar_otree (*Node _ x ts*) ↔ (∀ *t* ∈ *set ts*. *invar_otree t* ∧ *x ≤ root t*)

definition *invar_tree* *t* ↔ *invar_btree t* ∧ *invar_otree t*

Binomial Heap invariant

definition *invar ts* ↔ (∀ *t* ∈ *set ts*. *invar_tree t*) ∧ (*sorted_wrt* (<) (*map rank ts*))

The children of a node are a valid heap

lemma *invar_children*:
invar_tree (*Node r v ts*) ⇒ *invar* (*rev ts*)
by (*auto simp: invar_tree_def invar_def rev_map[symmetric]*)

47.2 Operations and Their Functional Correctness

47.2.1 *link*

context

includes *pattern_aliases*

begin

fun *link* :: ('*a*::*linorder*) tree ⇒ '*a* tree ⇒ '*a* tree **where**
link (*Node r x₁ ts₁ =: t₁*) (*Node r' x₂ ts₂ =: t₂*) =
 (*if x₁ ≤ x₂ then Node (r+1) x₁ (t₂#ts₁) else Node (r+1) x₂ (t₁#ts₂)*)

end

lemma *invar_link*:

assumes *invar_tree* t_1
assumes *invar_tree* t_2
assumes $\text{rank } t_1 = \text{rank } t_2$
shows *invar_tree* ($\text{link } t_1 t_2$)
using *assms* **unfolding** *invar_tree_def*
by (*cases* (t_1, t_2) *rule: link.cases*) *auto*

lemma *rank_link[simp]*: $\text{rank } (\text{link } t_1 t_2) = \text{rank } t_1 + 1$
by (*cases* (t_1, t_2) *rule: link.cases*) *simp*

lemma *mset_link[simp]*: $\text{mset_tree } (\text{link } t_1 t_2) = \text{mset_tree } t_1 + \text{mset_tree } t_2$
by (*cases* (t_1, t_2) *rule: link.cases*) *simp*

47.2.2 *ins_tree*

fun *ins_tree* :: '*a*::*linorder* *tree* \Rightarrow '*a* *heap* \Rightarrow '*a* *heap* **where**
ins_tree $t \ [] = [t]$
| *ins_tree* $t_1 (t_2 \# ts) =$
(if rank $t_1 < \text{rank } t_2$ *then* $t_1 \# t_2 \# ts$ *else* $\text{ins_tree } (\text{link } t_1 t_2) ts$ *)*

lemma *invar_tree0[simp]*: *invar_tree* (*Node* 0 $x \ []$)
unfolding *invar_tree_def* **by** *auto*

lemma *invar_Cons[simp]*:
invar ($t \# ts$)
 $\longleftrightarrow \text{invar_tree } t \wedge \text{invar } ts \wedge (\forall t' \in \text{set } ts. \text{rank } t < \text{rank } t')$
by (*auto simp: invar_def*)

lemma *invar_ins_tree*:
assumes *invar_tree* t
assumes *invar* ts
assumes $\forall t' \in \text{set } ts. \text{rank } t \leq \text{rank } t'$
shows *invar* ($\text{ins_tree } t ts$)
using *assms*
by (*induction* $t ts$ *rule: ins_tree.induct*) (*auto simp: invar_link less_eq_Suc_le[symmetric]*)

lemma *mset_heap_ins_tree[simp]*:
 $\text{mset_heap } (\text{ins_tree } t ts) = \text{mset_tree } t + \text{mset_heap } ts$
by (*induction* $t ts$ *rule: ins_tree.induct*) *auto*

lemma *ins_tree_rank_bound*:
assumes $t' \in \text{set } (\text{ins_tree } t ts)$
assumes $\forall t' \in \text{set } ts. \text{rank } t_0 < \text{rank } t'$

```

assumes rank  $t_0 < \text{rank } t$ 
shows rank  $t_0 < \text{rank } t'$ 
using assms
by (induction  $t$   $ts$  rule: ins_tree.induct) (auto split: if_splits)

```

47.2.3 insert

```

hide_const (open) insert

```

```

definition insert :: 'a::linorder  $\Rightarrow$  'a heap  $\Rightarrow$  'a heap where
insert  $x$   $ts = \text{ins\_tree } (\text{Node } 0\ x\ [])\ ts$ 

```

```

lemma invar_insert[simp]: invar  $t \Longrightarrow \text{invar } (\text{insert } x\ t)$ 
by (auto intro!: invar_ins_tree simp: insert_def)

```

```

lemma mset_heap_insert[simp]: mset_heap (insert  $x$   $t$ ) =  $\{\#x\# \} + \text{mset\_heap } t$ 
by(auto simp: insert_def)

```

47.2.4 merge

```

context
includes pattern_aliases
begin

```

```

fun merge :: 'a::linorder heap  $\Rightarrow$  'a heap  $\Rightarrow$  'a heap where
  merge  $ts_1\ [] = ts_1$ 
| merge  $[]\ ts_2 = ts_2$ 
| merge ( $t_1\ \#ts_1 =: h_1$ ) ( $t_2\ \#ts_2 =: h_2$ ) = (
  if rank  $t_1 < \text{rank } t_2$  then  $t_1\ \# \text{merge } ts_1\ h_2$  else
  if rank  $t_2 < \text{rank } t_1$  then  $t_2\ \# \text{merge } h_1\ ts_2$ 
  else ins_tree (link  $t_1\ t_2$ ) (merge  $ts_1\ ts_2$ )
)

```

```

end

```

```

lemma merge_simp2[simp]: merge  $[]\ ts_2 = ts_2$ 
by (cases  $ts_2$ ) auto

```

```

lemma merge_rank_bound:
  assumes  $t' \in \text{set } (\text{merge } ts_1\ ts_2)$ 
  assumes  $\forall t_1 \in \text{set } ts_1. \text{rank } t < \text{rank } t_1$ 
  assumes  $\forall t_2 \in \text{set } ts_2. \text{rank } t < \text{rank } t_2$ 
  shows rank  $t < \text{rank } t'$ 

```

```

using assms
by (induction  $ts_1$   $ts_2$  arbitrary:  $t'$  rule: merge.induct)
    (auto split: if_splits simp: ins_tree_rank_bound)

lemma invar_merge[simp]:
  assumes invar  $ts_1$ 
  assumes invar  $ts_2$ 
  shows invar (merge  $ts_1$   $ts_2$ )
using assms
by (induction  $ts_1$   $ts_2$  rule: merge.induct)
    (auto 0 3 simp: Suc_le_eq intro!: invar_ins_tree invar_link elim!: merge_rank_bound)

```

Longer, more explicit proof of *invar_merge*, to illustrate the application of the *merge_rank_bound* lemma.

```

lemma
  assumes invar  $ts_1$ 
  assumes invar  $ts_2$ 
  shows invar (merge  $ts_1$   $ts_2$ )
  using assms
proof (induction  $ts_1$   $ts_2$  rule: merge.induct)
  case ( $\exists t_1$   $ts_1$   $t_2$   $ts_2$ )
  — Invariants of the parts can be shown automatically
  from  $\exists$ .prems have [simp]:
    invar_tree  $t_1$  invar_tree  $t_2$ 

    by auto

  — These are the three cases of the merge function
  consider (LT) rank  $t_1 <$  rank  $t_2$ 
    | (GT) rank  $t_1 >$  rank  $t_2$ 
    | (EQ) rank  $t_1 =$  rank  $t_2$ 
  using antisym_conv3 by blast
  then show ?case proof cases
  case LT
  — merge takes the first tree from the left heap
  then have merge ( $t_1 \# ts_1$ ) ( $t_2 \# ts_2$ ) =  $t_1 \#$  merge  $ts_1$  ( $t_2 \# ts_2$ ) by
simp
  also have invar ... proof (simp, intro conjI)
  — Invariant follows from induction hypothesis
  show invar (merge  $ts_1$  ( $t_2 \# ts_2$ ))
    using LT 3.IH 3.prems by simp

  — It remains to show that  $t_1$  has smallest rank.
  show  $\forall t' \in set$  (merge  $ts_1$  ( $t_2 \# ts_2$ )). rank  $t_1 <$  rank  $t'$ 

```

```

    — Which is done by auxiliary lemma merge_rank_bound
    using LT 3.prems by (force elim!: merge_rank_bound)
  qed
  finally show ?thesis .
next
  — merge takes the first tree from the right heap
  case GT
  — The proof is anaologous to the LT case
  then show ?thesis using 3.prems 3.IH by (force elim!: merge_rank_bound)
next
  case [simp]: EQ
  — merge links both first trees, and inserts them into the merged remaining
heaps
  have merge (t1 # ts1) (t2 # ts2) = ins_tree (link t1 t2) (merge ts1 ts2)
by simp
  also have invar ... proof (intro invar_ins_tree invar_link)
  — Invariant of merged remaining heaps follows by IH
  show invar (merge ts1 ts2)
  using EQ 3.prems 3.IH by auto

  — For insertion, we have to show that the rank of the linked tree is ≤
the ranks in the merged remaining heaps
  show ∀ t' ∈ set (merge ts1 ts2). rank (link t1 t2) ≤ rank t'
  proof —
  — Which is, again, done with the help of merge_rank_bound
  have rank (link t1 t2) = Suc (rank t2) by simp
  thus ?thesis using 3.prems by (auto simp: Suc_le_eq elim!:
merge_rank_bound)
  qed
  qed simp_all
  finally show ?thesis .
qed
qed auto

```

```

lemma mset_heap_merge[simp]:
  mset_heap (merge ts1 ts2) = mset_heap ts1 + mset_heap ts2
by (induction ts1 ts2 rule: merge.induct) auto

```

47.2.5 *get_min*

```

fun get_min :: 'a::linorder heap ⇒ 'a where
  get_min [t] = root t
| get_min (t#ts) = min (root t) (get_min ts)

```

```

lemma invar_tree_root_min:
  assumes invar_tree t
  assumes  $x \in \# \text{ mset\_tree } t$ 
  shows  $\text{root } t \leq x$ 
using assms unfolding invar_tree_def
by (induction t arbitrary: x rule: mset_tree.induct) (fastforce simp: mset_heap_def)

```

```

lemma get_min_mset:
  assumes  $ts \neq []$ 
  assumes invar ts
  assumes  $x \in \# \text{ mset\_heap } ts$ 
  shows  $\text{get\_min } ts \leq x$ 
  using assms
apply (induction ts arbitrary: x rule: get_min.induct)
apply (auto
  simp: invar_tree_root_min min_def intro: order_trans;
  meson linear order_trans invar_tree_root_min
  )+
done

```

```

lemma get_min_member:
   $ts \neq [] \implies \text{get\_min } ts \in \# \text{ mset\_heap } ts$ 
by (induction ts rule: get_min.induct) (auto simp: min_def)

```

```

lemma get_min:
  assumes  $\text{mset\_heap } ts \neq \{\#\}$ 
  assumes invar ts
  shows  $\text{get\_min } ts = \text{Min\_mset } (\text{mset\_heap } ts)$ 
using assms get_min_member get_min_mset
by (auto simp: eq_Min_iff)

```

47.2.6 *get_min_rest*

```

fun get_min_rest :: 'a::linorder heap  $\Rightarrow$  'a tree  $\times$  'a heap where
  get_min_rest [t] = (t,[])
| get_min_rest (t#ts) = (let (t',ts') = get_min_rest ts
  in if  $\text{root } t \leq \text{root } t'$  then (t,ts) else (t',t#ts'))

```

```

lemma get_min_rest_get_min_same_root:
  assumes  $ts \neq []$ 
  assumes  $\text{get\_min\_rest } ts = (t',ts')$ 
  shows  $\text{root } t' = \text{get\_min } ts$ 
using assms

```

by (*induction ts arbitrary: t' ts' rule: get_min.induct*) (*auto simp: min_def split: prod.splits*)

lemma *mset_get_min_rest*:

assumes *get_min_rest ts = (t', ts')*

assumes *ts ≠ []*

shows *mset ts = {#t'#} + mset ts'*

using *assms*

by (*induction ts arbitrary: t' ts' rule: get_min.induct*) (*auto split: prod.splits if_splits*)

lemma *set_get_min_rest*:

assumes *get_min_rest ts = (t', ts')*

assumes *ts ≠ []*

shows *set ts = Set.insert t' (set ts')*

using *mset_get_min_rest[OF assms, THEN arg_cong[where f=set_mset]]*

by *auto*

lemma *invar_get_min_rest*:

assumes *get_min_rest ts = (t', ts')*

assumes *ts ≠ []*

assumes *invar ts*

shows *invar_tree t' and invar ts'*

proof –

have *invar_tree t' ∧ invar ts'*

using *assms*

proof (*induction ts arbitrary: t' ts' rule: get_min.induct*)

case (*2 t v va*)

then show *?case*

apply (*clarsimp split: prod.splits if_splits*)

apply (*drule set_get_min_rest; fastforce*)

done

qed *auto*

thus *invar_tree t' and invar ts'* **by** *auto*

qed

47.2.7 *del_min*

definition *del_min* :: *'a::linorder heap ⇒ 'a::linorder heap* **where**

del_min ts = (case get_min_rest ts of

(Node r x ts₁, ts₂) ⇒ merge (rev ts₁) ts₂)

lemma *invar_del_min[simp]*:

assumes *ts ≠ []*

```

assumes invar ts
shows invar (del_min ts)
using assms
unfolding del_min_def
by (auto
      split: prod.split tree.split
      intro!: invar_merge invar_children
      dest: invar_get_min_rest
    )

```

```

lemma mset_heap_del_min:
  assumes ts ≠ []
  shows mset_heap ts = mset_heap (del_min ts) + {# get_min ts #}
using assms
unfolding del_min_def
apply (clarsimp split: tree.split prod.split)
apply (frule (1) get_min_rest_get_min_same_root)
apply (frule (1) mset_get_min_rest)
apply (auto simp: mset_heap_def)
done

```

47.2.8 Instantiating the Priority Queue Locale

Last step of functional correctness proof: combine all the above lemmas to show that binomial heaps satisfy the specification of priority queues with merge.

```

interpretation binheap: Priority_Queue_Merge
  where empty = [] and is_empty = (=) [] and insert = insert
  and get_min = get_min and del_min = del_min and merge = merge
  and invar = invar and mset = mset_heap
proof (unfold locales, goal_cases)
  case 1 thus ?case by simp
next
  case 2 thus ?case by auto
next
  case 3 thus ?case by auto
next
  case (4 q)
  thus ?case using mset_heap_del_min[of q] get_min[OF _ <invar q>]
    by (auto simp: union_single_eq_diff)
next
  case (5 q) thus ?case using get_min[of q] by auto
next
  case 6 thus ?case by (auto simp add: invar_def)

```

```

next
  case 7 thus ?case by simp
next
  case 8 thus ?case by simp
next
  case 9 thus ?case by simp
next
  case 10 thus ?case by simp
qed

```

47.3 Complexity

The size of a binomial tree is determined by its rank

```

lemma size_mset_btree:
  assumes invar_btree t
  shows size (mset_tree t) = 2rank t
  using assms
proof (induction t)
  case (Node r v ts)
  hence IH: size (mset_tree t) = 2rank t if t ∈ set ts for t
  using that by auto

```

```

from Node have COMPL: map rank ts = rev [0..r] by auto

```

```

have size (mset_heap ts) = (∑ t←ts. size (mset_tree t))
  by (induction ts) auto
also have ... = (∑ t←ts. 2rank t) using IH
  by (auto cong: map_cong)
also have ... = (∑ r←map rank ts. 2r)
  by (induction ts) auto
also have ... = (∑ i∈{0..r}. 2i)
  unfolding COMPL
  by (auto simp: rev_map[symmetric] interv_sum_list_conv_sum_set_nat)
also have ... = 2r - 1
  by (induction r) auto
finally show ?case
  by (simp)
qed

```

```

lemma size_mset_tree:
  assumes invar_tree t
  shows size (mset_tree t) = 2rank t
using assms unfolding invar_tree_def
by (simp add: size_mset_btree)

```

The length of a binomial heap is bounded by the number of its elements

```

lemma size_mset_heap:
  assumes invar ts
  shows  $\text{length } ts \leq \log 2 (\text{size } (\text{mset\_heap } ts) + 1)$ 
proof -
  from  $\langle \text{invar } ts \rangle$  have
    ASC: sorted_wrt ( $<$ ) (map rank ts) and
    TINV:  $\forall t \in \text{set } ts. \text{invar\_tree } t$ 
    unfolding invar_def by auto

  have  $(2::\text{nat})^{\text{length } ts} = (\sum_{i \in \{0..<\text{length } ts\}} 2^i) + 1$ 
    by (simp add: sum_power2)
  also have  $\dots \leq (\sum_{t \leftarrow ts} 2^{\text{rank } t}) + 1$ 
    using sorted_wrt_less_sum_mono_lowerbound[OF _ ASC, of ( $\wedge$ )
    ( $2::\text{nat}$ )]
    using power_increasing[where  $a=2::\text{nat}$ ]
    by (auto simp: o_def)
  also have  $\dots = (\sum_{t \leftarrow ts} \text{size } (\text{mset\_tree } t)) + 1$  using TINV
    by (auto cong: map_cong simp: size_mset_tree)
  also have  $\dots = \text{size } (\text{mset\_heap } ts) + 1$ 
    unfolding mset_heap_def by (induction ts) auto
  finally have  $2^{\text{length } ts} \leq \text{size } (\text{mset\_heap } ts) + 1$  .
  then show ?thesis using le_log2_of_power by blast
qed

```

47.3.1 Timing Functions

We define timing functions for each operation, and provide estimations of their complexity.

```

definition T_link :: 'a::linorder tree  $\Rightarrow$  'a tree  $\Rightarrow$  nat where
  [simp]: T_link _ _ = 1

```

This function is non-canonical: we omitted a $+1$ in the *else*-part, to keep the following analysis simpler and more to the point.

```

fun T_ins_tree :: 'a::linorder tree  $\Rightarrow$  'a heap  $\Rightarrow$  nat where
  T_ins_tree t [] = 1
  | T_ins_tree t1 (t2 # ts) = (
    (if rank t1 < rank t2 then 1
     else T_link t1 t2 + T_ins_tree (link t1 t2) ts)
  )

```

```

definition T_insert :: 'a::linorder  $\Rightarrow$  'a heap  $\Rightarrow$  nat where
  T_insert x ts = T_ins_tree (Node 0 x []) ts + 1

```

lemma *T_ins_tree_simple_bound*: $T_ins_tree\ t\ ts \leq length\ ts + 1$
by (*induction t ts rule: T_ins_tree.induct*) *auto*

47.3.2 *T_insert*

lemma *T_insert_bound*:

assumes *invar ts*

shows $T_insert\ x\ ts \leq \log\ 2\ (size\ (mset_heap\ ts) + 1) + 2$

proof –

have $real\ (T_insert\ x\ ts) \leq real\ (length\ ts) + 2$

unfolding *T_insert_def* **using** *T_ins_tree_simple_bound*

using *of_nat_mono* **by** *fastforce*

also note *size_mset_heap[OF ‹invar ts›]*

finally show *?thesis* **by** *simp*

qed

47.3.3 *T_merge*

context

includes *pattern_aliases*

begin

fun *T_merge* :: '*a*::*linorder heap* \Rightarrow '*a heap* \Rightarrow *nat* **where**

T_merge *ts*₁ [] = 1

| *T_merge* [] *ts*₂ = 1

| *T_merge* (*t*₁#*ts*₁ =: *h*₁) (*t*₂#*ts*₂ =: *h*₂) = 1 + (
if *rank* *t*₁ < *rank* *t*₂ *then* *T_merge* *ts*₁ *h*₂
else if *rank* *t*₂ < *rank* *t*₁ *then* *T_merge* *h*₁ *ts*₂
else *T_ins_tree* (*link* *t*₁ *t*₂) (*merge* *ts*₁ *ts*₂) + *T_merge* *ts*₁ *ts*₂
)

end

A crucial idea is to estimate the time in correlation with the result length, as each carry reduces the length of the result.

lemma *T_ins_tree_length*:

$T_ins_tree\ t\ ts + length\ (ins_tree\ t\ ts) = 2 + length\ ts$

by (*induction t ts rule: ins_tree.induct*) *auto*

lemma *T_merge_length*:

$length\ (merge\ ts_1\ ts_2) + T_merge\ ts_1\ ts_2 \leq 2 * (length\ ts_1 + length\ ts_2) + 1$

by (*induction ts₁ ts₂ rule: T_merge.induct*)

(*auto simp: T_ins_tree_length algebra_simps*)

Finally, we get the desired logarithmic bound

lemma *T_merge_bound*:

fixes $ts_1\ ts_2$

defines $n_1 \equiv \text{size } (\text{mset_heap } ts_1)$

defines $n_2 \equiv \text{size } (\text{mset_heap } ts_2)$

assumes *invar* ts_1 *invar* ts_2

shows $T_merge\ ts_1\ ts_2 \leq 4 * \log 2\ (n_1 + n_2 + 1) + 1$

proof –

note $n_defs = \text{assms}(1,2)$

have $T_merge\ ts_1\ ts_2 \leq 2 * \text{real } (\text{length } ts_1) + 2 * \text{real } (\text{length } ts_2) + 1$

using T_merge_length [of $ts_1\ ts_2$] **by** *simp*

also note size_mset_heap [OF *invar* ts_1]

also note size_mset_heap [OF *invar* ts_2]

finally have $T_merge\ ts_1\ ts_2 \leq 2 * \log 2\ (n_1 + 1) + 2 * \log 2\ (n_2 + 1) + 1$

unfolding n_defs **by** (*simp add: algebra_simps*)

also have $\log 2\ (n_1 + 1) \leq \log 2\ (n_1 + n_2 + 1)$

unfolding n_defs **by** (*simp add: algebra_simps*)

also have $\log 2\ (n_2 + 1) \leq \log 2\ (n_1 + n_2 + 1)$

unfolding n_defs **by** (*simp add: algebra_simps*)

finally show *?thesis* **by** (*simp add: algebra_simps*)

qed

47.3.4 *T_get_min*

fun *T_get_min* :: '*a*::*linorder* *heap* \Rightarrow *nat* **where**

$T_get_min\ [t] = 1$

| $T_get_min\ (t\#\!ts) = 1 + T_get_min\ ts$

lemma *T_get_min_estimate*: $ts \neq [] \Longrightarrow T_get_min\ ts = \text{length } ts$

by (*induction* ts *rule: T_get_min.induct*) *auto*

lemma *T_get_min_bound*:

assumes *invar* ts

assumes $ts \neq []$

shows $T_get_min\ ts \leq \log 2\ (\text{size } (\text{mset_heap } ts) + 1)$

proof –

have $1: T_get_min\ ts = \text{length } ts$ **using** *assms* *T_get_min_estimate* **by** *auto*

also note size_mset_heap [OF *invar* ts]

finally show *?thesis* .

qed

47.3.5 T_del_min

fun $T_get_min_rest$:: 'a::linorder heap \Rightarrow nat **where**

$T_get_min_rest$ [t] = 1
| $T_get_min_rest$ (t#ts) = 1 + $T_get_min_rest$ ts

lemma $T_get_min_rest_estimate$: $ts \neq [] \implies T_get_min_rest$ ts = length ts

by (induction ts rule: $T_get_min_rest.induct$) auto

lemma $T_get_min_rest_bound$:

assumes invar ts

assumes $ts \neq []$

shows $T_get_min_rest$ ts \leq log 2 (size (mset_heap ts) + 1)

proof –

have 1: $T_get_min_rest$ ts = length ts **using** assms $T_get_min_rest_estimate$
by auto

also note size_mset_heap[OF <invar ts>]

finally show ?thesis .

qed

Note that although the definition of function rev has quadratic complexity, it can and is implemented (via suitable code lemmas) as a linear time function. Thus the following definition is justified:

definition T_rev xs = length xs + 1

definition T_del_min :: 'a::linorder heap \Rightarrow nat **where**

T_del_min ts = $T_get_min_rest$ ts + (case get_min_rest ts of (Node
_ x ts₁, ts₂)
 $\implies T_rev$ ts₁ + T_merge (rev ts₁) ts₂
) + 1

lemma $T_del_min_bound$:

fixes ts

defines n \equiv size (mset_heap ts)

assumes invar ts **and** $ts \neq []$

shows T_del_min ts \leq 6 * log 2 (n+1) + 3

proof –

obtain r x ts₁ ts₂ **where** GM: get_min_rest ts = (Node r x ts₁, ts₂)
by (metis surj_pair tree.exhaust_sel)

have I1: invar (rev ts₁) **and** I2: invar ts₂

using invar_get_min_rest[OF GM <ts \neq []> <invar ts>] invar_children

by auto

```

define  $n_1$  where  $n_1 = \text{size } (\text{mset\_heap } ts_1)$ 
define  $n_2$  where  $n_2 = \text{size } (\text{mset\_heap } ts_2)$ 

have  $n_1 \leq n$   $n_1 + n_2 \leq n$  unfolding  $n\_def$   $n_1\_def$   $n_2\_def$ 
  using  $\text{mset\_get\_min\_rest}[OF\ GM\ \langle ts \neq [] \rangle]$ 
  by (auto simp: mset_heap_def)

have  $T\_del\_min\ ts = \text{real } (T\_get\_min\_rest\ ts) + \text{real } (T\_rev\ ts_1) +$ 
 $\text{real } (T\_merge\ (rev\ ts_1)\ ts_2) + 1$ 
  unfolding  $T\_del\_min\_def\ GM$ 
  by simp
also have  $T\_get\_min\_rest\ ts \leq \log\ 2\ (n+1)$ 
  using  $T\_get\_min\_rest\_bound[OF\ \langle \text{invar } ts \rangle\ \langle ts \neq [] \rangle]$  unfolding  $n\_def$ 
by simp
also have  $T\_rev\ ts_1 \leq 1 + \log\ 2\ (n_1 + 1)$ 
  unfolding  $T\_rev\_def\ n_1\_def$  using  $\text{size\_mset\_heap}[OF\ I1]$  by simp
also have  $T\_merge\ (rev\ ts_1)\ ts_2 \leq 4 * \log\ 2\ (n_1 + n_2 + 1) + 1$ 
  unfolding  $n_1\_def\ n_2\_def$  using  $T\_merge\_bound[OF\ I1\ I2]$  by (simp
add: algebra_simps)
finally have  $T\_del\_min\ ts \leq \log\ 2\ (n+1) + \log\ 2\ (n_1 + 1) + 4 * \log\ 2$ 
 $(\text{real } (n_1 + n_2) + 1) + 3$ 
  by (simp add: algebra_simps)
also note  $\langle n_1 + n_2 \leq n \rangle$ 
also note  $\langle n_1 \leq n \rangle$ 
finally show ?thesis by (simp add: algebra_simps)
qed

end

```

48 Time functions for various standard library operations

```

theory Time_Funs
  imports Main
begin

```

```

fun  $T\_length :: 'a\ list \Rightarrow nat$  where
   $T\_length\ [] = 1$ 
|  $T\_length\ (x \# xs) = T\_length\ xs + 1$ 

```

```

lemma  $T\_length\_eq: T\_length\ xs = \text{length } xs + 1$ 
  by (induction xs) auto

```

lemmas [simp del] = T_length.simps

fun T_map :: ('a ⇒ nat) ⇒ 'a list ⇒ nat **where**
 T_map T_f [] = 1
| T_map T_f (x # xs) = T_f x + T_map T_f xs + 1

lemma T_map_eq: T_map T_f xs = (∑ x←xs. T_f x) + length xs + 1
 by (induction xs) auto

lemmas [simp del] = T_map.simps

fun T_filter :: ('a ⇒ nat) ⇒ 'a list ⇒ nat **where**
 T_filter T_p [] = 1
| T_filter T_p (x # xs) = T_p x + T_filter T_p xs + 1

lemma T_filter_eq: T_filter T_p xs = (∑ x←xs. T_p x) + length xs + 1
 by (induction xs) auto

lemmas [simp del] = T_filter.simps

fun T_nth :: 'a list ⇒ nat ⇒ nat **where**
 T_nth [] n = 1
| T_nth (x # xs) n = (case n of 0 ⇒ 1 | Suc n' ⇒ T_nth xs n' + 1)

lemma T_nth_eq: T_nth xs n = min n (length xs) + 1
 by (induction xs n rule: T_nth.induct) (auto split: nat.splits)

lemmas [simp del] = T_nth.simps

fun T_take :: nat ⇒ 'a list ⇒ nat **where**
 T_take n [] = 1
| T_take n (x # xs) = (case n of 0 ⇒ 1 | Suc n' ⇒ T_take n' xs + 1)

lemma T_take_eq: T_take n xs = min n (length xs) + 1
 by (induction xs arbitrary: n) (auto split: nat.splits)

fun T_drop :: nat ⇒ 'a list ⇒ nat **where**
 T_drop n [] = 1

| $T_drop\ n\ (x\ \#\ xs) = (case\ n\ of\ 0 \Rightarrow 1\ | \ Suc\ n' \Rightarrow T_drop\ n'\ xs + 1)$

lemma T_drop_eq : $T_drop\ n\ xs = min\ n\ (length\ xs) + 1$
by (*induction xs arbitrary: n*) (*auto split: nat.splits*)

end

49 The Median-of-Medians Selection Algorithm

theory *Selection*

imports *Complex_Main Sorting Time_Funs*

begin

Note that there is significant overlap between this theory (which is intended mostly for the Functional Data Structures book) and the Median-of-Medians AFP entry.

49.1 Auxiliary material

lemma $replicate_numeral$: $replicate\ (numeral\ n)\ x = x\ \# \ replicate\ (pred_numeral\ n)\ x$

by (*simp add: numeral_eq_Suc*)

lemma $isort_correct$: $isort\ xs = sort\ xs$

using $sorted_isort\ mset_isort$ **by** (*metis properties_for_sort*)

lemma $sum_list_replicate$ [*simp*]: $sum_list\ (replicate\ n\ x) = n * x$

by (*induction n*) *auto*

lemma $mset_concat$: $mset\ (concat\ xss) = sum_list\ (map\ mset\ xss)$

by (*induction xss*) *simp_all*

lemma $set_mset_sum_list$ [*simp*]: $set_mset\ (sum_list\ xs) = (\bigcup_{x \in set\ xs} set_mset\ x)$

by (*induction xs*) *auto*

lemma $filter_mset_image_mset$:

$filter_mset\ P\ (image_mset\ f\ A) = image_mset\ f\ (filter_mset\ (\lambda x. P\ (f\ x))\ A)$

by (*induction A*) *auto*

lemma $filter_mset_sum_list$: $filter_mset\ P\ (sum_list\ xs) = sum_list\ (map\ (filter_mset\ P)\ xs)$

by (*induction xs*) *simp_all*

lemma *sum_mset_mset_mono*:

assumes $(\bigwedge x. x \in\# A \implies f x \subseteq\# g x)$

shows $(\sum_{x \in\# A}. f x) \subseteq\# (\sum_{x \in\# A}. g x)$

using *assms* **by** (*induction A*) (*auto intro!*: *subset_mset.add_mono*)

lemma *mset_filter_mono*:

assumes $A \subseteq\# B \bigwedge x. x \in\# A \implies P x \implies Q x$

shows $\text{filter_mset } P A \subseteq\# \text{filter_mset } Q B$

by (*rule mset_subset_eqI*) (*insert assms, auto simp: mset_subset_eq_count count_eq_zero_iff*)

lemma *size_mset_sum_mset_distrib*: $\text{size } (\text{sum_mset } A :: 'a \text{ multiset}) = \text{sum_mset } (\text{image_mset } \text{size } A)$

by (*induction A*) *auto*

lemma *sum_mset_mono*:

assumes $\bigwedge x. x \in\# A \implies f x \leq (g x :: 'a :: \{\text{ordered_ab_semigroup_add, comm_monoid_add}\})$

shows $(\sum_{x \in\# A}. f x) \leq (\sum_{x \in\# A}. g x)$

using *assms* **by** (*induction A*) (*auto intro!*: *add_mono*)

lemma *filter_mset_is_empty_iff*: $\text{filter_mset } P A = \{\#\} \iff (\forall x. x \in\# A \longrightarrow \neg P x)$

by (*auto simp: multiset_eq_iff count_eq_zero_iff*)

lemma *sort_eq_Nil_iff* [*simp*]: $\text{sort } xs = [] \iff xs = []$

by (*metis set_empty set_sort*)

lemma *sort_mset_cong*: $\text{mset } xs = \text{mset } ys \implies \text{sort } xs = \text{sort } ys$

by (*metis sorted_list_of_multiset_mset*)

lemma *Min_set_sorted*: $\text{sorted } xs \implies xs \neq [] \implies \text{Min } (\text{set } xs) = \text{hd } xs$

by (*cases xs; force intro: Min_insert2*)

lemma *hd_sort*:

fixes $xs :: 'a :: \text{linorder } \text{list}$

shows $xs \neq [] \implies \text{hd } (\text{sort } xs) = \text{Min } (\text{set } xs)$

by (*subst Min_set_sorted [symmetric]*) *auto*

lemma *length_filter_conv_size_filter_mset*: $\text{length } (\text{filter } P xs) = \text{size } (\text{filter_mset } P (\text{mset } xs))$

by (*induction xs*) *auto*

```

lemma sorted_filter_less_subset_take:
  assumes sorted xs and i < length xs
  shows  $\{\#x \in\# \text{mset } xs. x < xs ! i\# \} \subseteq\# \text{mset } (\text{take } i \text{ } xs)$ 
  using assms
proof (induction xs arbitrary: i rule: list.induct)
  case (Cons x xs i)
  show ?case
  proof (cases i)
    case 0
    thus ?thesis using Cons.prem by (auto simp: filter_mset_is_empty_iff)
  next
    case (Suc i')
    have  $\{\#y \in\# \text{mset } (x \# xs). y < (x \# xs) ! i'\# \} \subseteq\# \text{add\_mset } x \{\#y \in\# \text{mset } xs. y < xs ! i'\#\}$ 
      using Suc Cons.prem by (auto)
    also have  $\dots \subseteq\# \text{add\_mset } x (\text{mset } (\text{take } i' \text{ } xs))$ 
      unfolding mset_subset_eq_add_mset_cancel using Cons.prem Suc
      by (intro Cons.IH) (auto)
    also have  $\dots = \text{mset } (\text{take } i \text{ } (x \# xs))$  by (simp add: Suc)
    finally show ?thesis .
  qed
qed auto

```

```

lemma sorted_filter_greater_subset_drop:
  assumes sorted xs and i < length xs
  shows  $\{\#x \in\# \text{mset } xs. x > xs ! i\# \} \subseteq\# \text{mset } (\text{drop } (\text{Suc } i) \text{ } xs)$ 
  using assms
proof (induction xs arbitrary: i rule: list.induct)
  case (Cons x xs i)
  show ?case
  proof (cases i)
    case 0
    thus ?thesis by (auto simp: sorted_append filter_mset_is_empty_iff)
  next
    case (Suc i')
    have  $\{\#y \in\# \text{mset } (x \# xs). y > (x \# xs) ! i'\# \} \subseteq\# \{\#y \in\# \text{mset } xs. y > xs ! i'\#\}$ 
      using Suc Cons.prem by (auto simp: set_conv_nth)
    also have  $\dots \subseteq\# \text{mset } (\text{drop } (\text{Suc } i') \text{ } xs)$ 
      using Cons.prem Suc by (intro Cons.IH) (auto)
    also have  $\dots = \text{mset } (\text{drop } (\text{Suc } i) \text{ } (x \# xs))$  by (simp add: Suc)
    finally show ?thesis .
  qed
qed auto

```

49.2 Chopping a list into equally-sized bits

```

fun chop :: nat ⇒ 'a list ⇒ 'a list list where
  chop 0 _ = []
| chop _ [] = []
| chop n xs = take n xs # chop n (drop n xs)

```

lemmas [simp del] = chop.simps

This is an alternative induction rule for *chop*, which is often nicer to use.

```

lemma chop_induct' [case_names trivial reduce]:
  assumes ∧ n xs. n = 0 ∨ xs = [] ⇒ P n xs
  assumes ∧ n xs. n > 0 ⇒ xs ≠ [] ⇒ P n (drop n xs) ⇒ P n xs
  shows P n xs
  using assms
proof induction_schema
  show wf (measure (length ∘ snd))
  by auto
qed (blast | simp)+

```

lemma chop_eq_Nil_iff [simp]: chop n xs = [] ⟷ n = 0 ∨ xs = []
by (induction n xs rule: chop.induct; subst chop.simps) auto

lemma chop_0 [simp]: chop 0 xs = []
by (simp add: chop.simps)

lemma chop_Nil [simp]: chop n [] = []
by (cases n) (auto simp: chop.simps)

lemma chop_reduce: n > 0 ⇒ xs ≠ [] ⇒ chop n xs = take n xs # chop n (drop n xs)
by (cases n; cases xs) (auto simp: chop.simps)

lemma concat_chop [simp]: n > 0 ⇒ concat (chop n xs) = xs
by (induction n xs rule: chop.induct; subst chop.simps) auto

lemma chop_elem_not_Nil [dest]: ys ∈ set (chop n xs) ⇒ ys ≠ []
by (induction n xs rule: chop.induct; subst (asm) chop.simps)
 (auto simp: eq_commute[of []] split: if_splits)

lemma length_chop_part_le: ys ∈ set (chop n xs) ⇒ length ys ≤ n
by (induction n xs rule: chop.induct; subst (asm) chop.simps) (auto split: if_splits)

lemma *length_chop*:
assumes $n > 0$
shows $\text{length } (\text{chop } n \text{ } xs) = \text{nat } \lceil \text{length } xs / n \rceil$
proof –
from $\langle n > 0 \rangle$ **have** $\text{real } n * \text{length } (\text{chop } n \text{ } xs) \geq \text{length } xs$
by (*induction* $n \text{ } xs$ *rule*: *chop.induct*; *subst chop.simps*) (*auto simp*:
field_simps)
moreover from $\langle n > 0 \rangle$ **have** $\text{real } n * \text{length } (\text{chop } n \text{ } xs) < \text{length } xs +$
 n
by (*induction* $n \text{ } xs$ *rule*: *chop.induct*; *subst chop.simps*)
(*auto simp*: *field_simps split*: *nat_diff_split_asm*)+
ultimately have $\text{length } (\text{chop } n \text{ } xs) \geq \text{length } xs / n$ **and** $\text{length } (\text{chop } n$
 $xs) < \text{length } xs / n + 1$
using *assms* **by** (*auto simp*: *field_simps*)
thus *?thesis* **by** *linarith*
qed

lemma *sum_msets_chop*: $n > 0 \implies (\sum_{ys \leftarrow \text{chop } n \text{ } xs. \text{mset } ys}) = \text{mset } xs$
by (*subst mset_concat [symmetric]*) *simp_all*

lemma *UN_sets_chop*: $n > 0 \implies (\bigcup_{ys \in \text{set } (\text{chop } n \text{ } xs). \text{set } ys}) = \text{set } xs$
by (*simp only*: *set_concat [symmetric] concat_chop*)

lemma *chop_append*: $d \text{ dvd } \text{length } xs \implies \text{chop } d \text{ } (xs @ ys) = \text{chop } d \text{ } xs @$
 $\text{chop } d \text{ } ys$
by (*induction* $d \text{ } xs$ *rule*: *chop_induct'*) (*auto simp*: *chop_reduce dvd_imp_le*)

lemma *chop_replicate [simp]*: $d > 0 \implies \text{chop } d \text{ } (\text{replicate } d \text{ } xs) = [\text{replicate } d \text{ } xs]$
by (*subst chop_reduce*) *auto*

lemma *chop_replicate_dvd [simp]*:
assumes $d \text{ dvd } n$
shows $\text{chop } d \text{ } (\text{replicate } n \text{ } x) = \text{replicate } (n \text{ div } d) \text{ } (\text{replicate } d \text{ } x)$
proof (*cases* $d = 0$)
case *False*
from *assms* **obtain** k **where** $n = d * k$
by *blast*
have $\text{chop } d \text{ } (\text{replicate } (d * k) \text{ } x) = \text{replicate } k \text{ } (\text{replicate } d \text{ } x)$
using *False* **by** (*induction* k) (*auto simp*: *replicate_add chop_append*)
thus *?thesis* **using** *False* **by** (*simp add*: k)
qed *auto*

lemma *chop_concat*:
assumes $\forall xs \in \text{set } xss. \text{length } xs = d \text{ and } d > 0$
shows $\text{chop } d (\text{concat } xss) = xss$
using *assms*
proof (*induction xss*)
case (*Cons xs xss*)
have $\text{chop } d (\text{concat } (xs \# xss)) = \text{chop } d (xs @ \text{concat } xss)$
by *simp*
also have $\dots = \text{chop } d xs @ \text{chop } d (\text{concat } xss)$
using *Cons.prem*s **by** (*intro chop_append*) *auto*
also have $\text{chop } d xs = [xs]$
using *Cons.prem*s **by** (*subst chop_reduce*) *auto*
also have $\text{chop } d (\text{concat } xss) = xss$
using *Cons.prem*s **by** (*intro Cons.IH*) *auto*
finally show *?case* **by** *simp*
qed *auto*

49.3 Selection

definition *select* :: $\text{nat} \Rightarrow ('a :: \text{linorder}) \text{list} \Rightarrow 'a$ **where**
select $k xs = \text{sort } xs ! k$

lemma *select_0*: $xs \neq [] \implies \text{select } 0 xs = \text{Min } (\text{set } xs)$
by (*simp add: hd_sort select_def flip: hd_conv_nth*)

lemma *select_mset_cong*: $\text{mset } xs = \text{mset } ys \implies \text{select } k xs = \text{select } k ys$
using *sort_mset_cong*[*of xs ys*] **unfolding** *select_def* **by** *auto*

lemma *select_in_set* [*intro, simp*]:
assumes $k < \text{length } xs$
shows $\text{select } k xs \in \text{set } xs$
proof –
from *assms* **have** $\text{sort } xs ! k \in \text{set } (\text{sort } xs)$ **by** (*intro nth_mem*) *auto*
also have $\text{set } (\text{sort } xs) = \text{set } xs$ **by** *simp*
finally show *?thesis* **by** (*simp add: select_def*)
qed

lemma
assumes $n < \text{length } xs$
shows *size_less_than_select*: $\text{size } \{\#y \in \# \text{mset } xs. y < \text{select } n xs\# \}$
 $\leq n$
and *size_greater_than_select*: $\text{size } \{\#y \in \# \text{mset } xs. y > \text{select } n xs\# \}$
 $< \text{length } xs - n$
proof –

```

have size {#y ∈# mset (sort xs). y < select n xs#} ≤ size (mset (take
n (sort xs)))
  unfolding select_def using assms
  by (intro size_mset_mono sorted_filter_less_subset_take) auto
thus size {#y ∈# mset xs. y < select n xs#} ≤ n
  by simp
have size {#y ∈# mset (sort xs). y > select n xs#} ≤ size (mset (drop
(Suc n) (sort xs)))
  unfolding select_def using assms
  by (intro size_mset_mono sorted_filter_greater_subset_drop) auto
thus size {#y ∈# mset xs. y > select n xs#} < length xs - n
  using assms by simp
qed

```

49.4 The designated median of a list

definition *median* where $median\ xs = select\ ((length\ xs - 1)\ div\ 2)\ xs$

lemma *median_in_set* [intro, simp]:

```

assumes xs ≠ []
shows median xs ∈ set xs

```

proof –

```

from assms have length xs > 0 by auto
hence (length xs - 1) div 2 < length xs by linarith
thus ?thesis by (simp add: median_def)

```

qed

lemma *size_less_than_median*: $size\ \{\#y \in\# mset\ xs.\ y < median\ xs\#\} \leq (length\ xs - 1)\ div\ 2$

proof (cases xs = [])

```

case False
hence length xs > 0
  by auto

```

```

hence less: (length xs - 1) div 2 < length xs
  by linarith

```

```

show size {#y ∈# mset xs. y < median xs#} ≤ (length xs - 1) div 2
  using size_less_than_select[OF less] by (simp add: median_def)

```

qed auto

lemma *size_greater_than_median*: $size\ \{\#y \in\# mset\ xs.\ y > median\ xs\#\} \leq length\ xs\ div\ 2$

proof (cases xs = [])

```

case False
hence length xs > 0

```

by *auto*
hence $\text{less: } (\text{length } xs - 1) \text{ div } 2 < \text{length } xs$
by *linarith*
have $\text{size } \{\#y \in \# \text{ mset } xs. y > \text{median } xs\# \} < \text{length } xs - (\text{length } xs - 1) \text{ div } 2$
using *size_greater_than_select[OF less]* **by** (*simp add: median_def*)
also have $\dots = \text{length } xs \text{ div } 2 + 1$
using $\langle \text{length } xs > 0 \rangle$ **by** *linarith*
finally show $\text{size } \{\#y \in \# \text{ mset } xs. y > \text{median } xs\# \} \leq \text{length } xs \text{ div } 2$
by *simp*
qed *auto*

lemmas *median_props = size_less_than_median size_greater_than_median*

49.5 A recurrence for selection

definition *partition3* :: 'a \Rightarrow 'a :: *linorder list* \Rightarrow 'a *list* \times 'a *list* \times 'a *list*
where

partition3 x $xs = (\text{filter } (\lambda y. y < x) xs, \text{filter } (\lambda y. y = x) xs, \text{filter } (\lambda y. y > x) xs)$

lemma *partition3_code* [*code*]:

partition3 x [] = ([], [], [])
partition3 x ($y \# ys$) =
 (case *partition3* x ys of (ls, es, gs) \Rightarrow
 if $y < x$ then ($y \# ls, es, gs$) else if $x = y$ then ($ls, y \# es, gs$) else
 ($ls, es, y \# gs$))
by (*auto simp: partition3_def*)

lemma *sort_append*:

assumes $\forall x \in \text{set } xs. \forall y \in \text{set } ys. x \leq y$
shows $\text{sort } (xs @ ys) = \text{sort } xs @ \text{sort } ys$
using *assms* **by** (*intro properties_for_sort*) (*auto simp: sorted_append*)

lemma *select_append*:

assumes $\forall y \in \text{set } ys. \forall z \in \text{set } zs. y \leq z$
shows $k < \text{length } ys \implies \text{select } k (ys @ zs) = \text{select } k ys$
and $k \in \{\text{length } ys..<\text{length } ys + \text{length } zs\} \implies$
 $\text{select } k (ys @ zs) = \text{select } (k - \text{length } ys) zs$
using *assms* **by** (*simp_all add: select_def sort_append nth_append*)

lemma *select_append'*:

assumes $\forall y \in \text{set } ys. \forall z \in \text{set } zs. y \leq z$ **and** $k < \text{length } ys + \text{length } zs$
shows $\text{select } k (ys @ zs) = (\text{if } k < \text{length } ys \text{ then } \text{select } k ys \text{ else } \text{select } (k - \text{length } ys) zs)$

$(k - \text{length } ys) \text{ } zs)$
using *assms* **by** (*auto intro!*: *select_append*)

theorem *select_rec_partition*:

assumes $k < \text{length } xs$
shows $\text{select } k \text{ } xs = ($
 $\text{let } (ls, es, gs) = \text{partition3 } x \text{ } xs$
 in
 $\text{if } k < \text{length } ls \text{ then } \text{select } k \text{ } ls$
 $\text{else if } k < \text{length } ls + \text{length } es \text{ then } x$
 $\text{else } \text{select } (k - \text{length } ls - \text{length } es) \text{ } gs$
 $)$ (**is** $_ = ?rhs$)

proof –

define $ls \ es \ gs$ **where** $ls = \text{filter } (\lambda y. y < x) \text{ } xs$ **and** $es = \text{filter } (\lambda y. y =$
 $x) \text{ } xs$

and $gs = \text{filter } (\lambda y. y > x) \text{ } xs$

define $nl \ ne$ **where** [*simp*]: $nl = \text{length } ls \ ne = \text{length } es$

have *mset_eq*: $\text{mset } xs = \text{mset } ls + \text{mset } es + \text{mset } gs$

unfolding *ls_def es_def gs_def* **by** (*induction xs*) *auto*

have *length_eq*: $\text{length } xs = \text{length } ls + \text{length } es + \text{length } gs$

unfolding *ls_def es_def gs_def*

using [[*simp_depth_limit* = 1]] **by** (*induction xs*) *auto*

have [*simp*]: $\text{select } i \text{ } es = x$ **if** $i < \text{length } es$ **for** i

proof –

have $\text{select } i \text{ } es \in \text{set } (\text{sort } es)$ **unfolding** *select_def*

using *that* **by** (*intro nth_mem*) *auto*

thus *?thesis*

by (*auto simp: es_def*)

qed

have $\text{select } k \text{ } xs = \text{select } k \text{ } (ls @ (es @ gs))$

by (*intro select_mset_cong*) (*simp_all add: mset_eq*)

also have $\dots = (\text{if } k < nl \text{ then } \text{select } k \text{ } ls \text{ else } \text{select } (k - nl) \text{ } (es @ gs))$

unfolding *nl_ne_def* **using** *assms*

by (*intro select_append'*) (*auto simp: ls_def es_def gs_def length_eq*)

also have $\dots = (\text{if } k < nl \text{ then } \text{select } k \text{ } ls \text{ else if } k < nl + ne \text{ then } x$
 $\text{else } \text{select } (k - nl - ne) \text{ } gs)$

proof (*rule if_cong*)

assume $\neg k < nl$

have $\text{select } (k - nl) \text{ } (es @ gs) =$

$(\text{if } k - nl < ne \text{ then } \text{select } (k - nl) \text{ } es \text{ else } \text{select } (k - nl -$

$ne) \text{ } gs)$

unfolding *nl_ne_def* **using** *assms* $\langle \neg k < nl \rangle$

by (*intro select_append'*) (*auto simp: ls_def es_def gs_def length_eq*)

```

    also have ... = (if k < nl + ne then x else select (k - nl - ne) gs)
      using ⟨¬k < nl⟩ by auto
    finally show select (k - nl) (es @ gs) = ... .
qed simp_all
also have ... = ?rhs
  by (simp add: partition3_def ls_def es_def gs_def)
finally show ?thesis .
qed

```

49.6 The size of the lists in the recursive calls

We now derive an upper bound for the number of elements of a list that are smaller (resp. bigger) than the median of medians with chopping size 5. To avoid having to do the same proof twice, we do it generically for an operation \prec that we will later instantiate with either $<$ or $>$.

context

```

  fixes xs :: 'a :: linorder list
  fixes M defines M ≡ median (map median (chop 5 xs))
begin

```

lemma *size_median_of_medians_aux*:

```

  fixes R :: 'a :: linorder ⇒ 'a ⇒ bool (infix < 50)
  assumes R: R ∈ {(<), (>)}
  shows size {#y ∈ # mset xs. y < M#} ≤ nat [0.7 * length xs + 3]

```

proof –

```

  define n and m where [simp]: n = length xs and m = length (chop 5 xs)

```

We define an abbreviation for the multiset of all the chopped-up groups:

We then split that multiset into those groups whose medians is less than M and the rest.

```

  define Y_small (Y_<) where Y_< = filter_mset (λys. median ys < M)
    (mset (chop 5 xs))
  define Y_big (Y_>) where Y_> = filter_mset (λys. ¬(median ys < M))
    (mset (chop 5 xs))
  have m = size (mset (chop 5 xs)) by (simp add: m_def)
  also have mset (chop 5 xs) = Y_< + Y_> unfolding Y_small_def Y_big_def
    by (rule multiset_partition)
  finally have m_eq: m = size Y_< + size Y_> by simp

```

At most half of the lists have a median that is smaller than the median of medians:

```

  have size Y_< = size (image_mset median Y_<) by simp

```

also have $image_mset\ median\ Y_{<} = \{\#y \in\# mset (map\ median\ (chop\ 5\ xs)).\ y < M\#\}$
unfolding Y_small_def **by** $(subst\ filter_mset_image_mset\ [symmetric])$
 $simp_all$
also have $size \dots \leq (length\ (map\ median\ (chop\ 5\ xs)))\ div\ 2$
unfolding M_def **using** $median_props[of\ map\ median\ (chop\ 5\ xs)]\ R$
by $auto$
also have $\dots = m\ div\ 2$ **by** $(simp\ add:\ m_def)$
finally have $size_Y_small:\ size\ Y_{<} \leq m\ div\ 2$.

We estimate the number of elements less than M by grouping them into elements coming from $Y_{<}$ and elements coming from Y_{\geq} :

have $\{\#y \in\# mset\ xs.\ y < M\#\} = \{\#y \in\# (\sum\ ys \leftarrow chop\ 5\ xs.\ mset\ ys).\ y < M\#\}$
by $(subst\ sum_msets_chop)\ simp_all$
also have $\dots = (\sum\ ys \leftarrow chop\ 5\ xs.\ \{\#y \in\# mset\ ys.\ y < M\#\})$
by $(subst\ filter_mset_sum_list)\ (simp\ add:\ o_def)$
also have $\dots = (\sum\ ys \in\# mset\ (chop\ 5\ xs).\ \{\#y \in\# mset\ ys.\ y < M\#\})$
by $(subst\ sum_mset_sum_list\ [symmetric])\ simp_all$
also have $mset\ (chop\ 5\ xs) = Y_{<} + Y_{\geq}$
by $(simp\ add:\ Y_small_def\ Y_big_def\ not_le)$
also have $(\sum\ ys \in\# \dots.\ \{\#y \in\# mset\ ys.\ y < M\#\}) =$
 $(\sum\ ys \in\# Y_{<}.\ \{\#y \in\# mset\ ys.\ y < M\#\}) + (\sum\ ys \in\# Y_{\geq}.\$
 $\{\#y \in\# mset\ ys.\ y < M\#\})$
by $simp$

Next, we overapproximate the elements contributed by $Y_{<}$: instead of those elements that are smaller than the median, we take *all* the elements of each group. For the elements contributed by Y_{\geq} , we overapproximate by taking all those that are less than their median instead of only those that are less than M .

also have $\dots \subseteq\# (\sum\ ys \in\# Y_{<}.\ mset\ ys) + (\sum\ ys \in\# Y_{\geq}.\ \{\#y \in\# mset\ ys.\ y < median\ ys\#\})$
using R
by $(intro\ subset_mset.add_mono\ sum_mset_mset_mono\ mset_filter_mono)$
 $(auto\ simp:\ Y_big_def)$
finally have $size\ \{\#y \in\# mset\ xs.\ y < M\#\} \leq size \dots$
by $(rule\ size_mset_mono)$
hence $size\ \{\#y \in\# mset\ xs.\ y < M\#\} \leq$
 $(\sum\ ys \in\# Y_{<}.\ length\ ys) + (\sum\ ys \in\# Y_{\geq}.\ size\ \{\#y \in\# mset\ ys.\ y$
 $< median\ ys\#\})$
by $(simp\ add:\ size_mset_sum_mset_distrib\ multiset.map_comp\ o_def)$

Next, we further overapproximate the first sum by noting that each group has at most size 5.

also have $(\sum ys \in \# Y_{\prec}. \text{length } ys) \leq (\sum ys \in \# Y_{\prec}. 5)$
by *(intro sum_mset_mono)* *(auto simp: Y_small_def length_chop_part_le)*
also have $\dots = 5 * \text{size } Y_{\prec}$ **by** *simp*

Next, we note that each group in Y_{\succeq} can have at most 2 elements that are smaller than its median.

also have $(\sum ys \in \# Y_{\succeq}. \text{size } \{\#y \in \# \text{mset } ys. y \prec \text{median } ys\}) \leq$
 $(\sum ys \in \# Y_{\succeq}. \text{length } ys \text{ div } 2)$

proof *(intro sum_mset_mono, goal_cases)*

fix *ys* **assume** $ys \in \# Y_{\succeq}$

hence $ys \neq []$

by *(auto simp: Y_big_def)*

thus $\text{size } \{\#y \in \# \text{mset } ys. y \prec \text{median } ys\} \leq \text{length } ys \text{ div } 2$

using *R median_props[of ys]* **by** *auto*

qed

also have $\dots \leq (\sum ys \in \# Y_{\succeq}. 2)$

by *(intro sum_mset_mono div_le_mono diff_le_mono)*

(auto simp: Y_big_def dest: length_chop_part_le)

also have $\dots = 2 * \text{size } Y_{\succeq}$ **by** *simp*

Simplifying gives us the main result.

also have $5 * \text{size } Y_{\prec} + 2 * \text{size } Y_{\succeq} = 2 * m + 3 * \text{size } Y_{\prec}$

by *(simp add: m_eq)*

also have $\dots \leq 3.5 * m$

using $\langle \text{size } Y_{\prec} \leq m \text{ div } 2 \rangle$ **by** *linarith*

also have $\dots = 3.5 * \lceil n / 5 \rceil$

by *(simp add: m_def length_chop)*

also have $\dots \leq 0.7 * n + 3.5$

by *linarith*

finally have $\text{size } \{\#y \in \# \text{mset } xs. y \prec M\} \leq 0.7 * n + 3.5$

by *simp*

thus $\text{size } \{\#y \in \# \text{mset } xs. y \prec M\} \leq \text{nat } \lceil 0.7 * n + 3 \rceil$

by *linarith*

qed

lemma *size_less_than_median_of_medians:*

$\text{size } \{\#y \in \# \text{mset } xs. y < M\} \leq \text{nat } \lceil 0.7 * \text{length } xs + 3 \rceil$

using *size_median_of_medians_aux[of (<)]* **by** *simp*

lemma *size_greater_than_median_of_medians:*

$\text{size } \{\#y \in \# \text{mset } xs. y > M\} \leq \text{nat } \lceil 0.7 * \text{length } xs + 3 \rceil$

using *size_median_of_medians_aux[of (>)]* **by** *simp*

end

49.7 Efficient algorithm

We handle the base cases and computing the median for the chopped-up sublists of size 5 using the naive selection algorithm where we sort the list using insertion sort.

definition *slow_select* **where**

slow_select k $xs = \text{isort } xs ! k$

definition *slow_median* **where**

slow_median $xs = \text{slow_select } ((\text{length } xs - 1) \text{ div } 2) \text{ } xs$

lemma *slow_select_correct*: *slow_select* k $xs = \text{select } k$ xs

by (*simp* *add*: *slow_select_def* *select_def* *isort_correct*)

lemma *slow_median_correct*: *slow_median* $xs = \text{median } xs$

by (*simp* *add*: *median_def* *slow_median_def* *slow_select_correct*)

The definition of the selection algorithm is complicated somewhat by the fact that its termination is contingent on its correctness: if the first recursive call were to return an element for x that is e.g. smaller than all list elements, the algorithm would not terminate.

Therefore, we first prove partial correctness, then termination, and then combine the two to obtain total correctness.

function *mom_select* **where**

mom_select k $xs =$ (
 if $\text{length } xs \leq 20$ *then*
 slow_select k xs
 else
 let $M = \text{mom_select } (((\text{length } xs + 4) \text{ div } 5 - 1) \text{ div } 2) (\text{map } \text{slow_median } (\text{chop } 5 \text{ } xs));$
 $(ls, es, gs) = \text{partition3 } M \text{ } xs$
 in
 if $k < \text{length } ls$ *then* *mom_select* k ls
 else if $k < \text{length } ls + \text{length } es$ *then* M
 else *mom_select* $(k - \text{length } ls - \text{length } es)$ gs
)
by *auto*

If *mom_select* terminates, it agrees with *select*:

lemma *mom_select_correct_aux*:

assumes *mom_select_dom* (k, xs) **and** $k < \text{length } xs$

shows *mom_select* k $xs = \text{select } k$ xs

using *assms*

proof (*induction* *rule*: *mom_select.pinduct*)

```

case (1 k xs)
show mom_select k xs = select k xs
proof (cases length xs ≤ 20)
  case True
    thus mom_select k xs = select k xs using 1.prem1 1.hyps
    by (subst mom_select.psimps) (auto simp: select_def slow_select_correct)
  next
    case False
      define x where
        x = mom_select (((length xs + 4) div 5 - 1) div 2) (map slow_median
(chop 5 xs))
      define ls es gs where ls = filter (λy. y < x) xs and es = filter (λy. y
= x) xs
        and gs = filter (λy. y > x) xs
      define nl ne where nl = length ls and ne = length es
      note defs = nl_def ne_def x_def ls_def es_def gs_def
      have tw: (ls, es, gs) = partition3 x xs
      unfolding partition3_def defs One_nat_def ..
      have length_eq: length xs = nl + ne + length gs
      unfolding nl_def ne_def ls_def es_def gs_def
      using [[simp_depth_limit = 1]] by (induction xs) auto
      note IH = 1.IH(2,3)[OF False x_def tw refl refl]

      have mom_select k xs = (if k < nl then mom_select k ls else if k < nl
+ ne then x
        else mom_select (k - nl - ne) gs) using 1.hyps
    False
      by (subst mom_select.psimps) (simp_all add: partition3_def flip: defs
One_nat_def)
      also have ... = (if k < nl then select k ls else if k < nl + ne then x
        else select (k - nl - ne) gs)
      using IH length_eq 1.prem1 by (simp add: ls_def es_def gs_def nl_def
ne_def)
      also have ... = select k xs using ⟨k < length xs⟩
      by (subst (3) select_rec_partition[of _ _ x]) (simp_all add: nl_def
ne_def flip: tw)
      finally show mom_select k xs = select k xs .
    qed
  qed

```

mom_select indeed terminates for all inputs:

lemma mom_select_termination: All mom_select_dom

proof (relation measure (length ∘ snd); (safe)?)

fix k :: nat **and** xs :: 'a list

```

assume  $\neg \text{length } xs \leq 20$ 
thus (((((length xs + 4) div 5 - 1) div 2, map slow_median (chop 5 xs)),
k, xs)
       $\in$  measure (length  $\circ$  snd)
  by (auto simp: length_chop nat_less_iff ceiling_less_iff)
next
  fix k :: nat and xs ls es gs :: 'a list
  define x where x = mom_select (((length xs + 4) div 5 - 1) div 2)
(map slow_median (chop 5 xs))
  assume A:  $\neg \text{length } xs \leq 20$ 
      (ls, es, gs) = partition3 x xs
      mom_select_dom (((length xs + 4) div 5 - 1) div 2,
map slow_median (chop 5 xs))
  have less: ((length xs + 4) div 5 - 1) div 2 < nat  $\lceil$ length xs / 5 $\rceil$ 
  using A(1) by linarith

  For termination, it suffices to prove that x is in the list.

  have x = select (((length xs + 4) div 5 - 1) div 2) (map slow_median
(chop 5 xs))
  using less unfolding x_def by (intro mom_select_correct_aux A)
(auto simp: length_chop)
  also have ...  $\in$  set (map slow_median (chop 5 xs))
  using less by (intro select_in_set) (simp_all add: length_chop)
  also have ...  $\subseteq$  set xs
  unfolding set_map
proof safe
  fix ys assume ys: ys  $\in$  set (chop 5 xs)
  hence median ys  $\in$  set ys
  by auto
  also have set ys  $\subseteq$   $\bigcup$  (set ' set (chop 5 xs))
  using ys by blast
  also have ... = set xs
  by (rule UN_sets_chop) simp_all
  finally show slow_median ys  $\in$  set xs
  by (simp add: slow_median_correct)
qed
finally have x  $\in$  set xs .
thus ((k, ls), k, xs)  $\in$  measure (length  $\circ$  snd)
  and ((k - length ls - length es, gs), k, xs)  $\in$  measure (length  $\circ$  snd)
  using A(1,2) by (auto simp: partition3_def intro!: length_filter_less[of
x])
qed

```

termination mom_select **by** (rule mom_select_termination)

lemmas $[simp\ del] = mom_select.simps$

lemma $mom_select_correct$: $k < length\ xs \implies mom_select\ k\ xs = select\ k\ xs$

using $mom_select_correct_aux$ **and** $mom_select_termination$ **by** $blast$

49.8 Running time analysis

fun $T_partition3$:: $'a \Rightarrow 'a\ list \Rightarrow nat$ **where**

$T_partition3\ x\ [] = 1$

| $T_partition3\ x\ (y\ \#\ ys) = T_partition3\ x\ ys + 1$

lemma $T_partition3_eq$: $T_partition3\ x\ xs = length\ xs + 1$

by $(induction\ x\ xs\ rule:\ T_partition3.induct)$ $auto$

definition T_slow_select :: $nat \Rightarrow 'a :: linorder\ list \Rightarrow nat$ **where**

$T_slow_select\ k\ xs = T_isort\ xs + T_nth\ (isort\ xs)\ k + 1$

definition T_slow_median :: $'a :: linorder\ list \Rightarrow nat$ **where**

$T_slow_median\ xs = T_slow_select\ ((length\ xs - 1)\ div\ 2)\ xs + 1$

lemma $T_slow_select_le$: $T_slow_select\ k\ xs \leq length\ xs^2 + 3 * length\ xs + 3$

proof $-$

have $T_slow_select\ k\ xs \leq (length\ xs + 1)^2 + (length\ (isort\ xs) + 1) + 1$

unfolding $T_slow_select_def$

by $(intro\ add_mono\ T_isort_length)$ $(auto\ simp:\ T_nth_eq)$

also have $\dots = length\ xs^2 + 3 * length\ xs + 3$

by $(simp\ add:\ isort_correct\ algebra_simps\ power2_eq_square)$

finally show $?thesis$.

qed

lemma $T_slow_median_le$: $T_slow_median\ xs \leq length\ xs^2 + 3 * length\ xs + 4$

unfolding $T_slow_median_def$ **using** $T_slow_select_le$ $[of\ (length\ xs - 1)\ div\ 2\ xs]$ **by** $simp$

fun T_chop :: $nat \Rightarrow 'a\ list \Rightarrow nat$ **where**

$T_chop\ 0\ _ = 1$

| $T_chop\ _ [] = 1$

| $T_chop\ n\ xs = T_take\ n\ xs + T_drop\ n\ xs + T_chop\ n\ (drop\ n\ xs)$

lemmas $[simp\ del] = T_chop.simps$

lemma $T_chop_Nil [simp]: T_chop\ d\ [] = 1$
by $(cases\ d)\ (auto\ simp:\ T_chop.simps)$

lemma $T_chop_0 [simp]: T_chop\ 0\ xs = 1$
by $(auto\ simp:\ T_chop.simps)$

lemma $T_chop_reduce:$
 $n > 0 \implies xs \neq [] \implies T_chop\ n\ xs = T_take\ n\ xs + T_drop\ n\ xs +$
 $T_chop\ n\ (drop\ n\ xs)$
by $(cases\ n;\ cases\ xs)\ (auto\ simp:\ T_chop.simps)$

lemma $T_chop_le: T_chop\ d\ xs \leq 5 * length\ xs + 1$
by $(induction\ d\ xs\ rule:\ T_chop.induct)\ (auto\ simp:\ T_chop_reduce\ T_take_eq\ T_drop_eq)$

The option *domintros* here allows us to explicitly reason about where the function does and does not terminate. With this, we can skip the termination proof this time because we can reuse the one for *mom_select*.

function $(domintros)\ T_mom_select :: nat \Rightarrow 'a :: linorder\ list \Rightarrow nat$
where

$T_mom_select\ k\ xs = ($
 $\quad if\ length\ xs \leq 20\ then$
 $\quad\quad T_slow_select\ k\ xs$
 $\quad else$
 $\quad\quad let\ xss = chop\ 5\ xs;$
 $\quad\quad\quad ms = map\ slow_median\ xss;$
 $\quad\quad\quad idx = (((length\ xs + 4)\ div\ 5 - 1)\ div\ 2);$
 $\quad\quad\quad x = mom_select\ idx\ ms;$
 $\quad\quad\quad (ls,\ es,\ gs) = partition3\ x\ xs;$
 $\quad\quad\quad nl = length\ ls;$
 $\quad\quad\quad ne = length\ es$
 $\quad\quad in$
 $\quad\quad\quad (if\ k < nl\ then\ T_mom_select\ k\ ls$
 $\quad\quad\quad\quad else\ if\ k < nl + ne\ then\ 0$
 $\quad\quad\quad\quad else\ T_mom_select\ (k - nl - ne)\ gs) +$
 $\quad\quad\quad T_mom_select\ idx\ ms + T_chop\ 5\ xs + T_map\ T_slow_median$
 $xss +$
 $\quad\quad\quad T_partition3\ x\ xs + T_length\ ls + T_length\ es + 1$
 $\quad\quad)$
by *auto*

```

termination  $T\_mom\_select$ 
proof (rule allI, safe)
  fix  $k :: nat$  and  $xs :: 'a :: linorder list$ 
  have  $mom\_select\_dom (k, xs)$ 
    using  $mom\_select\_termination$  by blast
  thus  $T\_mom\_select\_dom (k, xs)$ 
    by (induction k xs rule: mom_select.pinduct)
      (rule T_mom_select.domintros, simp_all)
qed

```

```

lemmas [simp del] =  $T\_mom\_select.simps$ 

```

```

function  $T'\_mom\_select :: nat \Rightarrow nat$  where
   $T'\_mom\_select n =$ 
    (if  $n \leq 20$  then
      463
    else
       $T'\_mom\_select (nat \lceil 0.2 * n \rceil) + T'\_mom\_select (nat \lceil 0.7 * n + 3 \rceil)$ 
    +  $17 * n + 50$ )
  by force+
termination by (relation measure id; simp; linarith)

```

```

lemmas [simp del] =  $T'\_mom\_select.simps$ 

```

```

lemma  $T'\_mom\_select\_ge: T'\_mom\_select n \geq 463$ 
  by (induction n rule: T'_mom_select.induct; subst T'_mom_select.simps)
auto

```

```

lemma  $T'\_mom\_select\_mono:$ 
   $m \leq n \implies T'\_mom\_select m \leq T'\_mom\_select n$ 
proof (induction n arbitrary: m rule: less_induct)
  case (less n m)
  show ?case
  proof (cases m ≤ 20)
    case True
    hence  $T'\_mom\_select m = 463$ 
      by (subst T'_mom_select.simps) auto
    also have  $\dots \leq T'\_mom\_select n$ 
      by (rule T'_mom_select_ge)
    finally show ?thesis .
  next
  case False

```

```

hence  $T'_mom\_select\ m =$ 
       $T'_mom\_select\ (nat\ [0.2*m]) + T'_mom\_select\ (nat\ [0.7*m$ 
+  $3]) + 17 * m + 50$ 
      by  $(subst\ T'_mom\_select.simps)\ auto$ 
also have  $\dots \leq T'_mom\_select\ (nat\ [0.2*n]) + T'_mom\_select\ (nat$ 
 $[0.7*n + 3]) + 17 * n + 50$ 
      using  $\langle m \leq n \rangle$  and False by  $(intro\ add\_mono\ less.IH;\ linearith)$ 
also have  $\dots = T'_mom\_select\ n$ 
      using  $\langle m \leq n \rangle$  and False by  $(subst\ T'_mom\_select.simps)\ auto$ 
finally show ?thesis .
qed
qed

```

lemma $T_mom_select_le_aux: T_mom_select\ k\ xs \leq T'_mom_select$
 $(length\ xs)$

proof $(induction\ k\ xs\ rule: T_mom_select.induct)$

case $(1\ k\ xs)$

define n **where** $[simp]: n = length\ xs$

define x **where**

$x = mom_select\ (((length\ xs + 4)\ div\ 5 - 1)\ div\ 2)\ (map\ slow_median$
 $(chop\ 5\ xs))$

define $ls\ es\ gs$ **where** $ls = filter\ (\lambda y. y < x)\ xs$ **and** $es = filter\ (\lambda y. y =$
 $x)\ xs$

and $gs = filter\ (\lambda y. y > x)\ xs$

define $nl\ ne$ **where** $nl = length\ ls$ **and** $ne = length\ es$

note $defs = nl_def\ ne_def\ x_def\ ls_def\ es_def\ gs_def$

have $tw: (ls,\ es,\ gs) = partition3\ x\ xs$

unfolding $partition3_def\ defs\ One_nat_def\ ..$

note $IH = 1.IH(1,2,3)[OF\ _ refl\ refl\ refl\ x_def\ tw\ refl\ refl\ refl\ refl]$

show *?case*

proof $(cases\ length\ xs \leq 20)$

case *True* — base case

hence $T_mom_select\ k\ xs \leq (length\ xs)^2 + 3 * length\ xs + 3$

using $T_slow_select_le[of\ k\ xs]$ **by** $(subst\ T_mom_select.simps)\ auto$

also have $\dots \leq 20^2 + 3 * 20 + 3$

using *True* **by** $(intro\ add_mono\ power_mono)\ auto$

also have $\dots \leq 463$

by *simp*

also have $\dots = T'_mom_select\ (length\ xs)$

using *True* **by** $(simp\ add: T'_mom_select.simps)$

finally show *?thesis* **by** *simp*

next

case *False* — recursive case

```

have  $((n + 4) \text{ div } 5 - 1) \text{ div } 2 < \text{nat } \lceil n / 5 \rceil$ 
  using False unfolding n_def by linarith
hence  $x = \text{select } (((n + 4) \text{ div } 5 - 1) \text{ div } 2) (\text{map } \text{slow\_median } (\text{chop } 5 \text{ xs}))$ 
  unfolding  $x\_def \ n\_def$  by (intro mom_select_correct) (auto simp: length_chop)
  also have  $((n + 4) \text{ div } 5 - 1) \text{ div } 2 = (\text{nat } \lceil n / 5 \rceil - 1) \text{ div } 2$ 
    by linarith
  also have  $\text{select } \dots (\text{map } \text{slow\_median } (\text{chop } 5 \text{ xs})) = \text{median } (\text{map } \text{slow\_median } (\text{chop } 5 \text{ xs}))$ 
    by (auto simp: median_def length_chop)
finally have  $x_{eq}: x = \text{median } (\text{map } \text{slow\_median } (\text{chop } 5 \text{ xs}))$  .

```

The cost of computing the medians of all the subgroups:

```

define  $T\_ms$  where  $T\_ms = T\_map \ T\_slow\_median (\text{chop } 5 \text{ xs})$ 
have  $T\_ms \leq 9 * n + 45$ 
proof -
  have  $T\_ms = (\sum_{ys \leftarrow \text{chop } 5 \text{ xs}} T\_slow\_median \ ys) + \text{length } (\text{chop } 5 \text{ xs}) + 1$ 
    by (simp add: T_ms_def T_map_eq)
  also have  $(\sum_{ys \leftarrow \text{chop } 5 \text{ xs}} T\_slow\_median \ ys) \leq (\sum_{ys \leftarrow \text{chop } 5 \text{ xs}} 44)$ 
    proof (intro sum_list_mono)
      fix  $ys$  assume  $ys \in \text{set } (\text{chop } 5 \text{ xs})$ 
      hence  $\text{length } ys \leq 5$ 
        using length_chop_part_le by blast
      have  $T\_slow\_median \ ys \leq (\text{length } ys) ^ 2 + 3 * \text{length } ys + 4$ 
        by (rule T_slow_median_le)
      also have  $\dots \leq 5 ^ 2 + 3 * 5 + 4$ 
        using (length ys ≤ 5) by (intro add_mono power_mono) auto
      finally show  $T\_slow\_median \ ys \leq 44$  by simp
    qed
  also have  $(\sum_{ys \leftarrow \text{chop } 5 \text{ xs}} 44) + \text{length } (\text{chop } 5 \text{ xs}) + 1 = 45 * \text{nat } \lceil \text{real } n / 5 \rceil + 1$ 
    by (simp add: map_replicate_const length_chop)
  also have  $\dots \leq 9 * n + 45$ 
    by linarith
finally show  $T\_ms \leq 9 * n + 45$  by simp
qed

```

The cost of the first recursive call (to compute the median of medians):

```

define  $T\_rec1$  where
   $T\_rec1 = T\_mom\_select (((\text{length } \text{xs} + 4) \text{ div } 5 - 1) \text{ div } 2) (\text{map } \text{slow\_median } (\text{chop } 5 \text{ xs}))$ 

```

```

have  $T\_rec1 \leq T'\_mom\_select$  (length (map slow_median (chop 5
xs)))
  using False unfolding  $T\_rec1\_def$  by (intro IH(3)) auto
hence  $T\_rec1 \leq T'\_mom\_select$  (nat  $\lceil 0.2 * n \rceil$ )
  by (simp add: length_chop)

The cost of the second recursive call (to compute the final result):

define  $T\_rec2$  where  $T\_rec2 =$  (if  $k < nl$  then  $T\_mom\_select$   $k$   $ls$ 
  else if  $k < nl + ne$  then 0
  else  $T\_mom\_select$  ( $k - nl - ne$ )  $gs$ )
consider  $k < nl \mid k \in \{nl..<nl+ne\} \mid k \geq nl+ne$ 
  by force
hence  $T\_rec2 \leq T'\_mom\_select$  (nat  $\lceil 0.7 * n + 3 \rceil$ )
proof cases
  assume  $k < nl$ 
  hence  $T\_rec2 = T\_mom\_select$   $k$   $ls$ 
    by (simp add:  $T\_rec2\_def$ )
  also have  $\dots \leq T'\_mom\_select$  (length  $ls$ )
    by (rule IH(1)) (use  $\langle k < nl \rangle$  False in  $\langle auto\ simp: defs \rangle$ )
  also have length  $ls \leq$  nat  $\lceil 0.7 * n + 3 \rceil$ 
    unfolding  $ls\_def$  using size_less_than_median_of_medians[ $of\ xs$ ]
    by (auto simp: length_filter_conv_size_filter_mset slow_median_correct[ $abs\_def$ ]
 $x\_eq$ )
  hence  $T'\_mom\_select$  (length  $ls$ )  $\leq T'\_mom\_select$  (nat  $\lceil 0.7 * n$ 
+  $3 \rceil$ )
    by (rule  $T'\_mom\_select\_mono$ )
  finally show ?thesis .
next
  assume  $k \in \{nl..<nl + ne\}$ 
  hence  $T\_rec2 = 0$ 
    by (simp add:  $T\_rec2\_def$ )
  thus ?thesis
    using  $T'\_mom\_select\_ge$ [ $of$  nat  $\lceil 0.7 * n + 3 \rceil$ ] by simp
next
  assume  $k \geq nl + ne$ 
  hence  $T\_rec2 = T\_mom\_select$  ( $k - nl - ne$ )  $gs$ 
    by (simp add:  $T\_rec2\_def$ )
  also have  $\dots \leq T'\_mom\_select$  (length  $gs$ )
    unfolding  $nl\_def$   $ne\_def$  by (rule IH(2)) (use  $\langle k \geq nl + ne \rangle$  False
in  $\langle auto\ simp: defs \rangle$ )
  also have length  $gs \leq$  nat  $\lceil 0.7 * n + 3 \rceil$ 
    unfolding  $gs\_def$  using size_greater_than_median_of_medians[ $of$ 
 $x_s$ ]
    by (auto simp: length_filter_conv_size_filter_mset slow_median_correct[ $abs\_def$ ]

```

x_eq)
hence $T'_mom_select (length\ gs) \leq T'_mom_select (nat \lceil 0.7 * n + 3 \rceil)$
by (rule $T'_mom_select_mono$)
finally show *?thesis* .
qed

Now for the final inequality chain:

have $T_mom_select\ k\ xs = T_rec2 + T_rec1 + T_ms + n + nl + ne + T_chop\ 5\ xs + 4$ **using** *False*
by (subst $T_mom_select.simps$, unfold $Let_def\ tw$ [*symmetric*] defs [*symmetric*])
(simp_all add: $nl_def\ ne_def\ T_rec1_def\ T_rec2_def\ T_partition3_eq\ T_length_eq\ T_ms_def$)
also have $nl \leq n$ **by** (simp add: $nl_def\ ls_def$)
also have $ne \leq n$ **by** (simp add: $ne_def\ es_def$)
also note $\langle T_ms \leq 9 * n + 45 \rangle$
also have $T_chop\ 5\ xs \leq 5 * n + 1$
using $T_chop_le[of\ 5\ xs]$ **by** *simp*
also note $\langle T_rec1 \leq T'_mom_select (nat \lceil 0.2*n \rceil) \rangle$
also note $\langle T_rec2 \leq T'_mom_select (nat \lceil 0.7*n + 3 \rceil) \rangle$
finally have $T_mom_select\ k\ xs \leq$
 $T'_mom_select (nat \lceil 0.7*n + 3 \rceil) + T'_mom_select (nat \lceil 0.2*n \rceil) + 17 * n + 50$
by *simp*
also have $\dots = T'_mom_select\ n$
using *False* **by** (subst $T'_mom_select.simps$) *auto*
finally show *?thesis* **by** *simp*
qed
qed

49.9 Akra–Bazzi Light

lemma *akra_bazzi_light_aux1*:
fixes $a\ b :: real$ **and** $n\ n0 :: nat$
assumes $ab: a > 0\ a < 1\ n > n0$
assumes $n0 \geq (max\ 0\ b + 1) / (1 - a)$
shows $nat \lceil a*n+b \rceil < n$
proof –
have $a * real\ n + max\ 0\ b \geq 0$
using ab **by** *simp*
hence $real (nat \lceil a*n+b \rceil) \leq a * n + max\ 0\ b + 1$
by *linarith*
also {

```

have n0 ≥ (max 0 b + 1) / (1 - a)
  by fact
also have ... < real n
  using assms by simp
finally have a * real n + max 0 b + 1 < real n
  using ab by (simp add: field_simps)
}
finally show nat [a*n+b] < n
  using ⟨n > n0⟩ by linarith
qed

```

lemma *akra_bazzi_light_aux2*:

```

fixes f :: nat ⇒ real
fixes n0 :: nat and a b c d :: real and C1 C2 C1 C2 :: real
assumes bounds: a > 0 c > 0 a + c < 1 C1 ≥ 0
assumes rec: ∀ n > n0. f n = f (nat [a*n+b]) + f (nat [c*n+d]) + C1 *
n + C2
assumes ineqs: n0 > (max 0 b + max 0 d + 2) / (1 - a - c)
              C3 ≥ C1 / (1 - a - c)
              C3 ≥ (C1 * n0 + C2 + C4) / ((1 - a - c) * n0 - max 0 b
- max 0 d - 2)
              ∀ n ≤ n0. f n ≤ C4
shows f n ≤ C3 * n + C4
proof (induction n rule: less_induct)
  case (less n)
  have 0 ≤ C1 / (1 - a - c)
    using bounds by auto
  also have ... ≤ C3
    by fact
  finally have C3 ≥ 0 .

```

show ?*case*

```

proof (cases n > n0)
  case False
  hence f n ≤ C4
    using ineqs(4) by auto
  also have ... ≤ C3 * real n + C4
    using bounds ⟨C3 ≥ 0⟩ by auto
  finally show ?thesis .

```

next

```

case True
have nonneg: a * n ≥ 0 c * n ≥ 0
  using bounds by simp_all

```

have $(\max 0 b + 1) / (1 - a) \leq (\max 0 b + \max 0 d + 2) / (1 - a - c)$
using *bounds* **by** (*intro frac_le*) *auto*
hence $n_0 \geq (\max 0 b + 1) / (1 - a)$
using *ineqs(1)* **by** *linarith*
hence *rec_less1*: $\text{nat } \lceil a*n+b \rceil < n$
using *bounds* $\langle n > n_0 \rangle$ **by** (*intro akra_bazzi_light_aux1* [*of_* n_0]) *auto*

have $(\max 0 d + 1) / (1 - c) \leq (\max 0 b + \max 0 d + 2) / (1 - a - c)$
using *bounds* **by** (*intro frac_le*) *auto*
hence $n_0 \geq (\max 0 d + 1) / (1 - c)$
using *ineqs(1)* **by** *linarith*
hence *rec_less2*: $\text{nat } \lceil c*n+d \rceil < n$
using *bounds* $\langle n > n_0 \rangle$ **by** (*intro akra_bazzi_light_aux1* [*of_* n_0]) *auto*

have $f n = f (\text{nat } \lceil a*n+b \rceil) + f (\text{nat } \lceil c*n+d \rceil) + C_1 * n + C_2$
using $\langle n > n_0 \rangle$ **by** (*subst rec*) *auto*
also have $\dots \leq (C_3 * \text{nat } \lceil a*n+b \rceil + C_4) + (C_3 * \text{nat } \lceil c*n+d \rceil + C_4) + C_1 * n + C_2$
using *rec_less1 rec_less2* **by** (*intro add_mono less.IH*) *auto*
also have $\dots \leq (C_3 * (a*n+\max 0 b+1) + C_4) + (C_3 * (c*n+\max 0 d+1) + C_4) + C_1 * n + C_2$
using *bounds* $\langle C_3 \geq 0 \rangle$ *nonneg* **by** (*intro add_mono mult_left_mono order.refl; linarith*)
also have $\dots = C_3 * n + ((C_3 * (\max 0 b + \max 0 d + 2) + 2 * C_4 + C_2) - (C_3 * (1 - a - c) - C_1) * n)$
by (*simp add: algebra_simps*)
also have $\dots \leq C_3 * n + ((C_3 * (\max 0 b + \max 0 d + 2) + 2 * C_4 + C_2) - (C_3 * (1 - a - c) - C_1) * n_0)$
using $\langle n > n_0 \rangle$ *ineqs(2)* *bounds*
by (*intro add_mono diff_mono order.refl mult_left_mono*) (*auto simp: field_simps*)
also have $(C_3 * (\max 0 b + \max 0 d + 2) + 2 * C_4 + C_2) - (C_3 * (1 - a - c) - C_1) * n_0 \leq C_4$
using *ineqs bounds* **by** (*simp add: field_simps*)
finally show $f n \leq C_3 * \text{real } n + C_4$
by (*simp add: mult_right_mono*)
qed
qed

lemma *akra_bazzi_light*:

```

fixes f :: nat ⇒ real
fixes n0 :: nat and a b c d C1 C2 :: real
assumes bounds: a > 0 c > 0 a + c < 1 C1 ≥ 0
assumes rec: ∀ n>n0. f n = f (nat [a*n+b]) + f (nat [c*n+d]) + C1 *
n + C2
shows ∃ C3 C4. ∀ n. f n ≤ C3 * real n + C4
proof -
  define n0' where n0' = max n0 (nat [(max 0 b + max 0 d + 2) / (1 -
a - c) + 1])
  define C4 where C4 = Max (f ' {..n0'})
  define C3 where C3 = max (C1 / (1 - a - c))
((C1 * n0' + C2 + C4) / ((1 - a - c) * n0' - max 0
b - max 0 d - 2))

  have f n ≤ C3 * n + C4 for n
  proof (rule akra_bazzi_light_aux2[OF bounds _])
    show ∀ n>n0'. f n = f (nat [a*n+b]) + f (nat [c*n+d]) + C1 * n +
C2
      using rec by (auto simp: n0'_def)
    next
      show C3 ≥ C1 / (1 - a - c)
      and C3 ≥ (C1 * n0' + C2 + C4) / ((1 - a - c) * n0' - max 0 b -
max 0 d - 2)
      by (simp_all add: C3_def)
    next
      have (max 0 b + max 0 d + 2) / (1 - a - c) < nat [(max 0 b + max
0 d + 2) / (1 - a - c) + 1]
      by linarith
      also have ... ≤ n0'
      by (simp add: n0'_def)
      finally show (max 0 b + max 0 d + 2) / (1 - a - c) < real n0'.
    next
      show ∀ n≤n0'. f n ≤ C4
      by (auto simp: C4_def)
  qed
thus ?thesis by blast
qed

```

```

lemma akra_bazzi_light_nat:
  fixes f :: nat ⇒ nat
  fixes n0 :: nat and a b c d :: real and C1 C2 :: nat
  assumes bounds: a > 0 c > 0 a + c < 1 C1 ≥ 0
  assumes rec: ∀ n>n0. f n = f (nat [a*n+b]) + f (nat [c*n+d]) + C1 *
n + C2

```

```

shows  $\exists C_3 C_4. \forall n. f\ n \leq C_3 * n + C_4$ 
proof -
  have  $\exists C_3 C_4. \forall n. \text{real } (f\ n) \leq C_3 * \text{real } n + C_4$ 
    using assms by (intro akra_bazzi_light[of a c C1 n0 f b d C2]) auto
  then obtain C3 C4 where le:  $\forall n. \text{real } (f\ n) \leq C_3 * \text{real } n + C_4$ 
    by blast
  have  $f\ n \leq \text{nat } \lceil C_3 \rceil * n + \text{nat } \lceil C_4 \rceil$  for n
  proof -
    have  $\text{real } (f\ n) \leq C_3 * \text{real } n + C_4$ 
      using le by blast
    also have  $\dots \leq \text{real } (\text{nat } \lceil C_3 \rceil) * \text{real } n + \text{real } (\text{nat } \lceil C_4 \rceil)$ 
      by (intro add_mono mult_right_mono; linarith)
    also have  $\dots = \text{real } (\text{nat } \lceil C_3 \rceil * n + \text{nat } \lceil C_4 \rceil)$ 
      by simp
    finally show ?thesis by linarith
  qed
  thus ?thesis by blast
qed

```

```

lemma T'_mom_select_le':  $\exists C_1 C_2. \forall n. T'_\text{mom\_select } n \leq C_1 * n + C_2$ 
proof (rule akra_bazzi_light_nat)
  show  $\forall n > 20. T'_\text{mom\_select } n = T'_\text{mom\_select } (\text{nat } \lceil 0.2 * n + 0 \rceil)$ 
  +
     $T'_\text{mom\_select } (\text{nat } \lceil 0.7 * n + 3 \rceil) + 17 * n + 50$ 
  using T'_mom_select.simps by auto
qed auto

```

end

50 Bibliographic Notes

Red-black trees The insert function follows Okasaki [15]. The delete function in theory *RBT_Set* follows Kahrs [11, 12], an alternative delete function is given in theory *RBT_Set2*.

2-3 trees Equational definitions were given by Hoffmann and O'Donnell [9] (only insertion) and Reade [19]. Our formalisation is based on the teaching material by Turbak [22] and the article by Hinze [8].

1-2 brother trees They were invented by Ottmann and Six [16, 17]. The functional version is due to Hinze [7].

AA trees They were invented by Arne Anderson [3]. Our formalisation follows Ragde [18] but fixes a number of mistakes.

Splay trees They were invented by Sleator and Tarjan [21]. Our formalisation follows Schoenmakers [20].

Join-based BSTs They were invented by Adams [1, 2] and analyzed by Blelloch *et al.* [4].

Leftist heaps They were invented by Crane [6]. A first functional implementation is due to Núñez *et al.* [14].

References

- [1] S. Adams. Implementing sets efficiently in a functional language. Technical Report CSTR 92-10, University of Southampton, Department of Electronics and Computer Science, 1992.
- [2] S. Adams. Efficient sets - A balancing act. *J. Funct. Program.*, 3(4):553–561, 1993.
- [3] A. Andersson. Balanced search trees made simple. In *Algorithms and Data Structures (WADS '93)*, volume 709 of *LNCS*, pages 60–71. Springer, 1993.
- [4] G. E. Blelloch, D. Ferizovic, and Y. Sun. Just join for parallel ordered sets. In *SPAA*, pages 253–264. ACM, 2016.
- [5] W. Braun and M. Rem. A logarithmic implementation of flexible arrays. Memorandum MR83/4. Eindhoven University of Technology, 1983.
- [6] C. A. Crane. *Linear Lists and Priority Queues as Balanced Binary Trees*. PhD thesis, Computer Science Department, Stanford University, 1972.
- [7] R. Hinze. Purely functional 1-2 brother trees. *J. Functional Programming*, 19(6):633–644, 2009.
- [8] R. Hinze. On constructing 2-3 trees. *J. Funct. Program.*, 28:e19, 2018.
- [9] C. M. Hoffmann and M. J. O’Donnell. Programming with equations. *ACM Trans. Program. Lang. Syst.*, 4(1):83–112, 1982.
- [10] R. R. Hoogerwoord. A logarithmic implementation of flexible arrays. In R. Bird, C. Morgan, and J. Woodcock, editors, *Mathematics of Program Construction, Second International Conference*, volume 669 of *LNCS*, pages 191–207. Springer, 1992.

- [11] S. Kahrs. Red black trees. <http://www.cs.ukc.ac.uk/people/staff/smk/redblack/rb.html>.
- [12] S. Kahrs. Red-black trees with types. *J. Functional Programming*, 11(4):425–432, 2001.
- [13] T. Nipkow. Automatic functional correctness proofs for functional search trees. <http://www.in.tum.de/~nipkow/pubs/trees.html>, Feb. 2016.
- [14] M. Núñez, P. Palao, and R. Pena. A second year course on data structures based on functional programming. In P. H. Hartel and M. J. Plasmeijer, editors, *Functional Programming Languages in Education*, volume 1022 of *LNCS*, pages 65–84. Springer, 1995.
- [15] C. Okasaki. *Purely Functional Data Structures*. Cambridge University Press, 1998.
- [16] T. Ottmann and H.-W. Six. Eine neue Klasse von ausgeglichenen Binärbäumen. *Angewandte Informatik*, 18(9):395–400, 1976.
- [17] T. Ottmann and D. Wood. 1-2 brother trees or AVL trees revisited. *Comput. J.*, 23(3):248–255, 1980.
- [18] P. Ragde. Simple balanced binary search trees. In Caldwell, Hölzenspies, and Achten, editors, *Trends in Functional Programming in Education*, volume 170 of *EPTCS*, pages 78–87, 2014.
- [19] C. Reade. Balanced trees with removals: An exercise in rewriting and proof. *Sci. Comput. Program.*, 18(2):181–204, 1992.
- [20] B. Schoenmakers. A systematic analysis of splaying. *Information Processing Letters*, 45:41–50, 1993.
- [21] D. D. Sleator and R. E. Tarjan. Self-adjusting binary search trees. *J. ACM*, 32(3):652–686, 1985.
- [22] F. Turbak. CS230 Handouts — Spring 2007, 2007. <http://cs.wellesley.edu/~cs230/spring07/handouts.html>.